

---

# **Sage Reference Manual: Cell complexes and their homology**

***Release 8.7***

**The Sage Development Team**

**Mar 25, 2019**



## CONTENTS

<b>1</b>	<b>Chain complexes</b>	<b>3</b>
<b>2</b>	<b>Chains and cochains</b>	<b>19</b>
<b>3</b>	<b>Morphisms of chain complexes</b>	<b>27</b>
<b>4</b>	<b>Chain homotopies and chain contractions</b>	<b>31</b>
<b>5</b>	<b>Homspaces between chain complexes</b>	<b>37</b>
<b>6</b>	<b>Finite simplicial complexes</b>	<b>41</b>
<b>7</b>	<b>Morphisms of simplicial complexes</b>	<b>79</b>
<b>8</b>	<b>Homsets between simplicial complexes</b>	<b>89</b>
<b>9</b>	<b>Examples of simplicial complexes</b>	<b>93</b>
<b>10</b>	<b>Finite Delta-complexes</b>	<b>107</b>
<b>11</b>	<b>Finite cubical complexes</b>	<b>123</b>
<b>12</b>	<b>Simplicial sets</b>	<b>139</b>
<b>13</b>	<b>Methods of constructing simplicial sets</b>	<b>177</b>
<b>14</b>	<b>Examples of simplicial sets.</b>	<b>201</b>
<b>15</b>	<b>Catalog of simplicial sets</b>	<b>207</b>
<b>16</b>	<b>Morphisms and homsets for simplicial sets</b>	<b>209</b>
<b>17</b>	<b>Generic cell complexes</b>	<b>223</b>
<b>18</b>	<b>Koszul Complexes</b>	<b>237</b>
<b>19</b>	<b>Hochschild Complexes</b>	<b>239</b>
<b>20</b>	<b>Homology Groups</b>	<b>245</b>
<b>21</b>	<b>Homology and cohomology with a basis</b>	<b>247</b>
<b>22</b>	<b>Algebraic topological model for a cell complex</b>	<b>257</b>

<b>23</b>	<b>Induced morphisms on homology</b>	<b>261</b>
<b>24</b>	<b>Utility Functions for Matrices</b>	<b>265</b>
<b>25</b>	<b>Interface to CHomP</b>	<b>267</b>
<b>26</b>	<b>Indices and Tables</b>	<b>275</b>
	<b>Python Module Index</b>	<b>277</b>
	<b>Index</b>	<b>279</b>

Sage includes some tools for algebraic topology: from the algebraic side, chain complexes and their homology, and from the topological side, simplicial complexes,  $\Delta$ -complexes, cubical complexes, and simplicial sets. A class of generic cell complexes is also available, mainly for developers who want to use it as a base for other types of cell complexes.



## CHAIN COMPLEXES

## AUTHORS:

- John H. Palmieri (2009-04)

This module implements bounded chain complexes of free  $R$ -modules, for any commutative ring  $R$  (although the interesting things, like homology, only work if  $R$  is the integers or a field).

Fix a ring  $R$ . A chain complex over  $R$  is a collection of  $R$ -modules  $\{C_n\}$  indexed by the integers, with  $R$ -module maps  $d_n : C_n \rightarrow C_{n+1}$  such that  $d_{n+1} \circ d_n = 0$  for all  $n$ . The maps  $d_n$  are called *differentials*.

One can vary this somewhat: the differentials may decrease degree by one instead of increasing it: sometimes a chain complex is defined with  $d_n : C_n \rightarrow C_{n-1}$  for each  $n$ . Indeed, the differentials may change dimension by any fixed integer.

Also, the modules may be indexed over an abelian group other than the integers, e.g.,  $\mathbf{Z}^m$  for some integer  $m \geq 1$ , in which case the differentials may change the grading by any element of that grading group. The elements of the grading group are generally called degrees, so  $C_n$  is the module in degree  $n$  and so on.

In this implementation, the ring  $R$  must be commutative and the modules  $C_n$  must be free  $R$ -modules. As noted above, homology calculations will only work if the ring  $R$  is either  $\mathbf{Z}$  or a field. The modules may be indexed by any free abelian group. The differentials may increase degree by 1 or decrease it, or indeed change it by any fixed amount: this is controlled by the `degree_of_differential` parameter used in defining the chain complex.

```
sage.homology.chain_complex.ChainComplex(data=None, base_ring=None, grading_group=None, degree_of_differential=1, degree=1, check=True)
```

Define a chain complex.

## INPUT:

- `data` – the data defining the chain complex; see below for more details.

The following keyword arguments are supported:

- `base_ring` – a commutative ring (optional), the ring over which the chain complex is defined. If this is not specified, it is determined by the data defining the chain complex.
- `grading_group` – a additive free abelian group (optional, default  $\mathbf{ZZ}$ ), the group over which the chain complex is indexed.
- `degree_of_differential` – element of `grading_group` (optional, default 1). The degree of the differential.
- `degree` – alias for `degree_of_differential`.
- `check` – boolean (optional, default `True`). If `True`, check that each consecutive pair of differentials are composable and have composite equal to zero.

OUTPUT:

A chain complex.

**Warning:** Right now, homology calculations will only work if the base ring is either  $\mathbf{Z}$  or a field, so please take this into account when defining a chain complex.

Use data to define the chain complex. This may be in any of the following forms.

1. a dictionary with integers (or more generally, elements of `grading_group`) for keys, and with `data[n]` a matrix representing (via left multiplication) the differential coming from degree  $n$ . (Note that the shape of the matrix then determines the rank of the free modules  $C_n$  and  $C_{n+d}$ .)
2. a list/tuple/iterable of the form  $[C_0, d_0, C_1, d_1, C_2, d_2, \dots]$ , where each  $C_i$  is a free module and each  $d_i$  is a matrix, as above. This only makes sense if `grading_group` is  $\mathbf{Z}$  and `degree` is 1.
3. a list/tuple/iterable of the form  $[r_0, d_0, r_1, d_1, r_2, d_2, \dots]$ , where  $r_i$  is the rank of the free module  $C_i$  and each  $d_i$  is a matrix, as above. This only makes sense if `grading_group` is  $\mathbf{Z}$  and `degree` is 1.
4. a list/tuple/iterable of the form  $[d_0, d_1, d_2, \dots]$  where each  $d_i$  is a matrix, as above. This only makes sense if `grading_group` is  $\mathbf{Z}$  and `degree` is 1.

**Note:** In fact, the free modules  $C_i$  in case 2 and the ranks  $r_i$  in case 3 are ignored: only the matrices are kept, and from their shapes, the ranks of the modules are determined. (Indeed, if `data` is a list or tuple, then any element which is not a matrix is discarded; thus the list may have any number of different things in it, and all of the non-matrices will be ignored.) No error checking is done to make sure, for instance, that the given modules have the appropriate ranks for the given matrices. However, as long as `check` is `True`, the code checks to see if the matrices are composable and that each appropriate composite is zero.

If the base ring is not specified, then the matrices are examined to determine a ring over which they are all naturally defined, and this becomes the base ring for the complex. If no such ring can be found, an error is raised. If the base ring is specified, then the matrices are converted automatically to this ring when defining the chain complex. If some matrix cannot be converted, then an error is raised.

EXAMPLES:

```
sage: ChainComplex()
Trivial chain complex over Integer Ring

sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: C
Chain complex with at most 2 nonzero terms over Integer Ring

sage: m = matrix(ZZ, 2, 2, [0, 1, 0, 0])
sage: D = ChainComplex([m, m], base_ring=GF(2)); D
Chain complex with at most 3 nonzero terms over Finite Field of size 2
sage: D == loads(dumps(D))
True
sage: D.differential(0)==m, m.is_immutable(), D.differential(0).is_immutable()
(True, False, True)
```

Note that when a chain complex is defined in Sage, new differentials may be created: every nonzero module in the chain complex must have a differential coming from it, even if that differential is zero:



```

sage: IZ = ChainComplex({0: identity_matrix(ZZ, 1)})
sage: diff = IZ.differential() # the differentials in the chain complex
sage: diff[-1], diff[0], diff[1]
([], [1], [])
sage: IZ.differential(1).parent()
Full MatrixSpace of 0 by 1 dense matrices over Integer Ring
sage: mat = ChainComplex({0: matrix(ZZ, 3, 4)}).differential(1)
sage: mat.nrows(), mat.ncols()
(0, 3)

```

Defining the base ring implicitly:

```

sage: ChainComplex([matrix(QQ, 3, 1), matrix(ZZ, 4, 3)])
Chain complex with at most 3 nonzero terms over Rational Field
sage: ChainComplex([matrix(GF(125, 'a'), 3, 1), matrix(ZZ, 4, 3)])
Chain complex with at most 3 nonzero terms over Finite Field in a of size 5^3

```

If the matrices are defined over incompatible rings, an error results:

```

sage: ChainComplex([matrix(GF(125, 'a'), 3, 1), matrix(QQ, 4, 3)])
Traceback (most recent call last):
...
TypeError: no common canonical parent for objects with parents: 'Finite Field in
↪a of size 5^3' and 'Rational Field'

```

If the base ring is given explicitly but is not compatible with the matrices, an error results:

```

sage: ChainComplex([matrix(GF(125, 'a'), 3, 1)], base_ring=QQ)
Traceback (most recent call last):
...
TypeError: unable to convert 0 to a rational

```

```

class sage.homology.chain_complex.ChainComplex_class(grading_group,
                                                       degree_of_differential, base_ring,
                                                       differentials)

```

Bases: `sage.structure.parent.Parent`

See `ChainComplex()` for full documentation.

The differentials are required to be in the following canonical form:

- All differentials that are not  $0 \times 0$  must be specified (even if they have zero rows or zero columns), and
- Differentials that are  $0 \times 0$  must not be specified.
- Immutable matrices over the `base_ring`

This and more is ensured by the assertions in the constructor. The `ChainComplex()` factory function must ensure that only valid input is passed.

EXAMPLES:

```

sage: C = ChainComplex(); C
Trivial chain complex over Integer Ring

sage: D = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: D
Chain complex with at most 2 nonzero terms over Integer Ring

```

**Element**alias of `Chain_class`**betti** (*deg=None, base\_ring=None*)

The Betti number the chain complex.

That is, write the homology in this degree as a direct sum of a free module and a torsion module; the Betti number is the rank of the free summand.

INPUT:

- *deg* – an element of the grading group for the chain complex or `None` (default `None`); if `None`, then return every Betti number, as a dictionary indexed by degree, or if an element of the grading group, then return the Betti number in that degree
- *base\_ring* – a commutative ring (optional, default is the base ring for the chain complex); compute homology with these coefficients – must be either the integers or a field

OUTPUT:

The Betti number in degree *deg* – the rank of the free part of the homology module in this degree.

EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: C.betti(0)
2
sage: [C.betti(n) for n in range(5)]
[2, 1, 0, 0, 0]
sage: C.betti()
{0: 2, 1: 1}

sage: D = ChainComplex({0: matrix(GF(5), [[3, 1], [1, 2]])})
sage: D.betti()
{0: 1, 1: 1}
```

**cartesian\_product** (*\*factors, \*\*kws*)Return the direct sum (Cartesian product) of *self* with *D*.

Let *C* and *D* be two chain complexes with differentials  $\partial_C$  and  $\partial_D$ , respectively, of the same degree (so they must also have the same grading group). The direct sum  $S = C \oplus D$  is a chain complex given by  $S_i = C_i \oplus D_i$  with differential  $\partial = \partial_C \oplus \partial_D$ .

INPUT:

- *subdivide* – (default: `False`) whether to subdivide the the differential matrices

EXAMPLES:

```
sage: R.<x,y> = QQ[]
sage: C = ChainComplex([matrix([-y],[x]), matrix([x, y])])
sage: D = ChainComplex([matrix([x-y]), matrix([0], [0])])
sage: ascii_art(C.cartesian_product(D))
      [x y 0]      [ -y      0]
      [0 0 0]      [  x      0]
      [0 0 0]      [  0 x - y]
0 <-- C_2 <----- C_1 <----- C_0 <-- 0

sage: D = ChainComplex({1: matrix([x-y]), 4: matrix([x], [y])})
sage: ascii_art(D)
[x]
```

(continues on next page)

(continued from previous page)

```

          [y]                      [x - y]
0 <-- C_5 <----- C_4 <-- 0 <-- C_2 <----- C_1 <-- 0
sage: ascii_art(cartesian_product([C, D]))

          [x]                      [  x      y      0]          [-y]
          [y]                      [  0      0 x - y]          [ x]
0 <-- C_5 <----- C_4 <-- 0 <-- C_2 <----- C_1 <----- C_0 <-- 0

```

The degrees of the differentials must agree:

```

sage: C = ChainComplex({1:matrix([[x]])}, degree_of_differential=-1)
sage: D = ChainComplex({1:matrix([[x]])}, degree_of_differential=1)
sage: C.cartesian_product(D)
Traceback (most recent call last):
...
ValueError: the degrees of the differentials must match

```

```

sage: R.<x,y,z> = QQ[]
sage: C1 = ChainComplex({1:matrix([[x]])})
sage: C2 = ChainComplex({1:matrix([[y]])})
sage: C3 = ChainComplex({1:matrix([[z]])})
sage: ascii_art(cartesian_product([C1, C2, C3]))

      [x 0 0]
      [0 y 0]
      [0 0 z]
0 <-- C_2 <----- C_1 <-- 0
sage: ascii_art(C1.cartesian_product([C2, C3], subdivide=True))

      [x|0|0]
      [-+--+]
      [0|y|0]
      [-+--+]
      [0|0|z]
0 <-- C_2 <----- C_1 <-- 0

```

```

sage: R.<x> = ZZ[]
sage: G = AdditiveAbelianGroup([0,7])
sage: d = {G(vector([1,1])):matrix([[x]])}
sage: C = ChainComplex(d, grading_group=G, degree=G(vector([2,1])))
sage: ascii_art(C.cartesian_product(C))

      [x 0]
      [0 x]
0 <-- C_(3, 2) <----- C_(1, 1) <-- 0

```

### **degree\_of\_differential()**

Return the degree of the differentials of the complex

OUTPUT:

An element of the grading group.

EXAMPLES:

```

sage: D = ChainComplex({0: matrix(ZZ, 2, 2, [1,0,0,2])})
sage: D.degree_of_differential()
1

```

### **differential(dim=None)**

The differentials which make up the chain complex.

INPUT:

- `dim` – element of the grading group (optional, default `None`); if this is `None`, return a dictionary of all of the differentials, or if this is a single element, return the differential starting in that dimension

OUTPUT:

Either a dictionary of all of the differentials or a single differential (i.e., a matrix).

EXAMPLES:

```
sage: D = ChainComplex({0: matrix(ZZ, 2, 2, [1, 0, 0, 2])})
sage: D.differential(0)
[1 0]
[0 2]
sage: D.differential(-1)
[]
sage: C = ChainComplex({0: identity_matrix(ZZ, 40)})
sage: diff = C.differential()
sage: diff[-1]
40 x 0 dense matrix over Integer Ring (use the '.str()' method to see the
↪entries)
sage: diff[0]
40 x 40 dense matrix over Integer Ring (use the '.str()' method to see the
↪entries)
sage: diff[1]
[]
```

**dual()**

The dual chain complex to `self`.

Since all modules in `self` are free of finite rank, the dual in dimension  $n$  is isomorphic to the original chain complex in dimension  $n$ , and the corresponding boundary matrix is the transpose of the matrix in the original complex. This converts a chain complex to a cochain complex and vice versa.

EXAMPLES:

```
sage: C = ChainComplex({2: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: C.degree_of_differential()
1
sage: C.differential(2)
[3 0 0]
[0 0 0]
sage: C.dual().degree_of_differential()
-1
sage: C.dual().differential(3)
[3 0]
[0 0]
[0 0]
```

**free\_module** (*degree=None*)

Return the free module at fixed degree, or their sum.

INPUT:

- `degree` – an element of the grading group or `None` (default).

OUTPUT:

The free module  $C_n$  at the given degree  $n$ . If the degree is not specified, the sum  $\bigoplus C_n$  is returned.

EXAMPLES:

```

sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0]), 1: matrix(ZZ,
↪ [[0, 1]])})
sage: C.free_module()
Ambient free module of rank 6 over the principal ideal domain Integer Ring
sage: C.free_module(0)
Ambient free module of rank 3 over the principal ideal domain Integer Ring
sage: C.free_module(1)
Ambient free module of rank 2 over the principal ideal domain Integer Ring
sage: C.free_module(2)
Ambient free module of rank 1 over the principal ideal domain Integer Ring

```

**free\_module\_rank(*degree*)**

Return the rank of the free module at the given degree.

INPUT:

- *degree* – an element of the grading group

OUTPUT:

Integer. The rank of the free module  $C_n$  at the given degree  $n$ .

EXAMPLES:

```

sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0]), 1: matrix(ZZ,
↪ [[0, 1]])})
sage: [C.free_module_rank(i) for i in range(-2, 5)]
[0, 0, 3, 2, 1, 0, 0]

```

**grading\_group()**

Return the grading group.

OUTPUT:

The discrete abelian group that indexes the individual modules of the complex. Usually  $\mathbf{Z}$ .

EXAMPLES:

```

sage: G = AdditiveAbelianGroup([0, 3])
sage: C = ChainComplex(grading_group=G, degree=G(vector([1, 2])))
sage: C.grading_group()
Additive abelian group isomorphic to  $\mathbf{Z} + \mathbf{Z}/3$ 
sage: C.degree_of_differential()
(1, 2)

```

**homology(*deg=None, base\_ring=None, generators=False, verbose=False, algorithm='pari'*)**

The homology of the chain complex.

INPUT:

- *deg* – an element of the grading group for the chain complex (default: `None`); the degree in which to compute homology – if this is `None`, return the homology in every degree in which the chain complex is possibly nonzero.
- *base\_ring* – a commutative ring (optional, default is the base ring for the chain complex); must be either the integers  $\mathbf{Z}$  or a field
- *generators* – boolean (optional, default `False`); if `True`, return generators for the homology groups along with the groups. See [trac ticket #6100](#)
- *verbose* – boolean (optional, default `False`); if `True`, print some messages as the homology is computed

- `algorithm` - string (optional, default `'pari'`); the options are:
  - `'auto'`
  - `'chomp'`
  - `'dhs'`
  - `'pari'`
  - `'no_chomp'`

see below for descriptions

#### OUTPUT:

If the degree is specified, the homology in degree `deg`. Otherwise, the homology in every dimension as a dictionary indexed by dimension.

#### ALGORITHM:

If `algorithm` is set to `'auto'`, then use CHomP if available. CHomP is available at the web page <http://chomp.rutgers.edu/>. It is also an optional package for Sage. If `algorithm` is `chomp`, always use `chomp`.

CHomP computes homology, not cohomology, and only works over the integers or finite prime fields. Therefore if any of these conditions fails, or if CHomP is not present, or if `algorithm` is set to `'no_chomp'`, go to plan B: if `self` has a `_homology` method – each simplicial complex has this, for example – then call that. Such a method implements specialized algorithms for the particular type of cell complex.

Otherwise, move on to plan C: compute the chain complex of `self` and compute its homology groups. To do this: over a field, just compute ranks and nullities, thus obtaining dimensions of the homology groups as vector spaces. Over the integers, compute Smith normal form of the boundary matrices defining the chain complex according to the value of `algorithm`. If `algorithm` is `'auto'` or `'no_chomp'`, then for each relatively small matrix, use the standard Sage method, which calls the Pari package. For any large matrix, reduce it using the Dumas, Heckenbach, Saunders, and Welker elimination algorithm [DHSW2003]: see `dhs_snf()` for details.

Finally, `algorithm` may also be `'pari'` or `'dhs'`, which forces the named algorithm to be used regardless of the size of the matrices and regardless of whether CHomP is available.

As of this writing, `'pari'` is the fastest standard option. The optional CHomP package may be better still.

**Warning:** This only works if the base ring is the integers or a field. Other values will return an error.

#### EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: C.homology()
{0: Z x Z, 1: Z x C3}
sage: C.homology(deg=1, base_ring = GF(3))
Vector space of dimension 2 over Finite Field of size 3
sage: D = ChainComplex({0: identity_matrix(ZZ, 4), 4: identity_matrix(ZZ, 30)}
↪)
sage: D.homology()
{0: 0, 1: 0, 4: 0, 5: 0}
```

Generators: generators are given as a list of cycles, each of which is an element in the appropriate free module, and hence is represented as a vector:

```
sage: C.homology(1, generators=True) # optional - CHomP
(Z x C3, [(0, 1), (1, 0)])
```

Tests for [trac ticket #6100](#), the Klein bottle with generators:

```
sage: d0 = matrix(ZZ, 0, 1)
sage: d1 = matrix(ZZ, 1, 3, [[0, 0, 0]])
sage: d2 = matrix(ZZ, 3, 2, [[1, 1], [1, -1], [-1, 1]])
sage: C_k = ChainComplex({0:d0, 1:d1, 2:d2}, degree=-1)
sage: C_k.homology(generators=True) # optional - CHomP
{0: (Z, [(1)]), 1: (Z x C2, [(0, 0, 1), (0, 1, -1)]), 2: 0}
```

From a torus using a field:

```
sage: T = simplicial_complexes.Torus()
sage: C_t = T.chain_complex()
sage: C_t.homology(base_ring=QQ, generators=True)
{0: [(Vector space of dimension 1 over Rational Field,
Chain(0:(0, 0, 0, 0, 0, 0, 0, 1)))],
1: [(Vector space of dimension 1 over Rational Field,
Chain(1:(0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, -1, 0, 0, 1))),
(Vector space of dimension 1 over Rational Field,
Chain(1:(0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -1, 0, 1, 0, -1,
↪0)))],
2: [(Vector space of dimension 1 over Rational Field,
Chain(2:(1, -1, 1, -1, 1, -1, -1, 1, -1, 1, -1, 1, 1, -1)))]}
```

### **nonzero\_degrees()**

Return the degrees in which the module is non-trivial.

See also [ordered\\_degrees\(\)](#).

OUTPUT:

The tuple containing all degrees  $n$  (grading group elements) such that the module  $C_n$  of the chain is non-trivial.

EXAMPLES:

```
sage: one = matrix(ZZ, [[1]])
sage: D = ChainComplex({0: one, 2: one, 6: one})
sage: ascii_art(D)
      [1]                      [1]      [0]      [1]
0 <-- C_7 <---- C_6 <-- 0 ... 0 <-- C_3 <---- C_2 <---- C_1 <---- C_0 <-- 0
sage: D.nonzero_degrees()
(0, 1, 2, 3, 6, 7)
```

### **ordered\_degrees(start=None, exclude\_first=False)**

Sort the degrees in the order determined by the differential

INPUT:

- `start` – (default: `None`) a degree (element of the grading group) or `None`
- `exclude_first` – boolean (optional; default: `False`); whether to exclude the lowest degree – this is a handy way to just get the degrees of the non-zero modules, as the domain of the first differential is zero.

OUTPUT:

If `start` has been specified, the longest tuple of degrees

- containing `start` (unless `start` would be the first and `exclude_first=True`),
- in ascending order relative to `degree_of_differential()`, and
- such that none of the corresponding differentials are  $0 \times 0$ .

If `start` has not been specified, a tuple of such tuples of degrees. One for each sequence of non-zero differentials. They are returned in sort order.

EXAMPLES:

```
sage: one = matrix(ZZ, [[1]])
sage: D = ChainComplex({0: one, 2: one, 6:one})
sage: ascii_art(D)
      [1]                [1]      [0]      [1]
0 <-- C_7 <---- C_6 <-- 0 ... 0 <-- C_3 <---- C_2 <---- C_1 <---- C_0 <-- 0
sage: D.ordered_degrees()
((-1, 0, 1, 2, 3), (5, 6, 7))
sage: D.ordered_degrees(exclude_first=True)
((0, 1, 2, 3), (6, 7))
sage: D.ordered_degrees(6)
(5, 6, 7)
sage: D.ordered_degrees(5, exclude_first=True)
(6, 7)
```

**random\_element()**

Return a random element.

EXAMPLES:

```
sage: D = ChainComplex({0: matrix(ZZ, 2, 2, [1,0,0,2])})
sage: D.random_element()      # random output
Chain with 1 nonzero terms over Integer Ring
```

**rank** (*degree*, *ring=None*)

Return the rank of a differential

INPUT:

- *degree* – an element  $\delta$  of the grading group. Which differential  $d_\delta$  we want to know the rank of
- *ring* – (optional) a commutative ring  $S$ ; if specified, the rank is computed after changing to this ring

OUTPUT:

The rank of the differential  $d_\delta \otimes_R S$ , where  $R$  is the base ring of the chain complex.

EXAMPLES:

```
sage: C = ChainComplex({0:matrix(ZZ, [[2]])})
sage: C.differential(0)
[2]
sage: C.rank(0)
1
sage: C.rank(0, ring=GF(2))
0
```

**shift** (*n=1*)

Shift this chain complex  $n$  times.

INPUT:

- *n* – an integer (optional, default 1)



The *shift* operation is also sometimes called *translation* or *suspension*.

To shift a chain complex by  $n$ , shift its entries up by  $n$  (if it is a chain complex) or down by  $n$  (if it is a cochain complex); that is, shifting by 1 always shifts in the opposite direction of the differential. In symbols, if  $C$  is a chain complex and  $C[n]$  is its  $n$ -th shift, then  $C[n]_j = C_{j-n}$ . The differential in the shift  $C[n]$  is obtained by multiplying each differential in  $C$  by  $(-1)^n$ .

Caveat: different sources use different conventions for shifting: what we call  $C[n]$  might be called  $C[-n]$  in some places. See for example. <https://ncatlab.org/nlab/show/suspension+of+a+chain+complex> (which uses  $C[n]$  as we do but acknowledges  $C[-n]$ ) or 1.2.8 in [Wei1994] (which uses  $C[-n]$ ).

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1).chain_complex()
sage: S1.shift(1).differential(2) == -S1.differential(1)
True
sage: S1.shift(2).differential(3) == S1.differential(1)
True
sage: S1.shift(3).homology(4)
Z
```

For cochain complexes, shifting goes in the other direction. Topologically, this makes sense if we grade the cochain complex for a space negatively:

```
sage: T = simplicial_complexes.Torus()
sage: co_T = T.chain_complex()._flip_()
sage: co_T.homology()
{-2: Z, -1: Z x Z, 0: Z}
sage: co_T.degree_of_differential()
1
sage: co_T.shift(2).homology()
{-4: Z, -3: Z x Z, -2: Z}
```

You can achieve the same result by tensoring (on the left, to get the signs right) with a rank one free module in degree  $-n * \deg$ , if  $\deg$  is the degree of the differential:

```
sage: C = ChainComplex({-2: matrix(ZZ, 0, 1)})
sage: C.tensor(co_T).homology()
{-4: Z, -3: Z x Z, -2: Z}
```

**tensor** (\*factors, \*\*kws)

Return the tensor product of `self` with `D`.

Let  $C$  and  $D$  be two chain complexes with differentials  $\partial_C$  and  $\partial_D$ , respectively, of the same degree (so they must also have the same grading group). The tensor product  $S = C \otimes D$  is a chain complex given by

$$S_i = \bigoplus_{a+b=i} C_a \otimes D_b$$

with differential

$$\partial(x \otimes y) = \partial_C x \otimes y + (-1)^{|a| \cdot |\partial_D|} x \otimes \partial_D y$$

for  $x \in C_a$  and  $y \in D_b$ , where  $|a|$  is the degree of  $a$  and  $|\partial_D|$  is the degree of  $\partial_D$ .

**Warning:** If the degree of the differential is even, then this may not result in a valid chain complex.

INPUT:

- `subdivide` – (default: `False`) whether to subdivide the the differential matrices

---

**Todo:** Make subdivision work correctly on multiple factors.

---

EXAMPLES:

```
sage: R.<x,y,z> = QQ[]
sage: C1 = ChainComplex({1:matrix([[x]]), degree_of_differential=-1})
sage: C2 = ChainComplex({1:matrix([[y]]), degree_of_differential=-1})
sage: C3 = ChainComplex({1:matrix([[z]]), degree_of_differential=-1})
sage: ascii_art(C1.tensor(C2))
      [ x]
      [y x]  [-y]
0 <-- C_0 <----- C_1 <----- C_2 <-- 0
sage: ascii_art(C1.tensor(C2).tensor(C3))
      [ y  x  0]      [ x]
      [-z  0  x]      [-y]
      [z y x]      [ 0 -z -y]      [ z]
0 <-- C_0 <----- C_1 <----- C_2 <----- C_3 <-- 0
```

```
sage: C = ChainComplex({2:matrix([[ -y],[x]]), 1:matrix([[x, y]]),
.....:                  degree_of_differential=-1); ascii_art(C)
      [-y]
      [x y]  [ x]
0 <-- C_0 <----- C_1 <----- C_2 <-- 0
sage: T = C.tensor(C)
sage: T.differential(1)
[x y x y]
sage: T.differential(2)
[-y x 0 y 0 0]
[ x 0 x 0 y 0]
[ 0 -x -y 0 0 -y]
[ 0 0 0 -x -y x]
sage: T.differential(3)
[ x y 0 0]
[ y 0 -y 0]
[-x 0 0 -y]
[ 0 y x 0]
[ 0 -x 0 x]
[ 0 0 x y]
sage: T.differential(4)
[-y]
[ x]
[-y]
[ x]
```

The degrees of the differentials must agree:

```
sage: C1p = ChainComplex({1:matrix([[x]]), degree_of_differential=1})
sage: C1.tensor(C1p)
Traceback (most recent call last):
...
ValueError: the degrees of the differentials must match
```

```
sage: R.<x,y> = ZZ[]
sage: G = AdditiveAbelianGroup([0,7])
```

(continues on next page)

(continued from previous page)

```

sage: d1 = {G(vector([1,1])):matrix([[x]])}
sage: C1 = ChainComplex(d1, grading_group=G, degree=G(vector([2,1])))
sage: d2 = {G(vector([3,0])):matrix([[y]])}
sage: C2 = ChainComplex(d2, grading_group=G, degree=G(vector([2,1])))
sage: ascii_art(C1.tensor(C2))

```

$$\begin{array}{ccccccc}
 & & & & [y] & & \\
 & & & & [x] & & \\
 & & [x & -y] & & & \\
 0 & \leftarrow & C_-(8, 3) & \leftarrow & \text{-----} & C_-(6, 2) & \leftarrow & \text{----} & C_-(4, 1) & \leftarrow & 0
 \end{array}$$

Check that [trac ticket #21760](#) is fixed:

```

sage: C = ChainComplex({0: matrix(ZZ, 0, 2)}, degree=-1)
sage: ascii_art(C)
0 <-- C_0 <-- 0
sage: T = C.tensor(C)
sage: ascii_art(T)
0 <-- C_0 <-- 0
sage: T.free_module_rank(0)
4

```

**torsion\_list** (*max\_prime*, *min\_prime*=2)

Look for torsion in this chain complex by computing its mod  $p$  homology for a range of primes  $p$ .

INPUT:

- *max\_prime* – prime number; search for torsion mod  $p$  for all  $p$  strictly less than this number
- *min\_prime* – prime (optional, default 2); search for torsion mod  $p$  for primes at least as big as this

Return a list of pairs  $(p, d)$  where  $p$  is a prime at which there is torsion and  $d$  is a list of dimensions in which this torsion occurs.

The base ring for the chain complex must be the integers; if not, an error is raised.

ALGORITHM:

let  $C$  denote the chain complex. Let  $P$  equal *max\_prime*. Compute the mod  $P$  homology of  $C$ , and use this as the base-line computation: the assumption is that this is isomorphic to the integral homology tensored with  $\mathbf{F}_P$ . Then compute the mod  $p$  homology for a range of primes  $p$ , and record whenever the answer differs from the base-line answer.

EXAMPLES:

```

sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: C.homology()
{0: Z x Z, 1: Z x C3}
sage: C.torsion_list(11)
[(3, [1])]
sage: C = ChainComplex([matrix(ZZ, 1, 1, [2]), matrix(ZZ, 1, 1), matrix(1, 1, [3])])
sage: C.homology(1)
C2
sage: C.homology(3)
C3
sage: C.torsion_list(5)
[(2, [1]), (3, [3])]

```

**class** sage.homology.chain\_complex.Chain\_class (*parent*, *vectors*, *check*=True)  
 Bases: sage.structure.element.ModuleElement

## A Chain in a Chain Complex

A chain is collection of module elements for each module  $C_n$  of the chain complex  $(C_n, d_n)$ . There is no restriction on how the differentials  $d_n$  act on the elements of the chain.

---

**Note:** You must use the chain complex to construct chains.

---

## EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])}, base_ring=GF(7))
sage: C.category()
Category of chain complexes over Finite Field of size 7
```

**is\_boundary()**

Return whether the chain is a boundary.

## OUTPUT:

Boolean. Whether the elements of the chain are in the image of the differentials.

## EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: c = C({0: vector([0, 1, 2]), 1: vector([3, 4])})
sage: c.is_boundary()
False
sage: z3 = C({1: (1, 0)})
sage: z3.is_cycle()
True
sage: (2*z3).is_boundary()
False
sage: (3*z3).is_boundary()
True
```

**is\_cycle()**

Return whether the chain is a cycle.

## OUTPUT:

Boolean. Whether the elements of the chain are in the kernel of the differentials.

## EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: c = C({0: vector([0, 1, 2]), 1: vector([3, 4])})
sage: c.is_cycle()
True
```

**vector (degree)**

Return the free module element in degree.

## EXAMPLES:

```
sage: C = ChainComplex({0: matrix(ZZ, 2, 3, [3, 0, 0, 0, 0, 0])})
sage: c = C({0: vector([1, 2, 3]), 1: vector([4, 5])})
sage: c.vector(0)
(1, 2, 3)
sage: c.vector(1)
(4, 5)
```

(continues on next page)

(continued from previous page)

```
sage: c.vector(2)
()
```



## CHAINS AND COCHAINS

This module implements formal linear combinations of cells of a given cell complex (*Chains*) and their dual (*Cochains*). It is closely related to the `sage.homology.chain_complex` module. The main differences are that chains and cochains here are of homogeneous dimension only, and that they reference their cell complex.

**class** `sage.homology.chains.CellComplexReference` (*cell\_complex, degree, cells=None*)

Bases: `object`

Auxiliary base class for chains and cochains

INPUT:

- `cell_complex` – The cell complex to reference
- `degree` – integer. The degree of the (co)chains
- `cells` – tuple of cells or `None`. Does not necessarily have to be the cells in the given degree, for computational purposes this could also be any collection that is in one-to-one correspondence with the cells. If `None`, the cells of the complex in the given degree are used.

EXAMPLES:

```
sage: X = simplicial_complexes.Simplex(2)
sage: from sage.homology.chains import CellComplexReference
sage: c = CellComplexReference(X, 1)
sage: c.cell_complex() is X
True
```

**cell\_complex()**

Return the underlying cell complex

OUTPUT:

A cell complex.

EXAMPLES:

```
sage: X = simplicial_complexes.Simplex(2)
sage: X.n_chains(1).cell_complex() is X
True
```

**degree()**

Return the dimension of the cells

OUTPUT:

Integer. The dimension of the cells.

EXAMPLES:

```
sage: X = simplicial_complexes.Simplex(2)
sage: X.n_chains(1).degree()
1
```

**class** sage.homology.chains.**Chains** (*cell\_complex, degree, cells=None, base\_ring=None*)  
 Bases: *sage.homology.chains.CellComplexReference, sage.combinat.free\_module.CombinatorialFreeModule*

Class for the free module of chains in a given degree.

INPUT:

- *n\_cells* – tuple of *n*-cells, which thus forms a basis for this module
- *base\_ring* – optional (default  $\mathbb{Z}$ )

One difference between chains and cochains is notation. In a simplicial complex, for example, a simplex  $(0, 1, 2)$  is written as “ $(0,1,2)$ ” in the group of chains but as “ $\chi(0,1,2)$ ” in the group of cochains.

Also, since the free modules of chains and cochains are dual, there is a pairing  $\langle c, z \rangle$ , sending a cochain *c* and a chain *z* to a scalar.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: C_2 = S2.n_chains(1)
sage: C_2_co = S2.n_chains(1, cochains=True)
sage: x = C_2.basis()[Simplex((0,2))]
sage: y = C_2.basis()[Simplex((1,3))]
sage: z = x+2*y
sage: a = C_2_co.basis()[Simplex((1,3))]
sage: b = C_2_co.basis()[Simplex((0,3))]
sage: c = 3*a-2*b
sage: z
(0, 2) + 2*(1, 3)
sage: c
-2*\chi_(0, 3) + 3*\chi_(1, 3)
sage: c.eval(z)
6
```

**class** **Element**

Bases: *sage.modules.with\_basis.indexed\_element.IndexedFreeModuleElement*

**boundary**()

Return the boundary of the chain

OUTPUT:

The boundary as a chain in one degree lower.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ)
sage: from sage.homology.cubical_complex import Cube
sage: chain = C1(Cube([[1, 1], [0, 1]])) - 2 * C1(Cube([[0, 1], [0, 0]]))
sage: chain
-2*[0,1] x [0,0] + [1,1] x [0,1]
sage: chain.boundary()
2*[0,0] x [0,0] - 3*[1,1] x [0,0] + [1,1] x [1,1]
```



**is\_boundary()**

Test whether the chain is a boundary

OUTPUT:

Boolean. Whether the chain is the *boundary()* of a chain in one degree higher.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ)
sage: from sage.homology.cubical_complex import Cube
sage: chain = C1(Cube([[1, 1], [0, 1]])) - C1(Cube([[0, 1], [0, 0]]))
sage: chain.is_boundary()
False
```

**is\_cycle()**

Test whether the chain is a cycle

OUTPUT:

Boolean. Whether the *boundary()* vanishes.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ)
sage: from sage.homology.cubical_complex import Cube
sage: chain = C1(Cube([[1, 1], [0, 1]])) - C1(Cube([[0, 1], [0, 0]]))
sage: chain.is_cycle()
False
```

**to\_complex()**

Return the corresponding chain complex element

OUTPUT:

An element of the chain complex, see *sage.homology.chain\_complex*.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ)
sage: from sage.homology.cubical_complex import Cube
sage: chain = C1(Cube([[1, 1], [0, 1]]))
sage: chain.to_complex()
Chain(1: (0, 0, 0, 1))
sage: ascii_art(_)
      d_0  [0]  d_1  [0]  d_2      d_3
0 <---- [0] <---- [0] <---- [0] <---- 0
      [0]      [0]
      [0]      [1]
```

**chain\_complex()**

Return the chain complex.

OUTPUT:

Chain complex, see *sage.homology.chain\_complex*.

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: CC = square.n_chains(2, QQ).chain_complex(); CC
Chain complex with at most 3 nonzero terms over Rational Field
sage: ascii_art(CC)
      [-1 -1  0  0]      [-1]
      [ 1  0 -1  0]      [ 1]
      [ 0  1  0 -1]      [-1]
      [ 0  0  1  1]      [ 1]
0 <-- C_0 <----- C_1 <----- C_2 <-- 0

```

**dual()**

Return the cochains.

OUTPUT:

The cochains of the same cells with the same base ring.

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: chains = square.n_chains(1, ZZ); chains
Free module generated by {[0,0] x [0,1], [0,1] x [0,0], [0,1] x [1,1], [1,1]
↪x [0,1]} over Integer Ring
sage: chains.dual()
Free module generated by {[0,0] x [0,1], [0,1] x [0,0], [0,1] x [1,1], [1,1]
↪x [0,1]} over Integer Ring
sage: type(chains)
<class 'sage.homology.chains.Chains_with_category'>
sage: type(chains.dual())
<class 'sage.homology.chains.Cochains_with_category'>

```

**class** sage.homology.chains.**Cochains**(cell\_complex, degree, cells=None, base\_ring=None)

Bases: `sage.homology.chains.CellComplexReference`, `sage.combinat.free_module.CombinatorialFreeModule`

Class for the free module of cochains in a given degree.

INPUT:

- `n_cells` – tuple of  $n$ -cells, which thus forms a basis for this module
- `base_ring` – optional (default  $\mathbf{Z}$ )

One difference between chains and cochains is notation. In a simplicial complex, for example, a simplex  $(0, 1, 2)$  is written as “ $(0,1,2)$ ” in the group of chains but as “ $\chi_i(0,1,2)$ ” in the group of cochains.

Also, since the free modules of chains and cochains are dual, there is a pairing  $\langle c, z \rangle$ , sending a cochain  $c$  and a chain  $z$  to a scalar.

EXAMPLES:

```

sage: S2 = simplicial_complexes.Sphere(2)
sage: C_2 = S2.n_chains(1)
sage: C_2_co = S2.n_chains(1, cochains=True)
sage: x = C_2.basis()[Simplex((0,2))]
sage: y = C_2.basis()[Simplex((1,3))]
sage: z = x+2*y
sage: a = C_2_co.basis()[Simplex((1,3))]
sage: b = C_2_co.basis()[Simplex((0,3))]
sage: c = 3*a-2*b

```

(continues on next page)

(continued from previous page)

```

sage: z
(0, 2) + 2*(1, 3)
sage: c
-2*\chi_(0, 3) + 3*\chi_(1, 3)
sage: c.eval(z)
6

```

**class Element**

Bases: sage.modules.with\_basis.indexed\_element.IndexedFreeModuleElement

**coboundary()**

Return the coboundary of this cochain

OUTPUT:

The coboundary as a cochain in one degree higher.

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ, cochains=True)
sage: from sage.homology.cubical_complex import Cube
sage: cochain = C1(Cube([[1, 1], [0, 1]])) - 2 * C1(Cube([[0, 1], [0, 1],
↪ 0]]))
sage: cochain
-2*\chi_[0,1] x [0,0] + \chi_[1,1] x [0,1]
sage: cochain.coboundary()
-\chi_[0,1] x [0,1]

```

**cup\_product (cochain)**

Return the cup product with another cochain.

INPUT:

- cochain – cochain over the same cell complex

EXAMPLES:

```

sage: T2 = simplicial_complexes.Torus()
sage: C1 = T2.n_chains(1, base_ring=ZZ, cochains=True)
sage: def l(i, j):
....:     return C1(Simplex([i, j]))
sage: l1 = l(1, 3) + l(1, 4) + l(1, 6) + l(2, 4) - l(4, 5) + l(5, 6)
sage: l2 = l(1, 6) - l(2, 3) - l(2, 5) + l(3, 6) - l(4, 5) + l(5, 6)

```

The two one-cocycles are cohomology generators:

```

sage: l1.is_cocycle(), l1.is_coboundary()
(True, False)
sage: l2.is_cocycle(), l2.is_coboundary()
(True, False)

```

Their cup product is a two-cocycle that is again non-trivial in cohomology:

```

sage: l12 = l1.cup_product(l2)
sage: l12
\chi_(1, 3, 6) - \chi_(2, 4, 5) - \chi_(4, 5, 6)
sage: l1.parent().degree(), l2.parent().degree(), l12.parent().degree()
(1, 1, 2)

```

(continues on next page)

(continued from previous page)

```
sage: l12.is_cocycle(), l12.is_coboundary()
(True, False)
```

**eval** (*other*)

Evaluate this cochain on the chain *other*.

INPUT:

- *other* – a chain for the same cell complex in the same dimension with the same base ring

OUTPUT: scalar

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: C_2 = S2.n_chains(1)
sage: C_2_co = S2.n_chains(1, cochains=True)
sage: x = C_2.basis()[Simplex((0,2))]
sage: y = C_2.basis()[Simplex((1,3))]
sage: z = x+2*y
sage: a = C_2_co.basis()[Simplex((1,3))]
sage: b = C_2_co.basis()[Simplex((0,3))]
sage: c = 3*a-2*b
sage: z
(0, 2) + 2*(1, 3)
sage: c
-2*\chi_(0, 3) + 3*\chi_(1, 3)
sage: c.eval(z)
6
```

**is\_coboundary** ()

Test whether the cochain is a coboundary

OUTPUT:

Boolean. Whether the cochain is the *coboundary* () of a cochain in one degree lower.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ, cochains=True)
sage: from sage.homology.cubical_complex import Cube
sage: cochain = C1(Cube([[1, 1], [0, 1]])) - C1(Cube([[0, 1], [0, 0]]))
sage: cochain.is_coboundary()
True
```

**is\_cocycle** ()

Test whether the cochain is a cocycle

OUTPUT:

Boolean. Whether the *coboundary* () vanishes.

EXAMPLES:

```
sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ, cochains=True)
sage: from sage.homology.cubical_complex import Cube
sage: cochain = C1(Cube([[1, 1], [0, 1]])) - C1(Cube([[0, 1], [0, 0]]))
sage: cochain.is_cocycle()
True
```

**to\_complex()**

Return the corresponding cochain complex element

OUTPUT:

An element of the cochain complex, see [sage.homology.chain\\_complex](#).

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: C1 = square.n_chains(1, QQ, cochains=True)
sage: from sage.homology.cubical_complex import Cube
sage: cochain = C1(Cube([[1, 1], [0, 1]]))
sage: cochain.to_complex()
Chain(1:(0, 0, 0, 1))
sage: ascii_art(_)
      d_2      d_1 [0] d_0 [0] d_-1
0 <---- [0] <---- [0] <---- [0] <----- 0
              [0]      [0]
              [1]      [0]

```

**cochain\_complex()**

Return the cochain complex.

OUTPUT:

Cochain complex, see [sage.homology.chain\\_complex](#).

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: C2 = square.n_chains(2, QQ, cochains=True)
sage: C2.cochain_complex()
Chain complex with at most 3 nonzero terms over Rational Field
sage: ascii_art(C2.cochain_complex())
              [-1  1  0  0]
              [-1  0  1  0]
              [ 0 -1  0  1]
          [-1  1 -1  1]      [ 0  0 -1  1]
0 <-- C_2 <----- C_1 <----- C_0 <-- 0

```

**dual()**

Return the chains

OUTPUT:

The chains of the same cells with the same base ring.

EXAMPLES:

```

sage: square = cubical_complexes.Cube(2)
sage: cochains = square.n_chains(1, ZZ, cochains=True); cochains
Free module generated by {[0,0] x [0,1], [0,1] x [0,0], [0,1] x [1,1], [1,1] x [0,1]} over Integer Ring
sage: cochains.dual()
Free module generated by {[0,0] x [0,1], [0,1] x [0,0], [0,1] x [1,1], [1,1] x [0,1]} over Integer Ring
sage: type(cochains)
<class 'sage.homology.chains.Cochains_with_category'>
sage: type(cochains.dual())
<class 'sage.homology.chains.Chains_with_category'>

```



## MORPHISMS OF CHAIN COMPLEXES

### AUTHORS:

- Benjamin Antieau <[d.ben.antieau@gmail.com](mailto:d.ben.antieau@gmail.com)> (2009.06)
- Travis Scrimshaw (2012-08-18): Made all simplicial complexes immutable to work with the homset cache.

This module implements morphisms of chain complexes. The input is a dictionary whose keys are in the grading group of the chain complex and whose values are matrix morphisms.

### EXAMPLES:

```
sage: S = simplicial_complexes.Sphere(1)
sage: S
Minimal triangulation of the 1-sphere
sage: C = S.chain_complex()
sage: C.differential()
{0: [], 1: [-1 -1  0]
 [ 1  0 -1]
 [ 0  1  1], 2: []}
sage: f = {0: zero_matrix(ZZ, 3, 3), 1: zero_matrix(ZZ, 3, 3)}
sage: G = Hom(C, C)
sage: x = G(f)
sage: x
Chain complex endomorphism of Chain complex with at most 2 nonzero terms over Integer_
↪Ring
sage: x._matrix_dictionary
{0: [0 0 0]
 [0 0 0]
 [0 0 0], 1: [0 0 0]
 [0 0 0]
 [0 0 0]}
```

```
class sage.homology.chain_complex_morphism.ChainComplexMorphism(matrices, C, D,
                                                                    check=True)
```

Bases: `sage.categories.morphism.Morphism`

An element of this class is a morphism of chain complexes.

**dual()**

The dual chain map to this one.

That is, the map from the dual of the codomain of this one to the dual of its domain, represented in each degree by the transpose of the corresponding matrix.

EXAMPLES:

```

sage: X = simplicial_complexes.Simplex(1)
sage: Y = simplicial_complexes.Simplex(0)
sage: g = Hom(X,Y)({0:0, 1:0})
sage: f = g.associated_chain_complex_morphism()
sage: f.in_degree(0)
[1 1]
sage: f.dual()
Chain complex morphism:
  From: Chain complex with at most 1 nonzero terms over Integer Ring
  To: Chain complex with at most 2 nonzero terms over Integer Ring
sage: f.dual().in_degree(0)
[1]
[1]
sage: ascii_art(f.domain())
      [-1]
      [ 1]
0 <-- C_0 <----- C_1 <-- 0
sage: ascii_art(f.dual().codomain())
      [-1  1]
0 <-- C_1 <----- C_0 <-- 0

```

**in\_degree(n)**

The matrix representing this morphism in degree n

INPUT:

- n – degree

EXAMPLES:

```

sage: C = ChainComplex({0: identity_matrix(ZZ, 1)})
sage: D = ChainComplex({0: zero_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: f = Hom(C,D)({0: identity_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: f.in_degree(0)
[1]

```

Note that if the matrix is not specified in the definition of the map, it is assumed to be zero:

```

sage: f.in_degree(2)
[]
sage: f.in_degree(2).nrows(), f.in_degree(2).ncols()
(1, 0)
sage: C.free_module(2)
Ambient free module of rank 0 over the principal ideal domain Integer Ring
sage: D.free_module(2)
Ambient free module of rank 1 over the principal ideal domain Integer Ring

```

**is\_identity()**

True if this is the identity map.

EXAMPLES:

```

sage: S = SimplicialComplex(is_mutable=False)
sage: H = Hom(S,S)
sage: i = H.identity()
sage: x = i.associated_chain_complex_morphism()
sage: x.is_identity()
True

```



**is\_injective()**

True if this map is injective.

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: H = Hom(S1, S1)
sage: flip = H({0:0, 1:2, 2:1})
sage: flip.associated_chain_complex_morphism().is_injective()
True

sage: pt = simplicial_complexes.Simplex(0)
sage: inclusion = Hom(pt, S1)({0:2})
sage: inclusion.associated_chain_complex_morphism().is_injective()
True
sage: inclusion.associated_chain_complex_morphism(cochain=True).is_injective()
False
```

**is\_surjective()**

True if this map is surjective.

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: H = Hom(S1, S1)
sage: flip = H({0:0, 1:2, 2:1})
sage: flip.associated_chain_complex_morphism().is_surjective()
True

sage: pt = simplicial_complexes.Simplex(0)
sage: inclusion = Hom(pt, S1)({0:2})
sage: inclusion.associated_chain_complex_morphism().is_surjective()
False
sage: inclusion.associated_chain_complex_morphism(cochain=True).is_
↪surjective()
True
```

**to\_matrix(deg=None)**

The matrix representing this chain map.

If the degree *deg* is specified, return the matrix in that degree; otherwise, return the (block) matrix for the whole chain map.

INPUT:

- *deg* – (optional, default None) the degree

EXAMPLES:

```
sage: C = ChainComplex({0: identity_matrix(ZZ, 1)})
sage: D = ChainComplex({0: zero_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: f = Hom(C,D)({0: identity_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: f.to_matrix(0)
[1]
sage: f.to_matrix()
[1|0|]
[-+ -+]
[0|0|]
[-+ -+]
[0|0|]
```

`sage.homology.chain_complex_morphism.is_ChainComplexMorphism(x)`

Returns True if and only if `x` is a chain complex morphism.

EXAMPLES:

```
sage: from sage.homology.chain_complex_morphism import is_ChainComplexMorphism
sage: S = simplicial_complexes.Sphere(14)
sage: H = Hom(S, S)
sage: i = H.identity() # long time (8s on sage.math, 2011)
sage: S = simplicial_complexes.Sphere(6)
sage: H = Hom(S, S)
sage: i = H.identity()
sage: x = i.associated_chain_complex_morphism()
sage: x # indirect doctest
Chain complex morphism:
  From: Chain complex with at most 7 nonzero terms over Integer Ring
  To: Chain complex with at most 7 nonzero terms over Integer Ring
sage: is_ChainComplexMorphism(x)
True
```

## CHAIN HOMOTOPIES AND CHAIN CONTRACTIONS

Chain homotopies are standard constructions in homological algebra: given chain complexes  $C$  and  $D$  and chain maps  $f, g : C \rightarrow D$ , say with differential of degree  $-1$ , a *chain homotopy*  $H$  between  $f$  and  $g$  is a collection of maps  $H_n : C_n \rightarrow D_{n+1}$  satisfying

$$\partial_D H + H \partial_C = f - g.$$

The presence of a chain homotopy defines an equivalence relation (*chain homotopic*) on chain maps. If  $f$  and  $g$  are chain homotopic, then one can show that  $f$  and  $g$  induce the same map on homology.

Chain contractions are not as well known. The papers [MAR2009], [RMA2009], and [PR2015] provide some references. Given two chain complexes  $C$  and  $D$ , a *chain contraction* is a chain homotopy  $H : C \rightarrow C$  for which there are chain maps  $\pi : C \rightarrow D$  (“projection”) and  $\iota : D \rightarrow C$  (“inclusion”) such that

- $H$  is a chain homotopy between  $1_C$  and  $\iota\pi$ ,
- $\pi\iota = 1_D$ ,
- $\pi H = 0$ ,
- $H\iota = 0$ ,
- $HH = 0$ .

Such a chain homotopy provides a strong relation between the chain complexes  $C$  and  $D$ ; for example, their homology groups are isomorphic.

**class** `sage.homology.chain_homotopy.ChainContraction` (*matrices, pi, iota*)  
 Bases: `sage.homology.chain_homotopy.ChainHomotopy`

A chain contraction.

An algebraic gradient vector field  $H : C \rightarrow C$  (that is a chain homotopy satisfying  $HH = 0$ ) for which there are chain maps  $\pi : C \rightarrow D$  (“projection”) and  $\iota : D \rightarrow C$  (“inclusion”) such that

- $H$  is a chain homotopy between  $1_C$  and  $\iota\pi$ ,
- $\pi\iota = 1_D$ ,
- $\pi H = 0$ ,
- $H\iota = 0$ .

$H$  is defined by a dictionary `matrices` of matrices.

INPUT:

- `matrices` – dictionary of matrices, keyed by dimension
- `pi` – a chain map  $C \rightarrow D$
- `iota` – a chain map  $D \rightarrow C$

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainContraction
sage: C = ChainComplex({0: zero_matrix(ZZ, 1), 1: identity_matrix(ZZ, 1)})
sage: D = ChainComplex({0: matrix(ZZ, 0, 1)})
```

The chain complex  $C$  is chain homotopy equivalent to  $D$ , which is just a copy of  $\mathbb{Z}$  in degree 0, and we construct a chain contraction:

```
sage: pi = Hom(C,D)({0: identity_matrix(ZZ, 1)})
sage: iota = Hom(D,C)({0: identity_matrix(ZZ, 1)})
sage: H = ChainContraction({0: zero_matrix(ZZ, 0, 1), 1: zero_matrix(ZZ, 1), 2:
↪ identity_matrix(ZZ, 1)}, pi, iota)
```

**dual()**

The chain contraction dual to this one.

This is useful when switching from homology to cohomology.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: phi, M = S2.algebraic_topological_model(QQ)
sage: phi.iota()
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Rational Field
  To: Chain complex with at most 3 nonzero terms over Rational Field
```

Lifting the degree zero homology class gives a single vertex, but the degree zero cohomology class needs to be detected on every vertex, and vice versa for degree 2:

```
sage: phi.iota().in_degree(0)
[0]
[0]
[0]
[1]
sage: phi.dual().iota().in_degree(0)
[1]
[1]
[1]
[1]
sage: phi.iota().in_degree(2)
[-1]
[ 1]
[-1]
[ 1]
sage: phi.dual().iota().in_degree(2)
[0]
[0]
[0]
[1]
```

**iota()**

The chain map  $\iota$  associated to this chain contraction.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: phi, M = S2.algebraic_topological_model(QQ)
```

(continues on next page)

(continued from previous page)

```
sage: phi.iota()
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Rational Field
  To: Chain complex with at most 3 nonzero terms over Rational Field
```

Lifting the degree zero homology class gives a single vertex:

```
sage: phi.iota().in_degree(0)
[0]
[0]
[0]
[1]
```

Lifting the degree two homology class gives the signed sum of all of the 2-simplices:

```
sage: phi.iota().in_degree(2)
[-1]
[ 1]
[-1]
[ 1]
```

**pi()**

The chain map  $\pi$  associated to this chain contraction.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: phi, M = S2.algebraic_topological_model(QQ)
sage: phi.pi()
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Rational Field
  To: Chain complex with at most 3 nonzero terms over Rational Field
sage: phi.pi().in_degree(0) # Every vertex represents a homology class.
[1 1 1 1]
sage: phi.pi().in_degree(1) # No homology in degree 1.
[]
```

The degree 2 homology generator is detected on a single simplex:

```
sage: phi.pi().in_degree(2)
[0 0 0 1]
```

**class** sage.homology.chain\_homotopy.ChainHomotopy(matrices, f, g=None)

Bases: sage.categories.morphism.Morphism

A chain homotopy.

A chain homotopy  $H$  between chain maps  $f, g : C \rightarrow D$  is a sequence of maps  $H_n : C_n \rightarrow D_{n+1}$  (if the chain complexes are graded homologically) satisfying

$$\partial_D H + H \partial_C = f - g.$$

INPUT:

- matrices – dictionary of matrices, keyed by dimension
- f – chain map  $C \rightarrow D$
- g (optional) – chain map  $C \rightarrow D$

The dictionary `matrices` defines  $H$  by specifying the matrix defining it in each degree: the entry  $m$  corresponding to key  $i$  gives the linear transformation  $C_i \rightarrow D_{i+1}$ .

If  $f$  is specified but not  $g$ , then  $g$  can be recovered from the defining formula. That is, if  $g$  is not specified, then it is defined to be  $f - \partial_D H - H \partial_C$ .

Note that the degree of the differential on the chain complex  $C$  must agree with that for  $D$ , and those degrees determine the “degree” of the chain homotopy map: if the degree of the differential is  $d$ , then the chain homotopy consists of a sequence of maps  $C_n \rightarrow C_{n-d}$ . The keys in the dictionary `matrices` specify the starting degrees.

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainHomotopy
sage: C = ChainComplex({0: identity_matrix(ZZ, 1)})
sage: D = ChainComplex({0: zero_matrix(ZZ, 1)})
sage: f = Hom(C,D)({0: identity_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: g = Hom(C,D)({0: zero_matrix(ZZ, 1), 1: zero_matrix(ZZ, 1)})
sage: H = ChainHomotopy({0: zero_matrix(ZZ, 0, 1), 1: identity_matrix(ZZ, 1)}, f,
↪g)
```

Note that the maps  $f$  and  $g$  are stored in the attributes `H._f` and `H._g`:

```
sage: H._f
Chain complex morphism:
  From: Chain complex with at most 2 nonzero terms over Integer Ring
  To: Chain complex with at most 2 nonzero terms over Integer Ring
sage: H._f.in_degree(0)
[1]
sage: H._g.in_degree(0)
[0]
```

A non-example:

```
sage: H = ChainHomotopy({0: zero_matrix(ZZ, 0, 1), 1: zero_matrix(ZZ, 1)}, f, g)
Traceback (most recent call last):
...
ValueError: the data do not define a valid chain homotopy
```

**dual()**

Dual chain homotopy to this one.

That is, if this one is a chain homotopy between chain maps  $f, g : C \rightarrow D$ , then its dual is a chain homotopy between the dual of  $f$  and the dual of  $g$ , from  $D^*$  to  $C^*$ . It is represented in each degree by the transpose of the corresponding matrix.

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainHomotopy
sage: C = ChainComplex({1: matrix(ZZ, 0, 2)}) # one nonzero term in degree 1
sage: D = ChainComplex({0: matrix(ZZ, 0, 1)}) # one nonzero term in degree 0
sage: f = Hom(C, D)({})
sage: H = ChainHomotopy({1: matrix(ZZ, 1, 2, (3,1))}, f, f)
sage: H.in_degree(1)
[3 1]
sage: H.dual().in_degree(0)
[3]
[1]
```

**in\_degree(n)**

The matrix representing this chain homotopy in degree  $n$ .

INPUT:

- $n$  – degree

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainHomotopy
sage: C = ChainComplex({1: matrix(ZZ, 0, 2)}) # one nonzero term in degree 1
sage: D = ChainComplex({0: matrix(ZZ, 0, 1)}) # one nonzero term in degree 0
sage: f = Hom(C, D)({})
sage: H = ChainHomotopy({1: matrix(ZZ, 1, 2, (3,1))}, f, f)
sage: H.in_degree(1)
[3 1]
```

This returns an appropriately sized zero matrix if the chain homotopy is not defined in degree  $n$ :

```
sage: H.in_degree(-3)
[]
```

### **is\_algebraic\_gradient\_vector\_field()**

An algebraic gradient vector field is a linear map  $H : C \rightarrow C$  such that  $HH = 0$ .

(Some authors also require that  $H\partial H = H$ , whereas some make this part of the definition of “homology gradient vector field. We have made the second choice.) See Molina-Abril and Réal [MAR2009] and Réal and Molina-Abril [RMA2009] for this and related terminology.

See also [`is\_homology\_gradient\_vector\_field\(\)`](#).

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainHomotopy
sage: C = ChainComplex({0: zero_matrix(ZZ, 1), 1: identity_matrix(ZZ, 1)})
```

The chain complex  $C$  is chain homotopy equivalent to a copy of  $\mathbb{Z}$  in degree 0. Two chain maps  $C \rightarrow C$  will be chain homotopic as long as they agree in degree 0.

```
sage: f = Hom(C,C)({0: identity_matrix(ZZ, 1), 1: matrix(ZZ, 1, 1, [3]), 2:
↪matrix(ZZ, 1, 1, [3])})
sage: g = Hom(C,C)({0: identity_matrix(ZZ, 1), 1: matrix(ZZ, 1, 1, [2]), 2:
↪matrix(ZZ, 1, 1, [2])})
sage: H = ChainHomotopy({0: zero_matrix(ZZ, 0, 1), 1: zero_matrix(ZZ, 1), 2:
↪identity_matrix(ZZ, 1)}, f, g)
sage: H.is_algebraic_gradient_vector_field()
True
```

A chain homotopy which is not an algebraic gradient vector field:

```
sage: H = ChainHomotopy({0: zero_matrix(ZZ, 0, 1), 1: identity_matrix(ZZ, 1),
↪2: identity_matrix(ZZ, 1)}, f, g)
sage: H.is_algebraic_gradient_vector_field()
False
```

### **is\_homology\_gradient\_vector\_field()**

A homology gradient vector field is an algebraic gradient vector field  $H : C \rightarrow C$  (i.e., a chain homotopy satisfying  $HH = 0$ ) such that  $\partial H \partial = \partial$  and  $H \partial H = H$ .

See Molina-Abril and Réal [MAR2009] and Réal and Molina-Abril [RMA2009] for this and related terminology.

See also [`is\_algebraic\_gradient\_vector\_field\(\)`](#).

EXAMPLES:

```
sage: from sage.homology.chain_homotopy import ChainHomotopy
sage: C = ChainComplex({0: zero_matrix(ZZ, 1), 1: identity_matrix(ZZ, 1)})

sage: f = Hom(C,C)({0: identity_matrix(ZZ, 1), 1: matrix(ZZ, 1, 1, [3]), 2:
↳matrix(ZZ, 1, 1, [3])})
sage: g = Hom(C,C)({0: identity_matrix(ZZ, 1), 1: matrix(ZZ, 1, 1, [2]), 2:
↳matrix(ZZ, 1, 1, [2])})
sage: H = ChainHomotopy({0: zero_matrix(ZZ, 0, 1), 1: zero_matrix(ZZ, 1), 2:
↳identity_matrix(ZZ, 1)}, f, g)
sage: H.is_homology_gradient_vector_field()
True
```



## HOMSPACES BETWEEN CHAIN COMPLEXES

Note that some significant functionality is lacking. Namely, the homspaces are not actually modules over the base ring. It will be necessary to enrich some of the structure of chain complexes for this to be naturally available. On other hand, there are various overloaded operators. `__mul__` acts as composition. One can `__add__`, and one can `__mul__` with a ring element on the right.

EXAMPLES:

```
sage: S = simplicial_complexes.Sphere(2)
sage: T = simplicial_complexes.Torus()
sage: C = S.chain_complex(augmented=True,cochain=True)
sage: D = T.chain_complex(augmented=True,cochain=True)
sage: G = Hom(C,D)
sage: G
Set of Morphisms from Chain complex with at most 4 nonzero terms over Integer Ring to
↳Chain complex with at most 4 nonzero terms over Integer Ring in Category of chain
↳complexes over Integer Ring

sage: S = simplicial_complexes.ChessboardComplex(3,3)
sage: H = Hom(S,S)
sage: i = H.identity()
sage: x = i.associated_chain_complex_morphism(augmented=True)
sage: x
Chain complex morphism:
  From: Chain complex with at most 4 nonzero terms over Integer Ring
  To: Chain complex with at most 4 nonzero terms over Integer Ring
sage: x._matrix_dictionary
{-1: [1], 0: [1 0 0 0 0 0 0 0 0]
 [0 1 0 0 0 0 0 0 0]
 [0 0 1 0 0 0 0 0 0]
 [0 0 0 1 0 0 0 0 0]
 [0 0 0 0 1 0 0 0 0]
 [0 0 0 0 0 1 0 0 0]
 [0 0 0 0 0 0 1 0 0]
 [0 0 0 0 0 0 0 1 0]
 [0 0 0 0 0 0 0 0 1], 1: [1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
 [0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
 [0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
 [0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0]
 [0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0]
 [0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0 0]
 [0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0]
 [0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0]
 [0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0]
 [0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0]
 [0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0]
```

(continues on next page)

(continued from previous page)

```

[0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1], 2: [1 0 0 0 0 0]
[0 1 0 0 0 0]
[0 0 1 0 0 0]
[0 0 0 1 0 0]
[0 0 0 0 1 0]
[0 0 0 0 0 1]}

sage: S = simplicial_complexes.Sphere(2)
sage: A = Hom(S,S)
sage: i = A.identity()
sage: x = i.associated_chain_complex_morphism()
sage: x
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Integer Ring
  To: Chain complex with at most 3 nonzero terms over Integer Ring
sage: y = x*4
sage: z = y*y
sage: (y+z)
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Integer Ring
  To: Chain complex with at most 3 nonzero terms over Integer Ring
sage: f = x._matrix_dictionary
sage: C = S.chain_complex()
sage: G = Hom(C,C)
sage: w = G(f)
sage: w == x
True

```

```

class sage.homology.chain_complex_homspace.ChainComplexHomspace(X, Y, category=None,
                                                                    base=None,
                                                                    check=True)

```

Bases: `sage.categories.homset.Homset`

Class of homspaces of chain complex morphisms.

EXAMPLES:

```

sage: T = SimplicialComplex([[1,2,3,4],[7,8,9]])
sage: C = T.chain_complex(augmented=True, cochain=True)
sage: G = Hom(C,C)
sage: G
Set of Morphisms from Chain complex with at most 5 nonzero terms over Integer_
↳ Ring to Chain complex with at most 5 nonzero terms over Integer Ring in_
↳ Category of chain complexes over Integer Ring

```

```
sage.homology.chain_complex_homspace.is_ChainComplexHomspace(x)
```

Returns True if and only if `x` is a morphism of chain complexes.

EXAMPLES:

```
sage: from sage.homology.chain_complex_homspace import is_ChainComplexHomspace
sage: T = SimplicialComplex([[1,2,3,4],[7,8,9]])
sage: C = T.chain_complex(augmented=True, cochain=True)
sage: G = Hom(C,C)
sage: is_ChainComplexHomspace(G)
True
```



## FINITE SIMPLICIAL COMPLEXES

### AUTHORS:

- John H. Palmieri (2009-04)
- D. Benjamin Antieau (2009-06): added `is_connected`, `generated_subcomplex`, `remove_facet`, and `is_flag_complex` methods; cached the output of the `graph()` method.
- Travis Scrimshaw (2012-08-17): Made `SimplicialComplex` have an immutable option, and added `__hash__()` function which checks to make sure it is immutable. Made `SimplicialComplex.remove_face()` into a mutator. Deprecated the `vertex_set` parameter.
- Christian Stump (2011-06): implementation of `is_cohen_macaulay`
- Travis Scrimshaw (2013-02-16): Allowed `SimplicialComplex` to make mutable copies.
- Simon King (2014-05-02): Let simplicial complexes be objects of the category of simplicial complexes.
- Jeremy Martin (2016-06-02): added `cone_vertices`, `decone`, `is_balanced`, `is_partitionable`, `intersection` methods

This module implements the basic structure of finite simplicial complexes. Given a set  $V$  of “vertices”, a simplicial complex on  $V$  is a collection  $K$  of subsets of  $V$  satisfying the condition that if  $S$  is one of the subsets in  $K$ , then so is every subset of  $S$ . The subsets  $S$  are called the ‘simplices’ of  $K$ .

---

**Note:** In Sage, the elements of the vertex set are determined automatically:  $V$  is defined to be the union of the sets in  $K$ . So in Sage’s implementation of simplicial complexes, every vertex is included in some face.

---

A simplicial complex  $K$  can be viewed as a purely combinatorial object, as described above, but it also gives rise to a topological space  $|K|$  (its *geometric realization*) as follows: first, the points of  $V$  should be in general position in euclidean space. Next, if  $\{v\}$  is in  $K$ , then the vertex  $v$  is in  $|K|$ . If  $\{v, w\}$  is in  $K$ , then the line segment from  $v$  to  $w$  is in  $|K|$ . If  $\{u, v, w\}$  is in  $K$ , then the triangle with vertices  $u, v$ , and  $w$  is in  $|K|$ . In general,  $|K|$  is the union of the convex hulls of simplices of  $K$ . Frequently, one abuses notation and uses  $K$  to denote both the simplicial complex and the associated topological space.



For any simplicial complex  $K$  and any commutative ring  $R$  there is an associated chain complex, with differential of degree  $-1$ . The  $n^{th}$  term is the free  $R$ -module with basis given by the  $n$ -simplices of  $K$ . The differential is determined by its value on any simplex: on the  $n$ -simplex with vertices  $(v_0, v_1, \dots, v_n)$ , the differential is the alternating sum with  $i^{th}$  summand  $(-1)^i$  multiplied by the  $(n-1)$ -simplex obtained by omitting vertex  $v_i$ .

In the implementation here, the vertex set must be finite. To define a simplicial complex, specify its *facets*: the maximal subsets (with respect to inclusion) of the vertex set belonging to  $K$ . Each facet can be specified as a list, a tuple, or a set.

**Note:** This class derives from *GenericCellComplex*, and so inherits its methods. Some of those methods are not listed here; see the *Generic Cell Complex* page instead.

#### EXAMPLES:

```
sage: SimplicialComplex([[1], [3, 7]])
Simplicial complex with vertex set (1, 3, 7) and facets {(1,), (3, 7)}
sage: SimplicialComplex() # the empty simplicial complex
Simplicial complex with vertex set () and facets {}
sage: X = SimplicialComplex([[0,1], [1,2], [2,3], [3,0]])
sage: X
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (0, 3), (1, 2),
↪ (2, 3)}
```

Sage can perform a number of operations on simplicial complexes, such as the join and the product, and it can also compute homology:

```
sage: S = SimplicialComplex([[0,1], [1,2], [0,2]]) # circle
sage: T = S.product(S) # torus
sage: T
Simplicial complex with 9 vertices and 18 facets
sage: T.homology() # this computes reduced homology
{0: 0, 1: Z x Z, 2: Z}
sage: T.euler_characteristic()
0
```

Sage knows about some basic combinatorial data associated to a simplicial complex:

```
sage: X = SimplicialComplex([[0,1], [1,2], [2,3], [0,3]])
sage: X.f_vector()
[1, 4, 4]
sage: X.face_poset()
Finite poset containing 8 elements
sage: x0, x1, x2, x3 = X.stanley_reisner_ring().gens()
sage: x0*x2 == x1*x3 == 0
True
sage: X.is_pure()
True
```

Mutability (see [trac ticket #12587](#)):

```
sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.add_face([1,3])
sage: S.remove_face([1,3]); S
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(3,), (1, 4), (2, 4)}
sage: hash(S)
Traceback (most recent call last):
...
ValueError: This simplicial complex must be immutable. Call set_immutable().
sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.set_immutable()
sage: S.add_face([1,3])
```

(continues on next page)

(continued from previous page)

```

Traceback (most recent call last):
...
ValueError: This simplicial complex is not mutable
sage: S.remove_face([1,3])
Traceback (most recent call last):
...
ValueError: This simplicial complex is not mutable
sage: hash(S) == hash(S)
True

sage: S2 = SimplicialComplex([[1,4], [2,4]], is_mutable=False)
sage: hash(S2) == hash(S)
True

```

We can also make mutable copies of an immutable simplicial complex (see [trac ticket #14142](#)):

```

sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.set_immutable()
sage: T = copy(S)
sage: T.is_mutable()
True
sage: S == T
True

```

**class** sage.homology.simplicial\_complex.**Simplex**(*X*)

Bases: `sage.structure.sage_object.SageObject`

Define a simplex.

Topologically, a simplex is the convex hull of a collection of vertices in general position. Combinatorially, it is defined just by specifying a set of vertices. It is represented in Sage by the tuple of the vertices.

**Parameters** *X* (*integer*, *list*, *other iterable*) – set of vertices

**Returns** simplex with those vertices

*X* may be a non-negative integer  $n$ , in which case the simplicial complex will have  $n + 1$  vertices  $(0, 1, \dots, n)$ , or it may be anything which may be converted to a tuple, in which case the vertices will be that tuple. In the second case, each vertex must be hashable, so it should be a number, a string, or a tuple, for instance, but not a list.

**Warning:** The vertices should be distinct, and no error checking is done to make sure this is the case.

EXAMPLES:

```

sage: Simplex(4)
(0, 1, 2, 3, 4)
sage: Simplex([3, 4, 1])
(3, 4, 1)
sage: X = Simplex((3, 'a', 'vertex')); X
(3, 'a', 'vertex')
sage: X == loads(dumps(X))
True

```

Vertices may be tuples but not lists:

```

sage: Simplex([(1,2), (3,4)])
((1, 2), (3, 4))
sage: Simplex([[1,2], [3,4]])
Traceback (most recent call last):
...
TypeError: unhashable type: 'list'

```

**alexander\_whitney**(*dim*)

Subdivide this simplex into a pair of simplices.

If this simplex has vertices  $v_0, v_1, \dots, v_n$ , then subdivide it into simplices  $(v_0, v_1, \dots, v_{dim})$  and  $(v_{dim}, v_{dim+1}, \dots, v_n)$ .

INPUT:

- *dim* – integer between 0 and one more than the dimension of this simplex

OUTPUT:

- a list containing just the triple  $(1, \text{left}, \text{right})$ , where *left* and *right* are the two simplices described above.

This method allows one to construct a coproduct from the  $p + q$ -chains to the tensor product of the  $p$ -chains and the  $q$ -chains. The number 1 (a Sage integer) is the coefficient of *left* tensor *right* in this coproduct. (The corresponding formula is more complicated for the cubes that make up a cubical complex, and the output format is intended to be consistent for both cubes and simplices.)

Calling this method `alexander_whitney` is an abuse of notation, since the actual Alexander-Whitney map goes from  $C(X \times Y) \rightarrow C(X) \otimes C(Y)$ , where  $C(-)$  denotes the chain complex of singular chains, but this subdivision of simplices is at the heart of it.

EXAMPLES:

```

sage: s = Simplex((0,1,3,4))
sage: s.alexander_whitney(0)
[(1, (0,), (0, 1, 3, 4))]
sage: s.alexander_whitney(2)
[(1, (0, 1, 3), (3, 4))]

```

**dimension**()

The dimension of this simplex.

The dimension of a simplex is the number of vertices minus 1.

EXAMPLES:

```

sage: Simplex(5).dimension() == 5
True
sage: Simplex(5).face(1).dimension()
4

```

**face**(*n*)

The  $n$ -th face of this simplex.

**Parameters** *n* (*integer*) – an integer between 0 and the dimension of this simplex

**Returns** the simplex obtained by removing the  $n$ -th vertex from this simplex

EXAMPLES:



```
sage: S = Simplex(4)
sage: S.face(0)
(1, 2, 3, 4)
sage: S.face(3)
(0, 1, 2, 4)
```

**faces()**

The list of faces (of codimension 1) of this simplex.

EXAMPLES:

```
sage: S = Simplex(4)
sage: S.faces()
[(1, 2, 3, 4), (0, 2, 3, 4), (0, 1, 3, 4), (0, 1, 2, 4), (0, 1, 2, 3)]
sage: len(Simplex(10).faces())
11
```

**is\_empty()**

Return True iff this simplex is the empty simplex.

EXAMPLES:

```
sage: [Simplex(n).is_empty() for n in range(-1,4)]
[True, False, False, False, False]
```

**is\_face(other)**

Return True iff this simplex is a face of other.

EXAMPLES:

```
sage: Simplex(3).is_face(Simplex(5))
True
sage: Simplex(5).is_face(Simplex(2))
False
sage: Simplex(['a', 'b', 'c']).is_face(Simplex(8))
False
```

**join(right, rename\_vertices=True)**

The join of this simplex with another one.

The join of two simplices  $[v_0, \dots, v_k]$  and  $[w_0, \dots, w_n]$  is the simplex  $[v_0, \dots, v_k, w_0, \dots, w_n]$ .

**Parameters**

- **right** – the other simplex (the right-hand factor)
- **rename\_vertices** (boolean; optional, default True) – If this is True, the vertices in the join will be renamed by this formula: vertex “v” in the left-hand factor → vertex “Lv” in the join, vertex “w” in the right-hand factor → vertex “Rw” in the join. If this is false, this tries to construct the join without renaming the vertices; this may cause problems if the two factors have any vertices with names in common.

EXAMPLES:

```
sage: Simplex(2).join(Simplex(3))
('L0', 'L1', 'L2', 'R0', 'R1', 'R2', 'R3')
sage: Simplex(['a', 'b']).join(Simplex(['x', 'y', 'z']))
('La', 'Lb', 'Rx', 'Ry', 'Rz')
```

(continues on next page)

(continued from previous page)

```
sage: Simplex(['a', 'b']).join(Simplex(['x', 'y', 'z']), rename_
↳vertices=False)
('a', 'b', 'x', 'y', 'z')
```

**product** (*other*, *rename\_vertices=True*)

The product of this simplex with another one, as a list of simplices.

#### Parameters

- **other** – the other simplex
- **rename\_vertices** (boolean; optional, default `True`) – If this is `False`, then the vertices in the product are the set of ordered pairs  $(v, w)$  where  $v$  is a vertex in the left-hand factor (`self`) and  $w$  is a vertex in the right-hand factor (`other`). If this is `True`, then the vertices are renamed as “LvRw” (e.g., the vertex (1,2) would become “L1R2”). This is useful if you want to define the Stanley-Reisner ring of the complex: vertex names like (0,1) are not suitable for that, while vertex names like “L0R1” are.

Algorithm: see Hatcher, p. 277-278 [Hat2002] (who in turn refers to Eilenberg-Steenrod, p. 68): given  $S = \text{Simplex}(m)$  and  $T = \text{Simplex}(n)$ , then  $S \times T$  can be triangulated as follows: for each path  $f$  from  $(0,0)$  to  $(m,n)$  along the integer grid in the plane, going up or right at each lattice point, associate an  $(m+n)$ -simplex with vertices  $v_0, v_1, \dots$ , where  $v_k$  is the  $k^{\text{th}}$  vertex in the path  $f$ .

Note that there are  $m+n$  choose  $n$  such paths. Note also that each vertex in the product is a pair of vertices  $(v, w)$  where  $v$  is a vertex in the left-hand factor and  $w$  is a vertex in the right-hand factor.

---

**Note:** This produces a list of simplices – not a *Simplex*, not a *SimplicialComplex*.

---

#### EXAMPLES:

```
sage: len(Simplex(2).product(Simplex(2)))
6
sage: Simplex(1).product(Simplex(1))
[('L0R0', 'L0R1', 'L1R1'), ('L0R0', 'L1R0', 'L1R1')]
sage: Simplex(1).product(Simplex(1), rename_vertices=False)
[((0, 0), (0, 1), (1, 1)), ((0, 0), (1, 0), (1, 1))]
```

**set** ()

The frozenset attached to this simplex.

#### EXAMPLES:

```
sage: Simplex(3).set()
frozenset({0, 1, 2, 3})
```

**tuple** ()

The tuple attached to this simplex.

#### EXAMPLES:

```
sage: Simplex(3).tuple()
(0, 1, 2, 3)
```

Although simplices are printed as if they were tuples, they are not the same type:

```
sage: type(Simplex(3).tuple())
<... 'tuple'>
sage: type(Simplex(3))
<class 'sage.homology.simplicial_complex.Simplex'>
```

```
class sage.homology.simplicial_complex.SimplicialComplex(maximal_faces=None,
                                                         from_characteristic_function=None,
                                                         maximality_check=True,
                                                         sort_facets=None,
                                                         name_check=False,
                                                         is_mutable=True,
                                                         is_immutable=False,
                                                         category=None)

Bases: sage.structure.parent.Parent, sage.homology.cell_complex.GenericCellComplex
```

Define a simplicial complex.

#### Parameters

- **maximal\_faces** – set of maximal faces
- **from\_characteristic\_function** – see below
- **maximality\_check** (boolean; optional, default `True`) – see below
- **sort\_facets** (*dict*) – see below
- **name\_check** (boolean; optional, default `False`) – see below
- **is\_mutable** (boolean; optional, default `True`) – Set to `False` to make this immutable
- **category** (*category; optional, default finite simplicial complexes*) – the category of the simplicial complex

#### Returns

a simplicial complex

`maximal_faces` should be a list or tuple or set (indeed, anything which may be converted to a set) whose elements are lists (or tuples, etc.) of vertices. Maximal faces are also known as ‘facets’. `maximal_faces` can also be a list containing a single non-negative integer  $n$ , in which case this constructs the simplicial complex with a single  $n$ -simplex as the only facet.

Alternatively, the maximal faces can be defined from a monotone boolean function on the subsets of a set  $X$ . While defining `maximal_faces=None`, you can thus set `from_characteristic_function=(f, X)` where  $X$  is the set of points and  $f$  a boolean monotone hereditary function that accepts a list of elements from  $X$  as input (see `subsets_with_hereditary_property()` for more information).

If `maximality_check` is `True`, check that each maximal face is, in fact, maximal. In this case, when producing the internal representation of the simplicial complex, omit those that are not. It is highly recommended that this be `True`; various methods for this class may fail if faces which are claimed to be maximal are in fact not.

`sort_facets`: if not set to `None`, the default, this should be a dictionary, used for sorting the vertices in each facet. The keys must be the vertices for the simplicial complex, and the values should be distinct sortable objects, for example integers. This should not need to be specified except in very special circumstances; currently the only use in the Sage library is when defining the product of a simplicial complex with itself: in this case, the vertices in the product must be sorted compatibly with the vertices in each factor so that the diagonal map is properly defined.

If `name_check` is `True`, check the names of the vertices to see if they can be easily converted to generators of a polynomial ring – use this if you plan to use the Stanley-Reisner ring for the simplicial complex.

## EXAMPLES:

```

sage: SimplicialComplex([[1,2], [1,4]])
Simplicial complex with vertex set (1, 2, 4) and facets {(1, 2), (1, 4)}
sage: SimplicialComplex([[0,2], [0,3], [0]])
Simplicial complex with vertex set (0, 2, 3) and facets {(0, 2), (0, 3)}
sage: SimplicialComplex([[0,2], [0,3], [0]], maximality_check=False)
Simplicial complex with vertex set (0, 2, 3) and facets {(0,), (0, 2), (0, 3)}

```

Finally, if the first argument is a simplicial complex, return that complex. If it is an object with a built-in conversion to simplicial complexes (via a `_simplicial_` method), then the resulting simplicial complex is returned:

```

sage: S = SimplicialComplex([[0,2], [0,3], [0,6]])
sage: SimplicialComplex(S) == S
True
sage: Tc = cubical_complexes.Torus(); Tc
Cubical complex with 16 vertices and 64 cubes
sage: Ts = SimplicialComplex(Tc); Ts
Simplicial complex with 16 vertices and 32 facets
sage: Ts.homology()
{0: 0, 1: Z x Z, 2: Z}

```

In the situation where the first argument is a simplicial complex or another object with a built-in conversion, most of the other arguments are ignored. The only exception is `is_mutable`:

```

sage: S.is_mutable()
True
sage: SimplicialComplex(S, is_mutable=False).is_mutable()
False

```

From a characteristic monotone boolean function, e.g. the simplicial complex of all subsets  $S \subseteq \{0, 1, 2, 3, 4\}$  such that  $\text{sum}(S) \leq 4$ :

```

sage: SimplicialComplex(from_characteristic_function=(lambda x: sum(x) <= 4,
↳ range(5)))
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 4), (0, 1, 2),
↳ (0, 1, 3)}

```

or e.g. the simplicial complex of all 168 hyperovals of the projective plane of order 4:

```

sage: l = designs.ProjectiveGeometryDesign(2,1,GF(4,name='a'))
sage: f = lambda S: not any(len(set(S).intersection(x))>2 for x in l)
sage: SimplicialComplex(from_characteristic_function=(f, l.ground_set()))
Simplicial complex with 21 vertices and 168 facets

```

**Warning:** Simplicial complexes are not proper parents as they do not possess element classes. In particular, parents are assumed to be hashable (and hence immutable) by the coercion framework. However this is close enough to being a parent with elements being the faces of `self` that we currently allow this abuse.

**add\_face** (*face*)

Add a face to this simplicial complex.

**Parameters** *face* – a subset of the vertex set

This *changes* the simplicial complex, adding a new face and all of its subfaces.

EXAMPLES:

```
sage: X = SimplicialComplex([[0,1], [0,2]])
sage: X.add_face([0,1,2,]); X
Simplicial complex with vertex set (0, 1, 2) and facets {(0, 1, 2)}
sage: Y = SimplicialComplex(); Y
Simplicial complex with vertex set () and facets {}
sage: Y.add_face([0,1])
sage: Y.add_face([1,2,3])
sage: Y
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (1, 2, 3)}
```

If you add a face which is already present, there is no effect:

```
sage: Y.add_face([1,3]); Y
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (1, 2, 3)}
```

**alexander\_dual** (*is\_mutable=True*)

The Alexander dual of this simplicial complex: according to the Macaulay2 documentation, this is the simplicial complex whose faces are the complements of its nonfaces.

Thus find the minimal nonfaces and take their complements to find the facets in the Alexander dual.

**Parameters** *is\_mutable* (boolean; optional, default True) – Determines if the output is mutable

EXAMPLES:

```
sage: Y = SimplicialComplex([i for i in range(5)]); Y
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0,), (1,), (2, 3), (3, 4), (4,)}
sage: Y.alexander_dual()
Simplicial complex with vertex set (0, 1, 2, 3, 4) and 10 facets
sage: X = SimplicialComplex([[0,1], [1,2], [2,3], [3,0]])
sage: X.alexander_dual()
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 2), (1, 3)}
```

**alexander\_whitney** (*simplex, dim\_left*)

Subdivide this simplex into a pair of simplices.

If this simplex has vertices  $v_0, v_1, \dots, v_n$ , then subdivide it into simplices  $(v_0, v_1, \dots, v_{dim})$  and  $(v_{dim}, v_{dim+1}, \dots, v_n)$ .

See `Simplex.alexander_whitney()` for more details. This method just calls that one.

INPUT:

- *simplex* – a simplex in this complex
- *dim* – integer between 0 and one more than the dimension of this simplex

OUTPUT: a list containing just the triple (1, left, right), where left and right are the two simplices described above.

EXAMPLES:

```
sage: s = Simplex((0,1,3,4))
sage: X = SimplicialComplex([s])
sage: X.alexander_whitney(s, 0)
[(1, (0,), (0, 1, 3, 4))]
```

(continues on next page)

(continued from previous page)

```
sage: X.alexander_whitney(s, 2)
[(1, (0, 1, 3), (3, 4))]
```

**algebraic\_topological\_model** (*base\_ring=None*)

Algebraic topological model for this simplicial complex with coefficients in *base\_ring*.

The term “algebraic topological model” is defined by Pilarczyk and Réal [PR2015].

INPUT:

- *base\_ring* - coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ). Must be a field.

Denote by  $C$  the chain complex associated to this simplicial complex. The algebraic topological model is a chain complex  $M$  with zero differential, with the same homology as  $C$ , along with chain maps  $\pi : C \rightarrow M$  and  $\iota : M \rightarrow C$  satisfying  $\iota\pi = 1_M$  and  $\pi\iota$  chain homotopic to  $1_C$ . The chain homotopy  $\phi$  must satisfy

- $\phi\phi = 0$ ,
- $\pi\phi = 0$ ,
- $\phi\iota = 0$ .

Such a chain homotopy is called a *chain contraction*.

OUTPUT: a pair consisting of

- chain contraction *phi* associated to  $C$ ,  $M$ ,  $\pi$ , and  $\iota$
- the chain complex  $M$

Note that from the chain contraction *phi*, one can recover the chain maps  $\pi$  and  $\iota$  via *phi*.*pi*() and *phi*.*iota*(). Then one can recover  $C$  and  $M$  from, for example, *phi*.*pi*()*.domain*() and *phi*.*pi*()*.codomain*(), respectively.

EXAMPLES:

```
sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: phi, M = RP2.algebraic_topological_model(GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = simplicial_complexes.Torus()
sage: phi, M = T.algebraic_topological_model(QQ)
sage: M.homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

**automorphism\_group** ()

Return the automorphism group of the simplicial complex.

This is done by creating a bipartite graph, whose vertices are vertices and facets of the simplicial complex, and computing its automorphism group.

**Warning:** Since [trac ticket #14319](#) the domain of the automorphism group is equal to the graph’s vertex set, and the *translation* argument has become useless.

EXAMPLES:

```

sage: S = simplicial_complexes.Simplex(3)
sage: S.automorphism_group().is_isomorphic(SymmetricGroup(4))
True

sage: P = simplicial_complexes.RealProjectivePlane()
sage: P.automorphism_group().is_isomorphic(AlternatingGroup(5))
True

sage: Z = SimplicialComplex([[1, 2], [2, 3, 'a']])
sage: Z.automorphism_group().is_isomorphic(CyclicPermutationGroup(2))
True
sage: group = Z.automorphism_group()
sage: sorted(group.domain())
[1, 2, 3, 'a']

```

Check that [trac ticket #17032](#) is fixed:

```

sage: s = SimplicialComplex([(0, 1), (2, 3)])
sage: s.automorphism_group().cardinality()
2

```

### **barycentric\_subdivision()**

The barycentric subdivision of this simplicial complex.

See [Wikipedia article Barycentric\\_subdivision](#) for a definition.

EXAMPLES:

```

sage: triangle = SimplicialComplex([[0, 1], [1, 2], [0, 2]])
sage: hexagon = triangle.barycentric_subdivision()
sage: hexagon
Simplicial complex with 6 vertices and 6 facets
sage: hexagon.homology(1) == triangle.homology(1)
True

```

Barycentric subdivisions can get quite large, since each  $n$ -dimensional facet in the original complex produces  $(n + 1)!$  facets in the subdivision:

```

sage: S4 = simplicial_complexes.Sphere(4)
sage: S4
Minimal triangulation of the 4-sphere
sage: S4.barycentric_subdivision()
Simplicial complex with 62 vertices and 720 facets

```

### **cells** (*subcomplex=None*)

The faces of this simplicial complex, in the form of a dictionary of sets keyed by dimension. If the optional argument *subcomplex* is present, then return only the faces which are *not* in the subcomplex.

**Parameters** *subcomplex* (optional, default None) – a subcomplex of this simplicial complex.

Return faces which are not in this subcomplex.

EXAMPLES:

```

sage: Y = SimplicialComplex([[1, 2], [1, 4]])
sage: Y.faces()
{-1: {}, 0: {(1, ), (2, ), (4, )}, 1: {(1, 2), (1, 4)}}
sage: L = SimplicialComplex([[1, 2]])
sage: Y.faces(subcomplex=L)
{-1: set(), 0: {(4, )}, 1: {(1, 4)}}

```

**chain\_complex**(*subcomplex=None*, *augmented=False*, *verbose=False*, *check=False*, *dimensions=None*, *base\_ring=Integer Ring*, *cochain=False*)

The chain complex associated to this simplicial complex.

#### Parameters

- **dimensions** – if `None`, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero.
- **base\_ring** (optional, default `ZZ`) – commutative ring
- **subcomplex** (optional, default `empty`) – a subcomplex of this simplicial complex. Compute the chain complex relative to this subcomplex.
- **augmented** (boolean; optional, default `False`) – If `True`, return the augmented chain complex (that is, include a class in dimension  $-1$  corresponding to the empty cell). This is ignored if `dimensions` is specified.
- **cochain** (boolean; optional, default `False`) – If `True`, return the cochain complex (that is, the dual of the chain complex).
- **verbose** (boolean; optional, default `False`) – If `True`, print some messages as the chain complex is computed.
- **check** (boolean; optional, default `False`) – If `True`, make sure that the chain complex is actually a chain complex: the differentials are composable and their product is zero.

**Note:** If `subcomplex` is nonempty, then the argument `augmented` has no effect: the chain complex relative to a nonempty subcomplex is zero in dimension  $-1$ .

The rows and columns of the boundary matrices are indexed by the lists given by the `_n_cells_sorted()` method, which by default are sorted.

#### EXAMPLES:

```
sage: circle = SimplicialComplex([[0,1], [1,2], [0, 2]])
sage: circle.chain_complex()
Chain complex with at most 2 nonzero terms over Integer Ring
sage: circle.chain_complex()._latex_()
'\Bold{Z}^{\{3\}} \xrightarrow{d_{\{1\}}} \Bold{Z}^{\{3\}}'
sage: circle.chain_complex(base_ring=QQ, augmented=True)
Chain complex with at most 3 nonzero terms over Rational Field
```

**cone**(*is\_mutable=True*)

The cone on this simplicial complex.

**Parameters** **is\_mutable** (boolean; optional, default `True`) – Determines if the output is mutable

The cone is the simplicial complex formed by adding a new vertex  $C$  and simplices of the form  $[C, v_0, \dots, v_k]$  for every simplex  $[v_0, \dots, v_k]$  in the original simplicial complex. That is, the cone is the join of the original complex with a one-point simplicial complex.

#### EXAMPLES:

```
sage: S = SimplicialComplex([[0], [1]])
sage: CS = S.cone()
sage: sorted(CS.vertices())
['L0', 'L1', 'R0']
```

(continues on next page)



(continued from previous page)

```

sage: len(CS.facets())
2
sage: CS.facets() == set([Simplex(['L0', 'R0']), Simplex(['L1', 'R0'])])
True

```

**cone\_vertices()**

Return the list of cone vertices of self.

A vertex is a cone vertex if and only if it appears in every facet.

EXAMPLES:

```

sage: SimplicialComplex([[1,2,3]]).cone_vertices()
[1, 2, 3]
sage: SimplicialComplex([[1,2,3], [1,3,4], [1,5,6]]).cone_vertices()
[1]
sage: SimplicialComplex([[1,2,3], [1,3,4], [2,5,6]]).cone_vertices()
[]

```

**connected\_component** (*simplex=None*)

Return the connected component of this simplicial complex containing simplex. If simplex is omitted, then return the connected component containing the zeroth vertex in the vertex list. (If the simplicial complex is empty, raise an error.)

EXAMPLES:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: S1 == S1.connected_component()
True
sage: X = S1.disjoint_union(S1)
sage: X == X.connected_component()
False
sage: X.connected_component(Simplex(['L0'])) == X.connected_
↪component(Simplex(['R0']))
False

sage: S0 = simplicial_complexes.Sphere(0)
sage: S0.vertices()
(0, 1)
sage: S0.connected_component()
Simplicial complex with vertex set (0,) and facets {(0,)}
sage: S0.connected_component(Simplex((1,)))
Simplicial complex with vertex set (1,) and facets {(1,)}

sage: SimplicialComplex([[]]).connected_component()
Traceback (most recent call last):
...
ValueError: the empty simplicial complex has no connected components

```

**connected\_sum** (*other, is\_mutable=True*)

The connected sum of this simplicial complex with another one.

**Parameters**

- **other** – another simplicial complex
- **is\_mutable** (boolean; optional, default True) – Determines if the output is mutable

**Returns** the connected sum self # other

**Warning:** This does not check that `self` and `other` are manifolds, only that their facets all have the same dimension. Since a (more or less) random facet is chosen from each complex and then glued together, this method may return random results if applied to non-manifolds, depending on which facet is chosen.

Algorithm: a facet is chosen from each surface, and removed. The vertices of these two facets are relabeled to  $(0, 1, \dots, \dim)$ . Of the remaining vertices, the ones from the left-hand factor are renamed by prepending an “L”, and similarly the remaining vertices in the right-hand factor are renamed by prepending an “R”.

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: S1.connected_sum(S1.connected_sum(S1)).homology()
{0: 0, 1: Z}
sage: P = simplicial_complexes.RealProjectivePlane(); P
Minimal triangulation of the real projective plane
sage: P.connected_sum(P)      # the Klein bottle
Simplicial complex with 9 vertices and 18 facets
```

The notation ‘+’ may be used for connected sum, also:

```
sage: P + P      # the Klein bottle
Simplicial complex with 9 vertices and 18 facets
sage: (P + P).homology()[1]
Z x C2
```

**decone()**

Return the subcomplex of `self` induced by the non-cone vertices.

EXAMPLES:

```
sage: SimplicialComplex([[1,2,3]]).decone()
Simplicial complex with vertex set {} and facets {}
sage: SimplicialComplex([[1,2,3], [1,3,4], [1,5,6]]).decone()
Simplicial complex with vertex set {2, 3, 4, 5, 6} and facets {(2, 3), (3, 4),
↪ (5, 6)}
sage: X = SimplicialComplex([[1,2,3], [1,3,4], [2,5,6]])
sage: X.decone() == X
True
```

**delta\_complex(sort\_simplices=False)**

Returns `self` as a  $\Delta$ -complex. The  $\Delta$ -complex is essentially identical to the simplicial complex: it has same simplices with the same boundaries.

**Parameters** `sort_simplices` (boolean; optional, default `False`) – if `True`, sort the list of simplices in each dimension

EXAMPLES:

```
sage: T = simplicial_complexes.Torus()
sage: Td = T.delta_complex()
sage: Td
Delta complex with 7 vertices and 43 simplices
sage: T.homology() == Td.homology()
True
```

**disjoint\_union** (*right*, *rename\_vertices=True*, *is\_mutable=True*)

The disjoint union of this simplicial complex with another one.

#### Parameters

- **right** – the other simplicial complex (the right-hand factor)
- **rename\_vertices** (*boolean; optional, default True*) – If this is True, the vertices in the disjoint union will be renamed by the formula: vertex “v” in the left-hand factor → vertex “Lv” in the disjoint union, vertex “w” in the right-hand factor → vertex “Rw” in the disjoint union. If this is false, this tries to construct the disjoint union without renaming the vertices; this will cause problems if the two factors have any vertices with names in common.

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: S2 = simplicial_complexes.Sphere(2)
sage: S1.disjoint_union(S2).homology()
{0: Z, 1: Z, 2: Z}
```

**f\_triangle**()

Compute the  $f$ -triangle of self.

The  $f$ -triangle is given by  $f_{i,j}$  being the number of faces  $F$  of size  $j$  such that  $i = \max_{G \subseteq F} |G|$ .

EXAMPLES:

```
sage: X = SimplicialComplex([[1,2,3], [3,4,5], [1,4], [1,5], [2,4], [2,5]])
sage: X.f_triangle()  ## this complex is not pure
[[0],
 [0, 0],
 [0, 0, 4],
 [1, 5, 6, 2]]
```

A complex is pure if and only if the last row is nonzero:

```
sage: X = SimplicialComplex([[1,2,3], [3,4,5], [1,4,5]])
sage: X.f_triangle()
[[0], [0, 0], [0, 0, 0], [1, 5, 8, 3]]
```

**face** (*simplex*, *i*)

The  $i$ -th face of *simplex* in this simplicial complex

INPUT:

- **simplex** – a simplex in this simplicial complex
- **i** – integer

EXAMPLES:

```
sage: S = SimplicialComplex([[0,1,4], [0,1,2]])
sage: S.face(Simplex((0,2)), 0)
(2,)

sage: S.face(Simplex((0,3)), 0)
Traceback (most recent call last):
...
ValueError: this simplex is not in this simplicial complex
```

**face\_iterator** (*increasing=True*)

An iterator for the faces in this simplicial complex.

INPUT:

- *increasing* – (optional, default `True`) if `True`, return faces in increasing order of dimension, thus starting with the empty face. Otherwise it returns faces in decreasing order of dimension.

---

**Note:** Among the faces of a fixed dimension, there is no sorting.
 

---

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: sorted(S1.face_iterator())
[(), (0,), (0, 1), (0, 2), (1,), (1, 2), (2,)]
```

**faces** (*subcomplex=None*)The faces of this simplicial complex, in the form of a dictionary of sets keyed by dimension. If the optional argument *subcomplex* is present, then return only the faces which are *not* in the subcomplex.**Parameters** *subcomplex* (optional, default `None`) – a subcomplex of this simplicial complex.

Return faces which are not in this subcomplex.

EXAMPLES:

```
sage: Y = SimplicialComplex([[1,2], [1,4]])
sage: Y.faces()
{-1: {()}, 0: {(1,), (2,), (4,)}, 1: {(1, 2), (1, 4)}}
sage: L = SimplicialComplex([[1,2]])
sage: Y.faces(subcomplex=L)
{-1: set(), 0: {(4,)}, 1: {(1, 4)}}
```

**facets** ()

The maximal faces (a.k.a. facets) of this simplicial complex.

This just returns the set of facets used in defining the simplicial complex, so if the simplicial complex was defined with no maximality checking, none is done here, either.

EXAMPLES:

```
sage: Y = SimplicialComplex([[0,2], [1,4]])
sage: sorted(Y.maximal_faces())
[(0, 2), (1, 4)]
```

*facets* is a synonym for *maximal\_faces*:

```
sage: S = SimplicialComplex([[0,1], [0,1,2]])
sage: S.facets()
{(0, 1, 2)}
```

**fixed\_complex** (*G*)Return the fixed simplicial complex  $Fix(G)$  for a subgroup *G*.

INPUT:

- *G* – a subgroup of the automorphism group of the simplicial complex or a list of elements of the automorphism group

OUTPUT:

- a simplicial complex  $\text{Fix}(G)$

Vertices in  $\text{Fix}(G)$  are the orbits of  $G$  (acting on vertices of `self`) that form a simplex in `self`. More generally, simplices in  $\text{Fix}(G)$  correspond to simplices in `self` that are union of such orbits.

A basic example:

```
sage: S4 = simplicial_complexes.Sphere(4)
sage: S3 = simplicial_complexes.Sphere(3)
sage: fix = S4.fixed_complex([S4.automorphism_group()((0,1))])
sage: fix
Simplicial complex with vertex set (0, 2, 3, 4, 5) and 5 facets
sage: fix.is_isomorphic(S3)
True
```

Another simple example:

```
sage: T = SimplicialComplex([[1,2,3],[2,3,4]])
sage: G = T.automorphism_group()
sage: T.fixed_complex([G([(1,4)])])
Simplicial complex with vertex set (2, 3) and facets {(2, 3)}
```

A more sophisticated example:

```
sage: RP2 = simplicial_complexes.ProjectivePlane()
sage: CP2 = simplicial_complexes.ComplexProjectivePlane()
sage: G = CP2.automorphism_group()
sage: H = G.subgroup([G([(2,3),(5,6),(8,9)])])
sage: CP2.fixed_complex(H).is_isomorphic(RP2)
True
```

### **flip\_graph()**

If `self` is pure, then it returns the flip graph of `self`, otherwise, it returns `None`.

The flip graph of a pure simplicial complex is the (undirected) graph with vertices being the facets, such that two facets are joined by an edge if they meet in a codimension 1 face.

The flip graph is used to detect if `self` is a pseudomanifold.

EXAMPLES:

```
sage: S0 = simplicial_complexes.Sphere(0)
sage: G = S0.flip_graph()
sage: G.vertices(); G.edges(labels=False)
[(0,)]
[(0,),(1,)]

sage: G = (S0.wedge(S0)).flip_graph()
sage: G.vertices(); G.edges(labels=False)
[(0,),( 'L1',),( 'R1',)]
[(0,),( 'L1',),(0,),( 'R1',),( 'L1',),( 'R1',)]

sage: S1 = simplicial_complexes.Sphere(1)
sage: S2 = simplicial_complexes.Sphere(2)
sage: G = (S1.wedge(S1)).flip_graph()
sage: len(G.vertices())
6
sage: len(G.edges())
10
```

(continues on next page)

(continued from previous page)

```

sage: (S1.wedge(S2)).flip_graph() is None
True

sage: G = S2.flip_graph()
sage: G.vertices(); G.edges(labels=False)
[(0, 1, 2), (0, 1, 3), (0, 2, 3), (1, 2, 3)]
[ ((0, 1, 2), (0, 1, 3)),
  ((0, 1, 2), (0, 2, 3)),
  ((0, 1, 2), (1, 2, 3)),
  ((0, 1, 3), (0, 2, 3)),
  ((0, 1, 3), (1, 2, 3)),
  ((0, 2, 3), (1, 2, 3))]

sage: T = simplicial_complexes.Torus()
sage: G = T.suspension(4).flip_graph()
sage: len(G.vertices()); len(G.edges(labels=False))
46
161

```

**fundamental\_group** (*base\_point=None, simplify=True*)

Return the fundamental group of this simplicial complex.

INPUT:

- *base\_point* (optional, default None) – if this complex is not path-connected, then specify a vertex; the fundamental group is computed with that vertex as a base point. If the complex is path-connected, then you may specify a vertex or leave this as its default setting of None. (If this complex is path-connected, then this argument is ignored.)
- *simplify* (bool, optional True) – if False, then return a presentation of the group in terms of generators and relations. If True, the default, simplify as much as GAP is able to.

Algorithm: we compute the edge-path group – see [Wikipedia article Fundamental\\_group](#). Choose a spanning tree for the 1-skeleton, and then the group’s generators are given by the edges in the 1-skeleton; there are two types of relations:  $e = 1$  if  $e$  is in the spanning tree, and for every 2-simplex, if its edges are  $e_0$ ,  $e_1$ , and  $e_2$ , then we impose the relation  $e_0 e_1^{-1} e_2 = 1$ .

EXAMPLES:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: S1.fundamental_group()
Finitely presented group < e | >

```

If we pass the argument *simplify=False*, we get generators and relations in a form which is not usually very helpful. Here is the cyclic group of order 2, for instance:

```

sage: RP2 = simplicial_complexes.RealProjectiveSpace(2)
sage: C2 = RP2.fundamental_group(simplify=False)
sage: C2
Finitely presented group < e0, e1, e2, e3, e4, e5, e6, e7, e8, e9 | e0, e3,
↪ e4, e7, e9, e5*e2^-1*e0, e7*e2^-1*e1, e8*e3^-1*e1, e8*e6^-1*e4, e9*e6^-1*e5
↪ >
sage: C2.simplified()
Finitely presented group < e1 | e1^2 >

```

This is the same answer given if the argument *simplify* is True (the default):

```
sage: RP2.fundamental_group()
Finitely presented group < e1 | e1^2 >
```

You must specify a base point to compute the fundamental group of a non-connected complex:

```
sage: K = S1.disjoint_union(RP2)
sage: K.fundamental_group()
Traceback (most recent call last):
...
ValueError: this complex is not connected, so you must specify a base point
sage: K.fundamental_group(base_point='L0')
Finitely presented group < e | >
sage: K.fundamental_group(base_point='R0').order()
2
```

Some other examples:

```
sage: S1.wedge(S1).fundamental_group()
Finitely presented group < e0, e1 | >
sage: simplicial_complexes.Torus().fundamental_group()
Finitely presented group < e1, e4 | e4^-1*e1^-1*e4*e1 >

sage: G = simplicial_complexes.MooreSpace(5).fundamental_group()
sage: G.ngens()
1
sage: x = G.gen(0)
sage: [(x**n).is_one() for n in range(1,6)]
[False, False, False, False, True]
```

### **g\_vector()**

The  $g$ -vector of this simplicial complex.

If the  $h$ -vector of the complex is  $(h_0, h_1, \dots, h_d, h_{d+1})$  – see [h\\_vector\(\)](#) – then its  $g$ -vector  $(g_0, g_1, \dots, g_{[(d+1)/2]})$  is defined by  $g_0 = 1$  and  $g_i = h_i - h_{i-1}$  for  $i > 0$ .

EXAMPLES:

```
sage: S3 = simplicial_complexes.Sphere(3).barycentric_subdivision()
sage: S3.f_vector()
[1, 30, 150, 240, 120]
sage: S3.h_vector()
[1, 26, 66, 26, 1]
sage: S3.g_vector()
[1, 25, 40]
```

### **generated\_subcomplex(sub\_vertex\_set, is\_mutable=True)**

Returns the largest sub-simplicial complex of `self` containing exactly `sub_vertex_set` as vertices.

#### **Parameters**

- **sub\_vertex\_set** – The sub-vertex set.
- **is\_mutable** (boolean; optional, default `True`) – Determines if the output is mutable

EXAMPLES:

```
sage: S = simplicial_complexes.Sphere(2)
sage: S
Minimal triangulation of the 2-sphere
```

(continues on next page)

(continued from previous page)

```
sage: S.generated_subcomplex([0,1,2])
Simplicial complex with vertex set (0, 1, 2) and facets {(0, 1, 2)}
```

**graph()**

The 1-skeleton of this simplicial complex, as a graph.

**Warning:** This may give the wrong answer if the simplicial complex was constructed with `maximality_check` set to `False`.

**EXAMPLES:**

```
sage: S = SimplicialComplex([[0,1,2,3]])
sage: G = S.graph(); G
Graph on 4 vertices
sage: G.edges()
[(0, 1, None), (0, 2, None), (0, 3, None), (1, 2, None), (1, 3, None), (2, 3, None)]
```

**h\_triangle()**

Compute the  $h$ -triangle of `self`.

The  $h$ -triangle of a simplicial complex  $\Delta$  is given by

$$h_{i,j} = \sum_{k=0}^j (-1)^{j-k} \binom{i-k}{j-k} f_{i,k},$$

where  $f_{i,k}$  is the  $f$ -triangle of  $\Delta$ .

**EXAMPLES:**

```
sage: X = SimplicialComplex([[1,2,3], [3,4,5], [1,4], [1,5], [2,4], [2,5]])
sage: X.h_triangle()
[[0],
 [0, 0],
 [0, 0, 4],
 [1, 2, -1, 0]]
```

**h\_vector()**

The  $h$ -vector of this simplicial complex.

If the complex has dimension  $d$  and  $(f_{-1}, f_0, f_1, \dots, f_d)$  is its  $f$ -vector (with  $f_{-1} = 1$ , representing the empty simplex), then the  $h$ -vector  $(h_0, h_1, \dots, h_d, h_{d+1})$  is defined by

$$\sum_{i=0}^{d+1} h_i x^{d+1-i} = \sum_{i=0}^{d+1} f_{i-1} (x-1)^{d+1-i}.$$

Alternatively,

$$h_j = \sum_{i=-1}^{j-1} (-1)^{j-i-1} \binom{d-i}{j-i-1} f_i.$$

**EXAMPLES:**

The  $f$ - and  $h$ -vectors of the boundary of an octahedron are computed in [Wikipedia article Simplicial complex](#):



```

sage: square = SimplicialComplex([[0,1], [1,2], [2,3], [0,3]])
sage: S0 = SimplicialComplex([[0], [1]])
sage: octa = square.join(S0) # boundary of an octahedron
sage: octa.f_vector()
[1, 6, 12, 8]
sage: octa.h_vector()
[1, 3, 3, 1]

```

**intersection** (*other*)

Calculate the intersection of two simplicial complexes.

EXAMPLES:

```

sage: X = SimplicialComplex([[1,2,3], [1,2,4]])
sage: Y = SimplicialComplex([[1,2,3], [1,4,5]])
sage: Z = SimplicialComplex([[1,2,3], [1,4], [2,4]])
sage: sorted(X.intersection(Y).facets())
[(1, 2, 3), (1, 4)]
sage: X.intersection(X) == X
True
sage: X.intersection(Z) == X
False
sage: X.intersection(Z) == Z
True

```

**is\_balanced** (*check\_purity=False, certificate=False*)

Determine whether *self* is balanced.

A simplicial complex  $X$  of dimension  $d - 1$  is balanced if and only if its vertices can be colored with  $d$  colors such that every face contains at most one vertex of each color. An equivalent condition is that the 1-skeleton of  $X$  is  $d$ -colorable. In some contexts, it is also required that  $X$  be pure (i.e., that all maximal faces of  $X$  have the same dimension).

INPUT:

- *check\_purity* – (default: *False*) if this is *True*, require that *self* be pure as well as balanced
- *certificate* – (default: *False*) if this is *True* and *self* is balanced, then return a  $d$ -coloring of the 1-skeleton.

EXAMPLES:

A 1-dim simplicial complex is balanced iff it is bipartite:

```

sage: X = SimplicialComplex([[1,2], [1,4], [3,4], [2,5]])
sage: X.is_balanced()
True
sage: sorted(X.is_balanced(certificate=True))
[[1, 3, 5], [2, 4]]
sage: X = SimplicialComplex([[1,2], [1,4], [3,4], [2,4]])
sage: X.is_balanced()
False

```

Any barycentric division is balanced:

```

sage: X = SimplicialComplex([[1,2,3], [1,2,4], [2,3,4]])
sage: X.is_balanced()
False
sage: X.barycentric_subdivision().is_balanced()
True

```

A non-pure balanced complex:

```
sage: X=SimplicialComplex([[1,2,3],[3,4]])
sage: X.is_balanced(check_purity=True)
False
sage: sorted(X.is_balanced(certificate=True))
[[1, 4], [2], [3]]
```

**is\_cohen\_macaulay** (*base\_ring=Rational Field, ncpus=0*)

Return True if self is Cohen-Macaulay.

A simplicial complex  $\Delta$  is Cohen-Macaulay over  $R$  iff  $\tilde{H}_i(\text{lk}_\Delta(F); R) = 0$  for all  $F \in \Delta$  and  $i < \dim \text{lk}_\Delta(F)$ . Here,  $\Delta$  is self and  $R$  is base\_ring, and lk denotes the link operator on self.

INPUT:

- base\_ring – (default: QQ) the base ring.
- ncpus – (default: 0) number of cpus used for the computation. If this is 0, determine the number of cpus automatically based on the hardware being used.

For finite simplicial complexes, this is equivalent to the statement that the Stanley-Reisner ring of self is Cohen-Macaulay.

EXAMPLES:

Spheres are Cohen-Macaulay:

```
sage: S = SimplicialComplex([[1,2],[2,3],[3,1]])
sage: S.is_cohen_macaulay(ncpus=3)
True
```

The following example is taken from Bruns, Herzog - Cohen-Macaulay rings, Figure 5.3:

```
sage: S = SimplicialComplex([[1,2,3],[1,4,5]])
sage: S.is_cohen_macaulay(ncpus=3)
False
```

The choice of base ring can matter. The real projective plane  $\mathbf{RP}^2$  has  $H_1(\mathbf{RP}^2) = \mathbf{Z}/2$ , hence is CM over  $\mathbf{Q}$  but not over  $\mathbf{Z}$ .

```
sage: X = simplicial_complexes.RealProjectivePlane()
sage: X.is_cohen_macaulay()
True
sage: X.is_cohen_macaulay(ZZ)
False
```

**is\_flag\_complex** ()

Returns True if and only if self is a flag complex.

A flag complex is a simplicial complex that is the largest simplicial complex on its 1-skeleton. Thus a flag complex is the clique complex of its graph.

EXAMPLES:

```
sage: h = Graph({0:[1,2,3,4],1:[2,3,4],2:[3]})
sage: x = h.clique_complex()
sage: x
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 4), (0, 1, 2, 3)}
sage: x.is_flag_complex()
```

(continues on next page)

(continued from previous page)

```

True

sage: X = simplicial_complexes.ChessboardComplex(3,3)
sage: X.is_flag_complex()
True

```

**is\_immutable()**

Return True if immutable.

EXAMPLES:

```

sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.is_immutable()
False
sage: S.set_immutable()
sage: S.is_immutable()
True

```

**is\_isomorphic(other, certificate=False)**

Check whether two simplicial complexes are isomorphic.

INPUT:

- `certificate` – if True, then output is (a, b), where a is a boolean and b is either a map or None

This is done by creating two graphs and checking whether they are isomorphic.

EXAMPLES:

```

sage: Z1 = SimplicialComplex([[0,1], [1,2], [2,3,4], [4,5]])
sage: Z2 = SimplicialComplex([[ 'a', 'b' ], [ 'b', 'c' ], [ 'c', 'd', 'e' ], [ 'e', 'f' ]])
sage: Z3 = SimplicialComplex([[1,2,3]])
sage: Z1.is_isomorphic(Z2)
True
sage: Z1.is_isomorphic(Z2, certificate=True)
(True, {0: 'a', 1: 'b', 2: 'c', 3: 'd', 4: 'e', 5: 'f'})
sage: Z3.is_isomorphic(Z2)
False

```

We check that [trac ticket #20751](#) is fixed:

```

sage: C1 = SimplicialComplex([[1,2,3], [2,4], [3,5], [5,6]])
sage: C2 = SimplicialComplex([[ 'a', 'b', 'c' ], [ 'b', 'd' ], [ 'c', 'e' ], [ 'e', 'f' ]])
sage: C1.is_isomorphic(C2, certificate=True)
(True, {1: 'a', 2: 'b', 3: 'c', 4: 'd', 5: 'e', 6: 'f'})

```

**is\_mutable()**

Return True if mutable.

EXAMPLES:

```

sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.is_mutable()
True
sage: S.set_immutable()
sage: S.is_mutable()
False

```

(continues on next page)

(continued from previous page)

```

sage: S2 = SimplicialComplex([[1,4], [2,4]], is_mutable=False)
sage: S2.is_mutable()
False
sage: S3 = SimplicialComplex([[1,4], [2,4]], is_mutable=False)
sage: S3.is_mutable()
False

```

**is\_partitionable** (*certificate=False*)Determine whether `self` is partitionable.

A partitioning of a simplicial complex  $X$  is a decomposition of its face poset into disjoint Boolean intervals  $[R, F]$ , where  $F$  ranges over all facets of  $X$ .

The method sets up an integer program with:

- a variable  $y_i$  for each pair  $(R, F)$ , where  $F$  is a facet of  $X$  and  $R$  is a subface of  $F$
- a constraint  $y_i + y_j \leq 1$  for each pair  $(R_i, F_i), (R_j, F_j)$  whose Boolean intervals intersect nontrivially (equivalent to  $(R_i \subseteq F_j \text{ and } R_j \subseteq F_i)$ )
- objective function equal to the sum of all  $y_i$

INPUT:

- `certificate` – (default: `False`) If `True`, and `self` is partitionable, then return a list of pairs  $(R, F)$  that form a partitioning.

EXAMPLES:

Simplices are trivially partitionable:

```

sage: X = SimplicialComplex([ [1,2,3,4] ])
sage: X.is_partitionable()
True
sage: X.is_partitionable(certificate=True)
[((), (1, 2, 3, 4), 4)]

```

Shellable complexes are partitionable:

```

sage: X = SimplicialComplex([ [1,3,5], [1,3,6], [1,4,5], [1,4,6], [2,3,5], [2,3,6],
↪ [2,4,5] ])
sage: X.is_partitionable()
True
sage: P = X.is_partitionable(certificate=True)
sage: n_intervals_containing = lambda f: len([ RF for RF in P if RF[0].is_
↪ face(f) and f.is_face(RF[1]) ])
sage: all( n_intervals_containing(f)==1 for k in X.faces().keys() for f in X.
↪ faces()[k] )
True

```

A non-shellable, non-Cohen-Macaulay, partitionable example, constructed by Björner:

```

sage: X = SimplicialComplex([ [1,2,3], [1,2,4], [1,3,4], [2,3,4], [1,5,6] ])
sage: X.is_partitionable()
True

```

The bowtie complex is not partitionable:

```
sage: X = SimplicialComplex([ [1,2,3], [1,4,5] ])
sage: X.is_partitionable()
False
```

**is\_pseudomanifold()**

Return True if self is a pseudomanifold.

A pseudomanifold is a simplicial complex with the following properties:

- it is pure of some dimension  $d$  (all of its facets are  $d$ -dimensional)
- every  $(d - 1)$ -dimensional simplex is the face of exactly two facets
- for every two facets  $S$  and  $T$ , there is a sequence of facets

$$S = f_0, f_1, \dots, f_n = T$$

such that for each  $i$ ,  $f_i$  and  $f_{i-1}$  intersect in a  $(d - 1)$ -simplex.

By convention,  $S^0$  is the only 0-dimensional pseudomanifold.

EXAMPLES:

```
sage: S0 = simplicial_complexes.Sphere(0)
sage: S0.is_pseudomanifold()
True
sage: (S0.wedge(S0)).is_pseudomanifold()
False
sage: S1 = simplicial_complexes.Sphere(1)
sage: S2 = simplicial_complexes.Sphere(2)
sage: (S1.wedge(S1)).is_pseudomanifold()
False
sage: (S1.wedge(S2)).is_pseudomanifold()
False
sage: S2.is_pseudomanifold()
True
sage: T = simplicial_complexes.Torus()
sage: T.suspension(4).is_pseudomanifold()
True
```

**is\_pure()**

Return True iff this simplicial complex is pure.

A simplicial complex is pure if and only if all of its maximal faces have the same dimension.

**Warning:** This may give the wrong answer if the simplicial complex was constructed with `maximality_check` set to `False`.

EXAMPLES:

```
sage: U = SimplicialComplex([[1,2], [1, 3, 4]])
sage: U.is_pure()
False
sage: X = SimplicialComplex([[0,1], [0,2], [1,2]])
sage: X.is_pure()
True
```

Demonstration of the warning:

```
sage: S = SimplicialComplex([[0,1], [0]], maximality_check=False)
sage: S.is_pure()
False
```

**is\_shellable** (*certificate=False*)

Return if `self` is shellable.

A simplicial complex is shellable if there exists a shelling order.

---

**Note:**

1. This method can check all orderings of the facets by brute force, hence can be very slow.
  2. This is shellability in the general (nonpure) sense of Björner and Wachs [BW1996]. This method does not check purity.
- 

**See also:**

[`is\_shelling\_order\(\)`](#)

**INPUT:**

- `certificate` – (default: `False`) if `True` then returns the shelling order (if it exists)

**EXAMPLES:**

```
sage: X = SimplicialComplex([[1,2,5], [2,3,5], [3,4,5], [1,4,5]])
sage: X.is_shellable()
True
sage: order = X.is_shellable(True); order
((1, 2, 5), (2, 3, 5), (1, 4, 5), (3, 4, 5))
sage: X.is_shelling_order(order)
True

sage: X = SimplicialComplex([[1,2,3], [3,4,5]])
sage: X.is_shellable()
False
```

Examples from Figure 1 in [BW1996]:

```
sage: X = SimplicialComplex([[1,2,3], [3,4], [4,5], [5,6], [4,6]])
sage: X.is_shellable()
True

sage: X = SimplicialComplex([[1,2,3], [3,4], [4,5,6]])
sage: X.is_shellable()
False
```

**REFERENCES:**

- [Wikipedia article Shelling\\_\(topology\)](#)

**is\_shelling\_order** (*shelling\_order, certificate=False*)

Return if the order of the facets given by `shelling_order` is a shelling order for `self`.

A sequence of facets  $(F_i)_{i=1}^N$  of a simplicial complex of dimension  $d$  is a *shelling order* if for all  $i =$

$2, 3, 4, \dots$ , the complex

$$X_i = \left( \bigcup_{j=1}^{i-1} F_j \right) \cap F_i$$

is pure and of dimension  $\dim F_i - 1$ .

INPUT:

- `shelling_order` – an ordering of the facets of `self`
- `certificate` – (default: `False`) if `True` then returns the index of the first facet that violate the condition

See also:

`is_shellable()`

EXAMPLES:

```
sage: facets = [[1,2,5],[2,3,5],[3,4,5],[1,4,5]]
sage: X = SimplicialComplex(facets)
sage: X.is_shelling_order(facets)
True

sage: b = [[1,2,5],[3,4,5],[2,3,5],[1,4,5]]
sage: X.is_shelling_order(b)
False
sage: X.is_shelling_order(b, True)
(False, 1)
```

A non-pure example:

```
sage: facets = [[1,2,3],[3,4],[4,5],[5,6],[4,6]]
sage: X = SimplicialComplex(facets)
sage: X.is_shelling_order(facets)
True
```

REFERENCES:

- [BW1996]

**join** (*right*, *rename\_vertices=True*, *is\_mutable=True*)

The join of this simplicial complex with another one.

The join of two simplicial complexes  $S$  and  $T$  is the simplicial complex  $S * T$  with simplices of the form  $[v_0, \dots, v_k, w_0, \dots, w_n]$  for all simplices  $[v_0, \dots, v_k]$  in  $S$  and  $[w_0, \dots, w_n]$  in  $T$ .

**Parameters**

- **right** – the other simplicial complex (the right-hand factor)
- **rename\_vertices** (boolean; optional, default `True`) – If this is `True`, the vertices in the join will be renamed by the formula: vertex “v” in the left-hand factor  $\rightarrow$  vertex “Lv” in the join, vertex “w” in the right-hand factor  $\rightarrow$  vertex “Rw” in the join. If this is false, this tries to construct the join without renaming the vertices; this will cause problems if the two factors have any vertices with names in common.
- **is\_mutable** (boolean; optional, default `True`) – Determines if the output is mutable

EXAMPLES:

```

sage: S = SimplicialComplex([[0], [1]])
sage: T = SimplicialComplex([[2], [3]])
sage: S.join(T)
Simplicial complex with vertex set ('L0', 'L1', 'R2', 'R3') and 4 facets
sage: S.join(T, rename_vertices=False)
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 2), (0, 3),
↪ (1, 2), (1, 3)}

```

The notation ‘\*’ may be used, as well:

```

sage: S * S
Simplicial complex with vertex set ('L0', 'L1', 'R0', 'R1') and 4 facets
sage: S * S * S * S * S * S * S * S * S
Simplicial complex with 16 vertices and 256 facets

```

**link** (*simplex*, *is\_mutable=True*)

The link of a simplex in this simplicial complex.

The link of a simplex  $F$  is the simplicial complex formed by all simplices  $G$  which are disjoint from  $F$  but for which  $F \cup G$  is a simplex.

#### Parameters

- **simplex** – a simplex in this simplicial complex.
- **is\_mutable** (boolean; optional, default True) – Determines if the output is mutable

EXAMPLES:

```

sage: X = SimplicialComplex([[0,1,2], [1,2,3]])
sage: X.link(Simplex([0]))
Simplicial complex with vertex set (1, 2) and facets {(1, 2)}
sage: X.link([1,2])
Simplicial complex with vertex set (0, 3) and facets {(0,), (3,)}
sage: Y = SimplicialComplex([[0,1,2,3]])
sage: Y.link([1])
Simplicial complex with vertex set (0, 2, 3) and facets {(0, 2, 3)}

```

**maximal\_faces** ()

The maximal faces (a.k.a. facets) of this simplicial complex.

This just returns the set of facets used in defining the simplicial complex, so if the simplicial complex was defined with no maximality checking, none is done here, either.

EXAMPLES:

```

sage: Y = SimplicialComplex([[0,2], [1,4]])
sage: sorted(Y.maximal_faces())
[(0, 2), (1, 4)]

```

facets is a synonym for maximal\_faces:

```

sage: S = SimplicialComplex([[0,1], [0,1,2]])
sage: S.facets()
{(0, 1, 2)}

```

**minimal\_nonfaces** ()

Set consisting of the minimal subsets of the vertex set of this simplicial complex which do not form faces.



Algorithm: Proceeds through the faces of the complex increasing the dimension, starting from dimension 0, and add the faces that are not contained in the complex and that are not already contained in a previously seen minimal non-face.

This is used in computing the *Stanley-Reisner ring* and the *Alexander dual*.

EXAMPLES:

```
sage: X = SimplicialComplex([[1,3],[1,2]])
sage: X.minimal_nonfaces()
{(2, 3)}
sage: Y = SimplicialComplex([[0,1],[1,2],[2,3],[3,0]])
sage: sorted(Y.minimal_nonfaces())
[(0, 2), (1, 3)]
```

**n\_faces** (*n*, *subcomplex=None*)

List of cells of dimension *n* of this cell complex. If the optional argument *subcomplex* is present, then return the *n*-dimensional cells which are *not* in the subcomplex.

#### Parameters

- **n** (*non-negative integer*) – the dimension
- **subcomplex** (optional, default *None*) – a subcomplex of this cell complex. Return the cells which are not in this subcomplex.

---

**Note:** The resulting list need not be sorted. If you want a sorted list of *n*-cells, use `_n_cells_sorted()`.

---

EXAMPLES:

```
sage: delta_complexes.Torus().n_cells(1)
[(0, 0), (0, 0), (0, 0)]
sage: cubical_complexes.Cube(1).n_cells(0)
[[1,1], [0,0]]
```

**n\_skeleton** (*n*)

The *n*-skeleton of this simplicial complex.

The *n*-skeleton of a simplicial complex is obtained by discarding all of the simplices in dimensions larger than *n*.

**Parameters** *n* – non-negative integer

EXAMPLES:

```
sage: X = SimplicialComplex([[0,1],[1,2,3],[0,2,3]])
sage: X.n_skeleton(1)
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (0, 2), ↪
↪ (0, 3), (1, 2), (1, 3), (2, 3)}
sage: X.set_immutable()
sage: X.n_skeleton(2)
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (0, 2, 3),
↪ (1, 2, 3)}
sage: X.n_skeleton(4)
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (0, 2, 3),
↪ (1, 2, 3)}
```

**product** (*right*, *rename\_vertices=True*, *is\_mutable=True*)

The product of this simplicial complex with another one.

#### Parameters

- **right** – the other simplicial complex (the right-hand factor)
- **rename\_vertices** (boolean; optional, default `True`) – If this is `False`, then the vertices in the product are the set of ordered pairs  $(v, w)$  where  $v$  is a vertex in `self` and  $w$  is a vertex in `right`. If this is `True`, then the vertices are renamed as “LvRw” (e.g., the vertex  $(1, 2)$  would become “L1R2”). This is useful if you want to define the Stanley-Reisner ring of the complex: vertex names like  $(0, 1)$  are not suitable for that, while vertex names like “L0R1” are.
- **is\_mutable** (boolean; optional, default `True`) – Determines if the output is mutable

The vertices in the product will be the set of ordered pairs  $(v, w)$  where  $v$  is a vertex in `self` and  $w$  is a vertex in `right`.

**Warning:** If  $X$  and  $Y$  are simplicial complexes, then  $X * Y$  returns their join, not their product.

#### EXAMPLES:

```
sage: S = SimplicialComplex([[0,1], [1,2], [0,2]]) # circle
sage: K = SimplicialComplex([[0,1]]) # edge
sage: Cyl = S.product(K) # cylinder
sage: sorted(Cyl.vertices())
['L0R0', 'L0R1', 'L1R0', 'L1R1', 'L2R0', 'L2R1']
sage: Cyl2 = S.product(K, rename_vertices=False)
sage: sorted(Cyl2.vertices())
[(0, 0), (0, 1), (1, 0), (1, 1), (2, 0), (2, 1)]
sage: T = S.product(S) # torus
sage: T
Simplicial complex with 9 vertices and 18 facets
sage: T.homology()
{0: 0, 1: Z x Z, 2: Z}
```

These can get large pretty quickly:

```
sage: T = simplicial_complexes.Torus(); T
Minimal triangulation of the torus
sage: K = simplicial_complexes.KleinBottle(); K
Minimal triangulation of the Klein bottle
sage: T.product(K) # long time: 5 or 6 seconds
Simplicial complex with 56 vertices and 1344 facets
```

**remove\_face** (*face*, *check=False*)

Remove a face from this simplicial complex.

#### Parameters

- **face** – a face of the simplicial complex
- **check** – boolean; optional, default `False`. If `True`, raise an error if `face` is not a face of this simplicial complex

This does not return anything; instead, it *changes* the simplicial complex.

ALGORITHM:

The facets of the new simplicial complex are the facets of the original complex not containing face, together with those of `link(face) * boundary(face)`.

EXAMPLES:

```
sage: S = range(1,5)
sage: Z = SimplicialComplex([S]); Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(1, 2, 3, 4)}
sage: Z.remove_face([1,2])
sage: Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(1, 3, 4), (2, 3, 4)}

sage: S = SimplicialComplex([[0,1,2],[2,3]])
sage: S
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(2, 3), (0, 1, 2)}
sage: S.remove_face([0,1,2])
sage: S
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1), (0, 2), (1, 2), (2, 3)}
```

**remove\_faces** (*faces*, *check=False*)

Remove a collection of faces from this simplicial complex.

**Parameters**

- **faces** – a list (or any iterable) of faces of the simplicial complex
- **check** – boolean; optional, default `False`. If `True`, raise an error if any element of `faces` is not a face of this simplicial complex

This does not return anything; instead, it *changes* the simplicial complex.

ALGORITHM:

Run `self.remove_face(f)` repeatedly, for `f` in `faces`.

EXAMPLES:

```
sage: S = range(1,5)
sage: Z = SimplicialComplex([S]); Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(1, 2, 3, 4)}
sage: Z.remove_faces([[1,2]])
sage: Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(1, 3, 4), (2, 3, 4)}

sage: Z = SimplicialComplex([S]); Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(1, 2, 3, 4)}
sage: Z.remove_faces([[1,2], [2,3]])
sage: Z
Simplicial complex with vertex set (1, 2, 3, 4) and facets {(2, 4), (1, 3, 4)}
```

**restriction\_sets** (*order*)

Return the restriction sets of the facets according to *order*.

A restriction set of a shelling order is the sequence of smallest new faces that are created during the shelling order.

**See also:**

`is_shelling_order()`

EXAMPLES:

```
sage: facets = [[1,2,5], [2,3,5], [3,4,5], [1,4,5]]
sage: X = SimplicialComplex(facets)
sage: X.restriction_sets(facets)
[(), (3,), (4,), (1, 4)]

sage: b = [[1,2,5], [3,4,5], [2,3,5], [1,4,5]]
sage: X.restriction_sets(b)
Traceback (most recent call last):
...
ValueError: not a shelling order
```

**set\_immutable()**

Make this simplicial complex immutable.

EXAMPLES:

```
sage: S = SimplicialComplex([[1,4], [2,4]])
sage: S.is_mutable()
True
sage: S.set_immutable()
sage: S.is_mutable()
False
```

**stanley\_reisner\_ring** (*base\_ring=Integer Ring*)

The Stanley-Reisner ring of this simplicial complex.

**Parameters** **base\_ring** (optional, default  $\mathbb{Z}\mathbb{Z}$ ) – a commutative ring

**Returns** a quotient of a polynomial algebra with coefficients in *base\_ring*, with one generator for each vertex in the simplicial complex, by the ideal generated by the products of those vertices which do not form faces in it.

Thus the ideal is generated by the products corresponding to the minimal nonfaces of the simplicial complex.

**Warning:** This may be quite slow!

Also, this may behave badly if the vertices have the ‘wrong’ names. To avoid this, define the simplicial complex at the start with the flag *name\_check* set to *True*.

More precisely, this is a quotient of a polynomial ring with one generator for each vertex. If the name of a vertex is a non-negative integer, then the corresponding polynomial generator is named ‘x’ followed by that integer (e.g., ‘x2’, ‘x3’, ‘x5’, ...). Otherwise, the polynomial generators are given the same names as the vertices. Thus if the vertex set is (2, ‘x2’), there will be problems.

EXAMPLES:

```
sage: X = SimplicialComplex([[0,1,2], [0,2,3]])
sage: X.stanley_reisner_ring()
Quotient of Multivariate Polynomial Ring in x0, x1, x2, x3 over Integer Ring
↳ by the ideal (x1*x3)
sage: Y = SimplicialComplex([[0,1,2,3,4]]); Y
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 2, 3,
↳ 4)}
sage: Y.add_face([0,1,2,3,4])
```

(continues on next page)

(continued from previous page)

```
sage: Y.stanley_reisner_ring(base_ring=QQ)
Multivariate Polynomial Ring in x0, x1, x2, x3, x4 over Rational Field
```

**star** (*simplex, is\_mutable=True*)

Return the star of a simplex in this simplicial complex.

The star of *simplex* is the simplicial complex formed by all simplices which contain *simplex*.

INPUT:

- *simplex* – a simplex in this simplicial complex
- *is\_mutable* – (default: True) boolean; determines if the output is mutable

EXAMPLES:

```
sage: X = SimplicialComplex([[0,1,2], [1,2,3]])
sage: X.star(Simplex([0]))
Simplicial complex with vertex set (0, 1, 2) and facets {(0, 1, 2)}
sage: X.star(Simplex([1]))
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1, 2), (1, 2, 3)}
sage: X.star(Simplex([1,2]))
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1, 2), (1, 2, 3)}
sage: X.star(Simplex([]))
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1, 2), (1, 2, 3)}
```

**stellar\_subdivision** (*simplex, inplace=False, is\_mutable=True*)

Return the stellar subdivision of a simplex in this simplicial complex.

The stellar subdivision of a face is obtained by adding a new vertex to the simplicial complex *self* joined to the star of the face and then deleting the face *simplex* to the result.

INPUT:

- *simplex* – a simplex face of *self*
- *inplace* – (default: False) boolean; determines if the operation is done on *self* or on a copy
- *is\_mutable* – (default: True) boolean; determines if the output is mutable

OUTPUT:

- A simplicial complex obtained by the stellar subdivision of the face *simplex*

EXAMPLES:

```
sage: SC = SimplicialComplex([[0,1,2], [1,2,3]])
sage: F1 = Simplex([1,2])
sage: F2 = Simplex([1,3])
sage: F3 = Simplex([1,2,3])
sage: SC.stellar_subdivision(F1)
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 4), (0, 2, 4), (1, 3, 4), (2, 3, 4)}
sage: SC.stellar_subdivision(F2)
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 2), (1, 2, 4), (2, 3, 4)}
sage: SC.stellar_subdivision(F3)
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 2), (1, 2, 4), (1, 3, 4), (2, 3, 4)}
```

(continues on next page)

(continued from previous page)

```
sage: SC.stellar_subdivision(F3, inplace=True); SC
Simplicial complex with vertex set (0, 1, 2, 3, 4) and facets {(0, 1, 2), (1, 2, 3), (2, 3, 4), (1, 3, 4), (2, 3, 4)}
```

The simplex to subdivide should be a face of self:

```
sage: SC = SimplicialComplex([[0,1,2],[1,2,3]])
sage: F4 = Simplex([3,4])
sage: SC.stellar_subdivision(F4)
Traceback (most recent call last):
...
ValueError: the face to subdivide is not a face of self
```

One can not modify an immutable simplicial complex:

```
sage: SC = SimplicialComplex([[0,1,2],[1,2,3]], is_mutable=False)
sage: SC.stellar_subdivision(F1, inplace=True)
Traceback (most recent call last):
...
ValueError: this simplicial complex is not mutable
```

**suspension** ( $n=1$ ,  $is\_mutable=True$ )

The suspension of this simplicial complex.

#### Parameters

- **n** (*optional*, *default 1*) – positive integer – suspend this many times.
- **is\_mutable** (boolean; *optional*, *default True*) – Determines if the output is mutable

The suspension is the simplicial complex formed by adding two new vertices  $S_0$  and  $S_1$  and simplices of the form  $[S_0, v_0, \dots, v_k]$  and  $[S_1, v_0, \dots, v_k]$  for every simplex  $[v_0, \dots, v_k]$  in the original simplicial complex. That is, the suspension is the join of the original complex with a two-point simplicial complex.

If the simplicial complex  $M$  happens to be a pseudomanifold (see `is_pseudomanifold()`), then this instead constructs Datta’s one-point suspension (see [Dat2007], p. 434): choose a vertex  $u$  in  $M$  and choose a new vertex  $w$  to add. Denote the join of simplices by “\*”. The facets in the one-point suspension are of the two forms

- $u * \alpha$  where  $\alpha$  is a facet of  $M$  not containing  $u$
- $w * \beta$  where  $\beta$  is any facet of  $M$ .

#### EXAMPLES:

```
sage: S0 = SimplicialComplex([[0], [1]])
sage: S0.suspension() == simplicial_complexes.Sphere(1)
True
sage: S3 = S0.suspension(3) # the 3-sphere
sage: S3.homology()
{0: 0, 1: 0, 2: 0, 3: Z}
```

For pseudomanifolds, the complex constructed here will be smaller than that obtained by taking the join with the 0-sphere: the join adds two vertices, while this construction only adds one.

```
sage: T = simplicial_complexes.Torus()
sage: sorted(T.join(S0).vertices()) # 9 vertices
['L0', 'L1', 'L2', 'L3', 'L4', 'L5', 'L6', 'R0', 'R1']
```

(continues on next page)

(continued from previous page)

```
sage: T.suspension().vertices() # 8 vertices
(0, 1, 2, 3, 4, 5, 6, 7)
```

**vertices()**

The vertex set, as a tuple, of this simplicial complex.

**EXAMPLES:**

```
sage: S = SimplicialComplex([[i] for i in range(16)] + [[0,1], [1,2]])
sage: S
Simplicial complex with 16 vertices and 15 facets
sage: sorted(S.vertices())
[0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15]
```

**wedge (right, rename\_vertices=True, is\_mutable=True)**

The wedge (one-point union) of this simplicial complex with another one.

**Parameters**

- **right** – the other simplicial complex (the right-hand factor)
- **rename\_vertices** (boolean; optional, default `True`) – If this is `True`, the vertices in the wedge will be renamed by the formula: first vertex in each are glued together and called “0”. Otherwise, each vertex “v” in the left-hand factor → vertex “Lv” in the wedge, vertex “w” in the right-hand factor → vertex “Rw” in the wedge. If this is `False`, this tries to construct the wedge without renaming the vertices; this will cause problems if the two factors have any vertices with names in common.
- **is\_mutable** (boolean; optional, default `True`) – Determines if the output is mutable

---

**Note:** This operation is not well-defined if `self` or `other` is not path-connected.

---

**EXAMPLES:**

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: S2 = simplicial_complexes.Sphere(2)
sage: S1.wedge(S2).homology()
{0: 0, 1: Z, 2: Z}
```

`sage.homology.simplicial_complex.facets_for_K3()`

Returns the facets for a minimal triangulation of the K3 surface.

This is a pure simplicial complex of dimension 4 with 16 vertices and 288 facets. The facets are obtained by constructing a few facets and a permutation group  $G$ , and then computing the  $G$ -orbit of those facets.

See Casella and Kühnel in [CK2001] and Spreer and Kühnel [SK2011]; the construction here uses the labeling from Spreer and Kühnel.

**EXAMPLES:**

```
sage: from sage.homology.simplicial_complex import facets_for_K3
sage: A = facets_for_K3() # long time (a few seconds)
sage: SimplicialComplex(A) == simplicial_complexes.K3Surface() # long time
True
```

`sage.homology.simplicial_complex.facets_for_RP4()`

Return the list of facets for a minimal triangulation of 4-dimensional real projective space.

We use vertices numbered 1 through 16, define two facets, and define a certain subgroup  $G$  of the symmetric group  $S_{16}$ . Then the set of all facets is the  $G$ -orbit of the two given facets.

See the description in Example 3.12 in Datta [Dat2007].

EXAMPLES:

```
sage: from sage.homology.simplicial_complex import facets_for_RP4
sage: A = facets_for_RP4()      # long time (1 or 2 seconds)
sage: SimplicialComplex(A) == simplicial_complexes.RealProjectiveSpace(4) # long_
↪time
True
```

`sage.homology.simplicial_complex.lattice_paths(t1, t2, length=None)`

Given lists (or tuples or ...) *t1* and *t2*, think of them as labelings for vertices: *t1* labeling points on the x-axis, *t2* labeling points on the y-axis, both increasing. Return the list of rectilinear paths along the grid defined by these points in the plane, starting from (*t1*[0], *t2*[0]), ending at (*t1*[last], *t2*[last]), and at each grid point, going either right or up. See the examples.

#### Parameters

- **t1** (*list, other iterable*) – labeling for vertices
- **t2** (*list, other iterable*) – labeling for vertices
- **length** (integer or None; optional, default None) – if not None, then an integer, the length of the desired path.

**Returns** list of lists of vertices making up the paths as described above

**Return type** list of lists

This is used when triangulating the product of simplices. The optional argument *length* is used for  $\Delta$ -complexes, to specify all simplices in a product: in the triangulation of a product of two simplices, there is a  $d$ -simplex for every path of length  $d+1$  in the lattice. The path must start at the bottom left and end at the upper right, and it must use at least one point in each row and in each column, so if *length* is too small, there will be no paths.

EXAMPLES:

```
sage: from sage.homology.simplicial_complex import lattice_paths
sage: lattice_paths([0,1,2], [0,1,2])
[[ (0, 0), (0, 1), (0, 2), (1, 2), (2, 2)],
  [(0, 0), (0, 1), (1, 1), (1, 2), (2, 2)],
  [(0, 0), (1, 0), (1, 1), (1, 2), (2, 2)],
  [(0, 0), (0, 1), (1, 1), (2, 1), (2, 2)],
  [(0, 0), (1, 0), (1, 1), (2, 1), (2, 2)],
  [(0, 0), (1, 0), (2, 0), (2, 1), (2, 2)]]
sage: lattice_paths('a', 'b', 'c'), (0, 3, 5)
[[ ('a', 0), ('a', 3), ('a', 5), ('b', 5), ('c', 5)],
  [('a', 0), ('a', 3), ('b', 3), ('b', 5), ('c', 5)],
  [('a', 0), ('b', 0), ('b', 3), ('b', 5), ('c', 5)],
  [('a', 0), ('a', 3), ('b', 3), ('c', 3), ('c', 5)],
  [('a', 0), ('b', 0), ('b', 3), ('c', 3), ('c', 5)],
  [('a', 0), ('b', 0), ('c', 0), ('c', 3), ('c', 5)]]
sage: lattice_paths(range(3), range(3), length=2)
[]
sage: lattice_paths(range(3), range(3), length=3)
[[ (0, 0), (1, 1), (2, 2)]]
sage: lattice_paths(range(3), range(3), length=4)
[[ (0, 0), (1, 1), (1, 2), (2, 2)]]
```

(continues on next page)



(continued from previous page)

```
[ (0, 0), (0, 1), (1, 2), (2, 2)],
[ (0, 0), (1, 1), (2, 1), (2, 2)],
[ (0, 0), (1, 0), (2, 1), (2, 2)],
[ (0, 0), (0, 1), (1, 1), (2, 2)],
[ (0, 0), (1, 0), (1, 1), (2, 2)]]
```

`sage.homology.simplicial_complex.rename_vertex(n, keep, left=True)`

Rename a vertex: the vertices from the list `keep` get relabeled 0, 1, 2, ..., in order. Any other vertex (e.g. 4) gets renamed to by prepending an 'L' or an 'R' (thus to either 'L4' or 'R4'), depending on whether the argument `left` is `True` or `False`.

#### Parameters

- **n** – a 'vertex': either an integer or a string
- **keep** – a list of three vertices
- **left** (boolean; optional, default `True`) – if `True`, rename for use in left factor

This is used by the `connected_sum()` method for simplicial complexes.

#### EXAMPLES:

```
sage: from sage.homology.simplicial_complex import rename_vertex
sage: rename_vertex(6, [5, 6, 7])
1
sage: rename_vertex(3, [5, 6, 7, 8, 9])
'L3'
sage: rename_vertex(3, [5, 6, 7], left=False)
'R3'
```



## MORPHISMS OF SIMPLICIAL COMPLEXES

### AUTHORS:

- Benjamin Antieau <[d.ben.antieau@gmail.com](mailto:d.ben.antieau@gmail.com)> (2009.06)
- Travis Scrimshaw (2012-08-18): Made all simplicial complexes immutable to work with the homset cache.

This module implements morphisms of simplicial complexes. The input is given by a dictionary on the vertex set of a simplicial complex. The initialization checks that faces are sent to faces.

There is also the capability to create the fiber product of two morphisms with the same codomain.

### EXAMPLES:

```
sage: S = SimplicialComplex([[0,2],[1,5],[3,4]], is_mutable=False)
sage: H = Hom(S,S.product(S, is_mutable=False))
sage: H.diagonal_morphism()
Simplicial complex morphism:
  From: Simplicial complex with vertex set (0, 1, 2, 3, 4, 5) and facets {(0, 2), (1, 5), (3, 4)}
  To: Simplicial complex with 36 vertices and 18 facets
  Defn: [0, 1, 2, 3, 4, 5] --> ['L0R0', 'L1R1', 'L2R2', 'L3R3', 'L4R4', 'L5R5']

sage: S = SimplicialComplex([[0,2],[1,5],[3,4]], is_mutable=False)
sage: T = SimplicialComplex([[0,2],[1,3]], is_mutable=False)
sage: f = {0:0,1:1,2:2,3:1,4:3,5:3}
sage: H = Hom(S,T)
sage: x = H(f)
sage: x.image()
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 2), (1, 3)}
sage: x.is_surjective()
True
sage: x.is_injective()
False
sage: x.is_identity()
False

sage: S = simplicial_complexes.Sphere(2)
sage: H = Hom(S,S)
sage: i = H.identity()
sage: i.image()
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(0, 1, 2), (0, 1, 3), (0, 2, 3), (1, 2, 3)}
sage: i.is_surjective()
True
sage: i.is_injective()
True
```

(continues on next page)

(continued from previous page)

```

sage: i.is_identity()
True

sage: S = simplicial_complexes.Sphere(2)
sage: H = Hom(S, S)
sage: i = H.identity()
sage: j = i.fiber_product(i)
sage: j
Simplicial complex morphism:
  From: Simplicial complex with 4 vertices and 4 facets
  To:   Minimal triangulation of the 2-sphere
  Defn: L0R0 |--> 0
        L1R1 |--> 1
        L2R2 |--> 2
        L3R3 |--> 3

sage: S = simplicial_complexes.Sphere(2)
sage: T = S.product(SimplicialComplex([[0, 1]]), rename_vertices = False, is_
↳mutable=False)
sage: H = Hom(T, S)
sage: T
Simplicial complex with 8 vertices and 12 facets
sage: sorted(T.vertices())
[(0, 0), (0, 1), (1, 0), (1, 1), (2, 0), (2, 1), (3, 0), (3, 1)]
sage: f = {(0, 0): 0, (0, 1): 0, (1, 0): 1, (1, 1): 1, (2, 0): 2, (2, 1): 2, (3, 0):
↳3, (3, 1): 3}
sage: x = H(f)
sage: U = simplicial_complexes.Sphere(1)
sage: G = Hom(U, S)
sage: U
Minimal triangulation of the 1-sphere
sage: g = {0:0, 1:1, 2:2}
sage: y = G(g)
sage: z = y.fiber_product(x)
sage: z
# this is the mapping path space

Simplicial complex morphism:
  From: Simplicial complex with 6 vertices and 4 facets
  To:   Minimal triangulation of the 2-sphere
  Defn: ['L0R(0, 0)', 'L0R(0, 1)', 'L1R(1, 0)', 'L1R(1, 1)', 'L2R(2, 0)', 'L2R(2, 1)
↳'] --> [0, 0, 1, 1, 2, 2]

```

```

class sage.homology.simplicial_complex_morphism.SimplicialComplexMorphism(f,
                                                                              X,
                                                                              Y)

```

Bases: `sage.categories.morphism.Morphism`

An element of this class is a morphism of simplicial complexes.

```

associated_chain_complex_morphism(base_ring=Integer Ring, augmented=False,
                                  cochain=False)

```

Returns the associated chain complex morphism of `self`.

EXAMPLES:

```

sage: S = simplicial_complexes.Sphere(1)
sage: T = simplicial_complexes.Sphere(2)
sage: H = Hom(S, T)
sage: f = {0:0, 1:1, 2:2}
sage: x = H(f)

```

(continues on next page)

(continued from previous page)

```

sage: x
Simplicial complex morphism:
  From: Minimal triangulation of the 1-sphere
  To:   Minimal triangulation of the 2-sphere
  Defn: 0 |--> 0
        1 |--> 1
        2 |--> 2
sage: a = x.associated_chain_complex_morphism()
sage: a
Chain complex morphism:
  From: Chain complex with at most 2 nonzero terms over Integer Ring
  To:   Chain complex with at most 3 nonzero terms over Integer Ring
sage: a._matrix_dictionary
{0: [1 0 0]
 [0 1 0]
 [0 0 1]
 [0 0 0], 1: [1 0 0]
 [0 1 0]
 [0 0 0]
 [0 0 1]
 [0 0 0]
 [0 0 0], 2: []}
sage: x.associated_chain_complex_morphism(augmented=True)
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Integer Ring
  To:   Chain complex with at most 4 nonzero terms over Integer Ring
sage: x.associated_chain_complex_morphism(cochain=True)
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Integer Ring
  To:   Chain complex with at most 2 nonzero terms over Integer Ring
sage: x.associated_chain_complex_morphism(augmented=True, cochain=True)
Chain complex morphism:
  From: Chain complex with at most 4 nonzero terms over Integer Ring
  To:   Chain complex with at most 3 nonzero terms over Integer Ring
sage: x.associated_chain_complex_morphism(base_ring=GF(11))
Chain complex morphism:
  From: Chain complex with at most 2 nonzero terms over Finite Field of size_
↪ 11
  To:   Chain complex with at most 3 nonzero terms over Finite Field of size_
↪ 11

```

Some simplicial maps which reverse the orientation of a few simplices:

```

sage: g = {0:1, 1:2, 2:0}
sage: H(g).associated_chain_complex_morphism()._matrix_dictionary
{0: [0 0 1]
 [1 0 0]
 [0 1 0]
 [0 0 0], 1: [ 0 -1  0]
 [ 0  0 -1]
 [ 0  0  0]
 [ 1  0  0]
 [ 0  0  0]
 [ 0  0  0], 2: []}
sage: X = SimplicialComplex([[0, 1]], is_mutable=False)
sage: Hom(X, X)({0:1, 1:0}).associated_chain_complex_morphism()._matrix_
↪ dictionary

```

(continues on next page)

(continued from previous page)

```
{0: [0 1]
 [1 0], 1: [-1]}
```

**fiber\_product** (*other*, *rename\_vertices=True*)

Fiber product of *self* and *other*. Both morphisms should have the same codomain. The method returns a morphism of simplicial complexes, which is the morphism from the space of the fiber product to the codomain.

## EXAMPLES:

```
sage: S = SimplicialComplex([[0,1],[1,2]], is_mutable=False)
sage: T = SimplicialComplex([[0,2],[1]], is_mutable=False)
sage: U = SimplicialComplex([[0,1],[2]], is_mutable=False)
sage: H = Hom(S,U)
sage: G = Hom(T,U)
sage: f = {0:0,1:1,2:0}
sage: g = {0:0,1:1,2:1}
sage: x = H(f)
sage: y = G(g)
sage: z = x.fiber_product(y)
sage: z
Simplicial complex morphism:
  From: Simplicial complex with 4 vertices and facets {...}
  To:   Simplicial complex with vertex set (0, 1, 2) and facets {(2,), (0, 1)}
  Defn: L0R0 |--> 0
        L1R1 |--> 1
        L1R2 |--> 1
        L2R0 |--> 0
```

**image** ()

Computes the image simplicial complex of *f*.

## EXAMPLES:

```
sage: S = SimplicialComplex([[0,1],[2,3]], is_mutable=False)
sage: T = SimplicialComplex([[0,1]], is_mutable=False)
sage: f = {0:0,1:1,2:0,3:1}
sage: H = Hom(S,T)
sage: x = H(f)
sage: x.image()
Simplicial complex with vertex set (0, 1) and facets {(0, 1)}

sage: S = SimplicialComplex(is_mutable=False)
sage: H = Hom(S,S)
sage: i = H.identity()
sage: i.image()
Simplicial complex with vertex set () and facets {}
sage: i.is_surjective()
True

sage: S = SimplicialComplex([[0,1]], is_mutable=False)
sage: T = SimplicialComplex([[0,1],[0,2]], is_mutable=False)
sage: f = {0:0,1:1}
sage: g = {0:0,1:1}
sage: k = {0:0,1:2}
sage: H = Hom(S,T)
sage: x = H(f)
sage: y = H(g)
```

(continues on next page)

(continued from previous page)

```

sage: z = H(k)
sage: x == y
True
sage: x == z
False
sage: x.image()
Simplicial complex with vertex set (0, 1) and facets {(0, 1)}
sage: y.image()
Simplicial complex with vertex set (0, 1) and facets {(0, 1)}
sage: z.image()
Simplicial complex with vertex set (0, 2) and facets {(0, 2)}

```

**induced\_homology\_morphism** (*base\_ring=None, cohomology=False*)

The map in (co)homology induced by this map

INPUT:

- *base\_ring* – must be a field (optional, default `QQ`)
- *cohomology* – boolean (optional, default `False`). If `True`, the map induced in cohomology rather than homology.

EXAMPLES:

```

sage: S = simplicial_complexes.Sphere(1)
sage: T = S.product(S, is_mutable=False)
sage: H = Hom(S, T)
sage: diag = H.diagonal_morphism()
sage: h = diag.induced_homology_morphism(QQ)
sage: h
Graded vector space morphism:
  From: Homology module of Minimal triangulation of the 1-sphere over_
↳ Rational Field
  To:   Homology module of Simplicial complex with 9 vertices and 18 facets_
↳ over Rational Field
Defn: induced by:
  Simplicial complex morphism:
    From: Minimal triangulation of the 1-sphere
    To:   Simplicial complex with 9 vertices and 18 facets
  Defn: 0 |--> L0R0
        1 |--> L1R1
        2 |--> L2R2

```

We can view the matrix form for the homomorphism:

```

sage: h.to_matrix(0) # in degree 0
[1]
sage: h.to_matrix(1) # in degree 1
[1]
[1]
sage: h.to_matrix() # the entire homomorphism
[1|0]
[-+-]
[0|1]
[0|1]
[-+-]
[0|0]

```

The map on cohomology should be dual to the map on homology:

```

sage: coh = diag.induced_homology_morphism(QQ, cohomology=True)
sage: coh.to_matrix(1)
[1 1]
sage: h.to_matrix() == coh.to_matrix().transpose()
True

```

We can evaluate the map on (co)homology classes:

```

sage: x,y = list(T.cohomology_ring(QQ).basis(1))
sage: coh(x)
h^{1,0}
sage: coh(2*x+3*y)
5*h^{1,0}

```

Note that the complexes must be immutable for this to work. Many, but not all, complexes are immutable when constructed:

```

sage: S.is_immutable()
True
sage: S.barycentric_subdivision().is_immutable()
False
sage: S2 = S.suspension()
sage: S2.is_immutable()
False
sage: h = Hom(S,S2)({0: 0, 1:1, 2:2}).induced_homology_morphism()
Traceback (most recent call last):
...
ValueError: the domain and codomain complexes must be immutable
sage: S2.set_immutable(); S2.is_immutable()
True
sage: h = Hom(S,S2)({0: 0, 1:1, 2:2}).induced_homology_morphism()

```

### `is_contiguous_to(other)`

Return True if self is contiguous to other.

Two morphisms  $f_0, f_1 : K \rightarrow L$  are *contiguous* if for any simplex  $\sigma \in K$ , the union  $f_0(\sigma) \cup f_1(\sigma)$  is a simplex in  $L$ . This is not a transitive relation, but it induces an equivalence relation on simplicial maps:  $f$  is equivalent to  $g$  if there is a finite sequence  $f_0 = f, f_1, \dots, f_n = g$  such that  $f_i$  and  $f_{i+1}$  are contiguous for each  $i$ .

This is related to maps being homotopic: if they are contiguous, then they induce homotopic maps on the geometric realizations. Given two homotopic maps on the geometric realizations, then after barycentrically subdividing  $n$  times for some  $n$ , the maps have simplicial approximations which are in the same contiguity class. (This last fact is only true if the domain is a *finite* simplicial complex, by the way.)

See Section 3.5 of Spanier [Spa1966] for details.

### ALGORITHM:

It is enough to check when  $\sigma$  ranges over the facets.

### INPUT:

- `other` – a simplicial complex morphism with the same domain and codomain as `self`

### EXAMPLES:

```

sage: K = simplicial_complexes.Simplex(1)
sage: L = simplicial_complexes.Sphere(1)

```

(continues on next page)



(continued from previous page)

```

sage: H = Hom(K, L)
sage: f = H({0: 0, 1: 1})
sage: g = H({0: 0, 1: 0})
sage: f.is_contiguous_to(f)
True
sage: f.is_contiguous_to(g)
True
sage: h = H({0: 1, 1: 2})
sage: f.is_contiguous_to(h)
False

```

**is\_identity()**

If self is an identity morphism, returns True. Otherwise, False.

EXAMPLES:

```

sage: T = simplicial_complexes.Sphere(1)
sage: G = Hom(T, T)
sage: T
Minimal triangulation of the 1-sphere
sage: j = G({0:0,1:1,2:2})
sage: j.is_identity()
True

sage: S = simplicial_complexes.Sphere(2)
sage: T = simplicial_complexes.Sphere(3)
sage: H = Hom(S, T)
sage: f = {0:0,1:1,2:2,3:3}
sage: x = H(f)
sage: x
Simplicial complex morphism:
  From: Minimal triangulation of the 2-sphere
  To:   Minimal triangulation of the 3-sphere
  Defn: 0 |--> 0
        1 |--> 1
        2 |--> 2
        3 |--> 3
sage: x.is_identity()
False

```

**is\_injective()**

Returns True if and only if self is injective.

EXAMPLES:

```

sage: S = simplicial_complexes.Sphere(1)
sage: T = simplicial_complexes.Sphere(2)
sage: U = simplicial_complexes.Sphere(3)
sage: H = Hom(T, S)
sage: G = Hom(T, U)
sage: f = {0:0,1:1,2:0,3:1}
sage: x = H(f)
sage: g = {0:0,1:1,2:2,3:3}
sage: y = G(g)
sage: x.is_injective()
False
sage: y.is_injective()

```

(continues on next page)

(continued from previous page)

True

**is\_surjective()**

Returns True if and only if self is surjective.

EXAMPLES:

```

sage: S = SimplicialComplex([(0,1,2)], is_mutable=False)
sage: S
Simplicial complex with vertex set (0, 1, 2) and facets {(0, 1, 2)}
sage: T = SimplicialComplex([(0,1)], is_mutable=False)
sage: T
Simplicial complex with vertex set (0, 1) and facets {(0, 1)}
sage: H = Hom(S,T)
sage: x = H({0:0,1:1,2:1})
sage: x.is_surjective()
True

sage: S = SimplicialComplex([[0,1],[2,3]], is_mutable=False)
sage: T = SimplicialComplex([[0,1]], is_mutable=False)
sage: f = {0:0,1:1,2:0,3:1}
sage: H = Hom(S,T)
sage: x = H(f)
sage: x.is_surjective()
True

```

**mapping\_torus()**

The mapping torus of a simplicial complex endomorphism

The mapping torus is the simplicial complex formed by taking the product of the domain of self with a 4 point interval  $[I_0, I_1, I_2, I_3]$  and identifying vertices of the form  $(I_0, v)$  with  $(I_3, w)$  where  $w$  is the image of  $v$  under the given morphism.

See [Wikipedia article Mapping torus](#)

EXAMPLES:

```

sage: C = simplicial_complexes.Sphere(1) # Circle
sage: T = Hom(C,C).identity().mapping_torus() ; T # Torus
Simplicial complex with 9 vertices and 18 facets
sage: T.homology() == simplicial_complexes.Torus().homology()
True

sage: f = Hom(C,C)({0:0,1:2,2:1})
sage: K = f.mapping_torus() ; K # Klein Bottle
Simplicial complex with 9 vertices and 18 facets
sage: K.homology() == simplicial_complexes.KleinBottle().homology()
True

```

sage.homology.simplicial\_complex\_morphism.is\_SimplicialComplexMorphism(x)

Returns True if and only if x is a morphism of simplicial complexes.

EXAMPLES:

```

sage: from sage.homology.simplicial_complex_morphism import is_
↪SimplicialComplexMorphism
sage: S = SimplicialComplex([[0,1],[3,4]], is_mutable=False)
sage: H = Hom(S,S)

```

(continues on next page)

(continued from previous page)

```
sage: f = {0:0,1:1,3:3,4:4}
sage: x = H(f)
sage: is_SimplicialComplexMorphism(x)
True
```



## HOMSETS BETWEEN SIMPLICIAL COMPLEXES

### AUTHORS:

- Travis Scrimshaw (2012-08-18): Made all simplicial complexes immutable to work with the homset cache.

### EXAMPLES:

```
sage: S = simplicial_complexes.Sphere(1)
sage: T = simplicial_complexes.Sphere(2)
sage: H = Hom(S, T)
sage: f = {0:0, 1:1, 2:3}
sage: x = H(f)
sage: x
Simplicial complex morphism:
  From: Minimal triangulation of the 1-sphere
  To: Minimal triangulation of the 2-sphere
  Defn: 0 |--> 0
        1 |--> 1
        2 |--> 3
sage: x.is_injective()
True
sage: x.is_surjective()
False
sage: x.image()
Simplicial complex with vertex set (0, 1, 3) and facets {(0, 1), (0, 3), (1, 3)}
sage: from sage.homology.simplicial_complex import Simplex
sage: s = Simplex([1, 2])
sage: x(s)
(1, 3)
```

```
class sage.homology.simplicial_complex_homset.SimplicialComplexHomset(X, Y,
                                                                    cate-
                                                                    gory=None,
                                                                    base=None,
                                                                    check=True)
```

Bases: `sage.categories.homset.Homset`

**an\_element()**

Return a (non-random) element of self.

### EXAMPLES:

```
sage: S = simplicial_complexes.KleinBottle()
sage: T = simplicial_complexes.Sphere(5)
sage: H = Hom(S, T)
sage: x = H.an_element()
```

(continues on next page)

(continued from previous page)

```

sage: x
Simplicial complex morphism:
  From: Minimal triangulation of the Klein bottle
  To:   Minimal triangulation of the 5-sphere
  Defn: [0, 1, 2, 3, 4, 5, 6, 7] --> [0, 0, 0, 0, 0, 0, 0, 0]

```

**diagonal\_morphism**(*rename\_vertices=True*)Return the diagonal morphism in  $\text{Hom}(S, S \times S)$ .

EXAMPLES:

```

sage: S = simplicial_complexes.Sphere(2)
sage: H = Hom(S, S.product(S, is_mutable=False))
sage: d = H.diagonal_morphism()
sage: d
Simplicial complex morphism:
  From: Minimal triangulation of the 2-sphere
  To:   Simplicial complex with 16 vertices and 96 facets
  Defn: 0 |--> L0R0
        1 |--> L1R1
        2 |--> L2R2
        3 |--> L3R3

sage: T = SimplicialComplex([[0], [1]], is_mutable=False)
sage: U = T.product(T, rename_vertices = False, is_mutable=False)
sage: G = Hom(T, U)
sage: e = G.diagonal_morphism(rename_vertices = False)
sage: e
Simplicial complex morphism:
  From: Simplicial complex with vertex set (0, 1) and facets {(0,), (1,)}
  To:   Simplicial complex with 4 vertices and facets {(0, 0), (0, 1), (1, 0), (1, 1)}
  Defn: 0 |--> (0, 0)
        1 |--> (1, 1)

```

**identity**()Return the identity morphism of  $\text{Hom}(S, S)$ .

EXAMPLES:

```

sage: S = simplicial_complexes.Sphere(2)
sage: H = Hom(S, S)
sage: i = H.identity()
sage: i.is_identity()
True

sage: T = SimplicialComplex([[0, 1]], is_mutable=False)
sage: G = Hom(T, T)
sage: G.identity()
Simplicial complex endomorphism of Simplicial complex with vertex set (0, 1)
and facets {(0, 1)}
  Defn: 0 |--> 0
        1 |--> 1

```

**sage.homology.simplicial\_complex\_homset.is\_SimplicialComplexHomset**(*x*)Return True if and only if *x* is a simplicial complex homspace.

EXAMPLES:

```
sage: S = SimplicialComplex(is_mutable=False)
sage: T = SimplicialComplex(is_mutable=False)
sage: H = Hom(S, T)
sage: H
Set of Morphisms from Simplicial complex with vertex set () and facets {}
to Simplicial complex with vertex set () and facets {}
in Category of finite simplicial complexes
sage: from sage.homology.simplicial_complex_homset import is_
↪SimplicialComplexHomset
sage: is_SimplicialComplexHomset(H)
True
```





## EXAMPLES OF SIMPLICIAL COMPLEXES

There are two main types: manifolds and examples related to graph theory.

For manifolds, there are functions defining the  $n$ -sphere for any  $n$ , the torus,  $n$ -dimensional real projective space for any  $n$ , the complex projective plane, surfaces of arbitrary genus, and some other manifolds, all as simplicial complexes.

Aside from surfaces, this file also provides functions for constructing some other simplicial complexes: the simplicial complex of not- $i$ -connected graphs on  $n$  vertices, the matching complex on  $n$  vertices, the chessboard complex for an  $n$  by  $i$  chessboard, and others. These provide examples of large simplicial complexes; for example, `simplicial_complexes.NotIConnectedGraphs(7,2)` has over a million simplices.

All of these examples are accessible by typing `simplicial_complexes.NAME`, where `NAME` is the name of the example.

- `BarnetteSphere()`
- `BrucknerGrunbaumSphere()`
- `ChessboardComplex()`
- `ComplexProjectivePlane()`
- `DunceHat()`
- `K3Surface()`
- `KleinBottle()`
- `MatchingComplex()`
- `MooreSpace()`
- `NotIConnectedGraphs()`
- `PoincareHomologyThreeSphere()`
- `PseudoQuaternionicProjectivePlane()`
- `RandomComplex()`
- `RandomTwoSphere()`
- `RealProjectivePlane()`
- `RealProjectiveSpace()`
- `RudinBall()`
- `ShiftedComplex()`
- `Simplex()`
- `Sphere()`

- `SumComplex()`
- `SurfaceOfGenus()`
- `Torus()`
- `ZieglerBall()`

You can also get a list by typing `simplicial_complexes.` and hitting the TAB key.

EXAMPLES:

```
sage: S = simplicial_complexes.Sphere(2) # the 2-sphere
sage: S.homology()
{0: 0, 1: 0, 2: Z}
sage: simplicial_complexes.SurfaceOfGenus(3)
Triangulation of an orientable surface of genus 3
sage: M4 = simplicial_complexes.MooreSpace(4)
sage: M4.homology()
{0: 0, 1: C4, 2: 0}
sage: simplicial_complexes.MatchingComplex(6).homology()
{0: 0, 1: Z^16, 2: 0}
```

`sage.homology.examples.BarnetteSphere()`

Returns Barnette's triangulation of the 3-sphere.

This is a pure simplicial complex of dimension 3 with 8 vertices and 19 facets, which is a non-polytopal triangulation of the 3-sphere. It was constructed by Barnette in [Bar1970]. The construction here uses the labeling from De Loera, Rambau and Santos [DLRS2010]. Another reference is chapter III.4 of Ewald [Ewa1996].

EXAMPLES:

```
sage: BS = simplicial_complexes.BarnetteSphere() ; BS
Barnette's triangulation of the 3-sphere
sage: BS.f_vector()
[1, 8, 27, 38, 19]
```

`sage.homology.examples.BrucknerGrunbaumSphere()`

Returns Bruckner and Grunbaum's triangulation of the 3-sphere.

This is a pure simplicial complex of dimension 3 with 8 vertices and 20 facets, which is a non-polytopal triangulation of the 3-sphere. It appeared first in [Br1910] and was studied in [GrS1967].

It is defined here as the link of any vertex in the unique minimal triangulation of the complex projective plane, see chapter 4 of [Kuh1995].

EXAMPLES:

```
sage: BGS = simplicial_complexes.BrucknerGrunbaumSphere() ; BGS
Bruckner and Grunbaum's triangulation of the 3-sphere
sage: BGS.f_vector()
[1, 8, 28, 40, 20]
```

`sage.homology.examples.ChessboardComplex(n, i)`

The chessboard complex for an  $n \times i$  chessboard.

Fix integers  $n, i > 0$  and consider sets  $V$  of  $n$  vertices and  $W$  of  $i$  vertices. A 'partial matching' between  $V$  and  $W$  is a graph formed by edges  $(v, w)$  with  $v \in V$  and  $w \in W$  so that each vertex is in at most one edge. If  $G$  is a partial matching, then so is any graph obtained by deleting edges from  $G$ . Thus the set of all partial matchings on  $V$  and  $W$ , viewed as a set of subsets of the  $n + i$  choose 2 possible edges, is closed under taking subsets, and thus forms a simplicial complex called the 'chessboard complex'. This function produces that simplicial

complex. (It is called the chessboard complex because such graphs also correspond to ways of placing rooks on an  $n$  by  $i$  chessboard so that none of them are attacking each other.)

INPUT:

- $n, i$  – positive integers.

See Dumas et al. [DHSW2003] for information on computing its homology by computer, and see Wachs [Wac2003] for an expository article about the theory.

EXAMPLES:

```
sage: C = simplicial_complexes.ChessboardComplex(5,5)
sage: C.f_vector()
[1, 25, 200, 600, 600, 120]
sage: simplicial_complexes.ChessboardComplex(3,3).homology()
{0: 0, 1: Z x Z x Z x Z, 2: 0}
```

sage.homology.examples.**ComplexProjectivePlane**()

A minimal triangulation of the complex projective plane.

This was constructed by Kühnel and Banchoff [KB1983].

EXAMPLES:

```
sage: C = simplicial_complexes.ComplexProjectivePlane()
sage: C.f_vector()
[1, 9, 36, 84, 90, 36]
sage: C.homology(2)
Z
sage: C.homology(4)
Z
```

sage.homology.examples.**DunceHat**()

Return the minimal triangulation of the dunce hat given by Hachimori [Hac2016].

This is a standard example of a space that is contractible but not collapsible.

EXAMPLES:

```
sage: D = simplicial_complexes.DunceHat(); D
Minimal triangulation of the dunce hat
sage: D.f_vector()
[1, 8, 24, 17]
sage: D.homology()
{0: 0, 1: 0, 2: 0}
sage: D.is_cohen_macaulay()
True
```

sage.homology.examples.**K3Surface**()

Returns a minimal triangulation of the K3 surface.

This is a pure simplicial complex of dimension 4 with 16 vertices and 288 facets. It was constructed by Casella and Kühnel in [CK2001]. The construction here uses the labeling from Spreer and Kühnel [SK2011].

EXAMPLES:

```
sage: K3=simplicial_complexes.K3Surface() ; K3
Minimal triangulation of the K3 surface
sage: K3.f_vector()
[1, 16, 120, 560, 720, 288]
```

This simplicial complex is implemented just by listing all 288 facets. The list of facets can be computed by the function `facets_for_K3()`, but running the function takes a few seconds.

`sage.homology.examples.KleinBottle()`

A minimal triangulation of the Klein bottle, as presented for example in Davide Cervone's thesis [Cer1994].

EXAMPLES:

```
sage: simplicial_complexes.KleinBottle()
Minimal triangulation of the Klein bottle
```

`sage.homology.examples.MatchingComplex(n)`

The matching complex of graphs on  $n$  vertices.

Fix an integer  $n > 0$  and consider a set  $V$  of  $n$  vertices. A 'partial matching' on  $V$  is a graph formed by edges so that each vertex is in at most one edge. If  $G$  is a partial matching, then so is any graph obtained by deleting edges from  $G$ . Thus the set of all partial matchings on  $n$  vertices, viewed as a set of subsets of the  $n$  choose 2 possible edges, is closed under taking subsets, and thus forms a simplicial complex called the 'matching complex'. This function produces that simplicial complex.

INPUT:

- $n$  – positive integer.

See Dumas et al. [DHSW2003] for information on computing its homology by computer, and see Wachs [Wac2003] for an expository article about the theory. For example, the homology of these complexes seems to have only mod 3 torsion, and this has been proved for the bottom non-vanishing homology group for the matching complex  $M_n$ .

EXAMPLES:

```
sage: M = simplicial_complexes.MatchingComplex(7)
sage: H = M.homology()
sage: H
{0: 0, 1: C3, 2: Z^20}
sage: H[2].ngens()
20
sage: simplicial_complexes.MatchingComplex(8).homology(2) # long time (6s on
↳ sage.math, 2012)
Z^132
```

`sage.homology.examples.MooreSpace(q)`

Triangulation of the mod  $q$  Moore space.

INPUT:

- $q$  -0 integer, at least 2

This is a simplicial complex with simplices of dimension 0, 1, and 2, such that its reduced homology is isomorphic to  $\mathbf{Z}/q\mathbf{Z}$  in dimension 1, zero otherwise.

If  $q = 2$ , this is the real projective plane. If  $q > 2$ , then construct it as follows: start with a triangle with vertices 1, 2, 3. We take a  $3q$ -gon forming a  $q$ -fold cover of the triangle, and we form the resulting complex as an identification space of the  $3q$ -gon. To triangulate this identification space, put  $q$  vertices  $A_0, \dots, A_{q-1}$ , in the interior, each of which is connected to 1, 2, 3 (two facets each:  $[1, 2, A_i], [2, 3, A_i]$ ). Put  $q$  more vertices in the interior:  $B_0, \dots, B_{q-1}$ , with facets  $[3, 1, B_i], [3, B_i, A_i], [1, B_i, A_{i+1}], [B_i, A_i, A_{i+1}]$ . Then triangulate the interior polygon with vertices  $A_0, A_1, \dots, A_{q-1}$ .

EXAMPLES:

```

sage: simplicial_complexes.MooreSpace(2)
Minimal triangulation of the real projective plane
sage: simplicial_complexes.MooreSpace(3).homology()[1]
C3
sage: simplicial_complexes.MooreSpace(4).suspension().homology()[2]
C4
sage: simplicial_complexes.MooreSpace(8)
Triangulation of the mod 8 Moore space

```

sage.homology.examples.**NotIConnectedGraphs**( $n, i$ )

The simplicial complex of all graphs on  $n$  vertices which are not  $i$ -connected.

Fix an integer  $n > 0$  and consider the set of graphs on  $n$  vertices. View each graph as its set of edges, so it is a subset of a set of size  $n$  choose 2. A graph is  $i$ -connected if, for any  $j < i$ , if any  $j$  vertices are removed along with the edges emanating from them, then the graph remains connected. Now fix  $i$ : it is clear that if  $G$  is not  $i$ -connected, then the same is true for any graph obtained from  $G$  by deleting edges. Thus the set of all graphs which are not  $i$ -connected, viewed as a set of subsets of the  $n$  choose 2 possible edges, is closed under taking subsets, and thus forms a simplicial complex. This function produces that simplicial complex.

INPUT:

- $n, i$  – non-negative integers with  $i$  at most  $n$

See Dumas et al. [DHSW2003] for information on computing its homology by computer, and see Babson et al. [BBLSW1999] for theory. For example, Babson et al. show that when  $i = 2$ , the reduced homology of this complex is nonzero only in dimension  $2n - 5$ , where it is free abelian of rank  $(n - 2)!$ .

EXAMPLES:

```

sage: simplicial_complexes.NotIConnectedGraphs(5,2).f_vector()
[1, 10, 45, 120, 210, 240, 140, 20]
sage: simplicial_complexes.NotIConnectedGraphs(5,2).homology(5).ngens()
6

```

sage.homology.examples.**PoincareHomologyThreeSphere**()

A triangulation of the Poincaré homology 3-sphere.

This is a manifold whose integral homology is identical to the ordinary 3-sphere, but it is not simply connected. In particular, its fundamental group is the binary icosahedral group, which has order 120. The triangulation given here has 16 vertices and is due to Björner and Lutz [BL2000].

EXAMPLES:

```

sage: S3 = simplicial_complexes.Sphere(3)
sage: Sigma3 = simplicial_complexes.PoincareHomologyThreeSphere()
sage: S3.homology() == Sigma3.homology()
True
sage: Sigma3.fundamental_group().cardinality() # long time
120

```

sage.homology.examples.**ProjectivePlane**()

A minimal triangulation of the real projective plane.

EXAMPLES:

```

sage: P = simplicial_complexes.RealProjectivePlane()
sage: Q = simplicial_complexes.ProjectivePlane()
sage: P == Q
True

```

(continues on next page)

(continued from previous page)

```

sage: P.cohomology(1)
0
sage: P.cohomology(2)
C2
sage: P.cohomology(1, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2
sage: P.cohomology(2, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2

```

sage.homology.examples.**PseudoQuaternionicProjectivePlane**()

Returns a pure simplicial complex of dimension 8 with 490 facets.

**Warning:** This is expected to be a triangulation of the projective plane  $HP^2$  over the ring of quaternions, but this has not been proved yet.

This simplicial complex has the same homology as  $HP^2$ . Its automorphism group is isomorphic to the alternating group  $A_5$  and acts transitively on vertices.

This is defined here using the description in [BK1992]. This article deals with three different triangulations. This procedure returns the only one which has a transitive group of automorphisms.

EXAMPLES:

```

sage: HP2 = simplicial_complexes.PseudoQuaternionicProjectivePlane() ; HP2
Simplicial complex with 15 vertices and 490 facets
sage: HP2.f_vector()
[1, 15, 105, 455, 1365, 3003, 4515, 4230, 2205, 490]

```

Checking its automorphism group:

```

sage: HP2.automorphism_group().is_isomorphic(AlternatingGroup(5))
True

```

sage.homology.examples.**RandomComplex**( $n, d, p=0.5$ )

A random  $d$ -dimensional simplicial complex on  $n$  vertices.

INPUT:

- $n$  – number of vertices
- $d$  – dimension of the complex
- $p$  – floating point number between 0 and 1 (optional, default 0.5)

A random  $d$ -dimensional simplicial complex on  $n$  vertices, as defined for example by Meshulam and Wallach [MW2009], is constructed as follows: take  $n$  vertices and include all of the simplices of dimension strictly less than  $d$ , and then for each possible simplex of dimension  $d$ , include it with probability  $p$ .

EXAMPLES:

```

sage: X = simplicial_complexes.RandomComplex(6, 2); X
Random 2-dimensional simplicial complex on 6 vertices
sage: len(list(X.vertices()))
6

```

If  $d$  is too large (if  $d + 1 > n$ , so that there are no  $d$ -dimensional simplices), then return the simplicial complex with a single  $(n + 1)$ -dimensional simplex:

```
sage: simplicial_complexes.RandomComplex(6, 12)
The 5-simplex
```

`sage.homology.examples.RandomTwoSphere(n)`

Return a random triangulation of the 2-dimensional sphere with  $n$  vertices.

INPUT:

$n$  – an integer

OUTPUT:

A random triangulation of the sphere chosen uniformly among the *rooted* triangulations on  $n$  vertices. Because some triangulations have nontrivial automorphism groups, this may not be equal to the uniform distribution among unrooted triangulations.

ALGORITHM:

The algorithm is taken from [PS2006], section 2.1.

Starting from a planar tree (represented by its contour as a sequence of vertices), one first performs local closures, until no one is possible. A local closure amounts to replace in the cyclic contour word a sequence `in1, in2, in3, lf, in3` by `in1, in3`. After all local closures are done, one has reached the partial closure, as in [PS2006], figure 5 (a).

Then one has to perform complete closure by adding two more vertices, in order to reach the situation of [PS2006], figure 5 (b). For this, it is necessary to find inside the final contour one of the two subsequences `lf, in, lf`.

At every step of the algorithm, newly created triangles are added in a simplicial complex.

This algorithm is implemented in `RandomTriangulation()`, which creates an embedded graph. The triangles of the simplicial complex are recovered from this embedded graph.

EXAMPLES:

```
sage: G = simplicial_complexes.RandomTwoSphere(6); G
Simplicial complex with vertex set (0, 1, 2, 3, 4, 5) and 8 facets
sage: G.homology()
{0: 0, 1: 0, 2: Z}
sage: G.is_pure()
True
sage: fg = G.flip_graph(); fg
Graph on 8 vertices
sage: fg.is_planar() and fg.is_regular(3)
True
```

`sage.homology.examples.RealProjectivePlane()`

A minimal triangulation of the real projective plane.

EXAMPLES:

```
sage: P = simplicial_complexes.RealProjectivePlane()
sage: Q = simplicial_complexes.ProjectivePlane()
sage: P == Q
True
sage: P.cohomology(1)
0
sage: P.cohomology(2)
C2
sage: P.cohomology(1, base_ring=GF(2))
```

(continues on next page)

(continued from previous page)

```

Vector space of dimension 1 over Finite Field of size 2
sage: P.cohomology(2, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2

```

`sage.homology.examples.RealProjectiveSpace(n)`

A triangulation of  $\mathbf{R}P^n$  for any  $n \geq 0$ .

INPUT:

- $n$  – integer, the dimension of the real projective space to construct

The first few cases are pretty trivial:

- $\mathbf{R}P^0$  is a point.
- $\mathbf{R}P^1$  is a circle, triangulated as the boundary of a single 2-simplex.
- $\mathbf{R}P^2$  is the real projective plane, here given its minimal triangulation with 6 vertices, 15 edges, and 10 triangles.
- $\mathbf{R}P^3$ : any triangulation has at least 11 vertices by a result of Walkup [Wal1970]; this function returns a triangulation with 11 vertices, as given by Lutz [Lut2005].
- $\mathbf{R}P^4$ : any triangulation has at least 16 vertices by a result of Walkup; this function returns a triangulation with 16 vertices as given by Lutz; see also Datta [Dat2007], Example 3.12.
- $\mathbf{R}P^n$ : Lutz has found a triangulation of  $\mathbf{R}P^5$  with 24 vertices, but it does not seem to have been published. Kühnel [Kuh1987] has described a triangulation of  $\mathbf{R}P^n$ , in general, with  $2^{n+1} - 1$  vertices; see also Datta, Example 3.21. This triangulation is presumably not minimal, but it seems to be the best in the published literature as of this writing. So this function returns it when  $n > 4$ .

ALGORITHM: For  $n < 4$ , these are constructed explicitly by listing the facets. For  $n = 4$ , this is constructed by specifying 16 vertices, two facets, and a certain subgroup  $G$  of the symmetric group  $S_{16}$ . Then the set of all facets is the  $G$ -orbit of the two given facets. This is implemented here by explicitly listing all of the facets; the facets can be computed by the function `facets_for_RP4()`, but running the function takes a few seconds.

For  $n > 4$ , the construction is as follows: let  $S$  denote the simplicial complex structure on the  $n$ -sphere given by the first barycentric subdivision of the boundary of an  $(n + 1)$ -simplex. This has a simplicial antipodal action: if  $V$  denotes the vertices in the boundary of the simplex, then the vertices in its barycentric subdivision  $S$  correspond to nonempty proper subsets  $U$  of  $V$ , and the antipodal action sends any subset  $U$  to its complement. One can show that modding out by this action results in a triangulation for  $\mathbf{R}P^n$ . To find the facets in this triangulation, find the facets in  $S$ . These are identified in pairs to form  $\mathbf{R}P^n$ , so choose a representative from each pair: for each facet in  $S$ , replace any vertex in  $S$  containing 0 with its complement.

Of course these complexes increase in size pretty quickly as  $n$  increases.

EXAMPLES:

```

sage: P3 = simplicial_complexes.RealProjectiveSpace(3)
sage: P3.f_vector()
[1, 11, 51, 80, 40]
sage: P3.homology()
{0: 0, 1: C2, 2: 0, 3: Z}
sage: P4 = simplicial_complexes.RealProjectiveSpace(4)
sage: P4.f_vector()
[1, 16, 120, 330, 375, 150]
sage: P4.homology() # long time
{0: 0, 1: C2, 2: 0, 3: C2, 4: 0}
sage: P5 = simplicial_complexes.RealProjectiveSpace(5) # long time (44s on sage.
↪math, 2012)

```

(continues on next page)



(continued from previous page)

```
sage: P5.f_vector() # long time
[1, 63, 903, 4200, 8400, 7560, 2520]
```

The following computation can take a long time – over half an hour – with Sage’s default computation of homology groups, but if you have CHomP installed, Sage will use that and the computation should only take a second or two. (You can download CHomP from <http://chomp.rutgers.edu/>, or you can install it as a Sage package using `sage -i chomp`).

```
sage: P5.homology() # long time # optional - CHomP
{0: 0, 1: C2, 2: 0, 3: C2, 4: 0, 5: Z}
sage: simplicial_complexes.RealProjectiveSpace(2).dimension()
2
sage: P3.dimension()
3
sage: P4.dimension() # long time
4
sage: P5.dimension() # long time
5
```

`sage.homology.examples.RudinBall()`

Return the non-shellable ball constructed by Rudin.

This complex is a non-shellable triangulation of the 3-ball with 14 vertices and 41 facets, constructed by Rudin in [Rud1958].

EXAMPLES:

```
sage: R = simplicial_complexes.RudinBall(); R
Rudin ball
sage: R.f_vector()
[1, 14, 66, 94, 41]
sage: R.homology()
{0: 0, 1: 0, 2: 0, 3: 0}
sage: R.is_cohen_macaulay()
True
```

`sage.homology.examples.ShiftedComplex(generators)`

Return the smallest shifted simplicial complex containing generators as faces.

Let  $V$  be a set of vertices equipped with a total order. The ‘componentwise partial ordering’ on  $k$ -subsets of  $V$  is defined as follows: if  $A = \{a_1 < \dots < a_k\}$  and  $B = \{b_1 < \dots < b_k\}$ , then  $A \leq_C B$  iff  $a_i \leq b_i$  for all  $i$ . A simplicial complex  $X$  on vertex set  $[n]$  is *shifted* if its faces form an order ideal under the componentwise partial ordering, i.e., if  $B \in X$  and  $A \leq_C B$  then  $A \in X$ . Shifted complexes of dimension 1 are also known as threshold graphs.

---

**Note:** This method assumes that  $V$  consists of positive integers with the natural ordering.

---

INPUT:

- `generators` – a list of generators of the order ideal, which may be lists, tuples or simplices

EXAMPLES:

```
sage: X = simplicial_complexes.ShiftedComplex([ Simplex([1,6]), (2,4), [8] ])
sage: sorted(X.facets())
[(1, 2), (1, 3), (1, 4), (1, 5), (1, 6), (2, 3), (2, 4), (7,), (8,)]
```

(continues on next page)

(continued from previous page)

```

sage: X = simplicial_complexes.ShiftedComplex([ [2,3,5] ])
sage: sorted(X.facets())
[(1, 2, 3), (1, 2, 4), (1, 2, 5), (1, 3, 4), (1, 3, 5), (2, 3, 4), (2, 3, 5)]
sage: X = simplicial_complexes.ShiftedComplex([ [1,3,5], [2,6] ])
sage: sorted(X.facets())
[(1, 2, 3), (1, 2, 4), (1, 2, 5), (1, 3, 4), (1, 3, 5), (1, 6), (2, 6)]

```

sage.homology.examples.**Simplex**( $n$ )

An  $n$ -dimensional simplex, as a simplicial complex.

INPUT:

- $n$  – a non-negative integer

OUTPUT: the simplicial complex consisting of the  $n$ -simplex on vertices  $(0, 1, \dots, n)$  and all of its faces.

EXAMPLES:

```

sage: simplicial_complexes.Simplex(3)
The 3-simplex
sage: simplicial_complexes.Simplex(5).euler_characteristic()
1

```

sage.homology.examples.**Sphere**( $n$ )

A minimal triangulation of the  $n$ -dimensional sphere.

INPUT:

- $n$  – positive integer

EXAMPLES:

```

sage: simplicial_complexes.Sphere(2)
Minimal triangulation of the 2-sphere
sage: simplicial_complexes.Sphere(5).homology()
{0: 0, 1: 0, 2: 0, 3: 0, 4: 0, 5: Z}
sage: [simplicial_complexes.Sphere(n).euler_characteristic() for n in range(6)]
[2, 0, 2, 0, 2, 0]
sage: [simplicial_complexes.Sphere(n).f_vector() for n in range(6)]
[[1, 2],
 [1, 3, 3],
 [1, 4, 6, 4],
 [1, 5, 10, 10, 5],
 [1, 6, 15, 20, 15, 6],
 [1, 7, 21, 35, 35, 21, 7]]

```

sage.homology.examples.**SumComplex**( $n, A$ )

The sum complexes of Linial, Meshulam, and Rosenthal [LMR2010].

If  $k + 1$  is the cardinality of  $A$ , then this returns a  $k$ -dimensional simplicial complex  $X_A$  with vertices  $\mathbf{Z}/(n)$ , and facets given by all  $k + 1$ -tuples  $(x_0, x_1, \dots, x_k)$  such that the sum  $\sum x_i$  is in  $A$ . See the paper by Linial, Meshulam, and Rosenthal [LMR2010], in which they prove various results about these complexes; for example, if  $n$  is prime, then  $X_A$  is rationally acyclic, and if in addition  $A$  forms an arithmetic progression in  $\mathbf{Z}/(n)$ , then  $X_A$  is  $\mathbf{Z}$ -acyclic. Throughout their paper, they assume that  $n$  and  $k$  are relatively prime, but the construction makes sense in general.

In addition to the results from the cited paper, these complexes can have large torsion, given the number of vertices; for example, if  $n = 10$ , and  $A = \{0, 1, 2, 3, 6\}$ , then  $H_3(X_A)$  is cyclic of order 2728, and there is a

4-dimensional complex on 13 vertices with  $H_3$  having a cyclic summand of order

$$706565607945 = 3 \cdot 5 \cdot 53 \cdot 79 \cdot 131 \cdot 157 \cdot 547.$$

See the examples.

INPUT:

- $n$  – a positive integer
- $A$  – a subset of  $\mathbf{Z}/(n)$

EXAMPLES:

```
sage: S = simplicial_complexes.SumComplex(10, [0,1,2,3,6]); S
Sum complex on vertices Z/10Z associated to {0, 1, 2, 3, 6}
sage: S.homology()
{0: 0, 1: 0, 2: 0, 3: C2728, 4: 0}
sage: factor(2728)
2^3 * 11 * 31

sage: S = simplicial_complexes.SumComplex(11, [0, 1, 3]); S
Sum complex on vertices Z/11Z associated to {0, 1, 3}
sage: S.homology(1)
C23
sage: S = simplicial_complexes.SumComplex(11, [0,1,2,3,4,7]); S
Sum complex on vertices Z/11Z associated to {0, 1, 2, 3, 4, 7}
sage: S.homology(algorithm='no_chomp') # long time
{0: 0, 1: 0, 2: 0, 3: 0, 4: C645679, 5: 0}
sage: factor(645679)
23 * 67 * 419

sage: S = simplicial_complexes.SumComplex(13, [0, 1, 3]); S
Sum complex on vertices Z/13Z associated to {0, 1, 3}
sage: S.homology(1)
C159
sage: factor(159)
3 * 53
sage: S = simplicial_complexes.SumComplex(13, [0,1,2,5]); S
Sum complex on vertices Z/13Z associated to {0, 1, 2, 5}
sage: S.homology(algorithm='no_chomp') # long time
{0: 0, 1: 0, 2: C146989209, 3: 0}
sage: factor(1648910295)
3^2 * 5 * 53 * 521 * 1327
sage: S = simplicial_complexes.SumComplex(13, [0,1,2,3,5]); S
Sum complex on vertices Z/13Z associated to {0, 1, 2, 3, 5}
sage: S.homology(algorithm='no_chomp') # long time
{0: 0, 1: 0, 2: 0, 3: C3 x C237 x C706565607945, 4: 0}
sage: factor(706565607945)
3 * 5 * 53 * 79 * 131 * 157 * 547

sage: S = simplicial_complexes.SumComplex(17, [0, 1, 4]); S
Sum complex on vertices Z/17Z associated to {0, 1, 4}
sage: S.homology(1, algorithm='no_chomp')
C140183
sage: factor(140183)
103 * 1361
sage: S = simplicial_complexes.SumComplex(19, [0, 1, 4]); S
Sum complex on vertices Z/19Z associated to {0, 1, 4}
sage: S.homology(1, algorithm='no_chomp')
```

(continues on next page)

(continued from previous page)

```

C5670599
sage: factor(5670599)
11 * 191 * 2699
sage: S = simplicial_complexes.SumComplex(31, [0, 1, 4]); S
Sum complex on vertices Z/31Z associated to {0, 1, 4}
sage: S.homology(1, algorithm='no_chomp') # long time
C5 x C5 x C5 x C5 x C26951480558170926865
sage: factor(26951480558170926865)
5 * 311 * 683 * 1117 * 11657 * 1948909

```

sage.homology.examples.**SurfaceOfGenus**(*g*, *orientable=True*)

A surface of genus *g*.

INPUT:

- *g* – a non-negative integer. The desired genus
- *orientable* – boolean (optional, default True). If True, return an orientable surface, and if False, return a non-orientable surface.

In the orientable case, return a sphere if *g* is zero, and otherwise return a *g*-fold connected sum of a torus with itself.

In the non-orientable case, raise an error if *g* is zero. If *g* is positive, return a *g*-fold connected sum of a real projective plane with itself.

EXAMPLES:

```

sage: simplicial_complexes.SurfaceOfGenus(2)
Triangulation of an orientable surface of genus 2
sage: simplicial_complexes.SurfaceOfGenus(1, orientable=False)
Triangulation of a non-orientable surface of genus 1

```

sage.homology.examples.**Torus**()

A minimal triangulation of the torus.

This is a simplicial complex with 7 vertices, 21 edges and 14 faces. It is the unique triangulation of the torus with 7 vertices, and has been found by Möbius in 1861.

This is also the combinatorial structure of the Császár polyhedron (see [Wikipedia article Császár polyhedron](#)).

EXAMPLES:

```

sage: T = simplicial_complexes.Torus(); T.homology(1)
Z x Z
sage: T.f_vector()
[1, 7, 21, 14]

```

REFERENCES:

- [Lut2002]

**class** sage.homology.examples.**UniqueSimplicialComplex**(*maximal\_faces=None*,  
*name=None, \*\*kws*)

Bases: `sage.homology.simplicial_complex.SimplicialComplex`, `sage.structure.unique_representation.UniqueRepresentation`

This combines `SimplicialComplex` and `UniqueRepresentation`. It is intended to be used to make standard examples of simplicial complexes unique. See [trac ticket #13566](#).

INPUT:

- the inputs are the same as for a `SimplicialComplex`, with one addition and two exceptions. The exceptions are that `is_mutable` and `is_immutable` are ignored: all instances of this class are immutable. The addition:
- `name` – string (optional), the string representation for this complex.

EXAMPLES:

```
sage: from sage.homology.examples import UniqueSimplicialComplex
sage: SimplicialComplex([[0,1]]) is SimplicialComplex([[0,1]])
False
sage: UniqueSimplicialComplex([[0,1]]) is UniqueSimplicialComplex([[0,1]])
True
sage: UniqueSimplicialComplex([[0,1]])
Simplicial complex with vertex set (0, 1) and facets {(0, 1)}
sage: UniqueSimplicialComplex([[0,1]], name='The 1-simplex')
The 1-simplex
```

`sage.homology.examples.ZieglerBall()`

Return the non-shellable ball constructed by Ziegler.

This complex is a non-shellable triangulation of the 3-ball with 10 vertices and 21 facets, constructed by Ziegler in [Zie1998] and the smallest such complex known.

EXAMPLES:

```
sage: Z = simplicial_complexes.ZieglerBall(); Z
Ziegler ball
sage: Z.f_vector()
[1, 10, 38, 50, 21]
sage: Z.homology()
{0: 0, 1: 0, 2: 0, 3: 0}
sage: Z.is_cohen_macaulay()
True
```

`sage.homology.examples.facets_for_K3()`

Returns the facets for a minimal triangulation of the  $K3$  surface.

This is a pure simplicial complex of dimension 4 with 16 vertices and 288 facets. The facets are obtained by constructing a few facets and a permutation group  $G$ , and then computing the  $G$ -orbit of those facets.

See Casella and Kühnel in [CK2001] and Spreer and Kühnel [SK2011]; the construction here uses the labeling from Spreer and Kühnel.

EXAMPLES:

```
sage: from sage.homology.examples import facets_for_K3
sage: A = facets_for_K3() # long time (a few seconds)
sage: SimplicialComplex(A) == simplicial_complexes.K3Surface() # long time
True
```

`sage.homology.examples.facets_for_RP4()`

Return the list of facets for a minimal triangulation of 4-dimensional real projective space.

We use vertices numbered 1 through 16, define two facets, and define a certain subgroup  $G$  of the symmetric group  $S_{16}$ . Then the set of all facets is the  $G$ -orbit of the two given facets.

See the description in Example 3.12 in Datta [Dat2007].

EXAMPLES:

```
sage: from sage.homology.examples import facets_for_RP4
sage: A = facets_for_RP4()      # long time (1 or 2 seconds)
sage: SimplicialComplex(A) == simplicial_complexes.RealProjectiveSpace(4) # long_
↪time
True
```

`sage.homology.examples.matching(A, B)`

List of maximal matchings between the sets A and B.

A matching is a set of pairs  $(a, b) \in A \times B$  where each  $a$  and  $b$  appears in at most one pair. A maximal matching is one which is maximal with respect to inclusion of subsets of  $A \times B$ .

INPUT:

- A, B – list, tuple, or indeed anything which can be converted to a set.

EXAMPLES:

```
sage: from sage.homology.examples import matching
sage: matching([1,2], [3,4])
[{(1, 3), (2, 4)}, {(1, 4), (2, 3)}]
sage: matching([0,2], [0])
[{(0, 0)}, {(2, 0)}]
```

## FINITE DELTA-COMPLEXES

AUTHORS:

- John H. Palmieri (2009-08)

This module implements the basic structure of finite  $\Delta$ -complexes. For full mathematical details, see Hatcher [Hat2002], especially Section 2.1 and the Appendix on “Simplicial CW Structures”. As Hatcher points out,  $\Delta$ -complexes were first introduced by Eilenberg and Zilber [EZ1950], although they called them “semi-simplicial complexes”.

A  $\Delta$ -complex is a generalization of a *simplicial complex*; a  $\Delta$ -complex  $X$  consists of sets  $X_n$  for each non-negative integer  $n$ , the elements of which are called *n-simplices*, along with *face maps* between these sets of simplices: for each  $n$  and for all  $0 \leq i \leq n$ , there are functions  $d_i$  from  $X_n$  to  $X_{n-1}$ , with  $d_i(s)$  equal to the  $i$ -th face of  $s$  for each simplex  $s \in X_n$ . These maps must satisfy the *simplicial identity*

$$d_i d_j = d_{j-1} d_i \text{ for all } i < j.$$

Given a  $\Delta$ -complex, it has a *geometric realization*: a topological space built by taking one topological  $n$ -simplex for each element of  $X_n$ , and gluing them together as determined by the face maps.

$\Delta$ -complexes are an alternative to simplicial complexes. Every simplicial complex is automatically a  $\Delta$ -complex; in the other direction, though, it seems in practice that one can often construct  $\Delta$ -complex representations for spaces with many fewer simplices than in a simplicial complex representation. For example, the minimal triangulation of a torus as a simplicial complex contains 14 triangles, 21 edges, and 7 vertices, while there is a  $\Delta$ -complex representation of a torus using only 2 triangles, 3 edges, and 1 vertex.

---

**Note:** This class derives from *GenericCellComplex*, and so inherits its methods. Some of those methods are not listed here; see the *Generic Cell Complex* page instead.

---

**class** sage.homology.delta\_complex.DeltaComplex (data=None, check\_validity=True)

Bases: *sage.homology.cell\_complex.GenericCellComplex*

Define a  $\Delta$ -complex.

### Parameters

- **data** – see below for a description of the options
- **check\_validity** (*boolean; optional, default True*) – If True, check that the simplicial identities hold.

**Returns** a  $\Delta$ -complex

Use *data* to define a  $\Delta$ -complex. It may be in any of three forms:

- data may be a dictionary indexed by simplices. The value associated to a  $d$ -simplex  $S$  can be any of:
  - a list or tuple of  $(d-1)$ -simplices, where the  $i$ th entry is the  $i$ th face of  $S$ , given as a simplex,
  - another  $d$ -simplex  $T$ , in which case the  $i$ th face of  $S$  is declared to be the same as the  $i$ th face of  $T$ :  $S$  and  $T$  are glued along their entire boundary,
  - None or True or False or anything other than the previous two options, in which case the faces are just the ordinary faces of  $S$ .

For example, consider the following:

```
sage: n = 5
sage: S5 = DeltaComplex({Simplex(n): True, Simplex(range(1, n+2)): Simplex(n)})
sage: S5
Delta complex with 6 vertices and 65 simplices
```

The first entry in dictionary forming the argument to `DeltaComplex` says that there is an  $n$ -dimensional simplex with its ordinary boundary. The second entry says that there is another simplex whose boundary is glued to that of the first one. The resulting  $\Delta$ -complex is, of course, homeomorphic to an  $n$ -sphere, or actually a 5-sphere, since we defined  $n$  to be 5. (Note that the second simplex here can be any  $n$ -dimensional simplex, as long as it is distinct from `Simplex(n)`.)

Let's compute its homology, and also compare it to the simplicial version:

```
sage: S5.homology()
{0: 0, 1: 0, 2: 0, 3: 0, 4: 0, 5: Z}
sage: S5.f_vector() # number of simplices in each dimension
[1, 6, 15, 20, 15, 6, 2]
sage: simplicial_complexes.Sphere(5).f_vector()
[1, 7, 21, 35, 35, 21, 7]
```


Both contain a single  $(-1)$ -simplex, the empty simplex; other than that, the  $\Delta$ -complex version contains fewer simplices than the simplicial one in each dimension.

To construct a torus, use:

```
sage: torus_dict = {Simplex([0,1,2]): True,
....:               Simplex([3,4,5]): (Simplex([0,1]), Simplex([0,2]), Simplex([1,
....:               ↪2]))),
....:               Simplex([0,1]): (Simplex(0), Simplex(0)),
....:               Simplex([0,2]): (Simplex(0), Simplex(0)),
....:               Simplex([1,2]): (Simplex(0), Simplex(0)),
....:               Simplex(0): ()}
sage: T = DeltaComplex(torus_dict); T
Delta complex with 1 vertex and 7 simplices
sage: T.cohomology(base_ring=QQ)
{0: Vector space of dimension 0 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

This  $\Delta$ -complex consists of two triangles (given by `Simplex([0,1,2])` and `Simplex([3,4,5])`); the boundary of the first is just its usual boundary: the 0th face is obtained by omitting the lowest numbered vertex, etc., and so the boundary consists of the edges  $[1,2]$ ,  $[0,2]$ , and  $[0,1]$ , in that order. The boundary of the second is, on the one hand, computed the same way: the  $n$ th face is obtained by omitting the  $n$ th vertex. On the other hand, the boundary is explicitly declared to be edges  $[0,1]$ ,  $[0,2]$ , and  $[1,2]$ , in that order. This glues the second triangle to the first in the prescribed way. The three edges each start and end at the single vertex, `Simplex(0)`.





- data may be nested lists or tuples. The  $n$ th entry in the list is a list of the  $n$ -simplices in the complex, and each  $n$ -simplex is encoded as a list, the  $i$ th entry of which is its  $i$ th face. Each face is represented by an integer, giving its index in the list of  $(n-1)$ -faces. For example, consider this:


```
sage: P = DeltaComplex( [ [()], [()], [(1,0), (1,0), (0,0)],
.....:                  [(1,0,2), (0,1,2)] ] )
```

The 0th entry in the list is  $[(), ()]$ : there are two 0-simplices, and their boundaries are empty.

The 1st entry in the list is  $[(1,0), (1,0), (0,0)]$ : there are three 1-simplices. Two of them have boundary  $(1,0)$ , which means that their 0th face is vertex 1 (in the list of vertices), and their 1st face is vertex 0. The other edge has boundary  $(0,0)$ , so it starts and ends at vertex 0.

The 2nd entry in the list is  $[(1,0,2), (0,1,2)]$ : there are two 2-simplices. The first 2-simplex has boundary  $(1,0,2)$ , meaning that its 0th face is edge 1 (in the list above), its 1st face is edge 0, and its 2nd face is edge 2; similarly for the 2nd 2-simplex.

If one draws two triangles and identifies them according to this description, the result is the real projective plane.



```
sage: P.homology(1)
C2
sage: P.cohomology(2)
C2
```

Closely related to this form for data is `X.cells()` for a  $\Delta$ -complex  $X$ : this is a dictionary, indexed by dimension  $d$ , whose  $d$ -th entry is a list of the  $d$ -simplices, as a list:

```
sage: P.cells()
{-1: [()],
 0: [(), ()],
 1: [(1,0), (1,0), (0,0)],
 2: [(1,0,2), (0,1,2)]}
```

- data may be a dictionary indexed by integers. For each integer  $n$ , the entry with key  $n$  is the list of  $n$ -simplices: this is the same format as is output by the `cells()` method.

```
sage: P = DeltaComplex( [ [()], [()], [(1,0), (1,0), (0,0)],
....:                    [(1,0,2), (0, 1, 2)] ])
sage: cells_dict = P.cells()
sage: cells_dict
{-1: ([()],),
 0: ([()], [()]),
 1: [(1, 0), (1, 0), (0, 0)],
 2: [(1, 0, 2), (0, 1, 2)]}
sage: DeltaComplex(cells_dict)
Delta complex with 2 vertices and 8 simplices
sage: P == DeltaComplex(cells_dict)
True
```

Since  $\Delta$ -complexes are generalizations of simplicial complexes, any simplicial complex may be viewed as a  $\Delta$ -complex:

```
sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: RP2_delta = RP2.delta_complex()
sage: RP2.f_vector()
[1, 6, 15, 10]
sage: RP2_delta.f_vector()
[1, 6, 15, 10]
```

Finally,  $\Delta$ -complex constructions for several familiar spaces are available as follows:

```
sage: delta_complexes.Sphere(4) # the 4-sphere
Delta complex with 5 vertices and 33 simplices
sage: delta_complexes.KleinBottle()
Delta complex with 1 vertex and 7 simplices
sage: delta_complexes.RealProjectivePlane()
Delta complex with 2 vertices and 8 simplices
```

Type `delta_complexes.` and then hit the TAB key to get the full list.

**alexander\_whitney**(*cell*, *dim\_left*)

Subdivide *cell* in this  $\Delta$ -complex into a pair of simplices.

For an abstract simplex with vertices  $v_0, v_1, \dots, v_n$ , then subdivide it into simplices  $(v_0, v_1, \dots, v_{dim\_left})$  and  $(v_{dim\_left}, v_{dim\_left+1}, \dots, v_n)$ . In a  $\Delta$ -complex, instead take iterated faces: take top faces to get the left factor, take bottom faces to get the right factor.

INPUT:

- *cell* – a simplex in this complex, given as a pair (*idx*, *tuple*), where *idx* is its index in the list of cells in the given dimension, and *tuple* is the tuple of its faces
- *dim\_left* – integer between 0 and one more than the dimension of this simplex

OUTPUT: a list containing just the triple (*1*, *left*, *right*), where *left* and *right* are the two cells described above, each given as pairs (*idx*, *tuple*).

EXAMPLES:

```
sage: X = delta_complexes.Torus()
sage: X.n_cells(2)
[(1, 2, 0), (0, 2, 1)]
sage: X.alexander_whitney((0, (1, 2, 0)), 1)
```

(continues on next page)

(continued from previous page)

```

[(1, (0, (0, 0)), (1, (0, 0)))]
sage: X.alexander_whitney((0, (1, 2, 0)), 0)
[(1, (0, ()), (0, (1, 2, 0)))]
sage: X.alexander_whitney((1, (0, 2, 1)), 2)
[(1, (1, (0, 2, 1)), (0, ()))]
```

**algebraic\_topological\_model** (*base\_ring=None*)Algebraic topological model for this  $\Delta$ -complex with coefficients in *base\_ring*.

The term “algebraic topological model” is defined by Pilarczyk and Réal [PR2015].

INPUT:

- *base\_ring* - coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ). Must be a field.

Denote by  $C$  the chain complex associated to this  $\Delta$ -complex. The algebraic topological model is a chain complex  $M$  with zero differential, with the same homology as  $C$ , along with chain maps  $\pi : C \rightarrow M$  and  $\iota : M \rightarrow C$  satisfying  $\iota\pi = 1_M$  and  $\pi\iota$  chain homotopic to  $1_C$ . The chain homotopy  $\phi$  must satisfy

- $\phi\phi = 0$ ,
- $\pi\phi = 0$ ,
- $\phi\iota = 0$ .

Such a chain homotopy is called a *chain contraction*.

OUTPUT: a pair consisting of

- chain contraction *phi* associated to  $C$ ,  $M$ ,  $\pi$ , and  $\iota$
- the chain complex  $M$

Note that from the chain contraction *phi*, one can recover the chain maps  $\pi$  and  $\iota$  via *phi*.*pi*() and *phi*.*iota*(). Then one can recover  $C$  and  $M$  from, for example, *phi*.*pi*().*domain*() and *phi*.*pi*().*codomain*(), respectively.

EXAMPLES:

```

sage: RP2 = delta_complexes.RealProjectivePlane()
sage: phi, M = RP2.algebraic_topological_model(GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = delta_complexes.Torus()
sage: phi, M = T.algebraic_topological_model(QQ)
sage: M.homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

**barycentric\_subdivision** ()

Not implemented.

EXAMPLES:

```

sage: K = delta_complexes.KleinBottle()
sage: K.barycentric_subdivision()
Traceback (most recent call last):
```

(continues on next page)

(continued from previous page)

```
...
NotImplementedError: Barycentric subdivisions are not implemented for Delta_
↳ complexes.
```

**cells** (*subcomplex=None*)

The cells of this  $\Delta$ -complex.

**Parameters** *subcomplex* (*optional, default None*) – a subcomplex of this complex

The cells of this  $\Delta$ -complex, in the form of a dictionary: the keys are integers, representing dimension, and the value associated to an integer  $d$  is the list of  $d$ -cells. Each  $d$ -cell is further represented by a list, the  $i$ th entry of which gives the index of its  $i$ th face in the list of  $(d-1)$ -cells.

If the optional argument *subcomplex* is present, then “return only the faces which are *not* in the subcomplex”. To preserve the indexing, which is necessary to compute the relative chain complex, this actually replaces the faces in *subcomplex* with *None*.

EXAMPLES:

```
sage: S2 = delta_complexes.Sphere(2)
sage: S2.cells()
{-1: (( ),),
 0: (( ), ( ), ( )),
 1: ((0, 1), (0, 2), (1, 2)),
 2: ((0, 1, 2), (0, 1, 2))}
sage: A = S2.subcomplex({1: [0,2]}) # one edge
sage: S2.cells(subcomplex=A)
{-1: (None,),
 0: (None, None, None),
 1: (None, (0, 2), None),
 2: ((0, 1, 2), (0, 1, 2))}
```

**chain\_complex** (*subcomplex=None, augmented=False, verbose=False, check=False, dimensions=None, base\_ring=Integer Ring, cochain=False*)

The chain complex associated to this  $\Delta$ -complex.

**Parameters**

- **dimensions** – if *None*, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero. NOT IMPLEMENTED YET: this function always returns the entire chain complex
- **base\_ring** (*optional, default ZZ*) – commutative ring
- **subcomplex** (*optional, default empty*) – a subcomplex of this simplicial complex. Compute the chain complex relative to this subcomplex.
- **augmented** (*boolean; optional, default False*) – If *True*, return the augmented chain complex (that is, include a class in dimension  $-1$  corresponding to the empty cell). This is ignored if *dimensions* is specified or if *subcomplex* is nonempty.
- **cochain** (*boolean; optional, default False*) – If *True*, return the cochain complex (that is, the dual of the chain complex).
- **verbose** (*boolean; optional, default False*) – If *True*, print some messages as the chain complex is computed.
- **check** (*boolean; optional, default False*) – If *True*, make sure that the chain complex is actually a chain complex: the differentials are composable and their

product is zero.

---

**Note:** If subcomplex is nonempty, then the argument augmented has no effect: the chain complex relative to a nonempty subcomplex is zero in dimension  $-1$ .

---

EXAMPLES:

```
sage: circle = delta_complexes.Sphere(1)
sage: circle.chain_complex()
Chain complex with at most 2 nonzero terms over Integer Ring
sage: circle.chain_complex()._latex_()
'\!\!\boldsymbol{Z}^{\!1} \!\!\xrightarrow{\!d_1\!} \!\!\boldsymbol{Z}^{\!1}\!'
sage: circle.chain_complex(base_ring=QQ, augmented=True)
Chain complex with at most 3 nonzero terms over Rational Field
sage: circle.homology(dim=1)
Z
sage: circle.cohomology(dim=1)
Z
sage: T = delta_complexes.Torus()
sage: T.chain_complex(subcomplex=T)
Trivial chain complex over Integer Ring
sage: T.homology(subcomplex=T, algorithm='no_chomp')
{0: 0, 1: 0, 2: 0}
sage: A = T.subcomplex({2: [1]}) # one of the two triangles forming T
sage: T.chain_complex(subcomplex=A)
Chain complex with at most 1 nonzero terms over Integer Ring
sage: T.homology(subcomplex=A)
{0: 0, 1: 0, 2: 0}
```

**cone()**

The cone on this  $\Delta$ -complex.

The cone is the complex formed by adding a new vertex  $C$  and simplices of the form  $[C, v_0, \dots, v_k]$  for every simplex  $[v_0, \dots, v_k]$  in the original complex. That is, the cone is the join of the original complex with a one-point complex.

EXAMPLES:

```
sage: K = delta_complexes.KleinBottle()
sage: K.cone()
Delta complex with 2 vertices and 14 simplices
sage: K.cone().homology()
{0: 0, 1: 0, 2: 0, 3: 0}
```

**connected\_sum(other)**

Return the connected sum of self with other.

**Parameters** *other* – another  $\Delta$ -complex

**Returns** the connected sum `self # other`

**Warning:** This does not check that self and other are manifolds. It doesn't even check that their facets all have the same dimension. It just chooses top-dimensional simplices from each complex, checks that they have the same dimension, removes them, and glues the remaining pieces together. Since a (more or less) random facet is chosen from each complex, this method may return random results if applied to non-manifolds, depending on which facet is chosen.

**ALGORITHM:**

Pick a top-dimensional simplex from each complex. Check to see if there are any identifications on either simplex, using the `_is_glued()` method. If there are no identifications, remove the simplices and glue the remaining parts of complexes along their boundary. If there are identifications on a simplex, subdivide it repeatedly (using `elementary_subdivision()`) until some piece has no identifications.

**EXAMPLES:**

```
sage: T = delta_complexes.Torus()
sage: S2 = delta_complexes.Sphere(2)
sage: T.connected_sum(S2).cohomology() == T.cohomology()
True
sage: RP2 = delta_complexes.RealProjectivePlane()
sage: T.connected_sum(RP2).homology(1)
Z x Z x C2
sage: T.connected_sum(RP2).homology(2)
0
sage: RP2.connected_sum(RP2).connected_sum(RP2).homology(1)
Z x Z x C2
```

**disjoint\_union** (*right*)

The disjoint union of this  $\Delta$ -complex with another one.

**Parameters** *right* – the other  $\Delta$ -complex (the right-hand factor)

**EXAMPLES:**

```
sage: S1 = delta_complexes.Sphere(1)
sage: S2 = delta_complexes.Sphere(2)
sage: S1.disjoint_union(S2).homology()
{0: Z, 1: Z, 2: Z}
```

**elementary\_subdivision** (*idx=-1*)

Perform an “elementary subdivision” on a top-dimensional simplex in this  $\Delta$ -complex. If the optional argument *idx* is present, it specifies the index (in the list of top-dimensional simplices) of the simplex to subdivide. If not present, subdivide the last entry in this list.

**Parameters** *idx* (*integer; optional, default -1*) – index specifying which simplex to subdivide

**Returns**  $\Delta$ -complex with one simplex subdivided.

*Elementary subdivision* of a simplex means replacing that simplex with the cone on its boundary. That is, given a  $\Delta$ -complex containing a  $d$ -simplex  $S$  with vertices  $v_0, \dots, v_d$ , form a new  $\Delta$ -complex by

- removing  $S$
- adding a vertex  $w$  (thought of as being in the interior of  $S$ )
- adding all simplices with vertices  $v_{i_0}, \dots, v_{i_k}, w$ , preserving any identifications present along the boundary of  $S$

The algorithm for achieving this uses `_epi_from_standard_simplex()` to keep track of simplices (with multiplicity) and what their faces are: this method defines a surjection  $\pi$  from the standard  $d$ -simplex to  $S$ . So first remove  $S$  and add a new vertex  $w$ , say at the end of the old list of vertices. Then for each vertex  $v$  in the standard  $d$ -simplex, add an edge from  $\pi(v)$  to  $w$ ; for each edge  $(v_0, v_1)$  in the standard  $d$ -simplex, add a triangle  $(\pi(v_0), \pi(v_1), w)$ , etc.

Note that given an  $n$ -simplex  $(v_0, v_1, \dots, v_n)$  in the standard  $d$ -simplex, the faces of the new  $(n+1)$ -simplex are given by removing vertices, one at a time, from  $(\pi(v_0), \dots, \pi(v_n), w)$ . These are either the image of

the old  $n$ -simplex (if  $w$  is removed) or the various new  $n$ -simplices added in the previous dimension. So keep track of what's added in dimension  $n$  for use in computing the faces in dimension  $n + 1$ .

In contrast with barycentric subdivision, note that only the interior of  $S$  has been changed; this allows for subdivision of a single top-dimensional simplex without subdividing every simplex in the complex.

The term “elementary subdivision” is taken from p. 112 in John M. Lee’s book [Lee2011].

EXAMPLES:

```
sage: T = delta_complexes.Torus()
sage: T.n_cells(2)
[(1, 2, 0), (0, 2, 1)]
sage: T.elementary_subdivision(0) # subdivide first triangle
Delta complex with 2 vertices and 13 simplices
sage: X = T.elementary_subdivision(); X # subdivide last triangle
Delta complex with 2 vertices and 13 simplices
sage: X.elementary_subdivision()
Delta complex with 3 vertices and 19 simplices
sage: X.homology() == T.homology()
True
```

**face\_poset()**

The face poset of this  $\Delta$ -complex, the poset of nonempty cells, ordered by inclusion.

EXAMPLES:

```
sage: T = delta_complexes.Torus()
sage: T.face_poset()
Finite poset containing 6 elements
```

**graph()**

The 1-skeleton of this  $\Delta$ -complex as a graph.

EXAMPLES:

```
sage: T = delta_complexes.Torus()
sage: T.graph()
Looped multi-graph on 1 vertex
sage: S = delta_complexes.Sphere(2)
sage: S.graph()
Graph on 3 vertices
sage: delta_complexes.Simplex(4).graph() == graphs.CompleteGraph(5)
True
```

**join(other)**

The join of this  $\Delta$ -complex with another one.

**Parameters** *other* – another  $\Delta$ -complex (the right-hand factor)

**Returns** the join  $\text{self} * \text{other}$

The join of two  $\Delta$ -complexes  $S$  and  $T$  is the  $\Delta$ -complex  $S * T$  with simplices of the form  $[v_0, \dots, v_k, w_0, \dots, w_n]$  for all simplices  $[v_0, \dots, v_k]$  in  $S$  and  $[w_0, \dots, w_n]$  in  $T$ . The faces are computed accordingly: the  $i$ th face of such a simplex is either  $(d_i S) * T$  if  $i \leq k$ , or  $S * (d_{i-k-1} T)$  if  $i > k$ .

EXAMPLES:

```
sage: T = delta_complexes.Torus()
sage: S0 = delta_complexes.Sphere(0)
```

(continues on next page)

(continued from previous page)

```
sage: T.join(S0) # the suspension of T
Delta complex with 3 vertices and 21 simplices
```

Compare to simplicial complexes:

```
sage: K = delta_complexes.KleinBottle()
sage: T_simp = simplicial_complexes.Torus()
sage: K_simp = simplicial_complexes.KleinBottle()
sage: T.join(K).homology()[3] == T_simp.join(K_simp).homology()[3] # long_
↳time (3 seconds)
True
```

The notation ‘\*’ may be used, as well:

```
sage: S1 = delta_complexes.Sphere(1)
sage: X = S1 * S1 # X is a 3-sphere
sage: X.homology()
{0: 0, 1: 0, 2: 0, 3: Z}
```

**n\_chains** (*n*, *base\_ring*=None, *cochains*=False)

Return the free module of chains in degree *n* over *base\_ring*.

INPUT:

- *n* – integer
- *base\_ring* – ring (optional, default **Z**)
- *cochains* – boolean (optional, default False); if True, return cochains instead

Since the list of *n*-cells for a  $\Delta$ -complex may have some ambiguity – for example, the list of edges may look like [(0, 0), (0, 0), (0, 0)] if each edge starts and ends at vertex 0 – we record the indices of the cells along with their tuples. So the basis of chains in such a case would look like [(0, (0, 0)), (1, (0, 0)), (2, (0, 0))].

The only difference between chains and cochains is notation: the dual cochain to the chain basis element *b* is written as  $\chi_b$ .

EXAMPLES:

```
sage: T = delta_complexes.Torus()
sage: T.n_chains(1, QQ)
Free module generated by {(0, (0, 0)), (1, (0, 0)), (2, (0, 0))} over_
↳Rational Field
sage: list(T.n_chains(1, QQ, cochains=False).basis())
[(0, (0, 0)), (1, (0, 0)), (2, (0, 0))]
sage: list(T.n_chains(1, QQ, cochains=True).basis())
[\chi_(0, (0, 0)), \chi_(1, (0, 0)), \chi_(2, (0, 0))]
```

**n\_skeleton** (*n*)

The *n*-skeleton of this  $\Delta$ -complex.

**Parameters** *n* (*non-negative integer*) – dimension

EXAMPLES:

```
sage: S3 = delta_complexes.Sphere(3)
sage: S3.n_skeleton(1) # 1-skeleton of a tetrahedron
Delta complex with 4 vertices and 11 simplices
```

(continues on next page)



(continued from previous page)

```
sage: S3.n_skeleton(1).dimension()
1
sage: S3.n_skeleton(1).homology()
{0: 0, 1: Z x Z x Z}
```

**product** (*other*)

The product of this  $\Delta$ -complex with another one.

**Parameters** *other* – another  $\Delta$ -complex (the right-hand factor)

**Returns** the product `self x other`

**Warning:** If  $X$  and  $Y$  are  $\Delta$ -complexes, then  $X*Y$  returns their join, not their product.

## EXAMPLES:

```
sage: K = delta_complexes.KleinBottle()
sage: X = K.product(K)
sage: X.homology(1)
Z x Z x C2 x C2
sage: X.homology(2)
Z x C2 x C2 x C2
sage: X.homology(3)
C2
sage: X.homology(4)
0
sage: X.homology(base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 4 over Finite Field of size 2,
 2: Vector space of dimension 6 over Finite Field of size 2,
 3: Vector space of dimension 4 over Finite Field of size 2,
 4: Vector space of dimension 1 over Finite Field of size 2}
sage: S1 = delta_complexes.Sphere(1)
sage: K.product(S1).homology() == S1.product(K).homology()
True
sage: S1.product(S1) == delta_complexes.Torus()
True
```

**subcomplex** (*data*)

Create a subcomplex.

**Parameters** *data* – a dictionary indexed by dimension or a list (or tuple); in either case, `data[n]` should be the list (or tuple or set) of the indices of the simplices to be included in the subcomplex.

This automatically includes all faces of the simplices in *data*, so you only have to specify the simplices which are maximal with respect to inclusion.

## EXAMPLES:

```
sage: X = delta_complexes.Torus()
sage: A = X.subcomplex({2: [0]}) # one of the triangles of X
sage: X.homology(subcomplex=A)
{0: 0, 1: 0, 2: Z}
```

In the following, `line` is a line segment and `ends` is the complex consisting of its two endpoints, so the relative homology of the two is isomorphic to the homology of a circle:

```

sage: line = delta_complexes.Simplex(1) # an edge
sage: line.cells()
{-1: ((,)), 0: ((, ()), 1: ((0, 1),))}
sage: ends = line.subcomplex({0: (0, 1)})
sage: ends.cells()
{-1: ((,)), 0: ((, ()))}
sage: line.homology(subcomplex=ends)
{0: 0, 1: Z}

```

**suspension** ( $n=1$ )

The suspension of this  $\Delta$ -complex.

**Parameters**  $n$  (*positive integer; optional, default 1*) – suspend this many times.

The suspension is the complex formed by adding two new vertices  $S_0$  and  $S_1$  and simplices of the form  $[S_0, v_0, \dots, v_k]$  and  $[S_1, v_0, \dots, v_k]$  for every simplex  $[v_0, \dots, v_k]$  in the original complex. That is, the suspension is the join of the original complex with a two-point complex (the 0-sphere).

EXAMPLES:

```

sage: S = delta_complexes.Sphere(0)
sage: S3 = S.suspension(3) # the 3-sphere
sage: S3.homology()
{0: 0, 1: 0, 2: 0, 3: Z}

```

**wedge** (*right*)

The wedge (one-point union) of this  $\Delta$ -complex with another one.

**Parameters** **right** – the other  $\Delta$ -complex (the right-hand factor)

---

**Note:** This operation is not well-defined if `self` or `other` is not path-connected.

---

EXAMPLES:

```

sage: S1 = delta_complexes.Sphere(1)
sage: S2 = delta_complexes.Sphere(2)
sage: S1.wedge(S2).homology()
{0: 0, 1: Z, 2: Z}

```

**class** `sage.homology.delta_complex.DeltaComplexExamples`

Some examples of  $\Delta$ -complexes.

Here are the available examples; you can also type `delta_complexes.` and hit TAB to get a list:

```

Sphere
Torus
RealProjectivePlane
KleinBottle
Simplex
SurfaceOfGenus

```

EXAMPLES:

```

sage: S = delta_complexes.Sphere(6) # the 6-sphere
sage: S.dimension()
6

```

(continues on next page)

(continued from previous page)


```

sage: S.cohomology(6)
Z
sage: delta_complexes.Torus() == delta_complexes.Sphere(3)
False

```

**KleinBottle()**

A  $\Delta$ -complex representation of the Klein bottle, consisting of one vertex, three edges, and two triangles.



EXAMPLES:


```

sage: delta_complexes.KleinBottle()
Delta complex with 1 vertex and 7 simplices

```

**RealProjectivePlane()**

A  $\Delta$ -complex representation of the real projective plane, consisting of two vertices, three edges, and two triangles.



EXAMPLES:

```

sage: P = delta_complexes.RealProjectivePlane()
sage: P.cohomology(1)
0
sage: P.cohomology(2)
C2
sage: P.cohomology(dim=1, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2
sage: P.cohomology(dim=2, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2

```

**Simplex(n)**

A  $\Delta$ -complex representation of an  $n$ -simplex, consisting of a single  $n$ -simplex and its faces. (This is the same as the simplicial complex representation available by using `simplicial_complexes.Simplex(n)`.)

EXAMPLES:

```
sage: delta_complexes.Simplex(3)
Delta complex with 4 vertices and 16 simplices
```

### **Sphere** ( $n$ )

A  $\Delta$ -complex representation of the  $n$ -dimensional sphere, formed by gluing two  $n$ -simplices along their boundary, except in dimension 1, in which case it is a single 1-simplex starting and ending at the same vertex.

**Parameters**  $n$  – dimension of the sphere

EXAMPLES:

```
sage: delta_complexes.Sphere(4).cohomology(4, base_ring=GF(3))
Vector space of dimension 1 over Finite Field of size 3
```

### **SurfaceOfGenus** ( $g$ , *orientable=True*)

A surface of genus  $g$  as a  $\Delta$ -complex.

**Parameters**

- $g$  (*non-negative integer*) – the genus
- **orientable** (bool, optional, default True) – whether the surface should be orientable

In the orientable case, return a sphere if  $g$  is zero, and otherwise return a  $g$ -fold connected sum of a torus with itself.

In the non-orientable case, raise an error if  $g$  is zero. If  $g$  is positive, return a  $g$ -fold connected sum of a real projective plane with itself.

EXAMPLES:


```
sage: delta_complexes.SurfaceOfGenus(1, orientable=False)
Delta complex with 2 vertices and 8 simplices
sage: delta_complexes.SurfaceOfGenus(3, orientable=False).homology(1)
Z x Z x C2
sage: delta_complexes.SurfaceOfGenus(3, orientable=False).homology(2)
0
```

Compare to simplicial complexes:

```
sage: delta_g4 = delta_complexes.SurfaceOfGenus(4)
sage: delta_g4.f_vector()
[1, 3, 27, 18]
sage: simpl_g4 = simplicial_complexes.SurfaceOfGenus(4)
sage: simpl_g4.f_vector()
[1, 19, 75, 50]
sage: delta_g4.homology() == simpl_g4.homology()
True
```

### **Torus** ()

A  $\Delta$ -complex representation of the torus, consisting of one vertex, three edges, and two triangles.



EXAMPLES:

```
sage: delta_complexes.Torus().homology(1)  
Z x Z
```



## FINITE CUBICAL COMPLEXES

### AUTHORS:

- John H. Palmieri (2009-08)

This module implements the basic structure of finite cubical complexes. For full mathematical details, see Kaczynski, Mischaikow, and Mrozek [KMM2004], for example.

Cubical complexes are topological spaces built from gluing together cubes of various dimensions; the collection of cubes must be closed under taking faces, just as with a simplicial complex. In this context, a “cube” means a product of intervals of length 1 or length 0 (degenerate intervals), with integer endpoints, and its faces are obtained by using the nondegenerate intervals: if  $C$  is a cube – a product of degenerate and nondegenerate intervals – and if  $[i, i + 1]$  is the  $k$ -th nondegenerate factor, then  $C$  has two faces indexed by  $k$ : the cubes obtained by replacing  $[i, i + 1]$  with  $[i, i]$  or  $[i + 1, i + 1]$ .

So to construct a space homeomorphic to a circle as a cubical complex, we could take for example the four line segments in the plane from  $(0, 2)$  to  $(0, 3)$  to  $(1, 3)$  to  $(1, 2)$  to  $(0, 2)$ . In Sage, this is done with the following command:

```
sage: S1 = CubicalComplex([([0,0], [2,3]), ([0,1], [3,3]), ([0,1], [2,2]), ([1,1], [2,
↪ 3])]); S1
Cubical complex with 4 vertices and 8 cubes
```

The argument to `CubicalComplex` is a list of the maximal “cubes” in the complex. Each “cube” can be an instance of the class `Cube` or a list (or tuple) of “intervals”, and an “interval” is a pair of integers, of one of the two forms  $[i, i]$  or  $[i, i + 1]$ . So the cubical complex `S1` above has four maximal cubes:

```
sage: len(S1.maximal_cells())
4
sage: sorted(S1.maximal_cells())
[[0,0] x [2,3], [0,1] x [2,2], [0,1] x [3,3], [1,1] x [2,3]]
```

The first of these, for instance, is the product of the degenerate interval  $[0, 0]$  with the unit interval  $[2, 3]$ : this is the line segment in the plane from  $(0, 2)$  to  $(0, 3)$ . We could form a topologically equivalent space by inserting some degenerate simplices:

```
sage: S1.homology()
{0: 0, 1: Z}
sage: X = CubicalComplex([([0,0], [2,3], [2]), ([0,1], [3,3], [2]), ([0,1], [2,2],
↪ [2]), ([1,1], [2,3], [2])])
sage: X.homology()
{0: 0, 1: Z}
```

Topologically, the cubical complex  $X$  consists of four edges of a square in  $\mathbf{R}^3$ : the same unit square as  $S1$ , but embedded in  $\mathbf{R}^3$  with  $z$ -coordinate equal to 2. Thus  $X$  is homeomorphic to  $S1$  (in fact, they’re “cubically equivalent”), and this is reflected in the fact that they have isomorphic homology groups.

**Note:** This class derives from *GenericCellComplex*, and so inherits its methods. Some of those methods are not listed here; see the *Generic Cell Complex* page instead.

**class** `sage.homology.cubical_complex.Cube` (*data*)  
 Bases: `sage.structure.sage_object.SageObject`

Define a cube for use in constructing a cubical complex.

“Elementary cubes” are products of intervals with integer endpoints, each of which is either a unit interval or a degenerate (length 0) interval; for example,

$$[0, 1] \times [3, 4] \times [2, 2] \times [1, 2]$$

is a 3-dimensional cube (since one of the intervals is degenerate) embedded in  $\mathbf{R}^4$ .

**Parameters** *data* – list or tuple of terms of the form  $(i, i+1)$  or  $(i, i)$  or  $(i, )$  – the last two are degenerate intervals.

**Returns** an elementary cube

Each cube is stored in a standard form: a tuple of tuples, with a nondegenerate interval  $[j, j]$  represented by  $(j, j)$ , not  $(j, )$ . (This is so that for any interval  $I$ ,  $I[1]$  will produce a value, not an `IndexError`.)

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [5,], [6,7], [-1, 0]]); C
[1,2] x [5,5] x [6,7] x [-1,0]
sage: C.dimension() # number of nondegenerate intervals
3
sage: C.nondegenerate_intervals() # indices of these intervals
[0, 2, 3]
sage: C.face(1, upper=False)
[1,2] x [5,5] x [6,6] x [-1,0]
sage: C.face(1, upper=True)
[1,2] x [5,5] x [7,7] x [-1,0]
sage: Cube().dimension() # empty cube has dimension -1
-1
```

**alexander\_whitney** (*dim*)

Subdivide this cube into pairs of cubes.

This provides a cubical approximation for the diagonal map  $K \rightarrow K \times K$ .

INPUT:

- *dim* – integer between 0 and one more than the dimension of this cube

OUTPUT:

- a list containing triples (*coeff*, *left*, *right*)

This uses the algorithm described by Pilarczyk and Réal [PR2015] on p. 267; the formula is originally due to Serre. Calling this method `alexander_whitney` is an abuse of notation, since the actual Alexander-Whitney map goes from  $C(K \times L) \rightarrow C(K) \otimes C(L)$ , where  $C(-)$  denotes the associated chain complex, but this subdivision of cubes is at the heart of it.

EXAMPLES:



```

sage: from sage.homology.cubical_complex import Cube
sage: C1 = Cube([[0,1], [3,4]])
sage: C1.alexander_whitney(0)
[(1, [0,0] x [3,3], [0,1] x [3,4])]
sage: C1.alexander_whitney(1)
[(1, [0,1] x [3,3], [1,1] x [3,4]), (-1, [0,0] x [3,4], [0,1] x [4,4])]
sage: C1.alexander_whitney(2)
[(1, [0,1] x [3,4], [1,1] x [4,4])]

```

**dimension()**

The dimension of this cube: the number of its nondegenerate intervals.

EXAMPLES:

```

sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [5,], [6,7], [-1, 0]])
sage: C.dimension()
3
sage: C = Cube([[1,], [5,], [6,], [-1,]])
sage: C.dimension()
0
sage: Cube([]).dimension() # empty cube has dimension -1
-1

```

**face** (*n*, *upper=True*)

The *n*th primary face of this cube.

**Parameters**

- **n** – an integer between 0 and one less than the dimension of this cube
- **upper** (*boolean; optional, default=True*) – if True, return the “upper” *n*th primary face; otherwise, return the “lower” *n*th primary face.

**Returns** the cube obtained by replacing the *n*th non-degenerate interval with either its upper or lower endpoint.

EXAMPLES:

```

sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [5,], [6,7], [-1, 0]]); C
[1,2] x [5,5] x [6,7] x [-1,0]
sage: C.face(0)
[2,2] x [5,5] x [6,7] x [-1,0]
sage: C.face(0, upper=False)
[1,1] x [5,5] x [6,7] x [-1,0]
sage: C.face(1)
[1,2] x [5,5] x [7,7] x [-1,0]
sage: C.face(2, upper=False)
[1,2] x [5,5] x [6,7] x [-1,-1]
sage: C.face(3)
Traceback (most recent call last):
...
ValueError: Can only compute the nth face if 0 <= n < dim.

```

**faces()**

The list of faces (of codimension 1) of this cube.

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [3,4]])
sage: C.faces()
[[2,2] x [3,4], [1,2] x [4,4], [1,1] x [3,4], [1,2] x [3,3]]
```

**faces\_as\_pairs()**

The list of faces (of codimension 1) of this cube, as pairs (upper, lower).

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [3,4]])
sage: C.faces_as_pairs()
[[([2,2] x [3,4], [1,1] x [3,4]), ([1,2] x [4,4], [1,2] x [3,3])]
```

**is\_face(*other*)**

Return True iff this cube is a face of other.

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C1 = Cube([[1,2], [5,], [6,7], [-1, 0]])
sage: C2 = Cube([[1,2], [5,], [6,], [-1, 0]])
sage: C1.is_face(C2)
False
sage: C1.is_face(C1)
True
sage: C2.is_face(C1)
True
```

**nondegenerate\_intervals()**

The list of indices of nondegenerate intervals of this cube.

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [5,], [6,7], [-1, 0]])
sage: C.nondegenerate_intervals()
[0, 2, 3]
sage: C = Cube([[1,], [5,], [6,], [-1,]])
sage: C.nondegenerate_intervals()
[]
```

**product(*other*)**

Cube obtained by concatenating the underlying tuples of the two arguments.

**Parameters** *other* – another cube

**Returns** the product of *self* and *other*, as a Cube

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [3,]])
sage: D = Cube([[4], [0,1]])
sage: C.product(D)
[1,2] x [3,3] x [4,4] x [0,1]
```

You can also use `__add__` or `+` or `__mul__` or `*`:

```
sage: D * C
[4,4] x [0,1] x [1,2] x [3,3]
sage: D + C * C
[4,4] x [0,1] x [1,2] x [3,3] x [1,2] x [3,3]
```

**tuple()**

The tuple attached to this cube.

EXAMPLES:

```
sage: from sage.homology.cubical_complex import Cube
sage: C = Cube([[1,2], [5,], [6,7], [-1, 0]])
sage: C.tuple()
(1, 2), (5, 5), (6, 7), (-1, 0))
```

```
class sage.homology.cubical_complex.CubicalComplex(maximal_faces=[], maximal-
                                                    ity_check=True)
Bases: sage.homology.cell_complex.GenericCellComplex
```

Define a cubical complex.

**Parameters**

- **maximal\_faces** – set of maximal faces
- **maximality\_check** (*boolean; optional, default True*) – see below

**Returns** a cubical complex

`maximal_faces` should be a list or tuple or set (or anything which may be converted to a set) of “cubes”: instances of the class `Cube`, or lists or tuples suitable for conversion to cubes. These cubes are the maximal cubes in the complex.

In addition, `maximal_faces` may be a cubical complex, in which case that complex is returned. Also, `maximal_faces` may instead be any object which has a `_cubical_` method (e.g., a simplicial complex); then that method is used to convert the object to a cubical complex.

If `maximality_check` is `True`, check that each maximal face is, in fact, maximal. In this case, when producing the internal representation of the cubical complex, omit those that are not. It is highly recommended that this be `True`; various methods for this class may fail if faces which are claimed to be maximal are in fact not.

EXAMPLES:

The empty complex, consisting of one cube, the empty cube:

```
sage: CubicalComplex()
Cubical complex with 0 vertices and 1 cube
```

A “circle” (four edges connecting the vertices (0,2), (0,3), (1,2), and (1,3)):

```
sage: S1 = CubicalComplex([([0,0], [2,3]), ([0,1], [3,3]), ([0,1], [2,2]), ([1,1],
↪ [2,3])])
sage: S1
Cubical complex with 4 vertices and 8 cubes
sage: S1.homology()
{0: 0, 1: Z}
```

A set of five points and its product with `S1`:

```

sage: pts = CubicalComplex([([0],), ([3],), ([6],), ([-12],), ([5],)])
sage: pts
Cubical complex with 5 vertices and 5 cubes
sage: pts.homology()
{0: Z x Z x Z x Z}
sage: X = S1.product(pts); X
Cubical complex with 20 vertices and 40 cubes
sage: X.homology()
{0: Z x Z x Z x Z, 1: Z^5}

```

Converting a simplicial complex to a cubical complex:

```

sage: S2 = simplicial_complexes.Sphere(2)
sage: C2 = CubicalComplex(S2)
sage: all([C2.homology(n) == S2.homology(n) for n in range(3)])
True

```

You can get the set of maximal cells or a dictionary of all cells:

```

sage: X.maximal_cells() # random: order may depend on the version of Python
{[0,0] x [2,3] x [-12,-12], [0,1] x [3,3] x [5,5], [0,1] x [2,2] x [3,3], [0,1] x
→ [2,2] x [0,0], [0,1] x [3,3] x [6,6], [1,1] x [2,3] x [0,0], [0,1] x [2,2] x [-
→ 12,-12], [0,0] x [2,3] x [6,6], [1,1] x [2,3] x [-12,-12], [1,1] x [2,3] x [5,
→ 5], [0,1] x [2,2] x [5,5], [0,1] x [3,3] x [3,3], [1,1] x [2,3] x [3,3], [0,0]
→ x [2,3] x [5,5], [0,1] x [3,3] x [0,0], [1,1] x [2,3] x [6,6], [0,1] x [2,2] x
→ [6,6], [0,0] x [2,3] x [0,0], [0,0] x [2,3] x [3,3], [0,1] x [3,3] x [-12,-12]}
sage: sorted(X.maximal_cells())
[[0,0] x [2,3] x [-12,-12],
 [0,0] x [2,3] x [0,0],
 [0,0] x [2,3] x [3,3],
 [0,0] x [2,3] x [5,5],
 [0,0] x [2,3] x [6,6],
 [0,1] x [2,2] x [-12,-12],
 [0,1] x [2,2] x [0,0],
 [0,1] x [2,2] x [3,3],
 [0,1] x [2,2] x [5,5],
 [0,1] x [2,2] x [6,6],
 [0,1] x [3,3] x [-12,-12],
 [0,1] x [3,3] x [0,0],
 [0,1] x [3,3] x [3,3],
 [0,1] x [3,3] x [5,5],
 [0,1] x [3,3] x [6,6],
 [1,1] x [2,3] x [-12,-12],
 [1,1] x [2,3] x [0,0],
 [1,1] x [2,3] x [3,3],
 [1,1] x [2,3] x [5,5],
 [1,1] x [2,3] x [6,6]]
sage: S1.cells()
{-1: set(),
 0: {[0,0] x [2,2], [0,0] x [3,3], [1,1] x [2,2], [1,1] x [3,3]},
 1: {[0,0] x [2,3], [0,1] x [2,2], [0,1] x [3,3], [1,1] x [2,3]}}

```

Chain complexes, homology, and cohomology:

```

sage: T = S1.product(S1); T
Cubical complex with 16 vertices and 64 cubes
sage: T.chain_complex()

```

(continues on next page)

(continued from previous page)

```
Chain complex with at most 3 nonzero terms over Integer Ring
sage: T.homology(base_ring=QQ)
{0: Vector space of dimension 0 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
sage: RP2 = cubical_complexes.RealProjectivePlane()
sage: RP2.cohomology(dim=[1, 2], base_ring=GF(2))
{1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
```

Joins are not implemented:

```
sage: S1.join(S1)
Traceback (most recent call last):
...
NotImplementedError: Joins are not implemented for cubical complexes.
```

Therefore, neither are cones or suspensions.

**alexander\_whitney** (*cube, dim\_left*)

Subdivide cube in this cubical complex into pairs of cubes.

See [Cube.alexander\\_whitney\(\)](#) for more details. This method just calls that one.

INPUT:

- *cube* – a cube in this cubical complex
- *dim* – integer between 0 and one more than the dimension of this cube

OUTPUT: a list containing triples (*coeff*, *left*, *right*)

EXAMPLES:

```
sage: C = cubical_complexes.Cube(3)
sage: c = list(C.n_cubes(3))[0]; c
[0,1] x [0,1] x [0,1]
sage: C.alexander_whitney(c, 1)
[(1, [0,1] x [0,0] x [0,0], [1,1] x [0,1] x [0,1]),
 (-1, [0,0] x [0,1] x [0,0], [0,1] x [1,1] x [0,1]),
 (1, [0,0] x [0,0] x [0,1], [0,1] x [0,1] x [1,1])]
```

**algebraic\_topological\_model** (*base\_ring=None*)

Algebraic topological model for this cubical complex with coefficients in *base\_ring*.

The term “algebraic topological model” is defined by Pilarczyk and Réal [PR2015].

INPUT:

- *base\_ring* - coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ). Must be a field.

Denote by  $C$  the chain complex associated to this cubical complex. The algebraic topological model is a chain complex  $M$  with zero differential, with the same homology as  $C$ , along with chain maps  $\pi : C \rightarrow M$  and  $\iota : M \rightarrow C$  satisfying  $\iota\pi = 1_M$  and  $\pi\iota$  chain homotopic to  $1_C$ . The chain homotopy  $\phi$  must satisfy

- $\phi\phi = 0$ ,
- $\pi\phi = 0$ ,
- $\phi\iota = 0$ .

Such a chain homotopy is called a *chain contraction*.

OUTPUT: a pair consisting of

- chain contraction  $\phi$  associated to  $C$ ,  $M$ ,  $\pi$ , and  $\iota$
- the chain complex  $M$

Note that from the chain contraction  $\phi$ , one can recover the chain maps  $\pi$  and  $\iota$  via  $\phi.\pi()$  and  $\phi.\iota()$ . Then one can recover  $C$  and  $M$  from, for example,  $\phi.\pi().\text{domain}()$  and  $\phi.\pi().\text{codomain}()$ , respectively.

EXAMPLES:

```
sage: RP2 = cubical_complexes.RealProjectivePlane()
sage: phi, M = RP2.algebraic_topological_model(GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = cubical_complexes.Torus()
sage: phi, M = T.algebraic_topological_model(QQ)
sage: M.homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

**cells** (*subcomplex=None*)

The cells of this cubical complex, in the form of a dictionary: the keys are integers, representing dimension, and the value associated to an integer  $d$  is the list of  $d$ -cells.

If the optional argument *subcomplex* is present, then return only the faces which are *not* in the subcomplex.

**Parameters** *subcomplex* (a cubical complex; optional, default None) – a subcomplex of this cubical complex

**Returns** cells of this complex not contained in *subcomplex*

**Return type** dictionary

EXAMPLES:

```
sage: S2 = cubical_complexes.Sphere(2)
sage: sorted(S2.cells()[2])
[[0,0] x [0,1] x [0,1],
 [0,1] x [0,0] x [0,1],
 [0,1] x [0,1] x [0,0],
 [0,1] x [0,1] x [1,1],
 [0,1] x [1,1] x [0,1],
 [1,1] x [0,1] x [0,1]]
```

**chain\_complex** (*subcomplex=None*, *augmented=False*, *verbose=False*, *check=False*, *dimensions=None*, *base\_ring=Integer Ring*, *cochain=False*)

The chain complex associated to this cubical complex.

**Parameters**

- **dimensions** – if None, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero. NOT IMPLEMENTED YET: this function always returns the entire chain complex

- **base\_ring**(*optional, default ZZ*) – commutative ring
- **subcomplex**(*optional, default empty*) – a subcomplex of this cubical complex. Compute the chain complex relative to this subcomplex.
- **augmented**(*boolean; optional, default False*) – If True, return the augmented chain complex (that is, include a class in dimension  $-1$  corresponding to the empty cell). This is ignored if *dimensions* is specified.
- **cochain**(*boolean; optional, default False*) – If True, return the cochain complex (that is, the dual of the chain complex).
- **verbose**(*boolean; optional, default False*) – If True, print some messages as the chain complex is computed.
- **check**(*boolean; optional, default False*) – If True, make sure that the chain complex is actually a chain complex: the differentials are composable and their product is zero.

---

**Note:** If *subcomplex* is nonempty, then the argument *augmented* has no effect: the chain complex relative to a nonempty subcomplex is zero in dimension  $-1$ .

---

EXAMPLES:

```
sage: S2 = cubical_complexes.Sphere(2)
sage: S2.chain_complex()
Chain complex with at most 3 nonzero terms over Integer Ring
sage: Prod = S2.product(S2); Prod
Cubical complex with 64 vertices and 676 cubes
sage: Prod.chain_complex()
Chain complex with at most 5 nonzero terms over Integer Ring
sage: Prod.chain_complex(base_ring=QQ)
Chain complex with at most 5 nonzero terms over Rational Field
sage: C1 = cubical_complexes.Cube(1)
sage: S0 = cubical_complexes.Sphere(0)
sage: C1.chain_complex(subcomplex=S0)
Chain complex with at most 1 nonzero terms over Integer Ring
sage: C1.homology(subcomplex=S0)
{0: 0, 1: Z}
```

**cone()**

The cone on this cubical complex.

NOT IMPLEMENTED

The cone is the complex formed by taking the join of the original complex with a one-point complex (that is, a 0-dimensional cube). Since joins are not implemented for cubical complexes, neither are cones.

EXAMPLES:

```
sage: C1 = cubical_complexes.Cube(1)
sage: C1.cone()
Traceback (most recent call last):
...
NotImplementedError: Cones are not implemented for cubical complexes.
```

**connected\_sum(*other*)**

Return the connected sum of self with *other*.

**Parameters** *other* – another cubical complex

**Returns** the connected sum `self # other`

**Warning:** This does not check that `self` and `other` are manifolds, only that their facets all have the same dimension. Since a (more or less) random facet is chosen from each complex and then glued together, this method may return random results if applied to non-manifolds, depending on which facet is chosen.

EXAMPLES:

```
sage: T = cubical_complexes.Torus()
sage: S2 = cubical_complexes.Sphere(2)
sage: T.connected_sum(S2).cohomology() == T.cohomology()
True
sage: RP2 = cubical_complexes.RealProjectivePlane()
sage: T.connected_sum(RP2).homology(1)
Z x Z x C2
sage: RP2.connected_sum(RP2).connected_sum(RP2).homology(1)
Z x Z x C2
```

**disjoint\_union** (*other*)

The disjoint union of this cubical complex with another one.

**Parameters** **right** – the other cubical complex (the right-hand factor)

Algorithm: first embed both complexes in  $d$ -dimensional Euclidean space. Then embed in  $(1+d)$ -dimensional space, calling the new axis  $x$ , and putting the first complex at  $x = 0$ , the second at  $x = 1$ .

EXAMPLES:

```
sage: S1 = cubical_complexes.Sphere(1)
sage: S2 = cubical_complexes.Sphere(2)
sage: S1.disjoint_union(S2).homology()
{0: Z, 1: Z, 2: Z}
```

**graph** ()

The 1-skeleton of this cubical complex, as a graph.

EXAMPLES:

```
sage: cubical_complexes.Sphere(2).graph()
Graph on 8 vertices
```

**is\_pure** ()

True iff this cubical complex is pure: that is, all of its maximal faces have the same dimension.

**Warning:** This may give the wrong answer if the cubical complex was constructed with `maximality_check` set to `False`.

EXAMPLES:

```
sage: S4 = cubical_complexes.Sphere(4)
sage: S4.is_pure()
True
sage: C = CubicalComplex([([0,0], [3,3]), ([1,2], [4,5])])
sage: C.is_pure()
False
```



**is\_subcomplex**(*other*)Return True if *self* is a subcomplex of *other*.**Parameters** *other* – a cubical complexEach maximal cube of *self* must be a face of a maximal cube of *other* for this to be True.

EXAMPLES:

```

sage: S1 = cubical_complexes.Sphere(1)
sage: C0 = cubical_complexes.Cube(0)
sage: C1 = cubical_complexes.Cube(1)
sage: cyl = S1.product(C1)
sage: end = S1.product(C0)
sage: end.is_subcomplex(cyl)
True
sage: cyl.is_subcomplex(end)
False

```

The embedding of the cubical complex is important here:

```

sage: C2 = cubical_complexes.Cube(2)
sage: C1.is_subcomplex(C2)
False
sage: C1.product(C0).is_subcomplex(C2)
True

```

$C1$  is not a subcomplex of  $C2$  because it's not embedded in  $\mathbf{R}^2$ . On the other hand,  $C1 \times C0$  is a face of  $C2$ . Look at their maximal cells:

```

sage: C1.maximal_cells()
{[0,1]}
sage: C2.maximal_cells()
{[0,1] x [0,1]}
sage: C1.product(C0).maximal_cells()
{[0,1] x [0,0]}

```

**join**(*other*)

The join of this cubical complex with another one.

NOT IMPLEMENTED.

**Parameters** *other* – another cubical complex

EXAMPLES:

```

sage: C1 = cubical_complexes.Cube(1)
sage: C1.join(C1)
Traceback (most recent call last):
...
NotImplementedError: Joins are not implemented for cubical complexes.

```

**maximal\_cells**()

The set of maximal cells (with respect to inclusion) of this cubical complex.

**Returns** Set of maximal cells

This just returns the set of cubes used in defining the cubical complex, so if the complex was defined with no maximality checking, none is done here, either.

EXAMPLES:

```

sage: interval = cubical_complexes.Cube(1)
sage: interval
Cubical complex with 2 vertices and 3 cubes
sage: interval.maximal_cells()
{[0,1]}
sage: interval.product(interval).maximal_cells()
{[0,1] x [0,1]}

```

**n\_cubes** (*n*, *subcomplex=None*)

The set of cubes of dimension *n* of this cubical complex. If the optional argument *subcomplex* is present, then return the *n*-dimensional cubes which are *not* in the subcomplex.

**Parameters**

- **n** (*integer*) – dimension
- **subcomplex** (*a cubical complex; optional, default None*) – a subcomplex of this cubical complex

**Returns** cells in dimension *n*

**Return type** set

EXAMPLES:

```

sage: C = cubical_complexes.Cube(3)
sage: C.n_cubes(3)
{[0,1] x [0,1] x [0,1]}
sage: sorted(C.n_cubes(2))
[[[0,0] x [0,1] x [0,1],
  [0,1] x [0,0] x [0,1],
  [0,1] x [0,1] x [0,0],
  [0,1] x [0,1] x [1,1],
  [0,1] x [1,1] x [0,1],
  [1,1] x [0,1] x [0,1]]

```

**n\_skeleton** (*n*)

The *n*-skeleton of this cubical complex.

**Parameters** **n** (*non-negative integer*) – dimension

**Returns** cubical complex

EXAMPLES:

```

sage: S2 = cubical_complexes.Sphere(2)
sage: C3 = cubical_complexes.Cube(3)
sage: S2 == C3.n_skeleton(2)
True

```

**product** (*other*)

The product of this cubical complex with another one.

**Parameters** **other** – another cubical complex

EXAMPLES:

```

sage: RP2 = cubical_complexes.RealProjectivePlane()
sage: S1 = cubical_complexes.Sphere(1)
sage: RP2.product(S1).homology()[1] # long time: 5 seconds
Z x C2

```

**suspension** ( $n=1$ )

The suspension of this cubical complex.

NOT IMPLEMENTED

**Parameters**  $n$  (*positive integer; optional, default 1*) – suspend this many times

The suspension is the complex formed by taking the join of the original complex with a two-point complex (the 0-sphere). Since joins are not implemented for cubical complexes, neither are suspensions.

EXAMPLES:

```
sage: C1 = cubical_complexes.Cube(1)
sage: C1.suspension()
Traceback (most recent call last):
...
NotImplementedError: Suspensions are not implemented for cubical complexes.
```

**wedge** (*other*)

The wedge (one-point union) of this cubical complex with another one.

**Parameters** **right** – the other cubical complex (the right-hand factor)

Algorithm: if `self` is embedded in  $d$  dimensions and `other` in  $n$  dimensions, embed them in  $d + n$  dimensions: `self` using the first  $d$  coordinates, `other` using the last  $n$ , translating them so that they have the origin as a common vertex.

---

**Note:** This operation is not well-defined if `self` or `other` is not path-connected.

---

EXAMPLES:

```
sage: S1 = cubical_complexes.Sphere(1)
sage: S2 = cubical_complexes.Sphere(2)
sage: S1.wedge(S2).homology()
{0: 0, 1: Z, 2: Z}
```

**class** `sage.homology.cubical_complex.CubicalComplexExamples`

Some examples of cubical complexes.

Here are the available examples; you can also type “`cubical_complexes.`” and hit TAB to get a list:

```
Sphere
Torus
RealProjectivePlane
KleinBottle
SurfaceOfGenus
Cube
```

EXAMPLES:

```
sage: cubical_complexes.Torus() # indirect doctest
Cubical complex with 16 vertices and 64 cubes
sage: cubical_complexes.Cube(7)
Cubical complex with 128 vertices and 2187 cubes
sage: cubical_complexes.Sphere(7)
Cubical complex with 256 vertices and 6560 cubes
```

**Cube** (*n*)

A cubical complex representation of an  $n$ -dimensional cube.

**Parameters** *n* (*non-negative integer*) – the dimension

EXAMPLES:

```
sage: cubical_complexes.Cube(0)
Cubical complex with 1 vertex and 1 cube
sage: cubical_complexes.Cube(3)
Cubical complex with 8 vertices and 27 cubes
```

**KleinBottle** ()

A cubical complex representation of the Klein bottle, formed by taking the connected sum of the real projective plane with itself.

EXAMPLES:

```
sage: cubical_complexes.KleinBottle()
Cubical complex with 42 vertices and 168 cubes
```

**RealProjectivePlane** ()

A cubical complex representation of the real projective plane. This is taken from the examples from CHomP, the Computational Homology Project: <http://chomp.rutgers.edu/>.

EXAMPLES:

```
sage: cubical_complexes.RealProjectivePlane()
Cubical complex with 21 vertices and 81 cubes
```

**Sphere** (*n*)

A cubical complex representation of the  $n$ -dimensional sphere, formed by taking the boundary of an  $(n + 1)$ -dimensional cube.

**Parameters** *n* (*non-negative integer*) – the dimension of the sphere

EXAMPLES:

```
sage: cubical_complexes.Sphere(7)
Cubical complex with 256 vertices and 6560 cubes
```

**SurfaceOfGenus** (*g*, *orientable=True*)

A surface of genus  $g$  as a cubical complex.

**Parameters**

- *g* (*non-negative integer*) – the genus
- **orientable** (*bool, optional, default True*) – whether the surface should be orientable

In the orientable case, return a sphere if  $g$  is zero, and otherwise return a  $g$ -fold connected sum of a torus with itself.

In the non-orientable case, raise an error if  $g$  is zero. If  $g$  is positive, return a  $g$ -fold connected sum of a real projective plane with itself.

EXAMPLES:

```
sage: cubical_complexes.SurfaceOfGenus(2)
Cubical complex with 32 vertices and 134 cubes
sage: cubical_complexes.SurfaceOfGenus(1, orientable=False)
Cubical complex with 21 vertices and 81 cubes
```

**Torus()**

A cubical complex representation of the torus, obtained by taking the product of the circle with itself.

**EXAMPLES:**

```
sage: cubical_complexes.Torus()
Cubical complex with 16 vertices and 64 cubes
```



## SIMPLICIAL SETS

AUTHORS:

- John H. Palmieri (2016-07)

This module implements simplicial sets.

A *simplicial set*  $X$  is a collection of sets  $X_n$  indexed by the non-negative integers; the set  $X_n$  is called the set of  $n$ -simplices. These sets are connected by maps

$$\begin{aligned}d_i : X_n &\rightarrow X_{n-1}, \quad 0 \leq i \leq n \quad (\text{face maps}) \\s_j : X_n &\rightarrow X_{n+1}, \quad 0 \leq j \leq n \quad (\text{degeneracy maps})\end{aligned}$$

satisfying the *simplicial identities*:

$$\begin{aligned}d_i d_j &= d_{j-1} d_i \quad \text{if } i < j \\d_i s_j &= s_{j-1} d_i \quad \text{if } i < j \\d_j s_j &= 1 = d_{j+1} s_j \\d_i s_j &= s_j d_{i-1} \quad \text{if } i > j + 1 \\s_i s_j &= s_{j+1} s_i \quad \text{if } i < j + 1\end{aligned}$$

See [Wikipedia article Simplicial\\_set](#), Peter May’s seminal book [May1967], or Greg Friedman’s “Illustrated introduction” [Arxiv 0809.4221](#) for more information.

Several simplicial sets are predefined, and users can construct others either by hand (using [SimplicialSet\\_finite](#)) or from existing ones using pushouts, pullbacks, etc.

EXAMPLES:

Some of the predefined simplicial sets:

```
sage: simplicial_sets.Torus()
Torus
sage: simplicial_sets.RealProjectiveSpace(7)
RP^7
sage: S5 = simplicial_sets.Sphere(5)
sage: S5
S^5
sage: S5.nondegenerate_simplices()
[v_0, sigma_5]
```

One class of infinite simplicial sets is available: classifying spaces of groups, or more generally, nerves of finite monoids:

```
sage: Sigma4 = groups.permutation.Symmetric(4)
sage: Sigma4.nerve()
Nerve of Symmetric group of order 4! as a permutation group
```

The same simplicial set (albeit with a different name) can also be constructed as

```
sage: simplicial_sets.ClassifyingSpace(Sigma4)
Classifying space of Symmetric group of order 4! as a permutation group
```

Type `simplicial_sets.` and hit the TAB key to get a full list of the predefined simplicial sets.

You can construct new simplicial sets from old by taking quotients, subsimplicial sets, disjoint unions, wedges (if they are pointed), smash products (if they are pointed and finite), products, pushouts, pullbacks, cones, and suspensions, most of which also have maps associated with them. Wedges, for example:

```
sage: T = simplicial_sets.Torus()
sage: S3 = simplicial_sets.Sphere(3)
sage: T.is_pointed() and S3.is_pointed()
True
sage: T.wedge(S3)
Wedge: (Torus v S^3)
sage: T.disjoint_union(S3) == T.coproduct(S3)
False

sage: W = T.wedge(S3)
sage: W.inclusion_map(0).domain()
Torus
sage: W.projection_map(1).codomain()
Quotient: (Wedge: (Torus v S^3)/Simplicial set with 6 non-degenerate simplices)
```

If the 1-sphere were not already available via `simplicial_sets.Sphere(1)`, you could construct it as follows:

```
sage: pt = simplicial_sets.Simplex(0)
sage: edge = pt.cone()
sage: S1 = edge.quotient(edge.n_skeleton(0))
sage: S1
Quotient: (Cone of 0-simplex/Simplicial set with 2 non-degenerate simplices)
```

At this point, `S1` is pointed: every quotient is automatically given a base point, namely the image of the subcomplex. So its suspension is the reduced suspension, and therefore is small:

```
sage: S5 = S1.suspension(4)
sage: S5
Sigma^4(Quotient: (Cone of 0-simplex/Simplicial set with 2 non-degenerate simplices))
sage: S5.f_vector()
[1, 0, 0, 0, 0, 1]
```

If we forget about the base point in `S1`, we would get the unreduced suspension instead:

```
sage: Z1 = S1.unset_base_point()
sage: Z1.suspension(4).f_vector()
[2, 2, 2, 2, 1, 1]
```

The cone on a pointed simplicial set is the reduced cone. The  $n$ -simplex in Sage is not pointed, but the simplicial set `Point` is.

```
sage: simplicial_sets.Simplex(0).cone().f_vector()
[2, 1]
sage: simplicial_sets.Point().cone().f_vector()
[1]
```

For most simplicial sets (the `Point` is the main exception), each time it is constructed, it gives a distinct copy, and two distinct simplicial sets are never equal:



```
sage: T = simplicial_sets.Torus()
sage: T == simplicial_sets.Torus()
False
sage: T == T
True
sage: simplicial_sets.Torus() == simplicial_sets.Torus()
False
sage: simplicial_sets.Point() == simplicial_sets.Point()
True
```

You can construct subsimplicial sets by specifying a list of simplices, and then you can define the quotient simplicial set:

```
sage: X = simplicial_sets.Simplex(2)
sage: e, f, g = X.n_cells(1)
sage: Y = X.subsimplicial_set([e, f, g])
sage: Z = X.quotient(Y)
```

Or equivalently:

```
sage: Y = X.n_skeleton(1)
sage: Z = X.quotient(Y)
sage: Z
Quotient: (2-simplex/Simplicial set with 6 non-degenerate simplices)
```

Note that subsimplicial sets and quotients come equipped with inclusion and quotient morphisms:

```
sage: inc = Y.inclusion_map()
sage: inc.domain() == Y and inc.codomain() == X
True
sage: quo = Z.quotient_map()
sage: quo.domain()
2-simplex
sage: quo.codomain() == Z
True
```

You can compute homology groups and the fundamental group of any simplicial set:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: eight = S1.wedge(S1)
sage: eight.fundamental_group()
Finitely presented group < e0, e1 | >

sage: Sigma3 = groups.permutation.Symmetric(3)
sage: BSigma3 = Sigma3.nerve()
sage: pi = BSigma3.fundamental_group(); pi
Finitely presented group < e0, e1 | e0^2, e1^3, (e0*e1^-1)^2 >
sage: pi.order()
6
sage: pi.is_abelian()
False

sage: RP6 = simplicial_sets.RealProjectiveSpace(6)
sage: RP6.homology(reduced=False, base_ring=GF(2))
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2,
```

(continues on next page)

(continued from previous page)

```

3: Vector space of dimension 1 over Finite Field of size 2,
4: Vector space of dimension 1 over Finite Field of size 2,
5: Vector space of dimension 1 over Finite Field of size 2,
6: Vector space of dimension 1 over Finite Field of size 2}
sage: RP6.homology(reduced=False, base_ring=QQ)
{0: Vector space of dimension 1 over Rational Field,
1: Vector space of dimension 0 over Rational Field,
2: Vector space of dimension 0 over Rational Field,
3: Vector space of dimension 0 over Rational Field,
4: Vector space of dimension 0 over Rational Field,
5: Vector space of dimension 0 over Rational Field,
6: Vector space of dimension 0 over Rational Field}

```

When infinite simplicial sets are involved, most computations are done by taking an  $n$ -skeleton for an appropriate  $n$ , either implicitly or explicitly:

```

sage: B3 = simplicial_sets.ClassifyingSpace(groups.misc.MultiplicativeAbelian([3]))
sage: B3.disjoint_union(B3).n_skeleton(3)
Disjoint union: (Simplicial set with 15 non-degenerate simplices u Simplicial set,
↳with 15 non-degenerate simplices)
sage: S1 = simplicial_sets.Sphere(1)
sage: B3.product(S1).homology(range(4))
{0: 0, 1: Z x C3, 2: C3, 3: C3}

```

Without the `range` argument, this would raise an error, since  $B3$  is infinite:

```

sage: B3.product(S1).homology()
Traceback (most recent call last):
...
NotImplementedError: this simplicial set may be infinite, so specify dimensions when
↳computing homology

```

It should be easy to construct many simplicial sets from the predefined ones using pushouts, pullbacks, etc., but they can also be constructed “by hand”: first define some simplices, then define a simplicial set by specifying their faces:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: f = AbstractSimplex(1, name='f')
sage: X = SimplicialSet({e: (v,w), f: (w,w)})

```

Now  $e$  is an edge from  $v$  to  $w$  and  $f$  is an edge starting and ending at  $w$ . Therefore the first homology group of  $X$  should be a copy of the integers:

```

sage: X.homology(1)
Z

```

```

sage.homology.simplicial_set.AbstractSimplex(dim, degeneracies=(), underlying=None,
name=None, latex_name=None)

```

An abstract simplex, a building block of a simplicial set.

In a simplicial set, a simplex either is non-degenerate or is obtained by applying degeneracy maps to a non-degenerate simplex.

INPUT:

- `dim` – a non-negative integer, the dimension of the underlying non-degenerate simplex.

- `degeneracies` (optional, default `None`) – a list or tuple of non-negative integers, the degeneracies to be applied.
- `underlying` (optional) – a non-degenerate simplex to which the degeneracies are being applied.
- `name` (optional) – string, a name for this simplex.
- `latex_name` (optional) – string, a name for this simplex to use in the LaTeX representation.

So to define a simplex formed by applying the degeneracy maps  $s_2s_1$  to a 1-simplex, call `AbstractSimplex(1, (2, 1))`.

Specify `underlying` if you need to keep explicit track of the underlying non-degenerate simplex, for example when computing faces of another simplex. This is mainly for use by the method `AbstractSimplex_class.apply_degeneracies()`.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: AbstractSimplex(3, (3, 1))
s_3 s_1 Delta^3
sage: AbstractSimplex(3, None)
Delta^3
sage: AbstractSimplex(3)
Delta^3
```

Simplices may be named (or renamed), affecting how they are printed:

```
sage: AbstractSimplex(0)
Delta^0
sage: v = AbstractSimplex(0, name='v')
sage: v
v
sage: v.rename('w_0')
sage: v
w_0
sage: latex(v)
w_0
sage: latex(AbstractSimplex(0, latex_name='\\sigma'))
\\sigma
```

The simplicial identities are used to put the degeneracies in standard decreasing form:

```
sage: x = AbstractSimplex(0, (0, 0, 0))
sage: x
s_2 s_1 s_0 Delta^0
sage: x.degeneracies()
[2, 1, 0]
```

Use of the `underlying` argument:

```
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(0, (0,), underlying=v)
sage: e
s_0 v
sage: e.nondegenerate() is v
True

sage: e.dimension()
1
```

(continues on next page)

(continued from previous page)

```
sage: e.is_degenerate()
True
```

Distinct non-degenerate simplices are never equal:

```
sage: AbstractSimplex(0, None) == AbstractSimplex(0, None)
False
sage: AbstractSimplex(0, (2,1,0)) == AbstractSimplex(0, (2,1,0))
False

sage: e = AbstractSimplex(0, ((0,)))
sage: f = AbstractSimplex(0, ((0,)))
sage: e == f
False
sage: e.nondegenerate() == f.nondegenerate()
False
```

This means that if, when defining a simplicial set, you specify the faces of a 2-simplex as:

```
(e, e, e)
```

then the faces are the same degenerate vertex, but if you specify the faces as:

```
(AbstractSimplex(0, ((0,))), AbstractSimplex(0, ((0,))), AbstractSimplex(0, ((0,
↪)))
```

then the faces are three different degenerate vertices.

View a command like `AbstractSimplex(0, (2,1,0))` as first constructing `AbstractSimplex(0)` and then applying degeneracies to it, and you always get distinct simplices from different calls to `AbstractSimplex(0)`. On the other hand, if you apply degeneracies to the same non-degenerate simplex, the resulting simplices are equal:

```
sage: v = AbstractSimplex(0)
sage: v.apply_degeneracies(1, 0) == v.apply_degeneracies(1, 0)
True
sage: AbstractSimplex(1, (0,), underlying=v) == AbstractSimplex(1, (0,),
↪underlying=v)
True
```

```
class sage.homology.simplicial_set.AbstractSimplex_class(dim, degeneracies=(),
                                                         underlying=None,
                                                         name=None, latex_name=None)
```

Bases: `sage.structure.sage_object.SageObject`

A simplex of dimension `dim`.

INPUT:

- `dim` – integer, the dimension
- `degeneracies` (optional) – iterable, the indices of the degeneracy maps
- `underlying` (optional) – a non-degenerate simplex
- `name` (optional) – string
- `latex_name` (optional) – string

Users should not call this directly, but instead use `AbstractSimplex()`. See that function for more documentation.

**apply\_degeneracies** (\*args)

Apply the degeneracies given by the arguments args to this simplex.

INPUT:

- args – integers

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: v = AbstractSimplex(0)
sage: e = v.apply_degeneracies(0)
sage: e.nondegenerate() == v
True
sage: f = e.apply_degeneracies(0)
sage: f
s_1 s_0 Delta^0
sage: f.degeneracies()
[1, 0]
sage: f.nondegenerate() == v
True
sage: v.apply_degeneracies(1, 0)
s_1 s_0 Delta^0
```

**degeneracies** ()

Return the list of indices for the degeneracy maps for this simplex.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: AbstractSimplex(4, (0,0,0)).degeneracies()
[2, 1, 0]
sage: AbstractSimplex(4, None).degeneracies()
[]
```

**dimension** ()

The dimension of this simplex.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: AbstractSimplex(3, (2,1)).dimension()
5
sage: AbstractSimplex(3, None).dimension()
3
sage: AbstractSimplex(7).dimension()
7
```

**is\_degenerate** ()

True if this simplex is degenerate.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: AbstractSimplex(3, (2,1)).is_degenerate()
True
```

(continues on next page)

(continued from previous page)

```
sage: AbstractSimplex(3, None).is_degenerate()
False
```

**is\_nondegenerate()**

True if this simplex is non-degenerate.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: AbstractSimplex(3, (2,1)).is_nondegenerate()
False
sage: AbstractSimplex(3, None).is_nondegenerate()
True
sage: AbstractSimplex(5).is_nondegenerate()
True
```

**nondegenerate()**

The non-degenerate simplex underlying this one.

Therefore return itself if this simplex is non-degenerate.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex
sage: v = AbstractSimplex(0, name='v')
sage: sigma = v.apply_degeneracies(1, 0)
sage: sigma.nondegenerate()
v
sage: tau = AbstractSimplex(1, (3,2,1))
sage: x = tau.nondegenerate(); x
Delta^1
sage: x == tau.nondegenerate()
True

sage: AbstractSimplex(1, None)
Delta^1
sage: AbstractSimplex(1, None) == x
False
sage: AbstractSimplex(1, None) == tau.nondegenerate()
False
```

```
class sage.homology.simplicial_set.NonDegenerateSimplex(dim, name=None, latex_name=None)
Bases: sage.homology.simplicial_set.AbstractSimplex_class, sage.misc.fast_methods.WithEqualityById
```

A nondegenerate simplex.

INPUT:

- `dim` – non-negative integer, the dimension
- `name` (optional) – string, a name for this simplex.
- `latex_name` (optional) – string, a name for this simplex to use in the LaTeX representation.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex
sage: v = AbstractSimplex(0, name='v')
sage: v
v
sage: type(v)
<class 'sage.homology.simplicial_set.NonDegenerateSimplex'>

```

Distinct non-degenerate simplices should never be equal, even if they have the same starting data.

```

sage: v == AbstractSimplex(0, name='v')
False
sage: AbstractSimplex(3) == AbstractSimplex(3)
False

sage: from sage.homology.simplicial_set import NonDegenerateSimplex
sage: x = NonDegenerateSimplex(0, name='x')
sage: x == NonDegenerateSimplex(0, name='x')
False

```

sage.homology.simplicial\_set.**SimplicialSet**  
 alias of `sage.homology.simplicial_set.SimplicialSet_finite`

**class** sage.homology.simplicial\_set.**SimplicialSet\_arbitrary**  
 Bases: `sage.structure.parent.Parent`

A simplicial set.

A simplicial set  $X$  is a collection of sets  $X_n$ , the  $n$ -simplices, indexed by the non-negative integers, together with maps

$$d_i : X_n \rightarrow X_{n-1}, \quad 0 \leq i \leq n \quad (\text{face maps})$$

$$s_j : X_n \rightarrow X_{n+1}, \quad 0 \leq j \leq n \quad (\text{degeneracy maps})$$

satisfying the *simplicial identities*:

$$d_i d_j = d_{j-1} d_i \quad \text{if } i < j$$

$$d_i s_j = s_{j-1} d_i \quad \text{if } i < j$$

$$d_j s_j = 1 = d_{j+1} s_j$$

$$d_i s_j = s_j d_{i-1} \quad \text{if } i > j + 1$$

$$s_i s_j = s_{j+1} s_i \quad \text{if } i < j + 1$$

This class is not fully implemented and is not intended to be called directly by users. It is intended instead to be used by other classes which inherit from this one. See `SimplicialSet_finite` and `Nerve` for two examples. In particular, any such class must implement a method `n_skeleton` – without this, most computations will be impossible. It must also implement an `__init__` method which should also set the category, so that methods defined at the category level, like `is_pointed` and `is_finite`, work correctly.

Note that the method `subsimplicial_set()` calls `n_skeleton()`, so to avoid circularity, the `n_skeleton()` method should call `simplicial_set_constructions.SubSimplicialSet` directly, not `subsimplicial_set()`.

**alexander\_whitney**(*simplex*, *dim\_left*)

Return the ‘subdivision’ of `simplex` in this simplicial set into a pair of simplices.

The left factor should have dimension `dim_left`, so the right factor should have dimension `dim - dim_left`, if `dim` is the dimension of the starting simplex. The results are obtained by applying iterated face maps to `simplex`. Writing  $d$  for `dim` and  $j$  for `dim_left`: apply  $d_{j+1}d_{j+2}\dots d_d$  to get the left factor,  $d_0\dots d_j$  to get the right factor.

INPUT:

- `dim_left` – integer, the dimension of the left-hand factor

OUTPUT: a list containing the triple  $(c, \text{left}, \text{right})$ , where `left` and `right` are the two simplices described above. If either `left` or `right` is degenerate, `c` is 0; otherwise, `c` is 1. This is so that, when used to compute cup products, it is easy to ignore terms which have degenerate factors.

EXAMPLES:

```
sage: S2 = simplicial_sets.Sphere(2)
sage: sigma = S2.n_cells(2)[0]
sage: S2.alexander_whitney(sigma, 0)
[(1, v_0, sigma_2)]
sage: S2.alexander_whitney(sigma, 1)
[(0, s_0 v_0, s_0 v_0)]
```

**all\_n\_simplices** (*n*)

Return a list of all simplices, non-degenerate and degenerate, in dimension *n*.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: degen = v.apply_degeneracies(0)
sage: tau = AbstractSimplex(2, name='tau')
sage: Y = SimplicialSet({tau: (degen, degen, degen), w: None})
```

*Y* is the disjoint union of a 2-sphere, with vertex *v* and non-degenerate 2-simplex *tau*, and a point *w*.

```
sage: Y.all_n_simplices(0)
[v, w]
sage: Y.all_n_simplices(1)
[s_0 v, s_0 w]
sage: Y.all_n_simplices(2)
[tau, s_1 s_0 v, s_1 s_0 w]
```

An example involving an infinite simplicial set:

```
sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.all_n_simplices(2)
[f * f,
 f * f^2,
 f^2 * f,
 f^2 * f^2, s_0 f, s_0 f^2, s_1 f, s_1 f^2, s_1 s_0 1]
```

**betti** (*dim=None, subcomplex=None*)

The Betti numbers of this simplicial complex as a dictionary (or a single Betti number, if only one dimension is given): the *i*th Betti number is the rank of the *i*th homology group.

INPUT:

- `dim` (optional, default `None`) – If `None`, then return the homology in every dimension. If `dim` is an integer or list, return the homology in the given dimensions. (Actually, if `dim` is a list, return the homology in the range from `min(dim)` to `max(dim)`.)
- **subcomplex** (optional, default `None`) – a subcomplex of this cell complex. Compute the Betti numbers of the homology relative to this subcomplex.



---

**Note:** If this simplicial set is not finite, you must specify dimensions in which to compute Betti numbers via the argument `dim`.

---

#### EXAMPLES:

Build the two-sphere as a three-fold join of a two-point space with itself:

```
sage: simplicial_sets.Sphere(5).betti()
{0: 1, 1: 0, 2: 0, 3: 0, 4: 0, 5: 1}

sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.betti(range(4))
{0: 1, 1: 0, 2: 0, 3: 0}
```

#### `cartesian_product` (\*others)

Return the product of this simplicial set with others.

##### INPUT:

- `others` – one or several simplicial sets

If  $X$  and  $Y$  are simplicial sets, then their product  $X \times Y$  is defined to be the simplicial set with  $n$ -simplices  $X_n \times Y_n$ . See [simplicial\\_set\\_constructions.ProductOfSimplicialSets](#) for more information.

If a simplicial set is constructed as a product, the factors are recorded and are accessible via the method [simplicial\\_set\\_constructions.Factors.factors\(\)](#). If each factor is finite, then you can also construct the projection maps onto each factor, the wedge as a subcomplex, and the fat wedge as a subcomplex.

#### EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, w)})
sage: square = X.product(X)
```

`square` is now the standard triangulation of the square: 4 vertices, 5 edges (the four on the border and the diagonal), 2 triangles:

```
sage: square.f_vector()
[4, 5, 2]

sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: T.homology(reduced=False)
{0: Z, 1: Z x Z, 2: Z}
```

Since `S1` is pointed, so is `T`:

```
sage: S1.is_pointed()
True
sage: S1.base_point()
v_0
```

(continues on next page)

(continued from previous page)

```

sage: T.is_pointed()
True
sage: T.base_point()
(v_0, v_0)

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: S2xS3 = S2.product(S3)
sage: S2xS3.homology(reduced=False)
{0: Z, 1: 0, 2: Z, 3: Z, 4: 0, 5: Z}

sage: S2xS3.factors() == (S2, S3)
True
sage: S2xS3.factors() == (S3, S2)
False

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: B.rename('RP^oo')
sage: X = B.product(B, S2)
sage: X
RP^oo x RP^oo x S^2
sage: X.factor(1)
RP^oo
sage: X.factors()
(RP^oo, RP^oo, S^2)

```

Projection maps and wedges:

```

sage: S2xS3.projection_map(0)
Simplicial set morphism:
  From: S^2 x S^3
  To:   S^2
  Defn: ...
sage: S2xS3.wedge_as_subset().homology()
{0: 0, 1: 0, 2: Z, 3: Z}

```

In the case of pointed simplicial sets, there is an inclusion of each factor into the product. These are not automatically defined in Sage, but they are easy to construct using identity maps and constant maps and the universal property of the product:

```

sage: one = S2.identity()
sage: const = S2.constant_map(codomain=S3)
sage: S2xS3.universal_property(one, const)
Simplicial set morphism:
  From: S^2
  To:   S^2 x S^3
  Defn: [v_0, sigma_2] --> [(v_0, v_0), (sigma_2, s_1 s_0 v_0)]

```

**cells** (*subcomplex=None*, *max\_dim=None*)

Return a dictionary of all non-degenerate simplices.

INPUT:

- *subcomplex* (optional) – a subsimplicial set of this simplicial set. If *subcomplex* is specified, then return the simplices in the quotient by the subcomplex.
- *max\_dim* – optional, default *None*. If specified, return the non-degenerate simplices of this dimension.

sion or smaller. This argument is required if this simplicial set is infinite.

Each key is a dimension, and the corresponding value is the list of simplices in that dimension.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0)
sage: w = AbstractSimplex(0)
sage: S0 = SimplicialSet({v: None, w: None})
sage: S0.cells()
{0: [Delta^0, Delta^0]}

sage: v.rename('v')
sage: w.rename('w')
sage: S0.cells()
{0: [v, w]}

sage: e = AbstractSimplex(1, name='e')
sage: S1 = SimplicialSet({e: (v, v)})
sage: S1.cells()
{0: [v], 1: [e]}

sage: S0.cells(S0.subsimplicial_set([v, w]))
{0: [*]}

sage: X = SimplicialSet({e: (v,w)})
sage: X.cells(X.subsimplicial_set([v, w]))
{0: [*], 1: [e]}
```

Test an infinite example:

```
sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.cells(max_dim=2)
{0: [1], 1: [f, f^2], 2: [f * f, f * f^2, f^2 * f, f^2 * f^2]}
sage: BC3.cells()
Traceback (most recent call last):
...
NotImplementedError: this simplicial set may be infinite, so specify max_dim
```

**chain\_complex** (*dimensions=None, base\_ring=Integer Ring, augmented=False, cochain=False, verbose=False, subcomplex=None, check=False*)

Return the normalized chain complex.

INPUT:

- *dimensions* – if *None*, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero.
- *base\_ring* (optional, default  $\mathbb{Z}\mathbb{Z}$ ) – commutative ring
- *augmented* (optional, default *False*) – if *True*, return the augmented chain complex (that is, include a class in dimension  $-1$  corresponding to the empty cell).
- *cochain* (optional, default *False*) – if *True*, return the cochain complex (that is, the dual of the chain complex).
- *verbose* (optional, default *False*) – ignored.

- `subcomplex` (optional, default `None`) – if present, compute the chain complex relative to this subcomplex.
- `check` (optional, default `False`) – If `True`, make sure that the chain complex is actually a chain complex: the differentials are composable and their product is zero.

---

**Note:** If this simplicial set is not finite, you must specify dimensions in which to compute its chain complex via the argument `dimensions`.

---

EXAMPLES:

```
sage: simplicial_sets.Sphere(5).chain_complex()
Chain complex with at most 3 nonzero terms over Integer Ring

sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.chain_complex(range(4), base_ring=GF(3))
Chain complex with at most 4 nonzero terms over Finite Field of size 3
```

**cohomology** (*dim=None, \*\*kws*)

Return the cohomology of this simplicial set.

INPUT:

- `dim` (optional, default `None`) – If `None`, then return the homology in every dimension. If `dim` is an integer or list, return the homology in the given dimensions. (Actually, if `dim` is a list, return the homology in the range from `min(dim)` to `max(dim)`.)
- `base_ring` (optional, default `ZZ`) – commutative ring, must be `ZZ` or a field.

Other arguments are also allowed, the same as for the `homology()` method – see `cell_complex.GenericCellComplex.homology()` for complete documentation – except that `homology()` accepts a `cohomology` key word, while this function does not: `cohomology` is automatically true here. Indeed, this function just calls `homology()` with argument `cohomology=True`.

---

**Note:** If this simplicial set is not finite, you must specify dimensions in which to compute homology via the argument `dim`.

---

EXAMPLES:

```
sage: simplicial_sets.KleinBottle().homology(1)
Z x C2
sage: simplicial_sets.KleinBottle().cohomology(1)
Z
sage: simplicial_sets.KleinBottle().cohomology(2)
C2
```

**cone** ()

Return the (reduced) cone on this simplicial set.

If this simplicial set  $X$  is not pointed, construct the ordinary cone: add a point  $v$  (which will become the base point) and for each simplex  $\sigma$  in  $X$ , add both  $\sigma$  and a simplex made up of  $v$  and  $\sigma$  (topologically, form the join of  $v$  and  $\sigma$ ).

If this simplicial set is pointed, then construct the reduced cone: take the quotient of the unreduced cone by the 1-simplex connecting the old base point to the new one.

In either case, as long as the simplicial set is finite, it comes equipped in Sage with a map from it into the cone.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, v)})
sage: CX = X.cone() # unreduced cone, since X not pointed
sage: CX.nondegenerate_simplices()
[* , v, (v,*), e, (e,*)]
sage: CX.base_point()
*
```

$X$  as a subset of the cone, and also the map from  $X$ , in the unreduced case:

```
sage: CX.base_as_subset()
Simplicial set with 2 non-degenerate simplices
sage: CX.map_from_base()
Simplicial set morphism:
From: Simplicial set with 2 non-degenerate simplices
To: Cone of Simplicial set with 2 non-degenerate simplices
Defn: [v, e] --> [v, e]
```

In the reduced case, only the map from  $X$  is available:

```
sage: X = X.set_base_point(v)
sage: CX = X.cone() # reduced cone
sage: CX.nondegenerate_simplices()
[* , e, (e,*)]
sage: CX.map_from_base()
Simplicial set morphism:
From: Simplicial set with 2 non-degenerate simplices
To: Reduced cone of Simplicial set with 2 non-degenerate simplices
Defn: [v, e] --> [* , e]
```

**constant\_map** (codomain=None, point=None)

Return a constant map with this simplicial set as its domain.

INPUT:

- **codomain** – optional, default None. If None, the codomain is the standard one-point space constructed by `Point()`. Otherwise, either the codomain must be a pointed simplicial set, in which case the map is constant at the base point, or **point** must be specified.
- **point** – optional, default None. If specified, it must be a 0-simplex in the codomain, and it will be the target of the constant map.

EXAMPLES:

```
sage: S4 = simplicial_sets.Sphere(4)
sage: S4.constant_map()
Simplicial set morphism:
From: S^4
To: Point
Defn: Constant map at *
sage: S0 = simplicial_sets.Sphere(0)
sage: S4.constant_map(codomain=S0)
```

(continues on next page)

(continued from previous page)

```

Simplicial set morphism:
  From: S^4
  To:   S^0
  Defn: Constant map at v_0

sage: Sigma3 = groups.permutation.Symmetric(3)
sage: Sigma3.nerve().constant_map()
Simplicial set morphism:
  From: Nerve of Symmetric group of order 3! as a permutation group
  To:   Point
  Defn: Constant map at *

```

**coproduct** (\*others)

Return the coproduct of this simplicial set with others.

INPUT:

- others – one or several simplicial sets

If these simplicial sets are pointed, return their wedge sum; if they are not, return their disjoint union. If some are pointed and some are not, raise an error: it is not clear in which category to work.

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: K = simplicial_sets.KleinBottle()
sage: D3 = simplicial_sets.Simplex(3)
sage: Y = S2.unset_base_point()
sage: Z = K.unset_base_point()

sage: S2.coproduct(K).is_pointed()
True
sage: S2.coproduct(K)
Wedge: (S^2 v Klein bottle)
sage: D3.coproduct(Y, Z).is_pointed()
False
sage: D3.coproduct(Y, Z)
Disjoint union: (3-simplex u Simplicial set with 2 non-degenerate simplices u
↳Simplicial set with 6 non-degenerate simplices)

```

The coproduct comes equipped with an inclusion map from each summand, as long as the summands are all finite:

```

sage: S2.coproduct(K).inclusion_map(0)
Simplicial set morphism:
  From: S^2
  To:   Wedge: (S^2 v Klein bottle)
  Defn: [v_0, sigma_2] --> [*, sigma_2]
sage: D3.coproduct(Y, Z).inclusion_map(2)
Simplicial set morphism:
  From: Simplicial set with 6 non-degenerate simplices
  To:   Disjoint union: (3-simplex u Simplicial set with 2 non-degenerate_
↳simplices u Simplicial set with 6 non-degenerate simplices)
  Defn: [Delta_{0,0}, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_{2,0},_
↳Delta_{2,1}] --> [Delta_{0,0}, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_
↳{2,0}, Delta_{2,1}]

```

**disjoint\_union** (\*others)

Return the disjoint union of this simplicial set with `others`.

INPUT:

- `others` – one or several simplicial sets

As long as the factors are all finite, the inclusion map from each factor is available. Any factors which are empty are ignored completely: they do not appear in the list of factors, etc.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: f = AbstractSimplex(1, name='f')
sage: X = SimplicialSet({e: (v, v)})
sage: Y = SimplicialSet({f: (v, w)})
sage: Z = X.disjoint_union(Y)
```

Since `X` and `Y` have simplices in common, Sage uses a copy of `Y` when constructing the disjoint union. Note the name conflict in the list of simplices: `v` appears twice:

```
sage: Z = X.disjoint_union(Y)
sage: Z.nondegenerate_simplices()
[v, v, w, e, f]
```

Factors and inclusion maps:

```
sage: T = simplicial_sets.Torus()
sage: S2 = simplicial_sets.Sphere(2)
sage: A = T.disjoint_union(S2)
sage: A.factors()
(Torus, S^2)
sage: i = A.inclusion_map(0)
sage: i.domain()
Torus
sage: i.codomain()
Disjoint union: (Torus u S^2)
```

Empty factors are ignored:

```
sage: from sage.homology.simplicial_set_examples import Empty
sage: E = Empty()
sage: K = S2.disjoint_union(S2, E, E, S2)
sage: K == S2.disjoint_union(S2, S2)
True
sage: K.factors()
(S^2, S^2, S^2)
```

**face** (*simplex*, *i*)

Return the *i*-th face of *simplex* in this simplicial set.

INPUT:

- `simplex` – a simplex in this simplicial set
- `i` – integer

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: sigma = S2.n_cells(2)[0]
sage: v_0 = S2.n_cells(0)[0]
sage: S2.face(sigma, 0)
s_0 v_0
sage: S2.face(sigma, 0) == v_0.apply_degeneracies(0)
True
sage: S2.face(S2.face(sigma, 0), 0) == v_0
True

```

**faces** (*simplex*)

Return the list of faces of simplex in this simplicial set.

INPUT:

- simplex – a simplex in this simplicial set, either degenerate or not

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: sigma = S2.n_cells(2)[0]
sage: S2.faces(sigma)
(s_0 v_0, s_0 v_0, s_0 v_0)
sage: S2.faces(sigma.apply_degeneracies(0))
[sigma_2, sigma_2, s_1 s_0 v_0, s_1 s_0 v_0]

sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: f2 = BC3.n_cells(1)[1]; f2
f^2
sage: BC3.faces(f2)
(1, 1)

```

**graph** ()

Return the 1-skeleton of this simplicial set, as a graph.

EXAMPLES:

```

sage: Delta3 = simplicial_sets.Simplex(3)
sage: G = Delta3.graph()
sage: G.edges()
[(0,), (1,), (0, 1)),
 ((0,), (2,), (0, 2)),
 ((0,), (3,), (0, 3)),
 ((1,), (2,), (1, 2)),
 ((1,), (3,), (1, 3)),
 ((2,), (3,), (2, 3))]

sage: T = simplicial_sets.Torus()
sage: T.graph()
Looped multi-graph on 1 vertex
sage: len(T.graph().edges())
3

sage: CP3 = simplicial_sets.ComplexProjectiveSpace(3)
sage: G = CP3.graph()
sage: len(G.vertices())
1
sage: len(G.edges())

```

(continues on next page)



(continued from previous page)

```

0

sage: Sigma3 = groups.permutation.Symmetric(3)
sage: Sigma3.nerve().is_connected()
True

```

**homology** (*dim=None, \*\*kws*)

Return the (reduced) homology of this simplicial set.

INPUT:

- *dim* (optional, default None – If None, then return the homology in every dimension. If *dim* is an integer or list, return the homology in the given dimensions. (Actually, if *dim* is a list, return the homology in the range from `min(dim)` to `max(dim)`.)
- *base\_ring* (optional, default ZZ) – commutative ring, must be ZZ or a field.

Other arguments are also allowed: see the documentation for `cell_complex.GenericCellComplex.homology()`.

**Note:** If this simplicial set is not finite, you must specify dimensions in which to compute homology via the argument *dim*.

EXAMPLES:

```

sage: simplicial_sets.Sphere(5).homology()
{0: 0, 1: 0, 2: 0, 3: 0, 4: 0, 5: Z}

sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.homology(range(4), base_ring=GF(3))
{0: Vector space of dimension 0 over Finite Field of size 3,
 1: Vector space of dimension 1 over Finite Field of size 3,
 2: Vector space of dimension 1 over Finite Field of size 3,
 3: Vector space of dimension 1 over Finite Field of size 3}

sage: BC2 = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: BK = BC2.product(BC2)
sage: BK.homology(range(4))
{0: 0, 1: C2 x C2, 2: C2, 3: C2 x C2 x C2}

```

**identity** ()

Return the identity map on this simplicial set.

EXAMPLES:

```

sage: S3 = simplicial_sets.Sphere(3)
sage: S3.identity()
Simplicial set endomorphism of S^3
Defn: Identity map

sage: BC3 = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([3]))
sage: one = BC3.identity()
sage: [(sigma, one(sigma)) for sigma in BC3.n_cells(2)]

```

(continues on next page)

(continued from previous page)

```
[(f * f, f * f),
 (f * f^2, f * f^2),
 (f^2 * f, f^2 * f),
 (f^2 * f^2, f^2 * f^2)]
```

**is\_connected()**

Return True if this simplicial set is connected.

EXAMPLES:

```
sage: T = simplicial_sets.Torus()
sage: K = simplicial_sets.KleinBottle()
sage: X = T.disjoint_union(K)
sage: T.is_connected()
True
sage: K.is_connected()
True
sage: X.is_connected()
False
sage: simplicial_sets.Sphere(0).is_connected()
False
```

**is\_reduced()**

Return True if this simplicial set has only one vertex.

EXAMPLES:

```
sage: simplicial_sets.Sphere(0).is_reduced()
False
sage: simplicial_sets.Sphere(3).is_reduced()
True
```

**join(\*others)**

The join of this simplicial set with others.

Not implemented. See <https://ncatlab.org/nlab/show/join+of+simplicial+sets> for a few descriptions, for anyone interested in implementing this. See also P. J. Ehlers and Tim Porter, Joins for (Augmented) Simplicial Sets, Jour. Pure Applied Algebra, 145 (2000) 37-44 [Arxiv 9904039](https://arxiv.org/abs/9904039).

- others – one or several simplicial sets

EXAMPLES:

```
sage: K = simplicial_sets.Simplex(2)
sage: K.join(K)
Traceback (most recent call last):
...
NotImplementedError: joins are not implemented for simplicial sets
```

**n\_cells(n, subcomplex=None)**

Return the list of cells of dimension  $n$  of this cell complex. If the optional argument `subcomplex` is present, then return the  $n$ -dimensional faces in the quotient by this subcomplex.

INPUT:

- $n$  – the dimension
- `subcomplex` (optional, default `None`) – a subcomplex of this cell complex. Return the cells which are in the quotient by this subcomplex.

EXAMPLES:

```
sage: simplicial_sets.Sphere(3).n_cells(3)
[sigma_3]
sage: simplicial_sets.Sphere(3).n_cells(2)
[]
sage: C2 = groups.misc.MultiplicativeAbelian([2])
sage: BC2 = C2.nerve()
sage: BC2.n_cells(3)
[f * f * f]
```

**n\_chains** (*n*, *base\_ring*=Integer Ring, *cochains*=False)

Return the free module of (normalized) chains in degree *n* over *base\_ring*.

This is the free module on the nondegenerate simplices in the given dimension.

INPUT:

- *n* – integer
- *base\_ring* – ring (optional, default  $\mathbb{Z}$ )
- *cochains* – boolean (optional, default False); if True, return cochains instead

The only difference between chains and cochains is notation: the generator corresponding to the dual of a simplex  $\sigma$  is written as " $\chi_\sigma$ " in the group of cochains.

EXAMPLES:

```
sage: S3 = simplicial_sets.Sphere(3)
sage: C = S3.n_chains(3, cochains=True)
sage: list(C.basis())
[\chi_sigma_3]
sage: Sigma3 = groups.permutation.Symmetric(3)
sage: BSigma3 = simplicial_sets.ClassifyingSpace(Sigma3)
sage: list(BSigma3.n_chains(1).basis())
[(1,2), (1,2,3), (1,3), (1,3,2), (2,3)]
sage: list(BSigma3.n_chains(1, cochains=True).basis())
[\chi_(1,2), \chi_(1,2,3), \chi_(1,3), \chi_(1,3,2), \chi_(2,3)]
```

**nondegenerate\_simplices** (*max\_dim*=None)

Return the sorted list of non-degenerate simplices in this simplicial set.

INPUT:

- *max\_dim* – optional, default None. If specified, return the non-degenerate simplices of this dimension or smaller. This argument is required if this simplicial set is infinite.

The sorting is in increasing order of dimension, and within each dimension, by the name (if present) of each simplex.

---

**Note:** The sorting is done when the simplicial set is constructed, so changing the name of a simplex after construction will not affect the ordering.

---

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0)
sage: w = AbstractSimplex(0)
```

(continues on next page)

(continued from previous page)

```
sage: S0 = SimplicialSet({v: None, w: None})
sage: S0.nondegenerate_simplices()
[Delta^0, Delta^0]
```

Name the vertices and reconstruct the simplicial set: they should be ordered alphabetically:

```
sage: v.rename('v')
sage: w.rename('w')
sage: S0 = SimplicialSet({v: None, w: None})
sage: S0.nondegenerate_simplices()
[v, w]
```

Rename but do not reconstruct the set; the ordering does not take the new names into account:

```
sage: v.rename('z')
sage: S0.nondegenerate_simplices() # old ordering is used
[z, w]

sage: X0 = SimplicialSet({v: None, w: None})
sage: X0.nondegenerate_simplices() # new ordering is used
[w, z]
```

Test an infinite example:

```
sage: C3 = groups.misc.MultiplicativeAbelian([3])
sage: BC3 = simplicial_sets.ClassifyingSpace(C3)
sage: BC3.nondegenerate_simplices(2)
[1, f, f^2, f * f, f * f^2, f^2 * f, f^2 * f^2]
sage: BC3.nondegenerate_simplices()
Traceback (most recent call last):
...
NotImplementedError: this simplicial set may be infinite, so specify max_dim
```

### product (\*others)

Return the product of this simplicial set with others.

INPUT:

- others – one or several simplicial sets

If  $X$  and  $Y$  are simplicial sets, then their product  $X \times Y$  is defined to be the simplicial set with  $n$ -simplices  $X_n \times Y_n$ . See [simplicial\\_set\\_constructions.ProductOfSimplicialSets](#) for more information.

If a simplicial set is constructed as a product, the factors are recorded and are accessible via the method [simplicial\\_set\\_constructions.Factors.factors\(\)](#). If each factor is finite, then you can also construct the projection maps onto each factor, the wedge as a subcomplex, and the fat wedge as a subcomplex.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, w)})
sage: square = X.product(X)
```

square is now the standard triangulation of the square: 4 vertices, 5 edges (the four on the border and the diagonal), 2 triangles:

```
sage: square.f_vector()
[4, 5, 2]

sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: T.homology(reduced=False)
{0: Z, 1: Z x Z, 2: Z}
```

Since  $S_1$  is pointed, so is  $T$ :

```
sage: S1.is_pointed()
True
sage: S1.base_point()
v_0
sage: T.is_pointed()
True
sage: T.base_point()
(v_0, v_0)

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: S2xS3 = S2.product(S3)
sage: S2xS3.homology(reduced=False)
{0: Z, 1: 0, 2: Z, 3: Z, 4: 0, 5: Z}

sage: S2xS3.factors() == (S2, S3)
True
sage: S2xS3.factors() == (S3, S2)
False

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: B.rename('RP^oo')
sage: X = B.product(B, S2)
sage: X
RP^oo x RP^oo x S^2
sage: X.factor(1)
RP^oo
sage: X.factors()
(RP^oo, RP^oo, S^2)
```

Projection maps and wedges:

```
sage: S2xS3.projection_map(0)
Simplicial set morphism:
  From: S^2 x S^3
  To:   S^2
  Defn: ...
sage: S2xS3.wedge_as_subset().homology()
{0: 0, 1: 0, 2: Z, 3: Z}
```

In the case of pointed simplicial sets, there is an inclusion of each factor into the product. These are not automatically defined in Sage, but they are easy to construct using identity maps and constant maps and the universal property of the product:

```

sage: one = S2.identity()
sage: const = S2.constant_map(codomain=S3)
sage: S2xS3.universal_property(one, const)
Simplicial set morphism:
  From: S^2
  To:   S^2 x S^3
  Defn: [v_0, sigma_2] --> [(v_0, v_0), (sigma_2, s_1 s_0 v_0)]

```

**pullback (\*maps)**

Return the pullback obtained from given maps.

INPUT:

- maps – several maps of simplicial sets, each of which has this simplicial set as its codomain

If only a single map  $f : X \rightarrow Y$  is given, then return  $X$ . If more than one map is given, say  $f_i : X_i \rightarrow Y$  for  $0 \leq i \leq m$ , then return the pullback defined by those maps. If no maps are given, return the one-point simplicial set.

In addition to the defining maps  $f_i$  used to construct the pullback  $P$ , there are also maps  $\bar{f}_i : P \rightarrow X_i$ , which we refer to as *structure maps* or *projection maps*. The pullback also has a universal property: given maps  $g_i : Z \rightarrow X_i$  such that  $f_i g_i = f_j g_j$  for all  $i, j$ , then there is a unique map  $g : Z \rightarrow P$  making the appropriate diagram commute: that is,  $\bar{f}_i g = g_i$  for all  $i$ . For example, given maps  $f : X \rightarrow Y$  and  $g : X \rightarrow Z$ , there is an induced map  $g : X \rightarrow Y \times Z$ .

In Sage, a pullback is equipped with its defining maps, and as long as the simplicial sets involved are finite, you can also access the structure maps and the universal property.

EXAMPLES:

Construct a product as a pullback:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: pt = simplicial_sets.Point()
sage: P = pt.pullback(S2.constant_map(), S2.constant_map())
sage: P.homology(2)
Z x Z

```

If the pullback is defined via maps  $f_i : X_i \rightarrow Y$ , then there are structure maps  $\bar{f}_i : Y_i \rightarrow P$ . The structure maps are only available in Sage when all of the maps involved have finite domains.

```

sage: S2 = simplicial_sets.Sphere(2)
sage: one = S2.Hom(S2).identity()
sage: P = S2.pullback(one, one)
sage: P.homology()
{0: 0, 1: 0, 2: Z}

sage: P.defining_map(0) == one
True
sage: P.structure_map(1)
Simplicial set morphism:
  From: Pullback of maps:
  Simplicial set endomorphism of S^2
  Defn: Identity map
  Simplicial set endomorphism of S^2
  Defn: Identity map
  To:   S^2
  Defn: [(v_0, v_0), (sigma_2, sigma_2)] --> [v_0, sigma_2]
sage: P.structure_map(0).domain() == P

```

(continues on next page)

(continued from previous page)

```
True
sage: P.structure_map(0).codomain() == S2
True
```

The universal property:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: K = T.factor(0, as_subset=True)
sage: f = S1.Hom(T)({S1.n_cells(0)[0]:K.n_cells(0)[0], S1.n_cells(1)[0]:K.n_
↳cells(1)[0]})
sage: D = S1.cone() # the cone C(S^1)
sage: g = D.map_from_base() # map from S^1 to C(S^1)
sage: P = T.product(D)
sage: h = P.universal_property(f, g)
sage: h.domain() == S1
True
sage: h.codomain() == P
True
```

### pushout (\*maps)

Return the pushout obtained from given maps.

INPUT:

- maps – several maps of simplicial sets, each of which has this simplicial set as its domain

If only a single map  $f : X \rightarrow Y$  is given, then return  $Y$ . If more than one map is given, say  $f_i : X \rightarrow Y_i$  for  $0 \leq i \leq m$ , then return the pushout defined by those maps. If no maps are given, return the empty simplicial set.

In addition to the defining maps  $f_i$  used to construct the pushout  $P$ , there are also maps  $\bar{f}_i : Y_i \rightarrow P$ , which we refer to as *structure maps*. The pushout also has a universal property: given maps  $g_i : Y_i \rightarrow Z$  such that  $g_i f_i = g_j f_j$  for all  $i, j$ , then there is a unique map  $g : P \rightarrow Z$  making the appropriate diagram commute: that is,  $g \bar{f}_i = g_i$  for all  $i$ .

In Sage, a pushout is equipped with its defining maps, and as long as the simplicial sets involved are finite, you can also access the structure maps and the universal property.

EXAMPLES:

Construct the 4-sphere as a quotient of a 4-simplex:

```
sage: K = simplicial_sets.Simplex(4)
sage: L = K.n_skeleton(3)
sage: S4 = L.pushout(L.constant_map(), L.inclusion_map())
sage: S4
Pushout of maps:
  Simplicial set morphism:
    From: Simplicial set with 30 non-degenerate simplices
    To:    Point
    Defn: Constant map at *
  Simplicial set morphism:
    From: Simplicial set with 30 non-degenerate simplices
    To:    4-simplex
    Defn: [(0,), (1,), (2,), (3,), (4,), (0, 1), (0, 2), (0, 3), (0, 4), (1, 2),
↳(1, 3), (1, 4), (2, 3), (2, 4), (3, 4), (0, 1, 2), (0, 1, 3), (0, 1, 4),
↳(0, 2, 3), (0, 2, 4), (0, 3, 4), (1, 2, 3), (1, 2, 4), (1, 3, 4), (2, 3,
↳4), (0, 1, 2, 3), (0, 1, 2, 4), (0, 1, 3, 4), (0, 2, 3, 4), (1, 2, 3, 4)] --
> [(0,), (1,), (2,), (3,), (4,), (0, 1), (0, 2), (0, 3), (0, 4), (1, 2),
↳(1, 3), (1, 4), (2, 3), (2, 4), (3, 4), (0, 1, 2), (0, 1, 3), (0, 1, 4),
↳(0, 2, 3), (0, 2, 4), (0, 3, 4), (1, 2, 3), (1, 2, 4), (1, 3, 4), (2, 3,
↳4), (0, 1, 2, 3), (0, 1, 2, 4), (0, 1, 3, 4), (0, 2, 3, 4), (1, 2, 3, 4)]
(continues on next page)
```

(continued from previous page)

```
sage: len(S4.nondegenerate_simplices())
2
sage: S4.homology(4)
Z
```

The associated maps:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: K = T.factor(0, as_subset=True)
sage: W = S1.wedge(T) # wedge, constructed as a pushout
sage: W.defined_map(1)
Simplicial set morphism:
  From: Point
  To:   S^1 x S^1
  Defn: Constant map at (v_0, v_0)
sage: W.structure_map(0)
Simplicial set morphism:
  From: S^1
  To:   Wedge: (S^1 v S^1 x S^1)
  Defn: [v_0, sigma_1] --> [*, sigma_1]

sage: f = S1.Hom(T)({S1.n_cells(0)[0]:K.n_cells(0)[0], S1.n_cells(1)[0]:K.n_
↪cells(1)[0]})
```

The maps  $f : S^1 \rightarrow T$  and  $1 : T \rightarrow T$  induce a map  $S^1 \vee T \rightarrow T$ :

```
sage: g = W.universal_property(f, Hom(T,T).identity())
sage: g.domain() == W
True
sage: g.codomain() == T
True
```

**quotient** (*subcomplex*, *vertex\_name*='\*')

Return the quotient of this simplicial set by subcomplex.

That is, subcomplex is replaced by a vertex.

INPUT:

- *subcomplex* – subsimplicial set of this simplicial set, or a list, tuple, or set of simplices defining a subsimplicial set.
- *vertex\_name* (optional) – string, name to be given to the new vertex. By default, use '\*'.

In Sage, from a quotient simplicial set, you can recover the ambient space, the subcomplex, and (if the ambient space is finite) the quotient map.

Base points: if the original simplicial set has a base point not contained in *subcomplex* and if the original simplicial set is finite, then use its image as the base point for the quotient. In all other cases, \* is the base point.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: f = AbstractSimplex(1, name='f')
```

(continues on next page)



(continued from previous page)

```

sage: X = SimplicialSet({e: (v, w), f: (v, w)})
sage: Y = X.quotient([f])
sage: Y.nondegenerate_simplices()
[* , e]
sage: Y.homology(1)
Z

sage: E = SimplicialSet({e: (v, w)})
sage: Z = E.quotient([v, w])
sage: Z.nondegenerate_simplices()
[* , e]
sage: Z.homology(1)
Z

sage: F = E.quotient([v])
sage: F.nondegenerate_simplices()
[* , w, e]
sage: F.base_point()
*

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.homology(base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 0 over Finite Field of size 2,
 2: Vector space of dimension 0 over Finite Field of size 2,
 3: Vector space of dimension 1 over Finite Field of size 2,
 4: Vector space of dimension 1 over Finite Field of size 2,
 5: Vector space of dimension 1 over Finite Field of size 2}

sage: RP5_2.ambient()
RP^5
sage: RP5_2.subcomplex()
Simplicial set with 3 non-degenerate simplices
sage: RP5_2.quotient_map()
Simplicial set morphism:
  From: RP^5
  To:   Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
  Defn: [1, f, f * f, f * f * f, f * f * f * f, f * f * f * f * f] --> [* , s_
↪ 0 * , s_1 s_0 * , f * f * f, f * f * f * f, f * f * f * f * f]

```

Behavior of base points:

```

sage: K = simplicial_sets.Simplex(3)
sage: K.is_pointed()
False
sage: L = K.subsimplicial_set([K.n_cells(1)[-1]])
sage: L.nondegenerate_simplices()
[(2, ), (3, ), (2, 3)]
sage: K.quotient([K.n_cells(1)[-1]]).base_point()
*

sage: K = K.set_base_point(K.n_cells(0)[0])
sage: K.base_point()
(0, )
sage: L = K.subsimplicial_set([K.n_cells(1)[-1]])

```

(continues on next page)

(continued from previous page)

```
sage: L.nondegenerate_simplices()
[(2,), (3,), (2, 3)]
sage: K.quotient(L).base_point()
(0,)
```

**reduce()**

Reduce this simplicial set.

That is, take the quotient by a spanning tree of the 1-skeleton, so that the resulting simplicial set has only one vertex. This only makes sense if the simplicial set is connected, so raise an error if not. If already reduced, return itself.

EXAMPLES:

```
sage: K = simplicial_sets.Simplex(2)
sage: K.is_reduced()
False
sage: X = K.reduce()
sage: X.is_reduced()
True
```

X is reduced, so calling reduce on it again returns X itself:

```
sage: X is X.reduce()
True
sage: K is K.reduce()
False
```

Raise an error for disconnected simplicial sets:

```
sage: S0 = simplicial_sets.Sphere(0)
sage: S0.reduce()
Traceback (most recent call last):
...
ValueError: this simplicial set is not connected
```

**rename\_latex(s)**

Rename or set the LaTeX name for this simplicial set.

INPUT:

- *s* – string, the LaTeX representation. Or *s* can be None, in which case the LaTeX name is unset.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0)
sage: X = SimplicialSet({v: None}, latex_name='*')
sage: latex(X)
*
sage: X.rename_latex('x_0')
sage: latex(X)
x_0
```

**subsimplicial\_set(simplices)**

Return the sub-simplicial set of this simplicial set determined by *simplices*, a set of nondegenerate simplices.

INPUT:

- `simplices` – set, list, or tuple of nondegenerate simplices in this simplicial set, or a simplicial complex – see below.

Each sub-simplicial set comes equipped with an inclusion map to its ambient space, and you can easily recover its ambient space.

If `simplices` is a simplicial complex, then the original simplicial set should itself have been converted from a simplicial complex, and `simplices` should be a subcomplex of that.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: f = AbstractSimplex(1, name='f')

sage: X = SimplicialSet({e: (v, w), f: (w, v)})
sage: Y = X.subsimplicial_set([e])
sage: Y
Simplicial set with 3 non-degenerate simplices
sage: Y.nondegenerate_simplices()
[v, w, e]

sage: S3 = simplicial_complexes.Sphere(3)
sage: K = SimplicialSet(S3)
sage: tau = K.n_cells(3)[0]
sage: tau.dimension()
3
sage: K.subsimplicial_set([tau])
Simplicial set with 15 non-degenerate simplices
```

A subsimplicial set knows about its ambient space and the inclusion map into it:

```
sage: RP4 = simplicial_sets.RealProjectiveSpace(4)
sage: M = RP4.n_skeleton(2)
sage: M
Simplicial set with 3 non-degenerate simplices
sage: M.ambient_space()
RP^4
sage: M.inclusion_map()
Simplicial set morphism:
  From: Simplicial set with 3 non-degenerate simplices
  To:    RP^4
  Defn: [1, f, f * f] --> [1, f, f * f]
```

An infinite ambient simplicial set:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: BxB = B.product(B)
sage: BxB.n_cells(2)[5:]
[(s_0 f, s_1 f), (s_1 f, f * f), (s_1 f, s_0 f), (s_1 s_0 1, f * f)]
sage: BxB.subsimplicial_set(BxB.n_cells(2)[5:])
Simplicial set with 8 non-degenerate simplices
```

**suspension** ( $n=1$ )

Return the (reduced)  $n$ -th suspension of this simplicial set.

INPUT:

- $n$  (optional, default 1) – integer, suspend this many times.

If this simplicial set  $X$  is not pointed, return the suspension: the quotient  $CX/X$ , where  $CX$  is the (ordinary, unreduced) cone on  $X$ . If  $X$  is pointed, then use the reduced cone instead, and so return the reduced suspension.

EXAMPLES:

```
sage: RP4 = simplicial_sets.RealProjectiveSpace(4)
sage: S1 = simplicial_sets.Sphere(1)
sage: SigmaRP4 = RP4.suspension()
sage: S1_smash_RP4 = S1.smash_product(RP4)
sage: SigmaRP4.homology() == S1_smash_RP4.homology()
True
```

The version of the suspension obtained by the smash product is typically less efficient than the reduced suspension produced here:

```
sage: SigmaRP4.f_vector()
[1, 0, 1, 1, 1, 1]
sage: S1_smash_RP4.f_vector()
[1, 1, 4, 6, 8, 5]
```

**wedge** (\*others)

Return the wedge sum of this pointed simplicial set with others.

- others – one or several simplicial sets

This constructs the quotient of the disjoint union in which the base points of all of the simplicial sets have been identified. This is the coproduct in the category of pointed simplicial sets.

This raises an error if any of the factors is not pointed.

From the wedge, you can access the factors, and if the simplicial sets involved are all finite, you can also access the inclusion map of each factor into the wedge, as well as the projection map onto each factor.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: w = AbstractSimplex(0, name='w')
sage: f = AbstractSimplex(1, name='f')
sage: X = SimplicialSet({e: (v, v)}, base_point=v)
sage: Y = SimplicialSet({f: (w, w)}, base_point=w)
sage: W = X.wedge(Y)
sage: W.nondegenerate_simplices()
[*, e, f]
sage: W.homology()
{0: 0, 1: Z x Z}
sage: S2 = simplicial_sets.Sphere(2)
sage: X.wedge(S2).homology(reduced=False)
{0: Z, 1: Z, 2: Z}
sage: X.wedge(X).nondegenerate_simplices()
[*, e, e]

sage: S3 = simplicial_sets.Sphere(3)
sage: W = S2.wedge(S3, S2)
sage: W.inclusion_map(2)
Simplicial set morphism:
```

(continues on next page)

(continued from previous page)

```

From: S^2
To:   Wedge: (S^2 v S^3 v S^2)
Defn: [v_0, sigma_2] --> [*, sigma_2]
sage: W.projection_map(1)
Simplicial set morphism:
From: Wedge: (S^2 v S^3 v S^2)
To:   Quotient: (Wedge: (S^2 v S^3 v S^2)/Simplicial set with 3 non-
↳degenerate simplices)
Defn: [*, sigma_2, sigma_2, sigma_3] --> [*, s_1 s_0 *, s_1 s_0 *, sigma_3]

```

Note that the codomain of the projection map is not identical to the original  $S^2$ , but is instead a quotient of the wedge which is isomorphic to  $S^2$ :

```

sage: S2.f_vector()
[1, 0, 1]
sage: W.projection_map(2).codomain().f_vector()
[1, 0, 1]
sage: (W.projection_map(2) * W.inclusion_map(2)).is_bijective()
True

```

```

class sage.homology.simplicial_set.SimplicialSet_finite(data, base_point=None,
                                                         name=None, check=True,
                                                         category=None, latex_name=None)
Bases: sage.homology.simplicial_set.SimplicialSet_arbitrary, sage.homology.
cell_complex.GenericCellComplex

```

A finite simplicial set.

A simplicial set  $X$  is a collection of sets  $X_n$ , the  $n$ -simplices, indexed by the non-negative integers, together with face maps  $d_i$  and degeneracy maps  $s_j$ . A simplex is *degenerate* if it is in the image of some  $s_j$ , and a simplicial set is *finite* if there are only finitely many non-degenerate simplices.

INPUT:

- `data` – the data defining the simplicial set. See below for details.
- `base_point` (optional, default `None`) – 0-simplex in this simplicial set, its base point
- `name` (optional, default `None`) – string, the name of the simplicial set
- `check` (optional, default `True`) – boolean. If `True`, check the simplicial identity on the face maps when defining the simplicial set.
- `category` (optional, default `None`) – the category in which to define this simplicial set. The default is either finite simplicial sets or finite pointed simplicial sets, depending on whether a base point is defined.
- `latex_name` (optional, default `None`) – string, the LaTeX representation of the simplicial set.

`data` should have one of the following forms: it could be a simplicial complex or  $\Delta$ -complex, in case it is converted to a simplicial set. Alternatively, it could be a dictionary. The keys are the nondegenerate simplices of the simplicial set, and the value corresponding to a simplex  $\sigma$  is a tuple listing the faces of  $\sigma$ . The 0-dimensional simplices may be omitted from `data` if they (or their degeneracies) are faces of other simplices; otherwise they must be included with value `None`.

See [simplicial\\_set](#) and the methods for simplicial sets for more information and examples.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: u = AbstractSimplex(0, name='u')
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: f = AbstractSimplex(1, name='f')

```

In the following simplicial set,  $u$  is an isolated vertex:

```

sage: X = SimplicialSet({e: (v,w), f: (w,w), u: None})
sage: X
Simplicial set with 5 non-degenerate simplices
sage: X.rename('X')
sage: X
X
sage: X = SimplicialSet({e: (v,w), f: (w,w), u: None}, name='Y')
sage: X
Y

```

**algebraic\_topological\_model** (*base\_ring=None*)

Return the algebraic topological model for this simplicial set with coefficients in *base\_ring*.

The term “algebraic topological model” is defined by Pilarczyk and Réal [PR2015].

INPUT:

- *base\_ring* - coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ). Must be a field.

Denote by  $C$  the chain complex associated to this simplicial set. The algebraic topological model is a chain complex  $M$  with zero differential, with the same homology as  $C$ , along with chain maps  $\pi : C \rightarrow M$  and  $\iota : M \rightarrow C$  satisfying  $\iota\pi = 1_M$  and  $\pi\iota$  chain homotopic to  $1_C$ . The chain homotopy  $\phi$  must satisfy

- $\phi\phi = 0$ ,
- $\pi\phi = 0$ ,
- $\phi\iota = 0$ .

Such a chain homotopy is called a *chain contraction*.

OUTPUT: a pair consisting of

- chain contraction  $\phi$  associated to  $C$ ,  $M$ ,  $\pi$ , and  $\iota$
- the chain complex  $M$

Note that from the chain contraction  $\phi$ , one can recover the chain maps  $\pi$  and  $\iota$  via  $\phi.\pi()$  and  $\phi.\iota()$ . Then one can recover  $C$  and  $M$  from, for example,  $\phi.\pi().\text{domain}()$  and  $\phi.\pi().\text{codomain}()$ , respectively.

EXAMPLES:

```

sage: RP2 = simplicial_sets.RealProjectiveSpace(2)
sage: phi, M = RP2.algebraic_topological_model(GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = simplicial_sets.Torus()
sage: phi, M = T.algebraic_topological_model(QQ)
sage: M.homology()

```

(continues on next page)

(continued from previous page)

```
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

**chain\_complex** (*dimensions=None, base\_ring=Integer Ring, augmented=False, cochain=False, verbose=False, subcomplex=None, check=False*)

Return the normalized chain complex.

INPUT:

- *dimensions* – if *None*, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero.
- *base\_ring* (optional, default  $\mathbb{Z}\mathbb{Z}$ ) – commutative ring
- *augmented* (optional, default *False*) – if *True*, return the augmented chain complex (that is, include a class in dimension  $-1$  corresponding to the empty cell).
- *cochain* (optional, default *False*) – if *True*, return the cochain complex (that is, the dual of the chain complex).
- *verbose* (optional, default *False*) – ignored.
- *subcomplex* (optional, default *None*) – if present, compute the chain complex relative to this subcomplex.
- *check* (optional, default *False*) – If *True*, make sure that the chain complex is actually a chain complex: the differentials are composable and their product is zero.

The normalized chain complex of a simplicial set is isomorphic to the chain complex obtained by modding out by degenerate simplices, and the latter is what is actually constructed here.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0)
sage: degen = v.apply_degeneracies(1, 0) # s_1 s_0 applied to v
sage: sigma = AbstractSimplex(3)
sage: S3 = SimplicialSet({sigma: (degen, degen, degen, degen)}) # the 3-sphere
sage: S3.chain_complex().homology()
{0: Z, 3: Z}
sage: S3.chain_complex(augmented=True).homology()
{-1: 0, 0: 0, 3: Z}
sage: S3.chain_complex(dimensions=range(3), base_ring=QQ).homology()
{0: Vector space of dimension 1 over Rational Field}

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP5.chain_complex(subcomplex=RP2).homology()
{0: Z, 3: C2, 4: 0, 5: Z}
```

**euler\_characteristic** ()

Return the Euler characteristic of this simplicial set: the alternating sum over  $n \geq 0$  of the number of nondegenerate  $n$ -simplices.

EXAMPLES:

```
sage: simplicial_sets.RealProjectiveSpace(4).euler_characteristic()
1
```

(continues on next page)

(continued from previous page)

```
sage: simplicial_sets.Sphere(6).euler_characteristic()
2
sage: simplicial_sets.KleinBottle().euler_characteristic()
0
```

**f\_vector()**

Return the list of the number of non-degenerate simplices in each dimension.

Unlike for some other cell complexes in Sage, this does not include the empty simplex in dimension  $-1$ ; thus its  $i$ -th entry is the number of  $i$ -dimensional simplices.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0)
sage: w = AbstractSimplex(0)
sage: S0 = SimplicialSet({v: None, w: None})
sage: S0.f_vector()
[2]

sage: e = AbstractSimplex(1)
sage: S1 = SimplicialSet({e: (v, v)})
sage: S1.f_vector()
[1, 1]
sage: simplicial_sets.Sphere(3).f_vector()
[1, 0, 0, 1]
```

**face\_data()**

Return the face-map data – a dictionary – defining this simplicial set.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, w)})
sage: X.face_data()[e]
(v, w)

sage: Y = SimplicialSet({v: None, w: None})
sage: v in Y.face_data()
True
sage: Y.face_data()[v] is None
True
```

**n\_skeleton(n)**

Return the  $n$ -skeleton of this simplicial set.

That is, the subsimplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
```

(continues on next page)



(continued from previous page)

```
sage: w = AbstractSimplex(0, name='w')
sage: degen = v.apply_degeneracies(0)
sage: tau = AbstractSimplex(2, name='tau')
sage: Y = SimplicialSet({tau: (degen, degen, degen), w: None})
```

$Y$  is the disjoint union of a 2-sphere, with vertex  $v$  and non-degenerate 2-simplex  $\tau$ , and a point  $w$ .

```
sage: Y.nondegenerate_simplices()
[v, w, tau]
sage: Y.n_skeleton(1).nondegenerate_simplices()
[v, w]
sage: Y.n_skeleton(2).nondegenerate_simplices()
[v, w, tau]
```

`sage.homology.simplicial_set.all_degeneracies` ( $n, l=1$ )

Return list of all composites of degeneracies (written in “admissible” form, i.e., as a strictly decreasing sequence) of length  $l$  on an  $n$ -simplex.

INPUT:

- $n, l$  – integers

On an  $n$ -simplex, one may apply the degeneracies  $s_i$  for  $0 \leq i \leq n$ . Then on the resulting  $n+1$ -simplex, one may apply  $s_i$  for  $0 \leq i \leq n+1$ , and so on. But one also has to take into account the simplicial identity

$$s_i s_j = s_{j+1} s_i \text{ if } i \leq j.$$

There are  $\binom{l+n}{n}$  such composites: each non-degenerate  $n$ -simplex leads to  $\binom{l+n}{n}$  degenerate  $l+n$  simplices.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import all_degeneracies
sage: all_degeneracies(0, 3)
{(2, 1, 0)}
sage: all_degeneracies(1, 1)
{(0,), (1,)}
sage: all_degeneracies(1, 3)
{(2, 1, 0), (3, 1, 0), (3, 2, 0), (3, 2, 1)}
```

`sage.homology.simplicial_set.face_degeneracies` ( $m, I$ )

Return the result of applying the face map  $d_m$  to the iterated degeneracy  $s_I = s_{i_1} s_{i_2} \dots s_{i_n}$ .

INPUT:

- $m$  – integer
- $I$  – tuple  $(i_1, i_2, \dots, i_n)$  of integers. We assume that this sequence is strictly decreasing.

Using the simplicial identities (see [simplicial\\_set](#)), we can rewrite

$$d_m s_{i_1} s_{i_2} \dots s_{i_n}$$

in one of the forms

$$s_{j_1} s_{j_2} \dots s_{j_n} d_t, \quad s_{j_1} s_{j_2} \dots s_{j_{n-1}}.$$

OUTPUT: the pair  $(J, t)$  or  $(J, \text{None})$ .  $J$  is returned as a list.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import face_degeneracies
sage: face_degeneracies(0, (1, 0))
([0], None)
sage: face_degeneracies(1, (1, 0))
([0], None)
sage: face_degeneracies(2, (1, 0))
([0], None)
sage: face_degeneracies(3, (1, 0))
([1, 0], 1)
sage: face_degeneracies(3, ())
([], 3)

```

`sage.homology.simplicial_set.shrink_simplicial_complex(K)`

Convert the simplicial complex  $K$  to a “small” simplicial set.

First convert  $K$  naively, then mod out by a large contractible subcomplex, as found by `simplicial_complex.SimplicialComplex._contractible_subcomplex()`. This will produce a simplicial set no larger than, and sometimes much smaller than, the initial simplicial complex.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import shrink_simplicial_complex
sage: K = simplicial_complexes.Simplex(3)
sage: X = shrink_simplicial_complex(K)
sage: X.f_vector()
[1]

sage: Y = simplicial_complexes.Sphere(2)
sage: S2 = shrink_simplicial_complex(Y)
sage: S2
Quotient: (Simplicial set with 14 non-degenerate simplices/Simplicial set with 13_
↳non-degenerate simplices)
sage: S2.f_vector()
[1, 0, 1]
sage: S2.homology()
{0: 0, 1: 0, 2: Z}

sage: Z = simplicial_complexes.SurfaceOfGenus(3)
sage: Z.f_vector()
[1, 15, 57, 38]
sage: Z.homology()
{0: 0, 1: Z^6, 2: Z}
sage: M = shrink_simplicial_complex(Z)
sage: M.f_vector() # random
[1, 32, 27]
sage: M.homology()
{0: 0, 1: Z^6, 2: Z}

```

`sage.homology.simplicial_set.standardize_degeneracies(*L)`

Return list of indices of degeneracy maps in standard (decreasing) order.

INPUT:

- $L$  – list of integers, representing a composition of degeneracies in a simplicial set.

OUTPUT: an equivalent list of degeneracies, standardized to be written in decreasing order, using the simplicial identity

$$s_i s_j = s_{j+1} s_i \text{ if } i \leq j.$$

For example,  $s_0 s_2 = s_3 s_0$  and  $s_0 s_0 = s_1 s_0$ .

EXAMPLES:

```
sage: from sage.homology.simplicial_set import standardize_degeneracies
sage: standardize_degeneracies(0, 0)
(1, 0)
sage: standardize_degeneracies(0, 0, 0, 0)
(3, 2, 1, 0)
sage: standardize_degeneracies(1, 2)
(3, 1)
```

`sage.homology.simplicial_set.standardize_face_maps(*L)`

Return list of indices of face maps in standard (non-increasing) order.

INPUT:

- `L` – list of integers, representing a composition of face maps in a simplicial set.

OUTPUT: an equivalent list of face maps, standardized to be written in non-increasing order, using the simplicial identity

$$d_i d_j = d_{j-1} d_i \text{ if } i < j.$$

For example,  $d_0 d_2 = d_1 d_0$  and  $d_0 d_1 = d_0 d_0$ .

EXAMPLES:

```
sage: from sage.homology.simplicial_set import standardize_face_maps
sage: standardize_face_maps(0, 1)
(0, 0)
sage: standardize_face_maps(0, 2)
(1, 0)
sage: standardize_face_maps(1, 3, 5)
(3, 2, 1)
```



## METHODS OF CONSTRUCTING SIMPLICIAL SETS

This implements various constructions on simplicial sets: subsimplicial sets, pullbacks, products, pushouts, quotients, wedges, disjoint unions, smash products, cones, and suspensions. The best way to access these is with methods attached to simplicial sets themselves, as in the following.

EXAMPLES:

```
sage: K = simplicial_sets.Simplex(1)
sage: square = K.product(K)

sage: K = simplicial_sets.Simplex(1)
sage: endpoints = K.n_skeleton(0)
sage: circle = K.quotient(endpoints)
```

The mapping cone of a morphism of simplicial sets is constructed as a pushout:

```
sage: eta = simplicial_sets.HopfMap()
sage: CP2 = eta.mapping_cone()
sage: type(CP2)
<class 'sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets_finite_
↳with_category'>
```

See the main documentation for simplicial sets, as well as for the classes for pushouts, pullbacks, etc., for more details.

Many of the classes defined here inherit from `sage.structure.unique_representation.UniqueRepresentation`. This means that they produce identical output if given the same input, so for example, if `K` is a simplicial set, calling `K.suspension()` twice returns the same result both times:

```
sage: CP2.suspension() is CP2.suspension()
True
```

So on one hand, a command like `simplicial_sets.Sphere(2)` constructs a distinct copy of a 2-sphere each time it is called; on the other, once you have constructed a 2-sphere, then constructing its cone, its suspension, its product with another simplicial set, etc., will give you the same result each time:

```
sage: simplicial_sets.Sphere(2) == simplicial_sets.Sphere(2)
False
sage: S2 = simplicial_sets.Sphere(2)
sage: S2.product(S2) == S2.product(S2)
True
sage: S2.disjoint_union(CP2, S2) == S2.disjoint_union(CP2, S2)
True
```

AUTHORS:

- John H. Palmieri (2016-07)

**class** sage.homology.simplicial\_set\_constructions.**ConeOfSimplicialSet**(base)  
 Bases: *sage.homology.simplicial\_set.SimplicialSet\_arbitrary*, *sage.structure.unique\_representation.UniqueRepresentation*

Return the unreduced cone on a finite simplicial set.

INPUT:

- base – return the cone on this simplicial set.

Add a point  $*$  (which will become the base point) and for each simplex  $\sigma$  in base, add both  $\sigma$  and a simplex made up of  $*$  and  $\sigma$  (topologically, form the join of  $*$  and  $\sigma$ ).

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, v)})
sage: CX = X.cone() # indirect doctest
sage: CX.nondegenerate_simplices()
[*, v, (v,*), e, (e,*)]
sage: CX.base_point()
*
```

**n\_skeleton**(n)

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- n – the dimension

In the case when the cone is infinite, the  $n$ -skeleton of the cone is computed as the  $n$ -skeleton of the cone of the  $n$ -skeleton.

EXAMPLES:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: X = B.disjoint_union(B)
sage: CX = B.cone()
sage: CX.n_skeleton(3).homology()
{0: 0, 1: 0, 2: 0, 3: Z}
```

**class** sage.homology.simplicial\_set\_constructions.**ConeOfSimplicialSet\_finite**(base)  
 Bases: *sage.homology.simplicial\_set\_constructions.ConeOfSimplicialSet*, *sage.homology.simplicial\_set.SimplicialSet\_finite*

Return the unreduced cone on a finite simplicial set.

INPUT:

- base – return the cone on this simplicial set.

Add a point  $*$  (which will become the base point) and for each simplex  $\sigma$  in base, add both  $\sigma$  and a simplex made up of  $*$  and  $\sigma$  (topologically, form the join of  $*$  and  $\sigma$ ).

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, v)})
sage: CX = X.cone() # indirect doctest
sage: CX.nondegenerate_simplices()
[*, v, (v,*), e, (e,*)]
sage: CX.base_point()
*
```

**base\_as\_subset()**

If this is the cone  $CX$  on  $X$ , return  $X$  as a subsimplicial set.

EXAMPLES:

```

sage: X = simplicial_sets.RealProjectiveSpace(4).unset_base_point()
sage: Y = X.cone()
sage: Y.base_as_subset()
Simplicial set with 5 non-degenerate simplices
sage: Y.base_as_subset() == X
True
```

**map\_from\_base()**

If this is the cone  $CX$  on  $X$ , return the inclusion map from  $X$ .

EXAMPLES:

```

sage: X = simplicial_sets.Simplex(2).n_skeleton(1)
sage: Y = X.cone()
sage: Y.map_from_base()
Simplicial set morphism:
  From: Simplicial set with 6 non-degenerate simplices
  To:   Cone of Simplicial set with 6 non-degenerate simplices
  Defn: [(0,), (1,), (2,), (0, 1), (0, 2), (1, 2)] --> [(0,), (1,), (2,), (0, 1), (0, 2), (1, 2)]
```

**class** sage.homology.simplicial\_set\_constructions.**DisjointUnionOfSimplicialSets** (factors=None)

Bases: `sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets`,  
`sage.homology.simplicial_set_constructions.Factors`

Return the disjoint union of simplicial sets.

INPUT:

- factors – a list or tuple of simplicial sets

Discard any factors which are empty and return the disjoint union of the remaining simplicial sets in `factors`. The disjoint union comes equipped with a map from each factor, as long as all of the factors are finite.

EXAMPLES:

```

sage: CP2 = simplicial_sets.ComplexProjectiveSpace(2)
sage: K = simplicial_sets.KleinBottle()
sage: W = CP2.disjoint_union(K)
sage: W.homology()
{0: Z, 1: Z x C2, 2: Z, 3: 0, 4: Z}

sage: W.inclusion_map(1)
Simplicial set morphism:
```

(continues on next page)

(continued from previous page)

```

From: Klein bottle
To:   Disjoint union: (CP^2 u Klein bottle)
Defn: [Delta_{0,0}, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_{2,0}, Delta_{
↪{2,1}} --> [Delta_{0,0}, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_{2,0}, ↪
↪Delta_{2,1}]

```

**n\_skeleton(n)**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

The  $n$ -skeleton of the disjoint union is computed as the disjoint union of the  $n$ -skeleta of the component simplicial sets.

EXAMPLES:

```

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪MultiplicativeAbelian([2]))
sage: T = simplicial_sets.Torus()
sage: X = B.disjoint_union(T)
sage: X.n_skeleton(3).homology()
{0: Z, 1: Z x Z x C2, 2: Z, 3: Z}

```

**summand(i)**

Return the  $i$ -th factor of this construction of simplicial sets.

INPUT:

- $i$  – integer, the index of the factor

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: K = S2.disjoint_union(S3)
sage: K.factor(0)
S^2
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪MultiplicativeAbelian([2]))
sage: X = B.wedge(S3, B)
sage: X.factor(1)
S^3
sage: X.factor(2)
Classifying space of Multiplicative Abelian group isomorphic to C2

```

**summands()**

Return the factors involved in this construction of simplicial sets.

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: S2.wedge(S3).factors() == (S2, S3)
True
sage: S2.product(S3).factors()[0]
S^2

```



**class** sage.homology.simplicial\_set\_constructions.**DisjointUnionOfSimplicialSets\_finite** (*factors*)  
 Bases: *sage.homology.simplicial\_set\_constructions.DisjointUnionOfSimplicialSets*,  
*sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets\_finite*

The disjoint union of finite simplicial sets.

**inclusion\_map** (*i*)

Return the inclusion map of the *i*-th factor.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: S2 = simplicial_sets.Sphere(2)
sage: W = S1.disjoint_union(S2, S1)
sage: W.inclusion_map(1)
Simplicial set morphism:
  From: S^2
  To:   Disjoint union: (S^1 u S^2 u S^1)
  Defn: [v_0, sigma_2] --> [v_0, sigma_2]
sage: W.inclusion_map(0).domain()
S^1
sage: W.inclusion_map(2).domain()
S^1
```

**class** sage.homology.simplicial\_set\_constructions.**Factors**

Bases: object

Classes which inherit from this should define a `_factors` attribute for their instances, and this class accesses that attribute. This is used by *ProductOfSimplicialSets*, *WedgeOfSimplicialSets*, and *DisjointUnionOfSimplicialSets*.

**factor** (*i*)

Return the *i*-th factor of this construction of simplicial sets.

INPUT:

- *i* – integer, the index of the factor

EXAMPLES:

```
sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: K = S2.disjoint_union(S3)
sage: K.factor(0)
S^2
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: X = B.wedge(S3, B)
sage: X.factor(1)
S^3
sage: X.factor(2)
Classifying space of Multiplicative Abelian group isomorphic to C2
```

**factors** ()

Return the factors involved in this construction of simplicial sets.

EXAMPLES:

```
sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
```

(continues on next page)

(continued from previous page)

```

sage: S2.wedge(S3).factors() == (S2, S3)
True
sage: S2.product(S3).factors()[0]
S^2

```

**class** sage.homology.simplicial\_set\_constructions.**ProductOfSimplicialSets** (*factors=None*)  
 Bases: *sage.homology.simplicial\_set\_constructions.PullbackOfSimplicialSets*,  
*sage.homology.simplicial\_set\_constructions.Factors*

Return the product of simplicial sets.

INPUT:

- *factors* – a list or tuple of simplicial sets

Return the product of the simplicial sets in *factors*.

If  $X$  and  $Y$  are simplicial sets, then their product  $X \times Y$  is defined to be the simplicial set with  $n$ -simplices  $X_n \times Y_n$ . Therefore the simplices in the product have the form  $(s_I \sigma, s_J \tau)$ , where  $s_I = s_{i_1} \dots s_{i_p}$  and  $s_J = s_{j_1} \dots s_{j_q}$  are composites of degeneracy maps, written in decreasing order. Such a simplex is nondegenerate if the indices  $I$  and  $J$  are disjoint. Therefore if  $\sigma$  and  $\tau$  are nondegenerate simplices of dimensions  $m$  and  $n$ , in the product they will lead to nondegenerate simplices up to dimension  $m + n$ , and no further.

This extends in the more or less obvious way to products with more than two factors: with three factors, a simplex  $(s_I \sigma, s_J \tau, s_K \rho)$  is nondegenerate if  $I \cap J \cap K$  is empty, etc.

If a simplicial set is constructed as a product, the factors are recorded and are accessible via the method *Factors.factors()*. If it is constructed as a product and then copied, this information is lost.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, w)})
sage: square = X.product(X)

```

*square* is now the standard triangulation of the square: 4 vertices, 5 edges (the four on the border plus the diagonal), 2 triangles:

```

sage: square.f_vector()
[4, 5, 2]

sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: T.homology(reduced=False)
{0: Z, 1: Z x Z, 2: Z}

```

Since *S1* is pointed, so is *T*:

```

sage: S1.is_pointed()
True
sage: S1.base_point()
v_0
sage: T.is_pointed()
True
sage: T.base_point()
(v_0, v_0)

```

(continues on next page)

(continued from previous page)

```

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: Z = S2.product(S3)
sage: Z.homology()
{0: 0, 1: 0, 2: Z, 3: Z, 4: 0, 5: Z}

```

Products involving infinite simplicial sets:

```

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.MultiplicativeAbelian([2]))
sage: B.rename('RP^oo')
sage: X = B.product(B)
sage: X
RP^oo x RP^oo
sage: X.n_cells(1)
[(f, f), (f, s_0 1), (s_0 1, f)]
sage: X.homology(range(3), base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 2 over Finite Field of size 2,
 2: Vector space of dimension 3 over Finite Field of size 2}
sage: Y = B.product(S2)
sage: Y.homology(range(5), base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 2 over Finite Field of size 2,
 3: Vector space of dimension 2 over Finite Field of size 2,
 4: Vector space of dimension 2 over Finite Field of size 2}

```

**factor** (*i*, *as\_subset=False*)

Return the *i*-th factor of the product.

INPUT:

- *i* – integer, the index of the factor
- *as\_subset* – boolean, optional (default False)

If *as\_subset* is True, return the *i*-th factor as a subsimplicial set of the product, identifying it with its product with the base point in each other factor. As a subsimplicial set, it comes equipped with an inclusion map. This option will raise an error if any factor does not have a base point.

If *as\_subset* is False, return the *i*-th factor in its original form as a simplicial set.

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: K = S2.product(S3)
sage: K.factor(0)
S^2

sage: K.factor(0, as_subset=True)
Simplicial set with 2 non-degenerate simplices
sage: K.factor(0, as_subset=True).homology()
{0: 0, 1: 0, 2: Z}

sage: K.factor(0) is S2
True

```

(continues on next page)

(continued from previous page)

```
sage: K.factor(0, as_subset=True) is S2
False
```

**n\_skeleton**(*n*)

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

In the finite case, this returns the ordinary  $n$ -skeleton. In the infinite case, it computes the  $n$ -skeleton of the product of the  $n$ -skeleta of the factors.

EXAMPLES:

```
sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: S2.product(S3).n_skeleton(2)
Simplicial set with 2 non-degenerate simplices
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: X = B.product(B)
sage: X.n_skeleton(2)
Simplicial set with 13 non-degenerate simplices
```

**class** sage.homology.simplicial\_set\_constructions.**ProductOfSimplicialSets\_finite**(*factors=None*)

Bases: `sage.homology.simplicial_set_constructions.ProductOfSimplicialSets`,  
`sage.homology.simplicial_set_constructions.PullbackOfSimplicialSets_finite`

The product of finite simplicial sets.

When the factors are all finite, there are more methods available for the resulting product, as compared to products with infinite factors: projection maps, the wedge as a subcomplex, and the fat wedge as a subcomplex. See `projection_map()`, `wedge_as_subset()`, and `fat_wedge_as_subset()`

**fat\_wedge\_as\_subset**()

Return the fat wedge as a subsimplicial set of this product of pointed simplicial sets.

The fat wedge consists of those terms where at least one factor is the base point. Thus with two factors this is the ordinary wedge, but with more factors, it is larger.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: X = S1.product(S1, S1)
sage: W = X.fat_wedge_as_subset()
sage: W.homology()
{0: 0, 1: Z x Z x Z, 2: Z x Z x Z}
```

**projection\_map**(*i*)

Return the map projecting onto the  $i$ -th factor.

INPUT:

- $i$  – integer, the index of the projection map

EXAMPLES:

```

sage: T = simplicial_sets.Torus()
sage: f_0 = T.projection_map(0)
sage: f_1 = T.projection_map(1)
sage: m_0 = f_0.induced_homology_morphism().to_matrix(1) # matrix in dim 1
sage: m_1 = f_1.induced_homology_morphism().to_matrix(1)
sage: m_0.rank()
1
sage: m_0 == m_1
False

```

**wedge\_as\_subset()**

Return the wedge as a subsimplicial set of this product of pointed simplicial sets.

This will raise an error if any factor is not pointed.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: w = AbstractSimplex(0, name='w')
sage: f = AbstractSimplex(1, name='f')
sage: X = SimplicialSet({e: (v, v)}, base_point=v)
sage: Y = SimplicialSet({f: (w, w)}, base_point=w)
sage: P = X.product(Y)
sage: W = P.wedge_as_subset()
sage: W.nondegenerate_simplices()
[(v, w), (e, s_0 w), (s_0 v, f)]
sage: W.homology()
{0: 0, 1: Z x Z}

```

**class** sage.homology.simplicial\_set\_constructions.**PullbackOfSimplicialSets**(maps=None)  
 Bases: sage.homology.simplicial\_set.SimplicialSet\_arbitrary, sage.structure.unique\_representation.UniqueRepresentation

Return the pullback obtained from the morphisms maps.

INPUT:

- maps – a list or tuple of morphisms of simplicial sets

If only a single map  $f : X \rightarrow Y$  is given, then return  $X$ . If no maps are given, return the one-point simplicial set. Otherwise, given a simplicial set  $Y$  and maps  $f_i : X_i \rightarrow Y$  for  $0 \leq i \leq m$ , construct the pullback  $P$ : see [Wikipedia article Pullback\\_\(category\\_theory\)](#). This is constructed as pullbacks of sets for each set of  $n$ -simplices, so  $P_n$  is the subset of the product  $\prod (X_i)_n$  consisting of those elements  $(x_i)$  for which  $f_i(x_i) = f_j(x_j)$  for all  $i, j$ .

This is pointed if the maps  $f_i$  are.

EXAMPLES:

The pullback of a quotient map by a subsimplicial set and the base point map gives a simplicial set isomorphic to the original subcomplex:

```

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: K = RP5.quotient(RP5.n_skeleton(2))
sage: X = K.pullback(K.quotient_map(), K.base_point_map())
sage: X.homology() == RP5.n_skeleton(2).homology()
True

```

Pullbacks of identity maps:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: one = S2.Hom(S2).identity()
sage: P = S2.pullback(one, one)
sage: P.homology()
{0: 0, 1: 0, 2: Z}

```

The pullback is constructed in terms of the product – of course, the product is a special case of the pullback – and the simplices are named appropriately:

```

sage: P.nondegenerate_simplices()
[(v_0, v_0), (sigma_2, sigma_2)]

```

#### **defining\_map(*i*)**

Return the  $i$ -th map defining the pullback.

INPUT:

- $i$  – integer

If this pullback was constructed as `Y.pullback(f_0, f_1, ...)`, this returns  $f_i$ .

EXAMPLES:

```

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: K = RP5.quotient(RP5.n_skeleton(2))
sage: Y = K.pullback(K.quotient_map(), K.base_point_map())
sage: Y.defining_map(1)
Simplicial set morphism:
  From: Point
  To:   Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
  Defn: Constant map at *
sage: Y.defining_map(0).domain()
RP^5

```

#### **n\_skeleton(*n*)**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

The  $n$ -skeleton of the pullback is computed as the pullback of the  $n$ -skeleta of the component simplicial sets.

EXAMPLES:

```

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪ MultiplicativeAbelian([2]))
sage: one = Hom(B,B).identity()
sage: c = Hom(B,B).constant_map()
sage: P = B.pullback(one, c)
sage: P.n_skeleton(2)
Pullback of maps:
  Simplicial set endomorphism of Simplicial set with 3 non-degenerate_
↪ simplices
  Defn: Identity map
  Simplicial set endomorphism of Simplicial set with 3 non-degenerate_
↪ simplices

```

(continues on next page)

(continued from previous page)

```
Defn: Constant map at 1
sage: P.n_skeleton(3).homology()
{0: 0, 1: C2, 2: 0, 3: Z}
```

**class** sage.homology.simplicial\_set\_constructions.**PullbackOfSimplicialSets\_finite**(maps=None)  
 Bases: *sage.homology.simplicial\_set\_constructions.PullbackOfSimplicialSets*,  
*sage.homology.simplicial\_set.SimplicialSet\_finite*

The pullback of finite simplicial sets obtained from maps.

When the simplicial sets involved are all finite, there are more methods available to the resulting pullback, as compared to case when some of the components are infinite: the structure maps from the pullback and the pullback's universal property: see *structure\_map()* and *universal\_property()*.

**projection\_map**(i)

Return the  $i$ -th projection map of the pullback.

INPUT:

- $i$  – integer

If this pullback  $P$  was constructed as  $Y.\text{pullback}(f_0, f_1, \dots)$ , where  $f_i : X_i \rightarrow Y$ , then there are structure maps  $\bar{f}_i : P \rightarrow X_i$ . This method constructs  $\bar{f}_i$ .

EXAMPLES:

```
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: K = RP5.quotient(RP5.n_skeleton(2))
sage: Y = K.pullback(K.quotient_map(), K.base_point_map())
sage: Y.structure_map(0)
Simplicial set morphism:
  From: Pullback of maps:
  Simplicial set morphism:
    From: RP^5
    To: Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
    Defn: [1, f, f * f, f * f * f, f * f * f * f, f * f * f * f * f] --> [*,
↪s_0 *, s_1 s_0 *, f * f * f, f * f * f * f, f * f * f * f * f]
  Simplicial set morphism:
    From: Point
    To: Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
    Defn: Constant map at *
  To: RP^5
  Defn: [(1, *), (f, s_0 *), (f * f, s_1 s_0 *)] --> [1, f, f * f]
sage: Y.structure_map(1).codomain()
Point
```

These maps are also accessible via *projection\_map*:

```
sage: Y.projection_map(1).codomain()
Point
```

**structure\_map**(i)

Return the  $i$ -th projection map of the pullback.

INPUT:

- $i$  – integer

If this pullback  $P$  was constructed as  $Y.\text{pullback}(f_0, f_1, \dots)$ , where  $f_i : X_i \rightarrow Y$ , then there are structure maps  $\bar{f}_i : P \rightarrow X_i$ . This method constructs  $\bar{f}_i$ .

EXAMPLES:

```
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: K = RP5.quotient(RP5.n_skeleton(2))
sage: Y = K.pullback(K.quotient_map(), K.base_point_map())
sage: Y.structure_map(0)
Simplicial set morphism:
  From: Pullback of maps:
  Simplicial set morphism:
    From: RP^5
    To:   Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
    Defn: [1, f, f * f, f * f * f, f * f * f * f, f * f * f * f * f] --> [*,
↪s_0 *, s_1 s_0 *, f * f * f, f * f * f * f, f * f * f * f * f]
  Simplicial set morphism:
    From: Point
    To:   Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
    Defn: Constant map at *
  To:   RP^5
  Defn: [(1, *), (f, s_0 *), (f * f, s_1 s_0 *)] --> [1, f, f * f]
sage: Y.structure_map(1).codomain()
Point
```

These maps are also accessible via `projection_map`:

```
sage: Y.projection_map(1).codomain()
Point
```

**universal\_property(\*maps)**

Return the map induced by maps.

INPUT:

- `maps` – maps from a simplicial set  $Z$  to the “factors”  $X_i$  forming the pullback.

If the pullback  $P$  is formed by maps  $f_i : X_i \rightarrow Y$ , then given maps  $g_i : Z \rightarrow X_i$  such that  $f_i g_i = f_j g_j$  for all  $i, j$ , then there is a unique map  $g : Z \rightarrow P$  making the appropriate diagram commute. This constructs that map.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: K = T.factor(0, as_subset=True)
sage: f = S1.Hom(T)({S1.n_cells(0)[0]:K.n_cells(0)[0], S1.n_cells(1)[0]:K.n_
↪cells(1)[0]})
sage: P = S1.product(T)
sage: P.universal_property(S1.Hom(S1).identity(), f)
Simplicial set morphism:
  From: S^1
  To:   S^1 x S^1 x S^1
  Defn: [v_0, sigma_1] --> [(v_0, (v_0, v_0)), (sigma_1, (sigma_1, s_0 v_0))]
```

```
class sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets(maps=None,
                                                                           ver-
                                                                           tex_name=None)
```

Bases: `sage.homology.simplicial_set.SimplicialSet_arbitrary`, `sage.structure.unique_representation.UniqueRepresentation`

Return the pushout obtained from the morphisms `maps`.

INPUT:



- `maps` – a list or tuple of morphisms of simplicial sets
- `vertex_name` – optional, default `None`

If only a single map  $f : X \rightarrow Y$  is given, then return  $Y$ . If no maps are given, return the empty simplicial set. Otherwise, given a simplicial set  $X$  and maps  $f_i : X \rightarrow Y_i$  for  $0 \leq i \leq m$ , construct the pushout  $P$ : see [Wikipedia article Pushout\\_\(category\\_theory\)](#). This is constructed as pushouts of sets for each set of  $n$ -simplices, so  $P_n$  is the disjoint union of the sets  $(Y_i)_n$ , with elements  $f_i(x)$  identified for  $n$ -simplex  $x$  in  $X$ .

Simplices in the pushout are given names as follows: if a simplex comes from a single  $Y_i$ , it inherits its name. Otherwise it must come from a simplex (or several) in  $X$ , and then it inherits one of those names, and it should be the first alphabetically. For example, if vertices  $v$ ,  $w$ , and  $z$  in  $X$  are glued together, then the resulting vertex in the pushout will be called  $v$ .

Base points are taken care of automatically: if each of the maps  $f_i$  is pointed, so is the pushout. If  $X$  is a point or if  $X$  is nonempty and any of the spaces  $Y_i$  is a point, use those for the base point. In all of these cases, if `vertex_name` is `None`, generate the name of the base point automatically; otherwise, use `vertex_name` for its name.

In all other cases, the pushout is not pointed.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: a = AbstractSimplex(0, name='a')
sage: b = AbstractSimplex(0, name='b')
sage: c = AbstractSimplex(0, name='c')
sage: e0 = AbstractSimplex(1, name='e_0')
sage: e1 = AbstractSimplex(1, name='e_1')
sage: e2 = AbstractSimplex(1, name='e_2')
sage: X = SimplicialSet({e2: (b, a)})
sage: Y0 = SimplicialSet({e2: (b,a), e0: (c,b), e1: (c,a)})
sage: Y1 = simplicial_sets.Simplex(0)
sage: f0_data = {a:a, b:b, e2: e2}
sage: v = Y1.n_cells(0)[0]
sage: f1_data = {a:v, b:v, e2:v.apply_degeneracies(0)}
sage: f0 = X.Hom(Y0)(f0_data)
sage: f1 = X.Hom(Y1)(f1_data)
sage: P = X.pushout(f0, f1)
sage: P.nondegenerate_simplices()
[a, c, e_0, e_1]
```

There are defining maps  $f_i : X \rightarrow Y_i$  and structure maps  $\bar{f}_i : Y_i \rightarrow P$ ; the latter are only implemented in Sage when each  $Y_i$  is finite.

```
sage: P.defining_map(0) == f0
True
sage: P.structure_map(1)
Simplicial set morphism:
  From: 0-simplex
  To:   Pushout of maps:
Simplicial set morphism:
  From: Simplicial set with 3 non-degenerate simplices
  To:   Simplicial set with 6 non-degenerate simplices
  Defn: [a, b, e_2] --> [a, b, e_2]
Simplicial set morphism:
  From: Simplicial set with 3 non-degenerate simplices
  To:   0-simplex
```

(continues on next page)

(continued from previous page)

```

Defn: Constant map at (0,)
Defn: Constant map at a
sage: P.structure_map(0).domain() == Y0
True
sage: P.structure_map(0).codomain() == P
True

```

An inefficient way of constructing a suspension for an unpointed set: take the pushout of two copies of the inclusion map  $X \rightarrow CX$ :

```

sage: T = simplicial_sets.Torus()
sage: T = T.unset_base_point()
sage: CT = T.cone()
sage: inc = CT.base_as_subset().inclusion_map()
sage: P = T.pushout(inc, inc)
sage: P.homology()
{0: 0, 1: 0, 2: Z x Z, 3: Z}
sage: len(P.nondegenerate_simplices())
20

```

It is more efficient to construct the suspension as the quotient  $CX/X$ :

```

sage: len(CT.quotient(CT.base_as_subset()).nondegenerate_simplices())
8

```

It is more efficient still if the original simplicial set has a base point:

```

sage: T = simplicial_sets.Torus()
sage: len(T.suspension().nondegenerate_simplices())
6

sage: S1 = simplicial_sets.Sphere(1)
sage: pt = simplicial_sets.Point()
sage: bouquet = pt.pushout(S1.base_point_map(), S1.base_point_map(), S1.base_
↪point_map())
sage: bouquet.homology(1)
Z x Z x Z

```

#### **defining\_map(i)**

Return the  $i$ -th map defining the pushout.

INPUT:

- $i$  – integer

If this pushout was constructed as  $X.\text{pushout}(f_0, f_1, \dots)$ , this returns  $f_i$ .

EXAMPLES:

```

sage: S1 = simplicial_sets.Sphere(1)
sage: T = simplicial_sets.Torus()
sage: X = S1.wedge(T) # a pushout
sage: X.defining_map(0)
Simplicial set morphism:
  From: Point
  To:   S^1
  Defn: Constant map at v_0
sage: X.defining_map(1).domain()

```

(continues on next page)

(continued from previous page)

```
Point
sage: X.defined_map(1).codomain()
Torus
```

**n\_skeleton**(*n*)

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

The  $n$ -skeleton of the pushout is computed as the pushout of the  $n$ -skeleta of the component simplicial sets.

EXAMPLES:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: K = B.n_skeleton(3)
sage: Q = K.pushout(K.inclusion_map(), K.constant_map())
sage: Q.n_skeleton(5).homology()
{0: 0, 1: 0, 2: 0, 3: 0, 4: Z, 5: Z}
```

Of course, computing the  $n$ -skeleton and then taking homology need not yield the same answer as asking for homology through dimension  $n$ , since the latter computation will use the  $(n + 1)$ -skeleton:

```
sage: Q.homology(range(6))
{0: 0, 1: 0, 2: 0, 3: 0, 4: Z, 5: C2}
```

**class** sage.homology.simplicial\_set\_constructions.**PushoutOfSimplicialSets\_finite** (*maps=None*,  
ver-  
tex\_name=None)

Bases: *sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets*,  
*sage.homology.simplicial\_set.SimplicialSet\_finite*

The pushout of finite simplicial sets obtained from maps.

When the simplicial sets involved are all finite, there are more methods available to the resulting pushout, as compared to case when some of the components are infinite: the structure maps to the pushout and the pushout's universal property: see *structure\_map()* and *universal\_property()*.

**structure\_map**(*i*)

Return the  $i$ -th structure map of the pushout.

INPUT:

- $i$  – integer

If this pushout  $Z$  was constructed as  $X.\text{pushout}(f_0, f_1, \dots)$ , where  $f_i : X \rightarrow Y_i$ , then there are structure maps  $\bar{f}_i : Y_i \rightarrow Z$ . This method constructs  $\bar{f}_i$ .

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: T = simplicial_sets.Torus()
sage: X = S1.disjoint_union(T) # a pushout
sage: X.structure_map(0)
Simplicial set morphism:
```

(continues on next page)

(continued from previous page)

```

From: S^1
To:   Disjoint union: (S^1 u Torus)
Defn: [v_0, sigma_1] --> [v_0, sigma_1]
sage: X.structure_map(1).domain()
Torus
sage: X.structure_map(1).codomain()
Disjoint union: (S^1 u Torus)

```

**universal\_property** (\*maps)

Return the map induced by maps

INPUT:

- maps – maps “factors”  $Y_i$  forming the pushout to a fixed simplicial set  $Z$ .

If the pushout  $P$  is formed by maps  $f_i : X \rightarrow Y_i$ , then given maps  $g_i : Y_i \rightarrow Z$  such that  $g_i f_i = g_j f_j$  for all  $i, j$ , then there is a unique map  $g : P \rightarrow Z$  making the appropriate diagram commute. This constructs that map.

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: x = AbstractSimplex(0, name='x')
sage: evw = AbstractSimplex(1, name='vw')
sage: evx = AbstractSimplex(1, name='vx')
sage: ewx = AbstractSimplex(1, name='wx')
sage: X = SimplicialSet({evw: (w, v), evx: (x, v)})
sage: Y_0 = SimplicialSet({evw: (w, v), evx: (x, v), ewx: (x, w)})
sage: Y_1 = SimplicialSet({evx: (x, v)})

sage: f_0 = Hom(X, Y_0)({v:v, w:w, x:x, evw:evw, evx:evx})
sage: f_1 = Hom(X, Y_1)({v:v, w:v, x:x, evw:v.apply_degeneracies(0), evx:evx})
sage: P = X.pushout(f_0, f_1)

sage: one = Hom(Y_1, Y_1).identity()
sage: g = Hom(Y_0, Y_1)({v:v, w:v, x:x, evw:v.apply_degeneracies(0), evx:evx,
↳ ewx:evx})
sage: P.universal_property(g, one)
Simplicial set morphism:
  From: Pushout of maps:
  Simplicial set morphism:
    From: Simplicial set with 5 non-degenerate simplices
    To:   Simplicial set with 6 non-degenerate simplices
    Defn: [v, w, x, vw, vx] --> [v, w, x, vw, vx]
  Simplicial set morphism:
    From: Simplicial set with 5 non-degenerate simplices
    To:   Simplicial set with 3 non-degenerate simplices
    Defn: [v, w, x, vw, vx] --> [v, v, x, s_0 v, vx]
  To:   Simplicial set with 3 non-degenerate simplices
  Defn: [v, x, vx, wx] --> [v, x, vx, vx]

```

**class** sage.homology.simplicial\_set\_constructions.**QuotientOfSimplicialSet** (inclusion, ver-  
tex\_name=’\*)

Bases: sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets

Return the quotient of a simplicial set by a subsimplicial set.

INPUT:

- `inclusion` – inclusion map of a subcomplex (= subsimplicial set) of a simplicial set
- `vertex_name` – optional, default `'*'`

A subcomplex  $A$  comes equipped with the inclusion map  $A \rightarrow X$  to its ambient complex  $X$ , and this constructs the quotient  $X/A$ , collapsing  $A$  to a point. The resulting point is called `vertex_name`, which is `'*'` by default.

When the simplicial sets involved are finite, there is a `QuotientOfSimplicialSet_finite.quotient_map()` method available.

EXAMPLES:

```
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2
Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
sage: RP5_2.quotient_map()
Simplicial set morphism:
  From: RP^5
  To:   Quotient: (RP^5/Simplicial set with 3 non-degenerate simplices)
Defn: [1, f, f * f, f * f * f, f * f * f * f, f * f * f * f * f] --> [*, s_0 *,
↪ s_1 s_0 *, f * f * f, f * f * f * f, f * f * f * f * f]
```

**ambient()**

Return the ambient space.

That is, if this quotient is  $K/L$ , return  $K$ .

EXAMPLES:

```
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.ambient()
RP^5

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪ MultiplicativeAbelian([2]))
sage: K = B.n_skeleton(3)
sage: Q = B.quotient(K)
sage: Q.ambient()
Classifying space of Multiplicative Abelian group isomorphic to C2
```

**n\_skeleton(n)**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

The  $n$ -skeleton of the quotient is computed as the quotient of the  $n$ -skeleta.

EXAMPLES:

```

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: K = B.n_skeleton(3)
sage: Q = B.quotient(K)
sage: Q.n_skeleton(6)
Quotient: (Simplicial set with 7 non-degenerate simplices/Simplicial set with
↳ 4 non-degenerate simplices)
sage: Q.n_skeleton(6).homology()
{0: 0, 1: 0, 2: 0, 3: 0, 4: Z, 5: C2, 6: 0}

```

**subcomplex()**

Return the subcomplex space associated to this quotient.

That is, if this quotient is  $K/L$ , return  $L$ .

EXAMPLES:

```

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.subcomplex()
Simplicial set with 3 non-degenerate simplices

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: K = B.n_skeleton(3)
sage: Q = B.quotient(K)
sage: Q.subcomplex()
Simplicial set with 4 non-degenerate simplices

```

```

class sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet_finite (inclusion,
ver-
tex_name= '*')

```

Bases: `sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet`,  
`sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets_finite`

The quotient of finite simplicial sets.

When the simplicial sets involved are finite, there is a `quotient_map()` method available.

**quotient\_map()**

Return the quotient map from the original simplicial set to the quotient.

EXAMPLES:

```

sage: K = simplicial_sets.Simplex(1)
sage: S1 = K.quotient(K.n_skeleton(0))
sage: q = S1.quotient_map()
sage: q
Simplicial set morphism:
  From: 1-simplex
  To:   Quotient: (1-simplex/Simplicial set with 2 non-degenerate simplices)
  Defn: [(0,), (1,), (0, 1)] --> [*, *, (0, 1)]
sage: q.domain() == K
True
sage: q.codomain() == S1
True

```

```

class sage.homology.simplicial_set_constructions.ReducedConeOfSimplicialSet (base)
  Bases: sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet

```

Return the reduced cone on a simplicial set.

INPUT:

- `base` – return the cone on this simplicial set.

Start with the unreduced cone: take `base` and add a point `*` (which will become the base point) and for each simplex  $\sigma$  in `base`, add both  $\sigma$  and a simplex made up of `*` and  $\sigma$  (topologically, form the join of `*` and  $\sigma$ ).

Now reduce: take the quotient by the 1-simplex connecting the old base point to the new one.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, v)})
sage: X = X.set_base_point(v)
sage: CX = X.cone() # indirect doctest
sage: CX.nondegenerate_simplices()
[*, e, (e,*)]
```

**`n_skeleton(n)`**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- `n` – the dimension

In the case when the cone is infinite, the  $n$ -skeleton of the cone is computed as the  $n$ -skeleton of the cone of the  $n$ -skeleton.

EXAMPLES:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: B.cone().n_skeleton(3).homology()
{0: 0, 1: 0, 2: 0, 3: Z}
```

```
class sage.homology.simplicial_set_constructions.ReducedConeOfSimplicialSet_finite(base)
Bases: sage.homology.simplicial_set_constructions.ReducedConeOfSimplicialSet,
sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet_finite
```

Return the reduced cone on a simplicial set.

INPUT:

- `base` – return the cone on this simplicial set.

Start with the unreduced cone: take `base` and add a point `*` (which will become the base point) and for each simplex  $\sigma$  in `base`, add both  $\sigma$  and a simplex made up of `*` and  $\sigma$  (topologically, form the join of `*` and  $\sigma$ ).

Now reduce: take the quotient by the 1-simplex connecting the old base point to the new one.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: e = AbstractSimplex(1, name='e')
sage: X = SimplicialSet({e: (v, v)})
sage: X = X.set_base_point(v)
```

(continues on next page)

(continued from previous page)

```
sage: CX = X.cone() # indirect doctest
sage: CX.nondegenerate_simplices()
[*, e, (e, *)]
```

**map\_from\_base()**

If this is the cone  $\tilde{C}X$  on  $X$ , return the map from  $X$ .

The map is defined to be the composite  $X \rightarrow CX \rightarrow \tilde{C}X$ . This is used by the `SuspensionOfSimplicialSet_finite` class to construct the reduced suspension: take the quotient of the reduced cone by the image of  $X$  therein.

EXAMPLES:

```
sage: S3 = simplicial_sets.Sphere(3)
sage: CS3 = S3.cone()
sage: CS3.map_from_base()
Simplicial set morphism:
  From: S^3
  To:   Reduced cone of S^3
  Defn: [v_0, sigma_3] --> [*, sigma_3]
```

**class** sage.homology.simplicial\_set\_constructions.**SmashProductOfSimplicialSets\_finite** (factors=

Bases: `sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet_finite`,  
`sage.homology.simplicial_set_constructions.Factors`

Return the smash product of finite pointed simplicial sets.

INPUT:

- `factors` – a list or tuple of simplicial sets

Return the smash product of the simplicial sets in `factors`: the smash product  $X \wedge Y$  is defined to be the quotient  $(X \times Y)/(X \vee Y)$ , where  $X \vee Y$  is the wedge sum.

Each element of `factors` must be finite and pointed. (As of July 2016, constructing the wedge as a subcomplex of the product is only possible in Sage for finite simplicial sets.)

EXAMPLES:

```
sage: T = simplicial_sets.Torus()
sage: S2 = simplicial_sets.Sphere(2)
sage: T.smash_product(S2).homology() == T.suspension(2).homology()
True
```

**class** sage.homology.simplicial\_set\_constructions.**SubSimplicialSet** (data, ambient=None)

Bases: `sage.homology.simplicial_set.SimplicialSet_finite`, `sage.structure.unique_representation.UniqueRepresentation`

Return a finite simplicial set as a subsimplicial set of another simplicial set.

This keeps track of the ambient simplicial set and the inclusion map from the subcomplex into it.

INPUT:

- `data` – the data defining the subset: a dictionary where the keys are simplices from the ambient simplicial set and the values are their faces.
- `ambient` – the ambient simplicial set. If omitted, use the same simplicial set as the subset and the ambient complex.

EXAMPLES:



```

sage: S3 = simplicial_sets.Sphere(3)
sage: K = simplicial_sets.KleinBottle()
sage: X = S3.disjoint_union(K)
sage: Y = X.structure_map(0).image() # the S3 summand
sage: Y.inclusion_map()
Simplicial set morphism:
  From: Simplicial set with 2 non-degenerate simplices
  To:   Disjoint union: (S^3 u Klein bottle)
  Defn: [v_0, sigma_3] --> [v_0, sigma_3]
sage: Y.ambient_space()
Disjoint union: (S^3 u Klein bottle)

```

**ambient\_space()**

Return the simplicial set of which this is a subsimplicial set.

EXAMPLES:

```

sage: T = simplicial_sets.Torus()
sage: eight = T.wedge_as_subset()
sage: eight
Simplicial set with 3 non-degenerate simplices
sage: eight.fundamental_group()
Finitely presented group < e0, e1 | >
sage: eight.ambient_space()
Torus

```

**inclusion\_map()**

Return the inclusion map from this subsimplicial set into its ambient space.

EXAMPLES:

```

sage: RP6 = simplicial_sets.RealProjectiveSpace(6)
sage: K = RP6.n_skeleton(2)
sage: K.inclusion_map()
Simplicial set morphism:
  From: Simplicial set with 3 non-degenerate simplices
  To:   RP^6
  Defn: [1, f, f * f] --> [1, f, f * f]

```

$RP^6$  itself is constructed as a subsimplicial set of  $RP^\infty$ :

```

sage: latex(RP6.inclusion_map())
RP^{6} \to RP^{\infty}

```

**class** sage.homology.simplicial\_set\_constructions.**SuspensionOfSimplicialSet** (*base*)  
 Bases: *sage.homology.simplicial\_set.SimplicialSet\_arbitrary*, *sage.structure.unique\_representation.UniqueRepresentation*

Return the (reduced) suspension of a simplicial set.

INPUT:

- *base* – return the suspension of this simplicial set.

If this simplicial set  $X=base$  is not pointed, or if it is itself an unreduced suspension, return the unreduced suspension: the quotient  $CX/X$ , where  $CX$  is the (ordinary, unreduced) cone on  $X$ . If  $X$  is pointed, then use the reduced cone instead, and so return the reduced suspension.

We use  $S$  to denote unreduced suspension,  $\Sigma$  for reduced suspension.

EXAMPLES:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.MultiplicativeAbelian([2]))
sage: B.suspension()
Sigma(Classifying space of Multiplicative Abelian group isomorphic to C2)
sage: B.suspension().n_skeleton(3).homology()
{0: 0, 1: 0, 2: C2, 3: 0}
```

If  $X$  is finite, the suspension comes with a quotient map from the cone:

```
sage: S3 = simplicial_sets.Sphere(3)
sage: S4 = S3.suspension()
sage: S4.quotient_map()
Simplicial set morphism:
  From: Reduced cone of  $S^3$ 
  To:   Sigma( $S^3$ )
  Defn: [ $*$ , sigma_3, (sigma_3,*)] --> [ $*$ , s_2 s_1 s_0 *, (sigma_3,*)]
```

**n\_skeleton( $n$ )**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

INPUT:

- $n$  – the dimension

In the case when the suspension is infinite, the  $n$ -skeleton of the suspension is computed as the  $n$ -skeleton of the suspension of the  $n$ -skeleton.

EXAMPLES:

```
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: SigmaB = B.suspension()
sage: SigmaB.n_skeleton(4).homology(base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 0 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2,
 3: Vector space of dimension 1 over Finite Field of size 2,
 4: Vector space of dimension 1 over Finite Field of size 2}
```

**class** sage.homology.simplicial\_set\_constructions.**SuspensionOfSimplicialSet\_finite**(base)

Bases: *sage.homology.simplicial\_set\_constructions.SuspensionOfSimplicialSet*,  
*sage.homology.simplicial\_set\_constructions.QuotientOfSimplicialSet\_finite*

The (reduced) suspension of a finite simplicial set.

See *SuspensionOfSimplicialSet* for more information.

**class** sage.homology.simplicial\_set\_constructions.**WedgeOfSimplicialSets**(factors=None)

Bases: *sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets*,  
*sage.homology.simplicial\_set\_constructions.Factors*

Return the wedge sum of pointed simplicial sets.

INPUT:

- factors – a list or tuple of simplicial sets

Return the wedge of the simplicial sets in factors: the wedge sum  $X \vee Y$  is formed by taking the disjoint union of  $X$  and  $Y$  and identifying their base points. In this construction, the new base point is renamed ‘\*’.

The wedge comes equipped with maps to and from each factor, or actually, maps from each factor, and maps to simplicial sets isomorphic to each factor. The codomains of the latter maps are quotients of the wedge, not identical to the original factors.

EXAMPLES:

```
sage: CP2 = simplicial_sets.ComplexProjectiveSpace(2)
sage: K = simplicial_sets.KleinBottle()
sage: W = CP2.wedge(K)
sage: W.homology()
{0: 0, 1: Z x C2, 2: Z, 3: 0, 4: Z}

sage: W.inclusion_map(1)
Simplicial set morphism:
  From: Klein bottle
  To:   Wedge: (CP^2 v Klein bottle)
  Defn: [Delta_{0,0}, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_{2,0}, Delta_{
↪{2,1}] --> [*, Delta_{1,0}, Delta_{1,1}, Delta_{1,2}, Delta_{2,0}, Delta_{2,1}]

sage: W.projection_map(0).domain()
Wedge: (CP^2 v Klein bottle)
sage: W.projection_map(0).codomain() # copy of CP^2
Quotient: (Wedge: (CP^2 v Klein bottle)/Simplicial set with 6 non-degenerate_
↪simplices)
sage: W.projection_map(0).codomain().homology()
{0: 0, 1: 0, 2: Z, 3: 0, 4: Z}
```

An error occurs if any of the factors is not pointed:

```
sage: CP2.wedge(simplicial_sets.Simplex(1))
Traceback (most recent call last):
...
ValueError: the simplicial sets must be pointed
```

**summand(*i*)**

Return the *i*-th factor of this construction of simplicial sets.

INPUT:

- *i* – integer, the index of the factor

EXAMPLES:

```
sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: K = S2.disjoint_union(S3)
sage: K.factor(0)
S^2
sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪MultiplicativeAbelian([2]))
sage: X = B.wedge(S3, B)
sage: X.factor(1)
S^3
sage: X.factor(2)
Classifying space of Multiplicative Abelian group isomorphic to C2
```

**summands()**

Return the factors involved in this construction of simplicial sets.

EXAMPLES:

```

sage: S2 = simplicial_sets.Sphere(2)
sage: S3 = simplicial_sets.Sphere(3)
sage: S2.wedge(S3).factors() == (S2, S3)
True
sage: S2.product(S3).factors()[0]
S^2

```

**class** sage.homology.simplicial\_set\_constructions.**WedgeOfSimplicialSets\_finite** (*factors=None*)  
 Bases: *sage.homology.simplicial\_set\_constructions.WedgeOfSimplicialSets*,  
*sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets\_finite*

The wedge sum of finite pointed simplicial sets.

**inclusion\_map** (*i*)

Return the inclusion map of the *i*-th factor.

EXAMPLES:

```

sage: S1 = simplicial_sets.Sphere(1)
sage: S2 = simplicial_sets.Sphere(2)
sage: W = S1.wedge(S2, S1)
sage: W.inclusion_map(1)
Simplicial set morphism:
  From: S^2
  To:   Wedge: (S^1 v S^2 v S^1)
  Defn: [v_0, sigma_2] --> [*, sigma_2]
sage: W.inclusion_map(0).domain()
S^1
sage: W.inclusion_map(2).domain()
S^1

```

**projection\_map** (*i*)

Return the projection map onto the *i*-th factor.

EXAMPLES:

```

sage: S1 = simplicial_sets.Sphere(1)
sage: S2 = simplicial_sets.Sphere(2)
sage: W = S1.wedge(S2, S1)
sage: W.projection_map(1)
Simplicial set morphism:
  From: Wedge: (S^1 v S^2 v S^1)
  To:   Quotient: (Wedge: (S^1 v S^2 v S^1)/Simplicial set with 3 non-
↳degenerate simplices)
  Defn: [*, sigma_1, sigma_1, sigma_2] --> [*, s_0 *, s_0 *, sigma_2]
sage: W.projection_map(1).image().homology(1)
0
sage: W.projection_map(1).image().homology(2)
Z

```

## EXAMPLES OF SIMPLICIAL SETS.

These are accessible via `simplicial_sets.Sphere(3)`, `simplicial_sets.Torus()`, etc. Type `simplicial_sets.[TAB]` to see a complete list.

AUTHORS:

- John H. Palmieri (2016-07)

`sage.homology.simplicial_set_examples.ClassifyingSpace(group)`

Return the classifying space of group, as a simplicial set.

INPUT:

- group – a finite group or finite monoid

See `sage.categories.finite_monoids.FiniteMonoids.ParentMethods.nerve()` for more details and more examples.

EXAMPLES:

```
sage: C2 = groups.misc.MultiplicativeAbelian([2])
sage: BC2 = simplicial_sets.ClassifyingSpace(C2)
sage: H = BC2.homology(range(9), base_ring=GF(2))
sage: [H[i].dimension() for i in range(9)]
[0, 1, 1, 1, 1, 1, 1, 1, 1]

sage: Klein4 = groups.misc.MultiplicativeAbelian([2, 2])
sage: BK = simplicial_sets.ClassifyingSpace(Klein4)
sage: BK
Classifying space of Multiplicative Abelian group isomorphic to C2 x C2
sage: BK.homology(range(5), base_ring=GF(2)) # long time (1 second)
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 2 over Finite Field of size 2,
 2: Vector space of dimension 3 over Finite Field of size 2,
 3: Vector space of dimension 4 over Finite Field of size 2,
 4: Vector space of dimension 5 over Finite Field of size 2}
```

`sage.homology.simplicial_set_examples.ComplexProjectiveSpace(n)`

Return complex  $n$ -dimensional projective space, as a simplicial set.

This is only defined when  $n$  is at most 4. It is constructed using the simplicial set decomposition provided by Kenzo, as described by Sergeraert [Ser2010]

EXAMPLES:

```
sage: simplicial_sets.ComplexProjectiveSpace(2).homology(reduced=False)
{0: Z, 1: 0, 2: Z, 3: 0, 4: Z}
sage: CP3 = simplicial_sets.ComplexProjectiveSpace(3)
```

(continues on next page)

(continued from previous page)

```

sage: CP3
CP^3
sage: latex(CP3)
CP^{3}
sage: CP3.f_vector()
[1, 0, 3, 10, 25, 30, 15]

sage: K = CP3.suspension() # long time (1 second)
sage: R = K.cohomology_ring(GF(2)) # long time
sage: R.gens() # long time
(h^{0,0}, h^{3,0}, h^{5,0}, h^{7,0})
sage: x = R.gens()[1] # long time
sage: x.Sq(2) # long time
h^{5,0}

sage: simplicial_sets.ComplexProjectiveSpace(4).f_vector()
[1, 0, 4, 22, 97, 255, 390, 315, 105]

sage: simplicial_sets.ComplexProjectiveSpace(5)
Traceback (most recent call last):
...
ValueError: complex projective spaces are only available in dimensions between 0_
↪and 4

```

`sage.homology.simplicial_set_examples.Empty()`

Return the empty simplicial set.

This should return the same simplicial set each time it is called.

EXAMPLES:

```

sage: from sage.homology.simplicial_set_examples import Empty
sage: E = Empty()
sage: E
Empty simplicial set
sage: E.nondegenerate_simplices()
[]
sage: E is Empty()
True

```

`sage.homology.simplicial_set_examples.HopfMap()`

Return a simplicial model of the Hopf map  $S^3 \rightarrow S^2$

This is taken from Exemple II.1.19 in the thesis of Clemens Berger [Ber1991].

The Hopf map is a fibration  $S^3 \rightarrow S^2$ . If it is viewed as attaching a 4-cell to the 2-sphere, the resulting adjunction space is 2-dimensional complex projective space. The resulting model is a bit larger than the one obtained from `simplicial_sets.ComplexProjectiveSpace(2)`.

EXAMPLES:

```

sage: g = simplicial_sets.HopfMap()
sage: g.domain()
Simplicial set with 20 non-degenerate simplices
sage: g.codomain()
S^2

```

Using the Hopf map to attach a cell:

```

sage: X = g.mapping_cone()
sage: CP2 = simplicial_sets.ComplexProjectiveSpace(2)
sage: X.homology() == CP2.homology()
True

sage: X.f_vector()
[1, 0, 5, 9, 6]
sage: CP2.f_vector()
[1, 0, 2, 3, 3]

```

`sage.homology.simplicial_set_examples.Horn(n, k)`

Return the horn  $\Lambda_k^n$ .

This is the subsimplicial set of the  $n$ -simplex obtained by removing its  $k$ -th face.

EXAMPLES:

```

sage: L = simplicial_sets.Horn(3, 0)
sage: L
(3, 0)-Horn
sage: L.n_cells(3)
[]
sage: L.n_cells(2)
[(0, 1, 2), (0, 1, 3), (0, 2, 3)]

sage: L20 = simplicial_sets.Horn(2, 0)
sage: latex(L20)
\Lambda^{2}_{0}
sage: L20.inclusion_map()
Simplicial set morphism:
  From: (2, 0)-Horn
  To:    2-simplex
  Defn: [(0,), (1,), (2,), (0, 1), (0, 2)] --> [(0,), (1,), (2,), (0, 1), (0, 2)]

```

`sage.homology.simplicial_set_examples.KleinBottle()`

Return the Klein bottle as a simplicial set.

This converts the  $\Delta$ -complex version to a simplicial set. It has one 0-simplex, three 1-simplices, and two 2-simplices.

EXAMPLES:

```

sage: K = simplicial_sets.KleinBottle()
sage: K.f_vector()
[1, 3, 2]
sage: K.homology(reduced=False)
{0: Z, 1: Z x C2, 2: 0}
sage: K
Klein bottle

```

**class** `sage.homology.simplicial_set_examples.Nerve(monoid)`

Bases: `sage.homology.simplicial_set.SimplicialSet_arbitrary`

The nerve of a multiplicative monoid.

INPUT:

- *monoid* – a multiplicative monoid

See `sage.categories.finite_monoids.FiniteMonoids.ParentMethods.nerve()` for full documentation.

## EXAMPLES:

```

sage: M = FiniteMonoids().example()
sage: M
An example of a finite multiplicative monoid: the integers modulo 12
sage: X = M.nerve()
sage: list(M)
[0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11]
sage: X.n_cells(0)
[1]
sage: X.n_cells(1)
[0, 10, 11, 2, 3, 4, 5, 6, 7, 8, 9]

```

**n\_skeleton(*n*)**

Return the  $n$ -skeleton of this simplicial set.

That is, the simplicial set generated by all nondegenerate simplices of dimension at most  $n$ .

## INPUT:

- $n$  – the dimension

## EXAMPLES:

```

sage: K4 = groups.misc.MultiplicativeAbelian([2,2])
sage: BK4 = simplicial_sets.ClassifyingSpace(K4)
sage: BK4.n_skeleton(3)
Simplicial set with 40 non-degenerate simplices
sage: BK4.n_cells(1) == BK4.n_skeleton(3).n_cells(1)
True
sage: BK4.n_cells(3) == BK4.n_skeleton(1).n_cells(3)
False

```

`sage.homology.simplicial_set_examples.Point()`

Return a single point called “\*” as a simplicial set.

This should return the same simplicial set each time it is called.

## EXAMPLES:

```

sage: P = simplicial_sets.Point()
sage: P.is_pointed()
True
sage: P.nondegenerate_simplices()
[*]

sage: Q = simplicial_sets.Point()
sage: P is Q
True
sage: P == Q
True

```

`sage.homology.simplicial_set_examples.RealProjectiveSpace(n)`

Return real  $n$ -dimensional projective space, as a simplicial set.

This is constructed as the  $n$ -skeleton of the nerve of the group of order 2, and therefore has a single non-degenerate simplex in each dimension up to  $n$ .

## EXAMPLES:



```

sage: simplicial_sets.RealProjectiveSpace(7)
RP^7
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP5.homology()
{0: 0, 1: C2, 2: 0, 3: C2, 4: 0, 5: Z}
sage: RP5
RP^5
sage: latex(RP5)
RP^{5}

sage: BC2 = simplicial_sets.RealProjectiveSpace(Infinity)
sage: latex(BC2)
RP^{\infty}

```

`sage.homology.simplicial_set_examples.Simplex(n)`  
Return the  $n$ -simplex as a simplicial set.

EXAMPLES:

```

sage: K = simplicial_sets.Simplex(2)
sage: K
2-simplex
sage: latex(K)
\Delta^{2}
sage: K.n_cells(0)
[(0,), (1,), (2,)]
sage: K.n_cells(1)
[(0, 1), (0, 2), (1, 2)]
sage: K.n_cells(2)
[(0, 1, 2)]

```

`sage.homology.simplicial_set_examples.Sphere(n)`  
Return the  $n$ -sphere as a simplicial set.

It is constructed with two non-degenerate simplices: a vertex  $v_0$  (which is the base point) and an  $n$ -simplex  $\sigma_n$ .

INPUT:

- $n$  – integer

EXAMPLES:

```

sage: S0 = simplicial_sets.Sphere(0)
sage: S0
S^0
sage: S0.nondegenerate_simplices()
[v_0, w_0]
sage: S0.is_pointed()
True
sage: simplicial_sets.Sphere(4)
S^4
sage: latex(simplicial_sets.Sphere(4))
S^{4}
sage: simplicial_sets.Sphere(4).nondegenerate_simplices()
[v_0, sigma_4]

```

`sage.homology.simplicial_set_examples.Torus()`  
Return the torus as a simplicial set.

This computes the product of the circle with itself, where the circle is represented using a single 0-simplex and a single 1-simplex. Thus it has one 0-simplex, three 1-simplices, and two 2-simplices.

EXAMPLES:

```
sage: T = simplicial_sets.Torus()
sage: T.f_vector()
[1, 3, 2]
sage: T.homology(reduced=False)
{0: Z, 1: Z x Z, 2: Z}
```

`sage.homology.simplicial_set_examples.simplicial_data_from_kenzo_output(filename)`  
Return data to construct a simplicial set, given Kenzo output.

INPUT:

- `filename` – name of file containing the output from Kenzo's `show-structure()` function

OUTPUT: data to construct a simplicial set from the Kenzo output

Several files with Kenzo output are in the directory `SAGE_EXTCODE/kenzo/`.

EXAMPLES:

```
sage: from sage.homology.simplicial_set_examples import simplicial_data_from_
      ↪kenzo_output
sage: from sage.homology.simplicial_set import SimplicialSet
sage: sphere = os.path.join(SAGE_ENV['SAGE_EXTCODE'], 'kenzo', 'S4.txt')
sage: S4 = SimplicialSet(simplicial_data_from_kenzo_output(sphere))
sage: S4.homology(reduced=False)
{0: Z, 1: 0, 2: 0, 3: 0, 4: Z}
```

## CATALOG OF SIMPLICIAL SETS

This provides pre-built simplicial sets:

- the  $n$ -sphere and  $n$ -dimensional real projective space, both (in theory) for any positive integer  $n$ . In practice, as  $n$  increases, it takes longer to construct these simplicial sets.
- the  $n$ -simplex and the horns obtained from it. As  $n$  increases, it takes *much* longer to construct these simplicial sets, because the number of nondegenerate simplices increases exponentially in  $n$ . For example, it is feasible to do `simplicial_sets.RealProjectiveSpace(100)` since it only has 101 nondegenerate simplices, but `simplicial_sets.Simplex(20)` is probably a bad idea.
- $n$ -dimensional complex projective space for  $n \leq 4$
- the classifying space of a finite multiplicative group or monoid
- the torus and the Klein bottle
- the point
- the Hopf map: this is a pre-built morphism, from which one can extract its domain, codomain, mapping cone, etc.

All of these examples are accessible by typing `simplicial_sets.NAME`, where `NAME` is the name of the example. Type `simplicial_sets.[TAB]` for a complete list.

EXAMPLES:

```
sage: RP10 = simplicial_sets.RealProjectiveSpace(8)
sage: RP10.homology()
{0: 0, 1: C2, 2: 0, 3: C2, 4: 0, 5: C2, 6: 0, 7: C2, 8: 0}

sage: eta = simplicial_sets.HopfMap()
sage: S3 = eta.domain()
sage: S2 = eta.codomain()
sage: S3.wedge(S2).homology()
{0: 0, 1: 0, 2: Z, 3: Z}
```



## MORPHISMS AND HOMSETS FOR SIMPLICIAL SETS

---

**Note:** Morphisms with infinite domain are not implemented in general: only constant maps and identity maps are currently implemented.

---

AUTHORS:

- John H. Palmieri (2016-07)

This module implements morphisms and homsets of simplicial sets.

```
class sage.homology.simplicial_set_morphism.SimplicialSetHomset(X, Y, cate-  
gory=None,  
base=None,  
check=True)
```

Bases: `sage.categories.homset.Homset`

A set of morphisms between simplicial sets.

Once a homset has been constructed in Sage, typically via `Hom(X, Y)` or `X.Hom(Y)`, one can use it to construct a morphism  $f$  by specifying a dictionary, the keys of which are the nondegenerate simplices in the domain, and the value corresponding to  $\sigma$  is the simplex  $f(\sigma)$  in the codomain.

EXAMPLES:

```
sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet  
sage: v = AbstractSimplex(0, name='v')  
sage: w = AbstractSimplex(0, name='w')  
sage: e = AbstractSimplex(1, name='e')  
sage: f = AbstractSimplex(1, name='f')  
sage: X = SimplicialSet({e: (v, w), f: (w, v)})  
sage: Y = SimplicialSet({e: (v, v)})
```

Define the homset:

```
sage: H = Hom(X, Y)
```

Now define a morphism by specifying a dictionary:

```
sage: H({v: v, w: v, e: e, f: e})  
Simplicial set morphism:  
From: Simplicial set with 4 non-degenerate simplices  
To: Simplicial set with 2 non-degenerate simplices  
Defn: [v, w, e, f] --> [v, v, e, e]
```

`an_element()`

Return an element of this homset: a constant map.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: S2 = simplicial_sets.Sphere(2)
sage: Hom(S2, S1).an_element()
Simplicial set morphism:
  From: S^2
  To:   S^1
  Defn: Constant map at v_0

sage: K = simplicial_sets.Simplex(3)
sage: L = simplicial_sets.Simplex(4)
sage: d = {K.n_cells(3)[0]: L.n_cells(0)[0].apply_degeneracies(2, 1, 0)}
sage: Hom(K, L)(d) == Hom(K, L).an_element()
True
```

**constant\_map** (*point=None*)

Return the constant map in this homset.

INPUT:

- *point* – optional, default None. If specified, it must be a 0-simplex in the codomain, and it will be the target of the constant map.

If *point* is specified, it is the target of the constant map. Otherwise, if the codomain is pointed, the target is its base point. If the codomain is not pointed and *point* is not specified, raise an error.

EXAMPLES:

```
sage: S3 = simplicial_sets.Sphere(3)
sage: T = simplicial_sets.Torus()
sage: T.n_cells(0)[0].rename('w')
sage: Hom(S3, T).constant_map()
Simplicial set morphism:
  From: S^3
  To:   Torus
  Defn: Constant map at w

sage: S0 = simplicial_sets.Sphere(0)
sage: v, w = S0.n_cells(0)
sage: Hom(S3, S0).constant_map(v)
Simplicial set morphism:
  From: S^3
  To:   S^0
  Defn: Constant map at v_0
sage: Hom(S3, S0).constant_map(w)
Simplicial set morphism:
  From: S^3
  To:   S^0
  Defn: Constant map at w_0
```

This constant map is not pointed, since it doesn't send the base point of  $S^3$  to the base point of  $S^0$ :

```
sage: Hom(S3, S0).constant_map(w).is_pointed()
False
```

**diagonal\_morphism** ()

Return the diagonal morphism in  $\text{Hom}(S, S \times S)$ .

EXAMPLES:

```

sage: RP2 = simplicial_sets.RealProjectiveSpace(2)
sage: Hom(RP2, RP2.product(RP2)).diagonal_morphism()
Simplicial set morphism:
  From: RP^2
  To:   RP^2 x RP^2
  Defn: [1, f, f * f] --> [(1, 1), (f, f), (f * f, f * f)]

```

**identity()**

Return the identity morphism in  $\text{Hom}(S, S)$ .

**EXAMPLES:**

```

sage: S1 = simplicial_sets.Sphere(1)
sage: Hom(S1, S1).identity()
Simplicial set endomorphism of S^1
  Defn: Identity map
sage: T = simplicial_sets.Torus()
sage: Hom(S1, T).identity()
Traceback (most recent call last):
...
TypeError: identity map is only defined for endomorphism sets

```

```

class sage.homology.simplicial_set_morphism.SimplicialSetMorphism (data=None,
                                                                    do-
                                                                    main=None,
                                                                    codomain=None,
                                                                    con-
                                                                    stant=None,
                                                                    iden-
                                                                    tity=False,
                                                                    check=True)

```

Bases: `sage.categories.morphism.Morphism`

Return a morphism of simplicial sets.

**INPUT:**

- `data` – optional. Dictionary defining the map.
- `domain` – simplicial set
- `codomain` – simplicial set
- `constant` – optional: if not `None`, then this should be a vertex in the codomain, in which case return the constant map with this vertex as the target.
- `identity` – optional: if `True`, return the identity morphism.
- `check` – optional, default `True`. If `True`, check that this is actually a morphism: it commutes with the face maps.

So to define a map, you must specify `domain` and `codomain`. If the map is constant, specify the target (a vertex in the codomain) as `constant`. If the map is the identity map, specify `identity=True`. Otherwise, pass a dictionary, `data`. The keys of the dictionary are the nondegenerate simplices of the domain, the corresponding values are simplices in the codomain.

In fact, the keys in `data` do not need to include all of the nondegenerate simplices, only those which are not faces of other nondegenerate simplices: if  $\sigma$  is a face of  $\tau$ , then the image of  $\sigma$  need not be specified.

**EXAMPLES:**

```

sage: from sage.homology.simplicial_set_morphism import SimplicialSetMorphism
sage: K = simplicial_sets.Simplex(1)
sage: S1 = simplicial_sets.Sphere(1)
sage: v0 = K.n_cells(0)[0]
sage: v1 = K.n_cells(0)[1]
sage: e01 = K.n_cells(1)[0]
sage: w = S1.n_cells(0)[0]
sage: sigma = S1.n_cells(1)[0]

sage: f = {v0: w, v1: w, e01: sigma}
sage: SimplicialSetMorphism(f, K, S1)
Simplicial set morphism:
  From: 1-simplex
  To:   S^1
  Defn: [(0,), (1,), (0, 1)] --> [v_0, v_0, sigma_1]

```

The same map can be defined as follows:

```

sage: H = Hom(K, S1)
sage: H(f)
Simplicial set morphism:
  From: 1-simplex
  To:   S^1
  Defn: [(0,), (1,), (0, 1)] --> [v_0, v_0, sigma_1]

```

Also, this map can be defined by specifying where the 1-simplex goes; the vertices then go where they have to, to satisfy the condition  $d_i \circ f = f \circ d_i$ :

```

sage: H = Hom(K, S1)
sage: H({e01: sigma})
Simplicial set morphism:
  From: 1-simplex
  To:   S^1
  Defn: [(0,), (1,), (0, 1)] --> [v_0, v_0, sigma_1]

```

A constant map:

```

sage: g = {e01: w.apply_degeneracies(0)}
sage: SimplicialSetMorphism(g, K, S1)
Simplicial set morphism:
  From: 1-simplex
  To:   S^1
  Defn: Constant map at v_0

```

The same constant map:

```

sage: SimplicialSetMorphism(domain=K, codomain=S1, constant=w)
Simplicial set morphism:
  From: 1-simplex
  To:   S^1
  Defn: Constant map at v_0

```

An identity map:

```

sage: SimplicialSetMorphism(domain=K, codomain=K, identity=True)
Simplicial set endomorphism of 1-simplex
  Defn: Identity map

```



Defining a map by specifying it on only some of the simplices in the domain:

```
sage: S5 = simplicial_sets.Sphere(5)
sage: s = S5.n_cells(5)[0]
sage: one = S5.Hom(S5)({s: s})
sage: one
Simplicial set endomorphism of S^5
Defn: Identity map
```

**associated\_chain\_complex\_morphism**(*base\_ring=Integer Ring, augmented=False, cochain=False*)  
Return the associated chain complex morphism of `self`.

INPUT:

- `base_ring` – default `ZZ`
- `augmented` – boolean, default `False`. If `True`, return the augmented complex.
- `cochain` – boolean, default `False`. If `True`, return the cochain complex.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: v0 = S1.n_cells(0)[0]
sage: e = S1.n_cells(1)[0]
sage: f = {v0: v0, e: v0.apply_degeneracies(0)} # constant map
sage: g = Hom(S1, S1)(f)
sage: g.associated_chain_complex_morphism().to_matrix()
[1|0]
[-+-]
[0|0]
```

**coequalizer**(*other*)

Return the coequalizer of this map with `other`.

INPUT:

- `other` – a morphism with the same domain and codomain as this map

If the two maps are  $f, g : X \rightarrow Y$ , then the coequalizer  $P$  is constructed as the pushout

```
X  v  Y  -->  Y
|          |
V          V
Y  ---->  P
```

where the upper left corner is the coproduct of  $X$  and  $Y$  (the wedge if they are pointed, the disjoint union otherwise), and the two maps  $X \amalg Y \rightarrow Y$  are  $f \amalg 1$  and  $g \amalg 1$ .

EXAMPLES:

```
sage: L = simplicial_sets.Simplex(2)
sage: pt = L.n_cells(0)[0]
sage: e = L.n_cells(1)[0]
sage: K = L.subsimplicial_set([e])
sage: f = K.inclusion_map()
sage: v, w = K.n_cells(0)
sage: g = Hom(K, L)({v: pt, w: pt, e: pt.apply_degeneracies(0)})
sage: P = f.coequalizer(g)
sage: P
```

(continues on next page)

(continued from previous page)

```

Pushout of maps:
  Simplicial set morphism:
    From: Disjoint union: (Simplicial set with 3 non-degenerate simplices u 2-
    ↪ simplex)
    To:    2-simplex
    Defn: ...
  Simplicial set morphism:
    From: Disjoint union: (Simplicial set with 3 non-degenerate simplices u 2-
    ↪ simplex)
    To:    2-simplex
    Defn: ...

```

**coproduct** (\*others)

Return the coproduct of this map with others.

- others – morphisms of simplicial sets.

If the relevant maps are  $f_i : X_i \rightarrow Y_i$ , this returns the natural map  $\coprod X_i \rightarrow \coprod Y_i$ .

EXAMPLES:

```

sage: S1 = simplicial_sets.Sphere(1)
sage: f = Hom(S1, S1).identity()
sage: f.coproduct(f).is_bijective()
True
sage: g = S1.constant_map(S1)
sage: g.coproduct(g).is_bijective()
False

```

**equalizer** (other)

Return the equalizer of this map with other.

INPUT:

- other – a morphism with the same domain and codomain as this map

If the two maps are  $f, g : X \rightarrow Y$ , then the equalizer  $P$  is constructed as the pullback

```

P ----> X
|         |
V         V
X --> X x Y

```

where the two maps  $X \rightarrow X \times Y$  are  $(1, f)$  and  $(1, g)$ .

EXAMPLES:

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: x = AbstractSimplex(0, name='x')
sage: evw = AbstractSimplex(1, name='vw')
sage: evx = AbstractSimplex(1, name='vx')
sage: ewx = AbstractSimplex(1, name='wx')
sage: X = SimplicialSet({evw: (w, v), evx: (x, v)})
sage: Y = SimplicialSet({evw: (w, v), evx: (x, v), ewx: (x, w)})

```

Here  $X$  is a wedge of two 1-simplices (a horn, that is), and  $Y$  is the boundary of a 2-simplex. The map  $f$  includes the two 1-simplices into  $Y$ , while the map  $g$  maps both 1-simplices to the same edge in  $Y$ .

```

sage: f = Hom(X, Y)({v:v, w:w, x:x, evw:evw, evx:evx})
sage: g = Hom(X, Y)({v:v, w:x, x:x, evw:evx, evx:evx})
sage: P = f.equalizer(g)
sage: P
Pullback of maps:
  Simplicial set morphism:
    From: Simplicial set with 5 non-degenerate simplices
    To:   Simplicial set with 5 non-degenerate simplices x Simplicial set
    ↪with 6 non-degenerate simplices
    Defn: [v, w, x, vw, vx] --> [(v, v), (w, w), (x, x), (vw, vw), (vx, vx)]
  Simplicial set morphism:
    From: Simplicial set with 5 non-degenerate simplices
    To:   Simplicial set with 5 non-degenerate simplices x Simplicial set
    ↪with 6 non-degenerate simplices
    Defn: [v, w, x, vw, vx] --> [(v, v), (w, x), (x, x), (vw, vx), (vx, vx)]

```

**image()**

Return the image of this morphism as a subsimplicial set of the codomain.

**EXAMPLES:**

```

sage: S1 = simplicial_sets.Sphere(1)
sage: T = S1.product(S1)
sage: K = T.factor(0, as_subset=True)
sage: f = S1.Hom(T)({S1.n_cells(0)[0]:K.n_cells(0)[0], S1.n_cells(1)[0]:K.n_
    ↪cells(1)[0]})
sage: f
Simplicial set morphism:
  From: S^1
  To:   S^1 x S^1
  Defn: [v_0, sigma_1] --> [(v_0, v_0), (sigma_1, s_0 v_0)]
sage: f.image()
Simplicial set with 2 non-degenerate simplices
sage: f.image().homology()
{0: 0, 1: Z}

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
    ↪MultiplicativeAbelian([2]))
sage: B.constant_map().image()
Point
sage: Hom(B,B).identity().image() == B
True

```

**induced\_homology\_morphism**(*base\_ring=None, cohomology=False*)

Return the map in (co)homology induced by this map

**INPUT:**

- *base\_ring* – must be a field (optional, default  $\mathbb{Q}\mathbb{Q}$ )
- *cohomology* – boolean (optional, default `False`). If `True`, the map induced in cohomology rather than homology.

**EXAMPLES:**

```

sage: from sage.homology.simplicial_set import AbstractSimplex, SimplicialSet
sage: v = AbstractSimplex(0, name='v')
sage: w = AbstractSimplex(0, name='w')
sage: e = AbstractSimplex(1, name='e')

```

(continues on next page)

(continued from previous page)

```

sage: f = AbstractSimplex(1, name='f')
sage: X = SimplicialSet({e: (v, w), f: (w, v)})
sage: Y = SimplicialSet({e: (v, v)})
sage: H = Hom(X, Y)
sage: f = H({v: v, w: v, e: e, f: e})
sage: g = f.induced_homology_morphism()
sage: g.to_matrix()
[1|0]
[-+-]
[0|2]
sage: g3 = f.induced_homology_morphism(base_ring=GF(3), cohomology=True)
sage: g3.to_matrix()
[1|0]
[-+-]
[0|2]

```

**is\_bijective()**

Return True if this map is bijective.

EXAMPLES:

```

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP2.inclusion_map().is_bijective()
False

sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.quotient_map().is_bijective()
False

sage: K = RP5_2.pullback(RP5_2.quotient_map(), RP5_2.base_point_map())
sage: f = K.universal_property(RP2.inclusion_map(), RP2.constant_map())
sage: f.is_bijective()
True

```

**is\_constant()**

Return True if this morphism is a constant map.

EXAMPLES:

```

sage: K = simplicial_sets.KleinBottle()
sage: S4 = simplicial_sets.Sphere(4)
sage: c = Hom(K, S4).constant_map()
sage: c.is_constant()
True
sage: X = S4.n_skeleton(3) # a point
sage: X.inclusion_map().is_constant()
True
sage: eta = simplicial_sets.HopfMap()
sage: eta.is_constant()
False

```

**is\_identity()**

Return True if this morphism is an identity map.

EXAMPLES:

```

sage: K = simplicial_sets.Simplex(1)
sage: v0 = K.n_cells(0)[0]
sage: v1 = K.n_cells(0)[1]
sage: e01 = K.n_cells(1)[0]
sage: L = simplicial_sets.Simplex(2).n_skeleton(1)
sage: w0 = L.n_cells(0)[0]
sage: w1 = L.n_cells(0)[1]
sage: w2 = L.n_cells(0)[2]
sage: f01 = L.n_cells(1)[0]
sage: f02 = L.n_cells(1)[1]
sage: f12 = L.n_cells(1)[2]

sage: d = {v0:w0, v1:w1, e01:f01}
sage: f = K.Hom(L)(d)
sage: f.is_identity()
False
sage: d = {w0:v0, w1:v1, w2:v1, f01:e01, f02:e01, f12: v1.apply_
↳ degeneracies(0,)}
sage: g = L.Hom(K)(d)
sage: (g*f).is_identity()
True
sage: (f*g).is_identity()
False
sage: (f*g).induced_homology_morphism().to_matrix(1)
[0]

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP5.n_skeleton(2).inclusion_map().is_identity()
False
sage: RP5.n_skeleton(5).inclusion_map().is_identity()
True

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↳ MultiplicativeAbelian([2]))
sage: Hom(B,B).identity().is_identity()
True
sage: Hom(B,B).constant_map().is_identity()
False

```

**is\_injective()**

Return True if this map is injective.

**EXAMPLES:**

```

sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP2.inclusion_map().is_injective()
True

sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.quotient_map().is_injective()
False

sage: K = RP5_2.pullback(RP5_2.quotient_map(), RP5_2.base_point_map())
sage: f = K.universal_property(RP2.inclusion_map(), RP2.constant_map())
sage: f.is_injective()
True

```

**is\_pointed()**

Return True if this is a pointed map.

That is, return True if the domain and codomain are pointed and this morphism preserves the base point.

EXAMPLES:

```
sage: S0 = simplicial_sets.Sphere(0)
sage: f = Hom(S0,S0).identity()
sage: f.is_pointed()
True
sage: v = S0.n_cells(0)[0]
sage: w = S0.n_cells(0)[1]
sage: g = Hom(S0,S0)({v:v, w:v})
sage: g.is_pointed()
True
sage: t = Hom(S0,S0)({v:w, w:v})
sage: t.is_pointed()
False
```

**is\_surjective()**

Return True if this map is surjective.

EXAMPLES:

```
sage: RP5 = simplicial_sets.RealProjectiveSpace(5)
sage: RP2 = RP5.n_skeleton(2)
sage: RP2.inclusion_map().is_surjective()
False

sage: RP5_2 = RP5.quotient(RP2)
sage: RP5_2.quotient_map().is_surjective()
True

sage: K = RP5_2.pullback(RP5_2.quotient_map(), RP5_2.base_point_map())
sage: f = K.universal_property(RP2.inclusion_map(), RP2.constant_map())
sage: f.is_surjective()
True
```

**mapping\_cone()**

Return the mapping cone defined by this map.

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: v_0, sigma_1 = S1.nondegenerate_simplices()
sage: K = simplicial_sets.Simplex(2).n_skeleton(1)
```

The mapping cone will be a little smaller if we use only pointed simplicial sets.  $S^1$  is already pointed, but not  $K$ .

```
sage: L = K.set_base_point(K.n_cells(0)[0])
sage: u,v,w = L.n_cells(0)
sage: e,f,g = L.n_cells(1)
sage: h = L.Hom(S1)({u:v_0, v:v_0, w:v_0, e:sigma_1, f:v_0.apply_
↪ degeneracies(0), g:sigma_1})
sage: h
Simplicial set morphism:
  From: Simplicial set with 6 non-degenerate simplices
```

(continues on next page)

(continued from previous page)

```

To: S^1
Defn: [(0,), (1,), (2,), (0, 1), (0, 2), (1, 2)] --> [v_0, v_0, v_0, sigma_
↪1, s_0 v_0, sigma_1]
sage: h.induced_homology_morphism().to_matrix()
[1|0]
[-+-]
[0|2]
sage: X = h.mapping_cone()
sage: X.homology() == simplicial_sets.RealProjectiveSpace(2).homology()
True

```

**n\_skeleton** (*n*, *domain=None*, *codomain=None*)Return the restriction of this morphism to the *n*-skeleta of the domain and codomain

INPUT:

- *n* – the dimension
- *domain* – optional, the domain. Specify this to explicitly specify the domain; otherwise, Sage will attempt to compute it. Specifying this can be useful if the domain is built as a pushout or pullback, so trying to compute it may lead to computing the *n*-skeleton of a map, causing an infinite recursion. (Users should not have to specify this, but it may be useful for developers.)
- *codomain* – optional, the codomain.

EXAMPLES:

```

sage: B = simplicial_sets.ClassifyingSpace(groups.misc.
↪MultiplicativeAbelian([2]))
sage: one = Hom(B,B).identity()
sage: one.n_skeleton(3)
Simplicial set endomorphism of Simplicial set with 4 non-degenerate simplices
Defn: Identity map
sage: c = Hom(B,B).constant_map()
sage: c.n_skeleton(3)
Simplicial set endomorphism of Simplicial set with 4 non-degenerate simplices
Defn: Constant map at 1

sage: K = simplicial_sets.Simplex(2)
sage: L = K.subsimplicial_set(K.n_cells(0)[:2])
sage: L.nondegenerate_simplices()
[(0,), (1,)]
sage: L.inclusion_map()
Simplicial set morphism:
From: Simplicial set with 2 non-degenerate simplices
To: 2-simplex
Defn: [(0,), (1,)] --> [(0,), (1,)]
sage: L.inclusion_map().n_skeleton(1)
Simplicial set morphism:
From: Simplicial set with 2 non-degenerate simplices
To: Simplicial set with 6 non-degenerate simplices
Defn: [(0,), (1,)] --> [(0,), (1,)]

```

**product** (*\*others*)Return the product of this map with *others*.

- *others* – morphisms of simplicial sets.

If the relevant maps are  $f_i : X_i \rightarrow Y_i$ , this returns the natural map  $\prod X_i \rightarrow \prod Y_i$ .

EXAMPLES:

```
sage: S1 = simplicial_sets.Sphere(1)
sage: f = Hom(S1, S1).identity()
sage: f.product(f).is_bijective()
True
sage: g = S1.constant_map(S1)
sage: g.product(g).is_bijective()
False
```

**pullback** (\*others)

Return the pullback of this morphism along with others.

INPUT:

- others – morphisms of simplicial sets, the codomains of which must all equal that of self.

This returns the pullback as a simplicial set. See [sage.homology.simplicial\\_set\\_constructions.PullbackOfSimplicialSets](#) for more documentation and examples.

EXAMPLES:

```
sage: T = simplicial_sets.Torus()
sage: K = simplicial_sets.KleinBottle()
sage: term_T = T.constant_map()
sage: term_K = K.constant_map()
sage: P = term_T.pullback(term_K) # the product as a pullback
sage: P
Pullback of maps:
  Simplicial set morphism:
    From: Torus
    To:   Point
    Defn: Constant map at *
  Simplicial set morphism:
    From: Klein bottle
    To:   Point
    Defn: Constant map at *
```

**pushout** (\*others)

Return the pushout of this morphism along with others.

INPUT:

- others – morphisms of simplicial sets, the domains of which must all equal that of self.

This returns the pushout as a simplicial set. See [sage.homology.simplicial\\_set\\_constructions.PushoutOfSimplicialSets](#) for more documentation and examples.

EXAMPLES:

```
sage: T = simplicial_sets.Torus()
sage: K = simplicial_sets.KleinBottle()
sage: init_T = T._map_from_empty_set()
sage: init_K = K._map_from_empty_set()
sage: D = init_T.pushout(init_K) # the disjoint union as a pushout
sage: D
Pushout of maps:
  Simplicial set morphism:
```

(continues on next page)



(continued from previous page)

```

From: Empty simplicial set
To:   Torus
Defn: [] --> []
Simplicial set morphism:
From: Empty simplicial set
To:   Klein bottle
Defn: [] --> []

```

**suspension** ( $n=1$ )

Return the  $n$ -th suspension of this morphism of simplicial sets.

INPUT:

- $n$  (optional) – non-negative integer, default 1

EXAMPLES:

```

sage: eta = simplicial_sets.HopfMap()
sage: susp_eta = eta.suspension()
sage: susp_eta.mapping_cone().homology() == eta.mapping_cone().suspension().
↪homology()
True

```

This uses reduced suspensions if the original morphism is pointed, unreduced otherwise. So for example, if a constant map is not pointed, its suspension is not a constant map:

```

sage: L = simplicial_sets.Simplex(1)
sage: L.constant_map().is_pointed()
False
sage: f = L.constant_map().suspension()
sage: f.is_constant()
False

sage: K = simplicial_sets.Sphere(3)
sage: K.constant_map().is_pointed()
True
sage: g = K.constant_map().suspension()
sage: g.is_constant()
True

sage: h = K.identity().suspension()
sage: h.is_identity()
True

```



## GENERIC CELL COMPLEXES

AUTHORS:

- John H. Palmieri (2009-08)

This module defines a class of abstract finite cell complexes. This is meant as a base class from which other classes (like *SimplicialComplex*, *CubicalComplex*, and *DeltaComplex*) should derive. As such, most of its properties are not implemented. It is meant for use by developers producing new classes, not casual users.

---

**Note:** Keywords for *chain\_complex()*, *homology()*, etc.: any keywords given to the *homology()* method get passed on to the *chain\_complex()* method and also to the constructor for chain complexes in *sage.homology.chain\_complex.ChainComplex\_class*, as well as its associated *homology()* method. This means that those keywords should have consistent meaning in all of those situations. It also means that it is easy to implement new keywords: for example, if you implement a new keyword for the *sage.homology.chain\_complex.ChainComplex\_class.homology()* method, then it will be automatically accessible through the *homology()* method for cell complexes – just make sure it gets documented.

---

**class** *sage.homology.cell\_complex.GenericCellComplex*

Bases: *sage.structure.sage\_object.SageObject*

Class of abstract cell complexes.

This is meant to be used by developers to produce new classes, not by casual users. Classes which derive from this are *SimplicialComplex*, *DeltaComplex*, and *CubicalComplex*.

Most of the methods here are not implemented, but probably should be implemented in a derived class. Most of the other methods call a non-implemented one; their docstrings contain examples from derived classes in which the various methods have been defined. For example, *homology()* calls *chain\_complex()*; the class *DeltaComplex* implements *chain\_complex()*, and so the *homology()* method here is illustrated with examples involving  $\Delta$ -complexes.

EXAMPLES:

It's hard to give informative examples of the base class, since essentially nothing is implemented.

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
```

**alexander\_whitney**(*cell*, *dim\_left*)

The decomposition of *cell* in this complex into left and right factors, suitable for computing cup products. This should provide a cellular approximation for the diagonal map  $K \rightarrow K \times K$ .

This method is not implemented for generic cell complexes, but must be implemented for any derived class to make cup products work in *self.cohomology\_ring()*.

INPUT:

- `cell` – a cell in this complex
- `dim_left` – the dimension of the left-hand factors in the decomposition

OUTPUT: a list containing triples  $(c, \text{left}, \text{right})$ . `left` and `right` should be cells in this complex, and `c` an integer. In the cellular approximation of the diagonal map, the chain represented by `cell` should get sent to the sum of terms  $c(\text{left} \otimes \text{right})$  in the tensor product  $C(K) \otimes C(K)$  of the chain complex for this complex with itself.

This gets used in the method `product_on_basis()` for the class of cohomology rings.

For simplicial and cubical complexes, the decomposition can be done at the level of individual cells: see `alexander_whitney()` and `alexander_whitney()`. Then the method for simplicial complexes just calls the method for individual simplices, and similarly for cubical complexes. For  $\Delta$ -complexes and simplicial sets, the method is instead defined at the level of the cell complex.

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.alexander_whitney(None, 2)
Traceback (most recent call last):
...
NotImplementedError: <abstract method alexander_whitney at ...>
```

**algebraic\_topological\_model** (*base\_ring=Rational Field*)

Algebraic topological model for this cell complex with coefficients in `base_ring`.

The term “algebraic topological model” is defined by Pilarczyk and Réal [PR2015].

This is not implemented for generic cell complexes. For any classes deriving from this one, when this method is implemented, it should essentially just call either `algebraic_topological_model()` or `algebraic_topological_model_delta_complex()`.

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.algebraic_topological_model(QQ)
Traceback (most recent call last):
...
NotImplementedError
```

**betti** (*dim=None, subcomplex=None*)

The Betti numbers of this simplicial complex as a dictionary (or a single Betti number, if only one dimension is given): the  $i$ th Betti number is the rank of the  $i$ th homology group.

#### Parameters

- **dim** (integer or list of integers or `None`; optional, default `None`) – If `None`, then return every Betti number, as a dictionary with keys the non-negative integers. If `dim` is an integer or list, return the Betti number for each given dimension. (Actually, if `dim` is a list, return the Betti numbers, as a dictionary, in the range from  $\min(\text{dim})$  to  $\max(\text{dim})$ . If `dim` is a number, return the Betti number in that dimension.)
- **subcomplex** (optional, default `None`) – a subcomplex of this cell complex. Compute the Betti numbers of the homology relative to this subcomplex.

EXAMPLES:

Build the two-sphere as a three-fold join of a two-point space with itself:

```

sage: S = SimplicialComplex([[0], [1]])
sage: (S*S*S).betti()
{0: 1, 1: 0, 2: 1}
sage: (S*S*S).betti([1,2])
{1: 0, 2: 1}
sage: (S*S*S).betti(2)
1

```

Or build the two-sphere as a  $\Delta$ -complex:

```

sage: S2 = delta_complexes.Sphere(2)
sage: S2.betti([1,2])
{1: 0, 2: 1}

```

Or as a cubical complex:

```

sage: S2c = cubical_complexes.Sphere(2)
sage: S2c.betti(2)
1

```

**cells** (*subcomplex=None*)

The cells of this cell complex, in the form of a dictionary: the keys are integers, representing dimension, and the value associated to an integer  $d$  is the set of  $d$ -cells. If the optional argument *subcomplex* is present, then return only the cells which are *not* in the subcomplex.

**Parameters** *subcomplex* (*optional, default None*) – a subcomplex of this cell complex. Return the cells which are not in this subcomplex.

This is not implemented in general; it should be implemented in any derived class. When implementing, see the warning in the *dimension()* method.

This method is used by various other methods, such as *n\_cells()* and *f\_vector()*.

EXAMPLES:

```

sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.cells()
Traceback (most recent call last):
...
NotImplementedError: <abstract method cells at ...>

```

**chain\_complex** (*subcomplex=None, augmented=False, verbose=False, check=True, dimensions=None, base\_ring='ZZ', cochain=False*)

This is not implemented for general cell complexes.

Some keywords to possibly implement in a derived class:

- *subcomplex* – a subcomplex: compute the relative chain complex
- *augmented* – a bool: whether to return the augmented complex
- *verbose* – a bool: whether to print informational messages as the chain complex is being computed
- *check* – a bool: whether to check that the each composite of two consecutive differentials is zero
- **dimensions** – if *None*, compute the chain complex in all dimensions. If a list or tuple of integers, compute the chain complex in those dimensions, setting the chain groups in all other dimensions to zero.

Definitely implement the following:

- `base_ring` – commutative ring (optional, default `ZZ`)
- `cochain` – a bool: whether to return the cochain complex

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.chain_complex()
Traceback (most recent call last):
...
NotImplementedError: <abstract method chain_complex at ...>
```

**cohomology**(*dim=None, base\_ring=Integer Ring, subcomplex=None, generators=False, algorithm='pari', verbose=False, reduced=True*)

The reduced cohomology of this cell complex.

The arguments are the same as for the `homology()` method, except that `homology()` accepts a cohomology key word, while this function does not: `cohomology` is automatically true here. Indeed, this function just calls `homology()` with `cohomology` set to True.

#### Parameters

- `dim` –
- `base_ring` –
- `subcomplex` –
- `algorithm` –
- `verbose` –
- `reduced` –

EXAMPLES:

```
sage: circle = SimplicialComplex([[0,1], [1,2], [0, 2]])
sage: circle.cohomology(0)
0
sage: circle.cohomology(1)
Z
sage: P2 = SimplicialComplex([[0,1,2], [0,2,3], [0,1,5], [0,4,5], [0,3,4], [1,
↪2,4], [1,3,4], [1,3,5], [2,3,5], [2,4,5]]) # projective plane
sage: P2.cohomology(2)
C2
sage: P2.cohomology(2, base_ring=GF(2))
Vector space of dimension 1 over Finite Field of size 2
sage: P2.cohomology(2, base_ring=GF(3))
Vector space of dimension 0 over Finite Field of size 3

sage: cubical_complexes.KleinBottle().cohomology(2)
C2
```

Relative cohomology:

```
sage: T = SimplicialComplex([[0,1]])
sage: U = SimplicialComplex([[0], [1]])
sage: T.cohomology(1, subcomplex=U)
Z
```

A  $\Delta$ -complex example:

```
sage: s5 = delta_complexes.Sphere(5)
sage: s5.cohomology(base_ring=GF(7))[5]
Vector space of dimension 1 over Finite Field of size 7
```

**cohomology\_ring** (*base\_ring=Rational Field*)

Return the unreduced cohomology with coefficients in *base\_ring* with a chosen basis.

This is implemented for simplicial, cubical, and  $\Delta$ -complexes, not for arbitrary generic cell complexes. The resulting elements are suitable for computing cup products. For simplicial complexes, they should be suitable for computing cohomology operations; so far, only mod 2 cohomology operations have been implemented.

INPUT:

- *base\_ring* – coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ); must be a field

The basis elements in dimension *dim* are named ' $h^{\{dim,i\}}$ ' where *i* ranges between 0 and *r* – 1, if *r* is the rank of the cohomology group.

---

**Note:** For all but the smallest complexes, this is likely to be slower than `cohomology()` (with field coefficients), possibly by several orders of magnitude. This and its companion `homology_with_basis()` carry extra information which allows computation of cup products, for example, but because of speed issues, you may only wish to use these if you need that extra information.

---

**EXAMPLES:**

```
sage: K = simplicial_complexes.KleinBottle()
sage: H = K.cohomology_ring(QQ); H
Cohomology ring of Minimal triangulation of the Klein bottle
over Rational Field
sage: sorted(H.basis(), key=str)
[h^{0,0}, h^{1,0}]
sage: H = K.cohomology_ring(GF(2)); H
Cohomology ring of Minimal triangulation of the Klein bottle
over Finite Field of size 2
sage: sorted(H.basis(), key=str)
[h^{0,0}, h^{1,0}, h^{1,1}, h^{2,0}]

sage: X = delta_complexes.SurfaceOfGenus(2)
sage: H = X.cohomology_ring(QQ); H
Cohomology ring of Delta complex with 3 vertices and 29 simplices
over Rational Field
sage: sorted(H.basis(1), key=str)
[h^{1,0}, h^{1,1}, h^{1,2}, h^{1,3}]

sage: H = simplicial_complexes.Torus().cohomology_ring(QQ); H
Cohomology ring of Minimal triangulation of the torus
over Rational Field
sage: x = H.basis()[1,0]; x
h^{1,0}
sage: y = H.basis()[1,1]; y
h^{1,1}
```

You can compute cup products of cohomology classes:

```
sage: x.cup_product(y)
-h^{2,0}
```

(continues on next page)

(continued from previous page)

```

sage: x * y # alternate notation
-h^{2,0}
sage: y.cup_product(x)
h^{2,0}
sage: x.cup_product(x)
0

```

Cohomology operations:

```

sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: K = RP2.suspension()
sage: K.set_immutable()
sage: y = K.cohomology_ring(GF(2)).basis()[2,0]; y
h^{2,0}
sage: y.Sq(1)
h^{3,0}

```

To compute the cohomology ring, the complex must be “immutable”. This is only relevant for simplicial complexes, and most simplicial complexes are immutable, but certain constructions make them mutable. The suspension is one example, and this is the reason for calling `K.set_immutable()` above. Another example:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: T = S1.product(S1)
sage: T.is_immutable()
False
sage: T.cohomology_ring()
Traceback (most recent call last):
...
ValueError: This simplicial complex must be immutable. Call set_immutable().
sage: T.set_immutable()
sage: T.cohomology_ring()
Cohomology ring of Simplicial complex with 9 vertices and
18 facets over Rational Field

```

**dimension()**

The dimension of this cell complex: the maximum dimension of its cells.

**Warning:** If the `cells()` method calls `dimension()`, then you’ll get an infinite loop. So either don’t use `dimension()` or override `dimension()`.

EXAMPLES:

```

sage: simplicial_complexes.RandomComplex(d=5, n=8).dimension()
5
sage: delta_complexes.Sphere(3).dimension()
3
sage: T = cubical_complexes.Torus()
sage: T.product(T).dimension()
4

```

**disjoint\_union(right)**

The disjoint union of this cell complex with another one.

**Parameters** `right` – the other cell complex (the right-hand factor)



Disjoint unions are not implemented for general cell complexes.

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex(); B = GenericCellComplex()
sage: A.disjoint_union(B)
Traceback (most recent call last):
...
NotImplementedError: <abstract method disjoint_union at ...>
```

**euler\_characteristic()**

The Euler characteristic of this cell complex: the alternating sum over  $n \geq 0$  of the number of  $n$ -cells.

EXAMPLES:

```
sage: simplicial_complexes.Simplex(5).euler_characteristic()
1
sage: delta_complexes.Sphere(6).euler_characteristic()
2
sage: cubical_complexes.KleinBottle().euler_characteristic()
0
```

**f\_vector()**

The  $f$ -vector of this cell complex: a list whose  $n^{\text{th}}$  item is the number of  $(n-1)$ -cells. Note that, like all lists in Sage, this is indexed starting at 0: the 0th element in this list is the number of  $(-1)$ -cells (which is 1: the empty cell is the only  $(-1)$ -cell).

EXAMPLES:

```
sage: simplicial_complexes.KleinBottle().f_vector()
[1, 8, 24, 16]
sage: delta_complexes.KleinBottle().f_vector()
[1, 1, 3, 2]
sage: cubical_complexes.KleinBottle().f_vector()
[1, 42, 84, 42]
```

**face\_poset()**

The face poset of this cell complex, the poset of nonempty cells, ordered by inclusion.

This uses the `cells()` method, and also assumes that for each cell  $f$ , all of `f.faces()`, `tuple(f)`, and `f.dimension()` make sense. (If this is not the case in some derived class, as happens with  $\Delta$ -complexes, then override this method.)

EXAMPLES:

```
sage: P = SimplicialComplex([[0, 1], [1, 2], [2, 3]]).face_poset(); P
Finite poset containing 7 elements
sage: sorted(P.list())
[(0,), (0, 1), (1,), (1, 2), (2,), (2, 3), (3,)]

sage: S2 = cubical_complexes.Sphere(2)
sage: S2.face_poset()
Finite poset containing 26 elements
```

**graph()**

The 1-skeleton of this cell complex, as a graph.

This is not implemented for general cell complexes.

EXAMPLES:

```

sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.graph()
Traceback (most recent call last):
...
NotImplementedError

```

**homology** (*dim=None, base\_ring=Integer Ring, subcomplex=None, generators=False, cohomology=False, algorithm='pari', verbose=False, reduced=True, \*\*kwds*)  
 The (reduced) homology of this cell complex.

#### Parameters

- **dim** (*integer or list of integers or None; optional, default None*) – If *None*, then return the homology in every dimension. If *dim* is an integer or list, return the homology in the given dimensions. (Actually, if *dim* is a list, return the homology in the range from `min(dim)` to `max(dim)`.)
- **base\_ring** (*optional, default ZZ*) – commutative ring, must be `ZZ` or a field.
- **subcomplex** (*optional, default empty*) – a subcomplex of this simplicial complex. Compute homology relative to this subcomplex.
- **generators** (*boolean; optional, default False*) – If *True*, return generators for the homology groups along with the groups. NOTE: Since [trac ticket #6100](#), the result may not be what you expect when not using `CHomP` since its return is in terms of the chain complex.
- **cohomology** (*boolean; optional, default False*) – If *True*, compute cohomology rather than homology.
- **algorithm** (*string; optional, default 'pari'*) – The options are 'auto', 'dhs', 'pari' or 'no\_chomp'. See below for a description of what they mean.
- **verbose** (*boolean; optional, default False*) – If *True*, print some messages as the homology is computed.
- **reduced** (*boolean; optional, default True*) – If *True*, return the reduced homology.

#### ALGORITHM:

If *algorithm* is set to 'auto', then use `CHomP` if available. (`CHomP` is available at the web page <http://chomp.rutgers.edu/>. It is also an optional package for Sage.)

`CHomP` computes homology, not cohomology, and only works over the integers or finite prime fields. Therefore if any of these conditions fails, or if `CHomP` is not present, or if *algorithm* is set to 'no\_chomp', go to plan B: if this complex has a `_homology` method – each simplicial complex has this, for example – then call that. Such a method implements specialized algorithms for the particular type of cell complex.

Otherwise, move on to plan C: compute the chain complex of this complex and compute its homology groups. To do this: over a field, just compute ranks and nullities, thus obtaining dimensions of the homology groups as vector spaces. Over the integers, compute Smith normal form of the boundary matrices defining the chain complex according to the value of *algorithm*. If *algorithm* is 'auto' or 'no\_chomp', then for each relatively small matrix, use the standard Sage method, which calls the Pari package. For any large matrix, reduce it using the Dumas, Heckenbach, Saunders, and Welker elimination algorithm: see `sage.homology.matrix_utils.dhs_snf()` for details.

Finally, *algorithm* may also be 'pari' or 'dhs', which forces the named algorithm to be used regardless of the size of the matrices and regardless of whether `CHomP` is available.

As of this writing, 'pari' is the fastest standard option. The optional CHomP package may be better still.

EXAMPLES:

```
sage: P = delta_complexes.RealProjectivePlane()
sage: P.homology()
{0: 0, 1: C2, 2: 0}
sage: P.homology(reduced=False)
{0: Z, 1: C2, 2: 0}
sage: P.homology(base_ring=GF(2))
{0: Vector space of dimension 0 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: S7 = delta_complexes.Sphere(7)
sage: S7.homology(7)
Z
sage: cubical_complexes.KleinBottle().homology(1, base_ring=GF(2))
Vector space of dimension 2 over Finite Field of size 2
```

If CHomP is installed, Sage can compute generators of homology groups:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: S2.homology(dim=2, generators=True, base_ring=GF(2)) # optional - CHomP
(Vector space of dimension 1 over Finite Field of size 2, [(0, 1, 2) + (0, 1,
↪3) + (0, 2, 3) + (1, 2, 3)])
```

When generators are computed, Sage returns a pair for each dimension: the group and the list of generators. For simplicial complexes, each generator is represented as a linear combination of simplices, as above, and for cubical complexes, each generator is a linear combination of cubes:

```
sage: S2_cub = cubical_complexes.Sphere(2)
sage: S2_cub.homology(dim=2, generators=True) # optional - CHomP
(Z, [-[[0,1] x [0,1] x [0,0]] + [[0,1] x [0,1] x [1,1]] - [[0,0] x [0,1] x [0,
↪1]] - [[0,1] x [1,1] x [0,1]] + [[0,1] x [0,0] x [0,1]] + [[1,1] x [0,1] x
↪[0,1]]])
```

**homology\_with\_basis** (*base\_ring=Rational Field, cohomology=False*)

Return the unreduced homology of this complex with coefficients in *base\_ring* with a chosen basis.

This is implemented for simplicial, cubical, and  $\Delta$ -complexes, not for arbitrary generic cell complexes.

INPUT:

- *base\_ring* – coefficient ring (optional, default  $\mathbb{Q}\mathbb{Q}$ ); must be a field
- *cohomology* – boolean (optional, default False); if True, return cohomology instead of homology

Homology basis elements are named ' $h_{\{\dim,i\}}$ ' where  $i$  ranges between 0 and  $r - 1$ , if  $r$  is the rank of the homology group. Cohomology basis elements are denoted  $h^{dim,i}$  instead.

**See also:**

If *cohomology* is True, this returns the cohomology as a graded module. For the ring structure, use *cohomology\_ring*().

EXAMPLES:

```

sage: K = simplicial_complexes.KleinBottle()
sage: H = K.homology_with_basis(QQ); H
Homology module of Minimal triangulation of the Klein bottle
over Rational Field
sage: sorted(H.basis(), key=str)
[h_{0,0}, h_{1,0}]
sage: H = K.homology_with_basis(GF(2)); H
Homology module of Minimal triangulation of the Klein bottle
over Finite Field of size 2
sage: sorted(H.basis(), key=str)
[h_{0,0}, h_{1,0}, h_{1,1}, h_{2,0}]

```

The homology is constructed as a graded object, so for example, you can ask for the basis in a single degree:

```

sage: H.basis(1)
Finite family {(1, 0): h_{1,0}, (1, 1): h_{1,1}}
sage: S3 = delta_complexes.Sphere(3)
sage: H = S3.homology_with_basis(QQ, cohomology=True)
sage: list(H.basis(3))
[h^{3,0}]

```

**is\_acyclic** (*base\_ring=Integer Ring*)

True if the reduced homology with coefficients in *base\_ring* of this cell complex is zero.

INPUT:

- *base\_ring* – optional, default ZZ. Compute homology with coefficients in this ring.

EXAMPLES:

```

sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: RP2.is_acyclic()
False
sage: RP2.is_acyclic(QQ)
True

```

This first computes the Euler characteristic: if it is not 1, the complex cannot be acyclic. So this should return False reasonably quickly on complexes with Euler characteristic not equal to 1:

```

sage: K = cubical_complexes.KleinBottle()
sage: C = cubical_complexes.Cube(2)
sage: P = K.product(C)
sage: P
Cubical complex with 168 vertices and 1512 cubes
sage: P.euler_characteristic()
0
sage: P.is_acyclic()
False

```

**is\_connected** ()

True if this cell complex is connected.

EXAMPLES:

```

sage: V = SimplicialComplex([[0,1,2],[3]])
sage: V
Simplicial complex with vertex set (0, 1, 2, 3) and facets {(3,), (0, 1, 2)}

```

(continues on next page)

(continued from previous page)

```

sage: V.is_connected()
False
sage: X = SimplicialComplex([[0,1,2]])
sage: X.is_connected()
True
sage: U = simplicial_complexes.ChessboardComplex(3,3)
sage: U.is_connected()
True
sage: W = simplicial_complexes.Sphere(3)
sage: W.is_connected()
True
sage: S = SimplicialComplex([[0,1],[2,3]])
sage: S.is_connected()
False

sage: cubical_complexes.Sphere(0).is_connected()
False
sage: cubical_complexes.Sphere(2).is_connected()
True

```

**join** (*right*)

The join of this cell complex with another one.

**Parameters** **right** – the other cell complex (the right-hand factor)

Joins are not implemented for general cell complexes. They may be implemented in some derived classes (like simplicial complexes).

EXAMPLES:

```

sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex(); B = GenericCellComplex()
sage: A.join(B)
Traceback (most recent call last):
...
NotImplementedError: <abstract method join at ...>

```

**n\_cells** (*n*, *subcomplex=None*)

List of cells of dimension *n* of this cell complex. If the optional argument *subcomplex* is present, then return the *n*-dimensional cells which are *not* in the subcomplex.

**Parameters**

- **n** (*non-negative integer*) – the dimension
- **subcomplex** (optional, default *None*) – a subcomplex of this cell complex. Return the cells which are not in this subcomplex.

---

**Note:** The resulting list need not be sorted. If you want a sorted list of *n*-cells, use `_n_cells_sorted()`.

---

EXAMPLES:

```

sage: delta_complexes.Torus().n_cells(1)
[(0, 0), (0, 0), (0, 0)]
sage: cubical_complexes.Cube(1).n_cells(0)
[[1,1], [0,0]]

```

**n\_chains** (*n*, *base\_ring=Integer Ring*, *cochains=False*)

Return the free module of chains in degree *n* over *base\_ring*.

INPUT:

- *n* – integer
- *base\_ring* – ring (optional, default  $\mathbb{Z}$ )
- *cochains* – boolean (optional, default `False`); if `True`, return cochains instead

The only difference between chains and cochains is notation. In a simplicial complex, for example, a simplex  $(0,1,2)$  is written as “ $(0,1,2)$ ” in the group of chains but as “ $\chi_{(0,1,2)}$ ” in the group of cochains.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: S2.n_chains(1, QQ)
Free module generated by {(0, 1), (0, 2), (0, 3), (1, 2), (1, 3), (2, 3)}_
↳over Rational Field
sage: list(simplicial_complexes.Sphere(2).n_chains(1, QQ, cochains=False).
↳basis())
[(0, 1), (0, 2), (0, 3), (1, 2), (1, 3), (2, 3)]
sage: list(simplicial_complexes.Sphere(2).n_chains(1, QQ, cochains=True).
↳basis())
[\chi_(0, 1), \chi_(0, 2), \chi_(0, 3), \chi_(1, 2), \chi_(1, 3), \chi_(2, 3)]
```

**n\_skeleton** (*n*)

The *n*-skeleton of this cell complex: the cell complex obtained by discarding all of the simplices in dimensions larger than *n*.

**Parameters** *n* – non-negative integer

This is not implemented for general cell complexes.

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex()
sage: A.n_skeleton(3)
Traceback (most recent call last):
...
NotImplementedError: <abstract method n_skeleton at ...>
```

**product** (*right*, *rename\_vertices=True*)

The (Cartesian) product of this cell complex with another one.

Products are not implemented for general cell complexes. They may be implemented in some derived classes (like simplicial complexes).

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex(); B = GenericCellComplex()
sage: A.product(B)
Traceback (most recent call last):
...
NotImplementedError: <abstract method product at ...>
```

**wedge** (*right*)

The wedge (one-point union) of this cell complex with another one.

Parameters **right** – the other cell complex (the right-hand factor)

Wedges are not implemented for general cell complexes.

EXAMPLES:

```
sage: from sage.homology.cell_complex import GenericCellComplex
sage: A = GenericCellComplex(); B = GenericCellComplex()
sage: A.wedge(B)
Traceback (most recent call last):
...
NotImplementedError: <abstract method wedge at ...>
```





## KOSZUL COMPLEXES

**class** sage.homology.koszul\_complex.**KoszulComplex**( $R$ ,  $elements$ )  
 Bases: `sage.homology.chain_complex.ChainComplex_class`, `sage.structure.unique_representation.UniqueRepresentation`

A Koszul complex.

Let  $R$  be a ring and consider  $x_1, x_2, \dots, x_n \in R$ . The Koszul complex  $K_*(x_1, \dots, x_n)$  is given by defining a chain complex structure on the exterior algebra  $\bigwedge^n R$  with the basis  $e_{i_1} \wedge \dots \wedge e_{i_a}$ . The differential is given by

$$\partial(e_{i_1} \wedge \dots \wedge e_{i_a}) = \sum_{r=1}^a (-1)^{r-1} x_{i_r} e_{i_1} \wedge \dots \wedge \hat{e}_{i_r} \wedge \dots \wedge e_{i_a},$$

where  $\hat{e}_{i_r}$  denotes the omitted factor.

Alternatively we can describe the Koszul complex by considering the basic complex  $K_{x_i}$

$$0 \rightarrow R \xrightarrow{x_i} R \rightarrow 0.$$

Then the Koszul complex is given by  $K_*(x_1, \dots, x_n) = \bigotimes_i K_{x_i}$ .

INPUT:

- $R$  – the base ring
- $elements$  – a tuple of elements of  $R$

EXAMPLES:

```
sage: R.<x,y,z> = QQ[]
sage: K = KoszulComplex(R, [x,y])
sage: ascii_art(K)
          [-y]
      [x y]  [ x]
0 <-- C_0 <----- C_1 <----- C_2 <-- 0
sage: K = KoszulComplex(R, [x,y,z])
sage: ascii_art(K)
          [-y -z  0]      [ z]
          [ x  0 -z]      [-y]
      [x y z]  [ 0  x  y]  [ x]
0 <-- C_0 <----- C_1 <----- C_2 <----- C_3 <-- 0
sage: K = KoszulComplex(R, [x+y*z,x+y-z])
sage: ascii_art(K)
          [-x - y + z]
      [ y*z + x x + y - z]  [ y*z + x]
0 <-- C_0 <----- C_1 <----- C_2 <-- 0
```

REFERENCES:

- [Wikipedia article Koszul\\_complex](#)

## HOCHSCHILD COMPLEXES

```
class sage.homology.hochschild_complex.HochschildComplex(A, M)
    Bases:      sage.structure.unique_representation.UniqueRepresentation,  sage.
               structure.parent.Parent
```

The Hochschild complex.

Let  $A$  be an algebra over a commutative ring  $R$  such that  $A$  a projective  $R$ -module, and  $M$  an  $A$ -bimodule. The *Hochschild complex* is the chain complex given by

$$C_n(A, M) := M \otimes A^{\otimes n}$$

with the boundary operators given as follows. For fixed  $n$ , define the face maps

$$f_{n,i}(m \otimes a_1 \otimes \cdots \otimes a_n) = \begin{cases} ma_1 \otimes \cdots \otimes a_n & \text{if } i = 0, \\ a_n m \otimes a_1 \otimes \cdots \otimes a_{n-1} & \text{if } i = n, \\ m \otimes a_1 \otimes \cdots \otimes a_i a_{i+1} \otimes \cdots \otimes a_n & \text{otherwise.} \end{cases}$$

We define the boundary operators as

$$d_n = \sum_{i=0}^n (-1)^i f_{n,i}.$$

The *Hochschild homology* of  $A$  is the homology of this complex. Alternatively, the Hochschild homology can be described by  $HH_n(A, M) = \text{Tor}_n^{A^e}(A, M)$ , where  $A^e = A \otimes A^o$  ( $A^o$  is the opposite algebra of  $A$ ) is the enveloping algebra of  $A$ .

*Hochschild cohomology* is the homology of the dual complex and can be described by  $HH^n(A, M) = \text{Ext}_{A^e}^n(A, M)$ .

Another perspective on Hochschild homology is that  $f_{n,i}$  make the family  $C_n(A, M)$  a simplicial object in the category of  $R$ -modules, and the degeneracy maps are

$$s_i(a_0 \otimes \cdots \otimes a_n) = a_0 \otimes \cdots \otimes a_i \otimes 1 \otimes a_{i+1} \otimes \cdots \otimes a_n$$

The Hochschild homology can also be constructed as the homology of this simplicial module.

REFERENCES:

- [Wikipedia article Hochschild\\_homology](#)
- <https://ncatlab.org/nlab/show/Hochschild+cohomology>
- [Red2001]

**class** `Element` (*parent, vectors*)

Bases: `sage.structure.element.ModuleElement`

A chain of the Hochschild complex.

INPUT:

Can be one of the following:

- A dictionary whose keys are the degree and whose  $d$ -th value is an element in the degree  $d$  module.
- An element in the coefficient module  $M$ .

EXAMPLES:

```
sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H(T.an_element())
Chain(0: 2*B['v'])
sage: H({0: T.an_element()})
Chain(0: 2*B['v'])
sage: H({1: H.module(1).an_element()})
Chain(1: 2*B['v'] # [1, 2, 3] + 2*B['v'] # [1, 3, 2] + 3*B['v'] # [2, 1, 3])
sage: H({0: H.module(0).an_element(), 3: H.module(3).an_element()})
Chain with 2 nonzero terms over Rational Field

sage: F.<x,y> = FreeAlgebra(ZZ)
sage: H = F.hochschild_complex(F)
sage: H(x + 2*y^2)
Chain(0: F[x] + 2*F[y^2])
sage: H({0: x*y - x})
Chain(0: -F[x] + F[x*y])
sage: H(2)
Chain(0: 2*F[1])
sage: H({0: x-y, 2: H.module(2).basis().an_element()})
Chain with 2 nonzero terms over Integer Ring
```

**vector** (*degree*)

Return the free module element in degree.

EXAMPLES:

```
sage: F.<x,y> = FreeAlgebra(ZZ)
sage: H = F.hochschild_complex(F)
sage: a = H({0: x-y, 2: H.module(2).basis().an_element()})
sage: [a.vector(i) for i in range(3)]
[F[x] - F[y], 0, F[1] # F[1] # F[1]]
```

**algebra** ()

Return the defining algebra of self.

EXAMPLES:

```
sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.algebra()
Symmetric group algebra of order 3 over Rational Field
```

**boundary** (*d*)

Return the boundary operator in degree *d*.

EXAMPLES:

```
sage: E.<x,y> = ExteriorAlgebra(QQ)
sage: H = E.hochschild_complex(E)
sage: d1 = H.boundary(1)
sage: z = d1.domain().an_element(); z
2*x # 1 + 2*x # x + 3*x # y
sage: d1(z)
0
sage: d1.matrix()
[ 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[ 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[ 0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[ 0 0 0 0 0 0 2 0 0 -2 0 0 0 0]

sage: s = SymmetricFunctions(QQ).s()
sage: H = s.hochschild_complex(s)
sage: d1 = H.boundary(1)
sage: x = d1.domain().an_element(); x
2*s[] # s[] + 2*s[] # s[1] + 3*s[] # s[2]
sage: d1(x)
0
sage: y = tensor([s.an_element(), s.an_element()])
sage: d1(y)
0
sage: z = tensor([s[2,1] + s[3], s.an_element()])
sage: d1(z)
0
```

**coboundary** (*d*)

Return the coboundary morphism of degree *d*.

EXAMPLES:

```
sage: E.<x,y> = ExteriorAlgebra(QQ)
sage: H = E.hochschild_complex(E)
sage: del1 = H.coboundary(1)
sage: z = del1.domain().an_element(); z
2 + 2*x + 3*y
sage: del1(z)
0
sage: del1.matrix()
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 2]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 -2]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
[ 0 0 0 0]
```

(continues on next page)

(continued from previous page)

```
[ 0  0  0  0]
[ 0  0  0  0]
```

**coefficients()**

Return the coefficients of self.

EXAMPLES:

```
sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.coefficients()
Trivial representation of Standard permutations of 3 over Rational Field
```

**cohomology(d)**

Return the d-th cohomology group.

EXAMPLES:

```
sage: E.<x,y> = ExteriorAlgebra(QQ)
sage: H = E.hochschild_complex(E)
sage: H.cohomology(0)
Vector space of dimension 3 over Rational Field
sage: H.cohomology(1)
Vector space of dimension 4 over Rational Field
sage: H.cohomology(2)
Vector space of dimension 6 over Rational Field

sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.cohomology(0)
Vector space of dimension 1 over Rational Field
sage: H.cohomology(1)
Vector space of dimension 0 over Rational Field
sage: H.cohomology(2)
Vector space of dimension 0 over Rational Field
```

When working over general rings (except  $\mathbb{Z}$ ) and we can construct a unitriangular basis for the image quotient, we fallback to a slower implementation using (combinatorial) free modules:

```
sage: R.<x,y> = QQ[]
sage: SGA = SymmetricGroupAlgebra(R, 2)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.cohomology(1)
Free module generated by {} over Multivariate Polynomial Ring in x, y over
↪Rational Field
```

**free\_module(\*args, \*\*kws)**Deprecated: Use `module()` instead. See [trac ticket #21386](#) for details.**homology(d)**

Return the d-th homology group.

EXAMPLES:

```

sage: E.<x,y> = ExteriorAlgebra(QQ)
sage: H = E.hochschild_complex(E)
sage: H.homology(0)
Vector space of dimension 3 over Rational Field
sage: H.homology(1)
Vector space of dimension 4 over Rational Field
sage: H.homology(2)
Vector space of dimension 6 over Rational Field

sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.homology(0)
Vector space of dimension 1 over Rational Field
sage: H.homology(1)
Vector space of dimension 0 over Rational Field
sage: H.homology(2)
Vector space of dimension 0 over Rational Field

```

When working over general rings (except  $\mathbf{Z}$ ) and we can construct a unitriangular basis for the image quotient, we fallback to a slower implementation using (combinatorial) free modules:

```

sage: R.<x,y> = QQ[]
sage: SGA = SymmetricGroupAlgebra(R, 2)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.homology(1)
Free module generated by {} over Multivariate Polynomial Ring in x, y over_
↪ Rational Field

```

#### **module(d)**

Return the module in degree d.

##### EXAMPLES:

```

sage: SGA = SymmetricGroupAlgebra(QQ, 3)
sage: T = SGA.trivial_representation()
sage: H = SGA.hochschild_complex(T)
sage: H.module(0)
Trivial representation of Standard permutations of 3 over Rational Field
sage: H.module(1)
Trivial representation of Standard permutations of 3 over Rational Field
# Symmetric group algebra of order 3 over Rational Field
sage: H.module(2)
Trivial representation of Standard permutations of 3 over Rational Field
# Symmetric group algebra of order 3 over Rational Field
# Symmetric group algebra of order 3 over Rational Field

```

#### **trivial\_module()**

Return the trivial module of self.

##### EXAMPLES:

```

sage: E.<x,y> = ExteriorAlgebra(QQ)
sage: H = E.hochschild_complex(E)
sage: H.trivial_module()
Free module generated by {} over Rational Field

```





## HOMOLOGY GROUPS

This module defines a `HomologyGroup()` class which is an abelian group that prints itself in a way that is suitable for homology groups.

`sage.homology.homology_group.HomologyGroup(n, base_ring, invfac=None)`  
 Abelian group on  $n$  generators which represents a homology group in a fixed degree.

INPUT:

- $n$  – integer; the number of generators
- `base_ring` – ring; the base ring over which the homology is computed
- `inv_fac` – list of integers; the invariant factors – ignored if the base ring is a field

OUTPUT:

A class that can represent the homology group in a fixed homological degree.

EXAMPLES:

```
sage: from sage.homology.homology_group import HomologyGroup
sage: G = AbelianGroup(5, [5,5,7,8,9]); G
Multiplicative Abelian group isomorphic to C5 x C5 x C7 x C8 x C9
sage: H = HomologyGroup(5, ZZ, [5,5,7,8,9]); H
C5 x C5 x C7 x C8 x C9
sage: AbelianGroup(4)
Multiplicative Abelian group isomorphic to Z x Z x Z x Z
sage: HomologyGroup(4, ZZ)
Z x Z x Z x Z
sage: HomologyGroup(100, ZZ)
Z^100
```

**class** `sage.homology.homology_group.HomologyGroup_class(n, invfac)`  
 Bases: `sage.groups.additive_abelian.additive_abelian_group.AdditiveAbelianGroup_fixed_gens`

Discrete Abelian group on  $n$  generators. This class inherits from `AdditiveAbelianGroup_fixed_gens`; see `sage.groups.additive_abelian.additive_abelian_group` for more documentation. The main difference between the classes is in the print representation.

EXAMPLES:

```
sage: from sage.homology.homology_group import HomologyGroup
sage: G = AbelianGroup(5, [5,5,7,8,9]); G
Multiplicative Abelian group isomorphic to C5 x C5 x C7 x C8 x C9
sage: H = HomologyGroup(5, ZZ, [5,5,7,8,9]); H
C5 x C5 x C7 x C8 x C9
```

(continues on next page)

(continued from previous page)

```
sage: G == loads(dumps(G))
True
sage: AbelianGroup(4)
Multiplicative Abelian group isomorphic to  $\mathbb{Z} \times \mathbb{Z} \times \mathbb{Z} \times \mathbb{Z}$ 
sage: HomologyGroup(4, ZZ)
 $\mathbb{Z} \times \mathbb{Z} \times \mathbb{Z} \times \mathbb{Z}$ 
sage: HomologyGroup(100, ZZ)
 $\mathbb{Z}^{100}$ 
```

## HOMOLOGY AND COHOMOLOGY WITH A BASIS

This module provides homology and cohomology vector spaces suitable for computing cup products and cohomology operations.

### REFERENCES:

- [GDR2003]
- [GDR1999]

### AUTHORS:

- John H. Palmieri, Travis Scrimshaw (2015-09)

**class** sage.homology.homology\_vector\_space\_with\_basis.**CohomologyRing**(*base\_ring*,  
*cell\_complex*)  
Bases: *sage.homology.homology\_vector\_space\_with\_basis*.  
*HomologyVectorSpaceWithBasis*

The cohomology ring.

---

**Note:** This is not intended to be created directly by the user, but instead via the *cohomology ring* of a *cell complex*.

---

### INPUT:

- *base\_ring* – must be a field
- *cell\_complex* – the cell complex whose homology we are computing

### EXAMPLES:

```
sage: CP2 = simplicial_complexes.ComplexProjectivePlane()
sage: H = CP2.cohomology_ring(QQ)
sage: H.basis(2)
Finite family {(2, 0): h^{2,0}}
sage: x = H.basis(2)[2,0]
```

The product structure is the cup product:

```
sage: x.cup_product(x)
-h^{4,0}
sage: x * x
-h^{4,0}
```

There are mod 2 cohomology operations defined, also, for simplicial complexes and simplicial sets:

```

sage: Hmod2 = CP2.cohomology_ring(GF(2))
sage: y = Hmod2.basis(2)[2,0]
sage: y.Sq(2)
h^{4,0}

sage: Y = simplicial_sets.RealProjectiveSpace(6).suspension()
sage: H_Y = Y.cohomology_ring(GF(2))
sage: b = H_Y.basis()[2,0]
sage: b.Sq(1)
h^{3,0}
sage: b.Sq(2)
0
sage: c = H_Y.basis()[4,0]
sage: c.Sq(1)
h^{5,0}
sage: c.Sq(2)
h^{6,0}
sage: c.Sq(3)
h^{7,0}
sage: c.Sq(4)
0

```

**class Element**

Bases: *sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis.Element*

**Sq(*i*)**

Return the result of applying  $Sq^i$  to this element.

INPUT:

- *i* – nonnegative integer

**Warning:** This is only implemented for simplicial complexes.

This cohomology operation is only defined in characteristic 2.

Algorithm: see González-Díaz and Réal [GDR1999], Corollary 3.2.

EXAMPLES:

```

sage: RP2 = simplicial_complexes.RealProjectiveSpace(2)
sage: x = RP2.cohomology_ring(GF(2)).basis()[1,0]
sage: x.Sq(1)
h^{2,0}

sage: K = RP2.suspension()
sage: K.set_immutable()
sage: y = K.cohomology_ring(GF(2)).basis()[2,0]
sage: y.Sq(1)
h^{3,0}

sage: RP4 = simplicial_complexes.RealProjectiveSpace(4)
sage: H = RP4.cohomology_ring(GF(2))
sage: x = H.basis()[1,0]
sage: y = H.basis()[2,0]
sage: z = H.basis()[3,0]
sage: x.Sq(1) == y

```

(continues on next page)

(continued from previous page)

```
True
sage: z.Sq(1) # long time
h^{4,0}
```

This calculation is much faster with simplicial sets (on one machine, 20 seconds with a simplicial complex, 4 ms with a simplicial set).

```
sage: RP4_ss = simplicial_sets.RealProjectiveSpace(4)
sage: z_ss = RP4_ss.cohomology_ring(GF(2)).basis()[3,0]
sage: z_ss.Sq(1)
h^{4,0}
```

### **cup\_product** (*other*)

Return the cup product of this element and *other*.

Algorithm: see González-Díaz and Réal [GDR2003], p. 88. Given two cohomology classes, lift them to cocycle representatives via the chain contraction for this complex, using `to_cycle()`. In the sum of their dimensions, look at all of the homology classes  $\gamma$ : lift each of those to a cycle representative, apply the Alexander-Whitney diagonal map to each cell in the cycle, evaluate the two cocycles on these factors, and multiply. The result is the value of the cup product cocycle on this homology class. After this has been done for all homology classes, since homology and cohomology are dual, one can tell which cohomology class corresponds to the cup product.

**See also:**

`CohomologyRing.product_on_basis()`

EXAMPLES:

```
sage: RP3 = simplicial_complexes.RealProjectiveSpace(3)
sage: H = RP3.cohomology_ring(GF(2))
sage: c = H.basis()[1,0]
sage: c.cup_product(c)
h^{2,0}
sage: c * c * c
h^{3,0}
```

We can also take powers:

```
sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: a = RP2.cohomology_ring(GF(2)).basis()[1,0]
sage: a**0
h^{0,0}
sage: a**1
h^{1,0}
sage: a**2
h^{2,0}
sage: a**3
0
```

A non-connected example:

```
sage: K = cubical_complexes.Torus().disjoint_union(cubical_complexes.
↳ Sphere(2))
sage: a,b = K.cohomology_ring(QQ).basis(2)
sage: a**0
h^{0,0} + h^{0,1}
```

**one()**

The multiplicative identity element.

EXAMPLES:

```

sage: H = simplicial_complexes.Torus().cohomology_ring(QQ)
sage: H.one()
h^{0,0}
sage: all(H.one() * x == x == x * H.one() for x in H.basis())
True

```

**product\_on\_basis**(*li*, *ri*)The cup product of the basis elements indexed by *li* and *ri* in this cohomology ring.

INPUT:

- *li*, *ri* – index of a cohomology class

See also:

[\*CohomologyRing.Element.cup\\_product\(\)\*](#) – the documentation for this method describes the algorithm.

EXAMPLES:

```

sage: RP3 = simplicial_complexes.RealProjectiveSpace(3)
sage: H = RP3.cohomology_ring(GF(2))
sage: c = H.basis()[1,0]
sage: c.cup_product(c).cup_product(c) # indirect doctest
h^{3,0}

sage: T = simplicial_complexes.Torus()
sage: x,y = T.cohomology_ring(QQ).basis(1)
sage: x.cup_product(y)
-h^{2,0}
sage: x.cup_product(x)
0

sage: one = T.cohomology_ring(QQ).basis()[0,0]
sage: x.cup_product(one)
h^{1,0}
sage: one.cup_product(y) == y
True
sage: one.cup_product(one)
h^{0,0}
sage: x.cup_product(y) + y.cup_product(x)
0

```

This also works with cubical complexes:

```

sage: T = cubical_complexes.Torus()
sage: x,y = T.cohomology_ring(QQ).basis(1)
sage: x.cup_product(y)
h^{2,0}
sage: x.cup_product(x)
0

```

 $\Delta$ -complexes:

```

sage: T_d = delta_complexes.Torus()
sage: a,b = T_d.cohomology_ring(QQ).basis(1)
sage: a.cup_product(b)
h^{2,0}
sage: b.cup_product(a)
-h^{2,0}
sage: RP2 = delta_complexes.RealProjectivePlane()
sage: w = RP2.cohomology_ring(GF(2)).basis()[1,0]
sage: w.cup_product(w)
h^{2,0}

```

and simplicial sets:

```

sage: from sage.homology.simplicial_set_examples import RealProjectiveSpace
sage: RP5 = RealProjectiveSpace(5)
sage: x = RP5.cohomology_ring(GF(2)).basis()[1,0]
sage: x**4
h^{4,0}

```

A non-connected example:

```

sage: K = cubical_complexes.Torus().disjoint_union(cubical_complexes.Torus())
sage: a,b,c,d = K.cohomology_ring(QQ).basis(1)
sage: x,y = K.cohomology_ring(QQ).basis(0)
sage: a.cup_product(x) == a
True
sage: a.cup_product(y)
0

```

```

class sage.homology.homology_vector_space_with_basis.HomologyVectorSpaceWithBasis (base_ring,
cell_complex,
co-
ho-
mol-
ogy=False,
cat-
e-
gory=None)

```

Bases: `sage.combinat.free_module.CombinatorialFreeModule`

Homology (or cohomology) vector space.

This provides enough structure to allow the computation of cup products and cohomology operations. See the class *CohomologyRing* (which derives from this) for examples.

It also requires field coefficients (hence the “VectorSpace” in the name of the class).

---

**Note:** This is not intended to be created directly by the user, but instead via the methods *homology\_with\_basis()* and *cohomology\_ring()* for the class of *cell complexes*.

---

INPUT:

- `base_ring` – must be a field
- `cell_complex` – the cell complex whose homology we are computing
- `cohomology` – (default: `False`) if `True`, return the cohomology as a module

- `category` – (optional) a subcategory of modules with basis

**EXAMPLES:**

Homology classes are denoted by  $h_{\{d, i\}}$  where  $d$  is the degree of the homology class and  $i$  is their index in the list of basis elements in that degree. Cohomology classes are denoted  $h^{\{1, 0\}}$ :

```
sage: RP2 = cubical_complexes.RealProjectivePlane()
sage: RP2.homology_with_basis(GF(2))
Homology module of Cubical complex with 21 vertices and 81 cubes
over Finite Field of size 2
sage: RP2.cohomology_ring(GF(2))
Cohomology ring of Cubical complex with 21 vertices and 81 cubes
over Finite Field of size 2
sage: simplicial_complexes.Torus().homology_with_basis(QQ)
Homology module of Minimal triangulation of the torus
over Rational Field
```

To access a basis element, use its degree and index (0 or 1 in the 1st cohomology group of a torus):

```
sage: H = simplicial_complexes.Torus().cohomology_ring(QQ)
sage: H.basis(1)
Finite family {(1, 0): h^{1,0}, (1, 1): h^{1,1}}
sage: x = H.basis()[1,0]; x
h^{1,0}
sage: y = H.basis()[1,1]; y
h^{1,1}
sage: 2*x-3*y
2*h^{1,0} - 3*h^{1,1}
```

You can compute cup products of cohomology classes:

```
sage: x.cup_product(y)
-h^{2,0}
sage: y.cup_product(x)
h^{2,0}
sage: x.cup_product(x)
0
```

This works with simplicial, cubical, and  $\Delta$ -complexes, and also simplicial sets:

```
sage: Klein_c = cubical_complexes.KleinBottle()
sage: H = Klein_c.cohomology_ring(GF(2))
sage: x,y = H.basis(1)
sage: x.cup_product(x)
0
sage: x.cup_product(y)
h^{2,0}
sage: y.cup_product(y)
h^{2,0}

sage: Klein_d = delta_complexes.KleinBottle()
sage: H = Klein_d.cohomology_ring(GF(2))
sage: u,v = H.basis(1)
sage: u.cup_product(u)
h^{2,0}
sage: u.cup_product(v)
0
```

(continues on next page)



(continued from previous page)

```
sage: v.cup_product(v)
h^{2,0}
```

An isomorphism between the rings for the cubical model and the  $\Delta$ -complex model can be obtained by sending  $x$  to  $u + v$ ,  $y$  to  $v$ .

```
sage: X = simplicial_sets.RealProjectiveSpace(6)
sage: H_X = X.cohomology_ring(GF(2))
sage: a = H_X.basis()[1,0]
sage: a**6
h^{6,0}
sage: a**7
0
```

All products of positive-dimensional elements in a suspension should be zero:

```
sage: Y = X.suspension()
sage: H_Y = Y.cohomology_ring(GF(2))
sage: b = H_Y.basis()[2,0]
sage: b**2
0
sage: B = sorted(H_Y.basis())[1:]
sage: B
[h^{2,0}, h^{3,0}, h^{4,0}, h^{5,0}, h^{6,0}, h^{7,0}]
sage: import itertools
sage: [a*b for (a,b) in itertools.combinations(B, 2)]
[0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0]
```

The basis elements in the simplicial complex case have been chosen differently; apply the change of basis  $x \mapsto a + b$ ,  $y \mapsto b$  to see the same product structure.

```
sage: Klein_s = simplicial_complexes.KleinBottle()
sage: H = Klein_s.cohomology_ring(GF(2))
sage: a,b = H.basis(1)
sage: a.cup_product(a)
0
sage: a.cup_product(b)
h^{2,0}
sage: (a+b).cup_product(a+b)
h^{2,0}
sage: b.cup_product(b)
h^{2,0}
```

### class Element

Bases: `sage.modules.with_basis.indexed_element.IndexedFreeModuleElement`

#### to\_cycle()

(Co)cycle representative of this homogeneous (co)homology class.

EXAMPLES:

```
sage: S2 = simplicial_complexes.Sphere(2)
sage: H = S2.homology_with_basis(QQ)
sage: h20 = H.basis()[2,0]; h20
h_{2,0}
sage: h20.to_cycle()
-(0, 1, 2) + (0, 1, 3) - (0, 2, 3) + (1, 2, 3)
```

Chains are written as linear combinations of simplices  $\sigma$ . Cochains are written as linear combinations of characteristic functions  $\chi_\sigma$  for those simplices:

```
sage: S2.cohomology_ring(QQ).basis()[2,0].to_cycle()
\chi_(1, 2, 3)
sage: S2.cohomology_ring(QQ).basis()[0,0].to_cycle()
\chi_(0,) + \chi_(1,) + \chi_(2,) + \chi_(3,)
```

**basis** ( $d=None$ )

Return (the degree  $d$  homogeneous component of) the basis of this graded vector space.

INPUT:

- $d$  – (optional) the degree

EXAMPLES:

```
sage: RP2 = simplicial_complexes.ProjectivePlane()
sage: H = RP2.homology_with_basis(QQ)
sage: H.basis()
Finite family {(0, 0): h_{0,0}}
sage: H.basis(0)
Finite family {(0, 0): h_{0,0}}
sage: H.basis(1)
Finite family {}
sage: H.basis(2)
Finite family {}
```

**complex** ()

The cell complex whose homology is being computed.

EXAMPLES:

```
sage: H = simplicial_complexes.Simplex(2).homology_with_basis(QQ)
sage: H.complex()
The 2-simplex
```

**contraction** ()

The chain contraction associated to this homology computation.

That is, to work with chain representatives of homology classes, we need the chain complex  $C$  associated to the cell complex, the chain complex  $H$  of its homology (with trivial differential), chain maps  $\pi : C \rightarrow H$  and  $\iota : H \rightarrow C$ , and a chain contraction  $\phi$  giving a chain homotopy between  $1_C$  and  $\iota \circ \pi$ .

OUTPUT:  $\phi$

See [ChainContraction](#) for information about chain contractions, and see [algebraic\\_topological\\_model\(\)](#) for the construction of this particular chain contraction  $\phi$ .

EXAMPLES:

```
sage: H = simplicial_complexes.Simplex(2).homology_with_basis(QQ)
sage: H.contraction()
Chain homotopy between:
  Chain complex endomorphism of Chain complex with at most 3 nonzero terms_
↪over Rational Field
  and Chain complex endomorphism of Chain complex with at most 3 nonzero_
↪terms over Rational Field
```

From the chain contraction, one can also recover the maps  $\pi$  and  $\iota$ :

```

sage: phi = H.contraction()
sage: phi.pi()
Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Rational Field
  To: Chain complex with at most 1 nonzero terms over Rational Field
sage: phi.iota()
Chain complex morphism:
  From: Chain complex with at most 1 nonzero terms over Rational Field
  To: Chain complex with at most 3 nonzero terms over Rational Field

```

**degree\_on\_basis**(*i*)

Return the degree of the basis element indexed by *i*.

EXAMPLES:

```

sage: H = simplicial_complexes.Torus().homology_with_basis(GF(7))
sage: H.degree_on_basis((2, 0))
2

```

`sage.homology.homology_vector_space_with_basis.sum_indices`(*k*, *i\_k\_plus\_one*, *S\_k\_plus\_one*)

This is a recursive function for computing the indices for the nested sums in González-Díaz and Réal [GDR1999], Corollary 3.2.

In the paper, given indices  $i_n, i_{n-1}, \dots, i_{k+1}$ , given  $k$ , and given  $S(k+1)$ , the number  $S(k)$  is defined to be

$$S(k) = -S(k+1) + \text{floor}(k/2) + \text{floor}((k+1)/2) + i_{k+1},$$

and  $i_k$  ranges from  $S(k)$  to  $i_{k+1} - 1$ . There are two special cases: if  $k = 0$ , then  $i_0 = S(0)$ . Also, the initial case of  $S(k)$  is  $S(n)$ , which is set in the method `Sq()` before calling this function. For this function, given  $k$ ,  $i_{k+1}$ , and  $S(k+1)$ , return a list consisting of the allowable possible indices  $[i_k, i_{k-1}, \dots, i_1, i_0]$  given by the above formula.

INPUT:

- *k* – non-negative integer
- *i\_k\_plus\_one* – the positive integer  $i_{k+1}$
- *S\_k\_plus\_one* – the integer  $S(k+1)$

EXAMPLES:

```

sage: from sage.homology.homology_vector_space_with_basis import sum_indices
sage: sum_indices(1, 3, 3)
[[1, 0], [2, 1]]
sage: sum_indices(0, 4, 2)
[[2]]

```



## ALGEBRAIC TOPOLOGICAL MODEL FOR A CELL COMPLEX

This file contains two functions, `algebraic_topological_model()` and `algebraic_topological_model_delta_complex()`. The second works more generally: for all simplicial, cubical, and  $\Delta$ -complexes. The first only works for simplicial and cubical complexes, but it is faster in those cases.

AUTHORS:

- John H. Palmieri (2015-09)

```
sage.homology.algebraic_topological_model.algebraic_topological_model(K,  
Algebraic topological model for cell complex K with coefficients in the field base_ring.  
base_ring=None)
```

INPUT:

- $K$  – either a simplicial complex or a cubical complex
- `base_ring` – coefficient ring; must be a field

OUTPUT: a pair  $(\phi, M)$  consisting of

- chain contraction  $\phi$
- chain complex  $M$

This construction appears in a paper by Pilarczyk and Réal [PR2015]. Given a cell complex  $K$  and a field  $F$ , there is a chain complex  $C$  associated to  $K$  with coefficients in  $F$ . The *algebraic topological model* for  $K$  is a chain complex  $M$  with trivial differential, along with chain maps  $\pi : C \rightarrow M$  and  $\iota : M \rightarrow C$  such that

- $\pi\iota = 1_M$ , and
- there is a chain homotopy  $\phi$  between  $1_C$  and  $\iota\pi$ .

In particular,  $\pi$  and  $\iota$  induce isomorphisms on homology, and since  $M$  has trivial differential, it is its own homology, and thus also the homology of  $C$ . Thus  $\iota$  lifts homology classes to their cycle representatives.

The chain homotopy  $\phi$  satisfies some additional properties, making it a *chain contraction*:

- $\phi\phi = 0$ ,
- $\pi\phi = 0$ ,
- $\phi\iota = 0$ .

Given an algebraic topological model for  $K$ , it is then easy to compute cup products and cohomology operations on the cohomology of  $K$ , as described in [GDR2003] and [PR2015].

Implementation details: the cell complex  $K$  must have an `n_cells()` method from which we can extract a list of cells in each dimension. Combining the lists in increasing order of dimension then defines a filtration of the complex: a list of cells in which the boundary of each cell consists of cells earlier in the list. This is required

by Pilarczyk and Réal's algorithm. There must also be a `chain_complex()` method, to construct the chain complex  $C$  associated to this chain complex.

In particular, this works for simplicial complexes and cubical complexes. It doesn't work for  $\Delta$ -complexes, though: the list of their  $n$ -cells has the wrong format.

Note that from the chain contraction `phi`, one can recover the chain maps  $\pi$  and  $\iota$  via `phi.pi()` and `phi.iota()`. Then one can recover  $C$  and  $M$  from, for example, `phi.pi().domain()` and `phi.pi().codomain()`, respectively.

EXAMPLES:

```
sage: from sage.homology.algebraic_topological_model import algebraic_topological_
      ↪model
sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: phi, M = algebraic_topological_model(RP2, GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = cubical_complexes.Torus()
sage: phi, M = algebraic_topological_model(T, QQ)
sage: M.homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

If you want to work with cohomology rather than homology, just dualize the outputs of this function:

```
sage: M.dual().homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
sage: M.dual().degree_of_differential()
1
sage: phi.dual()
Chain homotopy between:
  Chain complex endomorphism of Chain complex with at most 3 nonzero terms over_
  ↪Rational Field
and Chain complex morphism:
  From: Chain complex with at most 3 nonzero terms over Rational Field
  To:   Chain complex with at most 3 nonzero terms over Rational Field
```

In degree 0, the inclusion of the homology  $M$  into the chain complex  $C$  sends the homology generator to a single vertex:

```
sage: K = simplicial_complexes.Simplex(2)
sage: phi, M = algebraic_topological_model(K, QQ)
sage: phi.iota().in_degree(0)
[0]
[0]
[1]
```

In cohomology, though, one needs the dual of every degree 0 cell to detect the degree 0 cohomology generator:

```
sage: phi.dual().iota().in_degree(0)
[1]
[1]
[1]
```

`sage.homology.algebraic_topological_model.algebraic_topological_model_delta_complex(K, base_ring)`

Algebraic topological model for cell complex  $K$  with coefficients in the field `base_ring`.

This has the same basic functionality as `algebraic_topological_model()`, but it also works for  $\Delta$ -complexes. For simplicial and cubical complexes it is somewhat slower, though.

INPUT:

- $K$  – a simplicial complex, a cubical complex, or a  $\Delta$ -complex
- `base_ring` – coefficient ring; must be a field

OUTPUT: a pair  $(\phi, M)$  consisting of

- chain contraction  $\phi$
- chain complex  $M$

See `algebraic_topological_model()` for the main documentation. The difference in implementation between the two: this uses matrix and vector algebra. The other function does more of the computations “by hand” and uses cells (given as simplices or cubes) to index various dictionaries. Since the cells in  $\Delta$ -complexes are not as nice, the other function does not work for them, while this function relies almost entirely on the structure of the associated chain complex.

EXAMPLES:

```
sage: from sage.homology.algebraic_topological_model import algebraic_topological_
      ↪model_delta_complex as AT_model
sage: RP2 = simplicial_complexes.RealProjectivePlane()
sage: phi, M = AT_model(RP2, GF(2))
sage: M.homology()
{0: Vector space of dimension 1 over Finite Field of size 2,
 1: Vector space of dimension 1 over Finite Field of size 2,
 2: Vector space of dimension 1 over Finite Field of size 2}
sage: T = delta_complexes.Torus()
sage: phi, M = AT_model(T, QQ)
sage: M.homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
```

If you want to work with cohomology rather than homology, just dualize the outputs of this function:

```
sage: M.dual().homology()
{0: Vector space of dimension 1 over Rational Field,
 1: Vector space of dimension 2 over Rational Field,
 2: Vector space of dimension 1 over Rational Field}
sage: M.dual().degree_of_differential()
1
sage: phi.dual()
Chain homotopy between:
  Chain complex endomorphism of Chain complex with at most 3 nonzero terms over_
  ↪Rational Field
  and Chain complex morphism:
    From: Chain complex with at most 3 nonzero terms over Rational Field
    To:   Chain complex with at most 3 nonzero terms over Rational Field
```

In degree 0, the inclusion of the homology  $M$  into the chain complex  $C$  sends the homology generator to a single vertex:

```
sage: K = delta_complexes.Simplex(2)
sage: phi, M = AT_model(K, QQ)
sage: phi.iota().in_degree(0)
[0]
[0]
[1]
```

In cohomology, though, one needs the dual of every degree 0 cell to detect the degree 0 cohomology generator:

```
sage: phi.dual().iota().in_degree(0)
[1]
[1]
[1]
```



## INDUCED MORPHISMS ON HOMOLOGY

This module implements morphisms on homology induced by morphisms of simplicial complexes. It requires working with field coefficients.

See [InducedHomologyMorphism](#) for documentation.

AUTHORS:

- John H. Palmieri (2015.09)

```
class sage.homology.homology_morphism.InducedHomologyMorphism(map,  
                                                                base_ring=None,  
                                                                cohomol-  
                                                                ogy=False)
```

Bases: `sage.categories.morphism.Morphism`

An element of this class is a morphism of (co)homology groups induced by a map of simplicial complexes. It requires working with field coefficients.

INPUT:

- `map` – the map of simplicial complexes
- `base_ring` – a field (optional, default `QQ`)
- `cohomology` – boolean (optional, default `False`). If `True`, return the induced map in cohomology rather than homology.

---

**Note:** This is not intended to be used directly by the user, but instead via the method `induced_homology_morphism()`.

---

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: H = Hom(S1, S1)
sage: f = H({0:0, 1:2, 2:1}) # f switches two vertices
sage: f_star = f.induced_homology_morphism(QQ, cohomology=True)
sage: f_star
Graded algebra endomorphism of Cohomology ring of Minimal triangulation of the 1-
→sphere over Rational Field
Defn: induced by:
      Simplicial complex endomorphism of Minimal triangulation of the 1-sphere
      Defn: 0 |--> 0
            1 |--> 2
            2 |--> 1
sage: f_star.to_matrix(1)
```

(continues on next page)

(continued from previous page)

```

[-1]
sage: f_star.to_matrix()
[ 1| 0]
[--+-]
[ 0|-1]

sage: T = simplicial_complexes.Torus()
sage: y = T.homology_with_basis(QQ).basis()[ (1,1) ]
sage: y.to_cycle()
(0, 5) - (0, 6) + (5, 6)

```

Since  $(0, 2) - (0, 5) + (2, 5)$  is a cycle representing a homology class in the torus, we can define a map  $S^1 \rightarrow T$  inducing an inclusion on  $H_1$ :

```

sage: Hom(S1, T)({0:0, 1:2, 2:5})
Simplicial complex morphism:
  From: Minimal triangulation of the 1-sphere
  To: Minimal triangulation of the torus
  Defn: 0 |--> 0
        1 |--> 2
        2 |--> 5
sage: g = Hom(S1, T)({0:0, 1:2, 2:5})
sage: g_star = g.induced_homology_morphism(QQ)
sage: g_star.to_matrix(0)
[1]
sage: g_star.to_matrix(1)
[-1]
[ 0]
sage: g_star.to_matrix()
[ 1| 0]
[--+-]
[ 0|-1]
[ 0| 0]
[--+-]
[ 0| 0]

```

We can evaluate such a map on (co)homology classes:

```

sage: H = S1.homology_with_basis(QQ)
sage: a = H.basis()[ (1,0) ]
sage: g_star(a)
-h_{1,0}

sage: T = S1.product(S1, is_mutable=False)
sage: diag = Hom(S1, T).diagonal_morphism()
sage: b, c = list(T.cohomology_ring().basis(1))
sage: diag_c = diag.induced_homology_morphism(cohomology=True)
sage: diag_c(b)
h^{1,0}
sage: diag_c(c)
h^{1,0}

```

**base\_ring()**

The base ring for this map

EXAMPLES:

```

sage: K = simplicial_complexes.Simplex(2)
sage: H = Hom(K, K)
sage: id = H.identity()
sage: id.induced_homology_morphism(QQ).base_ring()
Rational Field
sage: id.induced_homology_morphism(GF(13)).base_ring()
Finite Field of size 13

```

**is\_identity()**

True if this is the identity map on (co)homology.

EXAMPLES:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: H = Hom(S1, S1)
sage: flip = H({0:0, 1:2, 2:1})
sage: flip.induced_homology_morphism(QQ).is_identity()
False
sage: flip.induced_homology_morphism(GF(2)).is_identity()
True
sage: rotate = H({0:1, 1:2, 2:0})
sage: rotate.induced_homology_morphism(QQ).is_identity()
True

```

**is\_injective()**

True if this map is injective on (co)homology.

EXAMPLES:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: K = simplicial_complexes.Simplex(2)
sage: H = Hom(S1, K)
sage: f = H({0:0, 1:1, 2:2})
sage: f.induced_homology_morphism().is_injective()
False
sage: f.induced_homology_morphism(cohomology=True).is_injective()
True

sage: T = simplicial_complexes.Torus()
sage: g = Hom(S1, T)({0:0, 1:3, 2:6})
sage: g_star = g.induced_homology_morphism(QQ)
sage: g_star.is_injective()
True

```

**is\_surjective()**

True if this map is surjective on (co)homology.

EXAMPLES:

```

sage: S1 = simplicial_complexes.Sphere(1)
sage: K = simplicial_complexes.Simplex(2)
sage: H = Hom(S1, K)
sage: f = H({0:0, 1:1, 2:2})
sage: f.induced_homology_morphism().is_surjective()
True
sage: f.induced_homology_morphism(cohomology=True).is_surjective()
False

```

**to\_matrix**(*deg=None*)

The matrix for this map.

If degree *deg* is specified, return the matrix just in that degree; otherwise, return the block matrix representing the entire map.

INPUT:

- *deg* – (optional, default *None*) the degree

EXAMPLES:

```
sage: S1 = simplicial_complexes.Sphere(1)
sage: S1_b = S1.barycentric_subdivision()
sage: S1_b.set_immutable()
sage: d = {(0,): 0, (0,1): 1, (1,): 2, (1,2): 0, (2,): 1, (0,2): 2}
sage: f = Hom(S1_b, S1)(d)
sage: h = f.induced_homology_morphism(QQ)
sage: h.to_matrix(1)
[2]
sage: h.to_matrix()
[1|0]
[-+-]
[0|2]
```

## UTILITY FUNCTIONS FOR MATRICES

The actual computation of homology groups ends up being linear algebra with the differentials thought of as matrices. This module contains some utility functions for this purpose.

`sage.homology.matrix_utils.dhsw_snf(mat, verbose=False)`

Preprocess a matrix using the “Elimination algorithm” described by Dumas et al. [DHSW2003], and then call `elementary_divisors` on the resulting (smaller) matrix.

---

**Note:** ‘snf’ stands for ‘Smith Normal Form’.

---

INPUT:

- `mat` – an integer matrix, either sparse or dense.

(They use the transpose of the matrix considered here, so they use rows instead of columns.)

ALGORITHM:

Go through `mat` one column at a time. For each column, add multiples of previous columns to it until either

- it’s zero, in which case it should be deleted.
- its first nonzero entry is 1 or -1, in which case it should be kept.
- its first nonzero entry is something else, in which case it is deferred until the second pass.

Then do a second pass on the deferred columns.

At this point, the columns with 1 or -1 in the first entry contribute to the rank of the matrix, and these can be counted and then deleted (after using the 1 or -1 entry to clear out its row). Suppose that there were  $N$  of these.

The resulting matrix should be much smaller; we then feed it to Sage’s `elementary_divisors` function, and prepend  $N$  1’s to account for the rows deleted in the previous step.

EXAMPLES:

```
sage: from sage.homology.matrix_utils import dhsw_snf
sage: mat = matrix(ZZ, 3, 4, range(12))
sage: dhsw_snf(mat)
[1, 4, 0]
sage: mat = random_matrix(ZZ, 20, 20, x=-1, y=2)
sage: mat.elementary_divisors() == dhsw_snf(mat)
True
```



## INTERFACE TO CHOMP

CHomP stands for “Computation Homology Program”, and is good at computing homology of simplicial complexes, cubical complexes, and chain complexes. It can also compute homomorphisms induced on homology by maps. See the CHomP web page <http://chomp.rutgers.edu/> for more information.

AUTHOR:

- John H. Palmieri

**class** `sage.interfaces.chomp.CHomP`  
Interface to the CHomP package.

### Parameters

- **program** (*string*) – which CHomP program to use
- **complex** – a simplicial or cubical complex
- **subcomplex** – a subcomplex of `complex` or `None` (the default)
- **base\_ring** (ring; optional, default  $\mathbf{Z}$ ) – ring over which to perform computations – must be  $\mathbf{Z}$  or  $\mathbf{F}_p$ .
- **generators** (*boolean; optional, default False*) – if `True`, also return list of generators
- **verbose** (*boolean; optional, default False*) – if `True`, print helpful messages as the computation progresses
- **extra\_opts** (*string*) – options passed directly to `program`

**Returns** homology groups as a dictionary indexed by dimension

The programs `homsimpl`, `homcubes`, and `homchain` are available through this interface. `homsimpl` computes the relative or absolute homology groups of simplicial complexes. `homcubes` computes the relative or absolute homology groups of cubical complexes. `homchain` computes the homology groups of chain complexes. For consistency with Sage’s other homology computations, the answers produced by `homsimpl` and `homcubes` in the absolute case are converted to reduced homology.

Note also that CHomP can only compute over the integers or  $\mathbf{F}_p$ . CHomP is fast enough, though, that if you want rational information, you should consider using CHomP with integer coefficients, or with mod  $p$  coefficients for a sufficiently large  $p$ , rather than using Sage’s built-in homology algorithms.

See also the documentation for the functions `homchain()`, `homcubes()`, and `homsimpl()` for more examples, including illustrations of some of the optional parameters.

EXAMPLES:

```
sage: from sage.interfaces.chomp import CHomP
sage: T = cubical_complexes.Torus()
sage: CHomP()('homcubes', T) # optional - CHomP
{0: 0, 1:  $\mathbb{Z} \times \mathbb{Z}$ , 2:  $\mathbb{Z}$ }
```

Relative homology of a segment relative to its endpoints:

```
sage: edge = simplicial_complexes.Simplex(1)
sage: ends = edge.n_skeleton(0)
sage: CHomP()('homsimpl', edge) # optional - CHomP
{0: 0}
sage: CHomP()('homsimpl', edge, ends) # optional - CHomP
{0: 0, 1:  $\mathbb{Z}$ }
```

Homology of a chain complex:

```
sage: C = ChainComplex({3: 2 * identity_matrix(ZZ, 2)}, degree=-1)
sage: CHomP()('homchain', C) # optional - CHomP
{2:  $\mathbb{C}^2 \times \mathbb{C}^2$ }
```

**help** (*program*)

Print a help message for *program*, a program from the CHomP suite.

**Parameters** *program* (*string*) – which CHomP program to use

**Returns** nothing – just print a message

EXAMPLES:

```
sage: from sage.interfaces.chomp import CHomP
sage: CHomP().help('homcubes') # optional - CHomP
HOMCUBES, ver. ... Copyright (C) ... by Pawel Pilarczyk...
```

`sage.interfaces.chomp.have_chomp` (*program*='homsimpl')

Return True if this computer has *program* installed.

The first time it is run, this function caches its result in the variable `_have_chomp` – a dictionary indexed by program name – and any subsequent time, it just checks the value of the variable.

This program is used in the routine `CHomP.__call__`.

If this computer doesn't have CHomP installed, you may obtain it from <http://chomp.rutgers.edu/>.

EXAMPLES:

```
sage: from sage.interfaces.chomp import have_chomp
sage: have_chomp() # random -- depends on whether CHomP is installed
True
sage: 'homsimpl' in sage.interfaces.chomp._have_chomp
True
sage: sage.interfaces.chomp._have_chomp['homsimpl'] == have_chomp()
True
```

`sage.interfaces.chomp.homchain` (*complex*=None, **\*\*kws**)

Compute the homology of a chain complex using the CHomP program *homchain*.

**Parameters**

- **complex** – a chain complex



- **generators** (*boolean; optional, default False*) – if True, also return list of generators
- **verbose** (*boolean; optional, default False*) – if True, print helpful messages as the computation progresses
- **help** (*boolean; optional, default False*) – if True, just print a help message and exit
- **extra\_opts** (*string*) – options passed directly to `homchain`

**Returns** homology groups as a dictionary indexed by dimension

EXAMPLES:

```
sage: from sage.interfaces.chomp import homchain
sage: C = cubical_complexes.Sphere(3).chain_complex()
sage: homchain(C)[3] # optional - CHomP
Z
```

Generators: these are given as a list after the homology group. Each generator is specified as a cycle, an element in the appropriate free module over the base ring:

```
sage: C2 = delta_complexes.Sphere(2).chain_complex()
sage: homchain(C2, generators=True)[2] # optional - CHomP
(Z, [(1, -1)])
sage: homchain(C2, generators=True, base_ring=GF(2))[2] # optional - CHomP
(Vector space of dimension 1 over Finite Field of size 2, [(1, 1)])
```

`sage.interfaces.chomp.homcubes` (*complex=None, subcomplex=None, \*\*kws*)

Compute the homology of a cubical complex using the CHomP program `homcubes`. If the argument `subcomplex` is present, compute homology of complex relative to `subcomplex`.

**Parameters**

- **complex** – a cubical complex
- **subcomplex** – a subcomplex of `complex` or `None` (the default)
- **base\_ring** (*ring; optional, default Z*) – ring over which to perform computations – must be `Z` or `Fp`.
- **generators** (*boolean; optional, default False*) – if True, also return list of generators
- **verbose** (*boolean; optional, default False*) – if True, print helpful messages as the computation progresses
- **help** (*boolean; optional, default False*) – if True, just print a help message and exit
- **extra\_opts** (*string*) – options passed directly to `homcubes`

**Returns** homology groups as a dictionary indexed by dimension

EXAMPLES:

```
sage: from sage.interfaces.chomp import homcubes
sage: S = cubical_complexes.Sphere(3)
sage: homcubes(S)[3] # optional - CHomP
Z
```

Relative homology:

```
sage: C3 = cubical_complexes.Cube(3)
sage: bdry = C3.n_skeleton(2)
sage: homcubes(C3, bdry) # optional - CHomP
{0: 0, 1: 0, 2: 0, 3: Z}
```

Generators: these are given as a list after the homology group. Each generator is specified as a linear combination of cubes:

```
sage: homcubes(cubical_complexes.Sphere(1), generators=True, base_
↪ring=GF(2))[1][1] # optional - CHomP
[[[1,1] x [0,1]] + [[0,1] x [1,1]] + [[0,1] x [0,0]] + [[0,0] x [0,1]]]
```

`sage.interfaces.chomp.homsimpl(complex=None, subcomplex=None, **kws)`

Compute the homology of a simplicial complex using the CHomP program `homsimpl`. If the argument `subcomplex` is present, compute homology of complex relative to `subcomplex`.

#### Parameters

- **complex** – a simplicial complex
- **subcomplex** – a subcomplex of `complex` or `None` (the default)
- **base\_ring** (ring; optional, default  $\mathbf{Z}$ ) – ring over which to perform computations – must be  $\mathbf{Z}$  or  $\mathbf{F}_p$ .
- **generators** (boolean; optional, default `False`) – if `True`, also return list of generators
- **verbose** (boolean; optional, default `False`) – if `True`, print helpful messages as the computation progresses
- **help** (boolean; optional, default `False`) – if `True`, just print a help message and exit
- **extra\_opts** (string) – options passed directly to program

**Returns** homology groups as a dictionary indexed by dimension

EXAMPLES:

```
sage: from sage.interfaces.chomp import homsimpl
sage: T = simplicial_complexes.Torus()
sage: M8 = simplicial_complexes.MooreSpace(8)
sage: M4 = simplicial_complexes.MooreSpace(4)
sage: X = T.disjoint_union(T).disjoint_union(T).disjoint_union(M8).disjoint_
↪union(M4)
sage: homsimpl(X)[1] # optional - CHomP
Z^6 x C4 x C8
```

Relative homology:

```
sage: S = simplicial_complexes.Simplex(3)
sage: bdry = S.n_skeleton(2)
sage: homsimpl(S, bdry)[3] # optional - CHomP
Z
```

Generators: these are given as a list after the homology group. Each generator is specified as a linear combination of simplices:

```

sage: homsimpl(S, bdry, generators=True)[3] # optional - CHomP
(Z, [(0, 1, 2, 3)])

sage: homsimpl(simplicial_complexes.Sphere(1), generators=True) # optional - CHomP
{0: 0, 1: (Z, [(0, 1) - (0, 2) + (1, 2)])}

```

`sage.interfaces.chomp.process_generators_chain(gen_string, dim, base_ring=None)`  
 Process CHomP generator information for simplicial complexes.

#### Parameters

- **gen\_string**(*string*) – generator output from CHomP
- **dim**(*integer*) – dimension in which to find generators
- **base\_ring**(*optional*, *default* ZZ) – base ring over which to do the computations

**Returns** list of generators in each dimension, as described below

*gen\_string* has the form

```

[H_0]
a1

[H_1]
a2
a3

[H_2]
a1 - a2

```

For each homology group, each line lists a homology generator as a linear combination of generators  $a_i$  of the group of chains in the appropriate dimension. The elements  $a_i$  are indexed starting with  $i = 1$ . Each generator is converted, using regular expressions, from a string to a vector (an element in the free module over *base\_ring*), with  $a_i$  representing the unit vector in coordinate  $i - 1$ . For example, the string  $a1 - a2$  gets converted to the vector  $(1, -1)$ .

Therefore the return value is a list of vectors.

#### EXAMPLES:

```

sage: from sage.interfaces.chomp import process_generators_chain
sage: s = "[H_0]\na1\n\n[H_1]\na2\na3\n"
sage: process_generators_chain(s, 1)
[(0, 1), (0, 0, 1)]
sage: s = "[H_0]\na1\n\n[H_1]\n5 * a2 - a1\na3\n"
sage: process_generators_chain(s, 1, base_ring=ZZ)
[(-1, 5), (0, 0, 1)]
sage: process_generators_chain(s, 1, base_ring=GF(2))
[(1, 1), (0, 0, 1)]

```

`sage.interfaces.chomp.process_generators_cubical(gen_string, dim)`  
 Process CHomP generator information for cubical complexes.

#### Parameters

- **gen\_string**(*string*) – generator output from CHomP
- **dim**(*integer*) – dimension in which to find generators

**Returns** list of generators in each dimension, as described below

`gen_string` has the form

```
The 2 generators of H_1 follow:
generator 1
-1 * [(0,0,0,0,0) (0,0,0,0,1)]
1 * [(0,0,0,0,0) (0,0,1,0,0)]
...
generator 2
-1 * [(0,1,0,1,1) (1,1,0,1,1)]
-1 * [(0,1,0,0,1) (0,1,0,1,1)]
...
```

Each line consists of a coefficient multiplied by a cube; the cube is specified by its “bottom left” and “upper right” corners.

For technical reasons, we remove the first coordinate of each tuple, and using regular expressions, the remaining parts get converted from a string to a pair (`coefficient`, `Cube`), with the cube represented as a product of tuples. For example, the first line in “generator 1” gets turned into

```
(-1, [0,0] x [0,0] x [0,0] x [0,1])
```

representing an element in the free abelian group with basis given by cubes. Each generator is a list of such pairs, representing the sum of such elements. These are reassembled in `CHomP.__call__()` to actual elements in the free module generated by the cubes of the cubical complex in the appropriate dimension.

Therefore the return value is a list of lists of pairs, one list of pairs for each generator.

EXAMPLES:

```
sage: from sage.interfaces.chomp import process_generators_cubical
sage: s = "The 2 generators of H_1 follow:\n generator 1:\n-1 * [(0,0,0,0,0) (0,0,0,\n↪0,1)]\n1 * [(0,0,0,0,0) (0,0,1,0,0)]"
sage: process_generators_cubical(s, 1)
[[-1, [0,0] x [0,0] x [0,0] x [0,1]), (1, [0,0] x [0,1] x [0,0] x [0,0])]
sage: len(process_generators_cubical(s, 1)) # only one generator
1
```

`sage.interfaces.chomp.process_generators_simplicial(gen_string, dim, complex)`  
Process CHomP generator information for simplicial complexes.

#### Parameters

- **gen\_string** (*string*) – generator output from CHomP
- **dim** (*integer*) – dimension in which to find generators
- **complex** – simplicial complex under consideration

**Returns** list of generators in each dimension, as described below

`gen_string` has the form

```
The 2 generators of H_1 follow:
generator 1
-1 * (1,6)
1 * (1,4)
...
generator 2
-1 * (1,6)
1 * (1,4)
...
```

where each line contains a coefficient and a simplex. Each line is converted, using regular expressions, from a string to a pair (coefficient, Simplex), like

```
(-1, (1, 6))
```

representing an element in the free abelian group with basis given by simplices. Each generator is a list of such pairs, representing the sum of such elements. These are reassembled in `CHomP.__call__()` to actual elements in the free module generated by the simplices of the simplicial complex in the appropriate dimension.

Therefore the return value is a list of lists of pairs, one list of pairs for each generator.

EXAMPLES:

```
sage: from sage.interfaces.chomp import process_generators_simplicial
sage: s = "The 2 generators of H_1 follow:\n generator 1:\n -1 * (1, 6)\n n1 * (1, 4) "
sage: process_generators_simplicial(s, 1, simplicial_complexes.Torus())
[[-1, (1, 6)), (1, (1, 4))]]
```



## INDICES AND TABLES

- [Index](#)
- [Module Index](#)
- [Search Page](#)





## PYTHON MODULE INDEX

### h

`sage.homology.algebraic_topological_model`, 257  
`sage.homology.cell_complex`, 223  
`sage.homology.chain_complex`, 3  
`sage.homology.chain_complex_homspace`, 37  
`sage.homology.chain_complex_morphism`, 27  
`sage.homology.chain_homotopy`, 31  
`sage.homology.chains`, 19  
`sage.homology.cubical_complex`, 123  
`sage.homology.delta_complex`, 107  
`sage.homology.examples`, 93  
`sage.homology.hochschild_complex`, 239  
`sage.homology.homology_group`, 245  
`sage.homology.homology_morphism`, 261  
`sage.homology.homology_vector_space_with_basis`, 247  
`sage.homology.koszul_complex`, 237  
`sage.homology.matrix_utils`, 265  
`sage.homology.simplicial_complex`, 41  
`sage.homology.simplicial_complex_homset`, 89  
`sage.homology.simplicial_complex_morphism`, 79  
`sage.homology.simplicial_set`, 139  
`sage.homology.simplicial_set_catalog`, 207  
`sage.homology.simplicial_set_constructions`, 177  
`sage.homology.simplicial_set_examples`, 201  
`sage.homology.simplicial_set_morphism`, 209

### i

`sage.interfaces.chomp`, 267



## A

AbstractSimplex() (in module sage.homology.simplicial\_set), 142  
 AbstractSimplex\_class (class in sage.homology.simplicial\_set), 144  
 add\_face() (sage.homology.simplicial\_complex.SimplicialComplex method), 48  
 alexander\_dual() (sage.homology.simplicial\_complex.SimplicialComplex method), 49  
 alexander\_whitney() (sage.homology.cell\_complex.GenericCellComplex method), 223  
 alexander\_whitney() (sage.homology.cubical\_complex.Cube method), 124  
 alexander\_whitney() (sage.homology.cubical\_complex.CubicalComplex method), 129  
 alexander\_whitney() (sage.homology.delta\_complex.DeltaComplex method), 110  
 alexander\_whitney() (sage.homology.simplicial\_complex.Simplex method), 44  
 alexander\_whitney() (sage.homology.simplicial\_complex.SimplicialComplex method), 49  
 alexander\_whitney() (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 147  
 algebra() (sage.homology.hochschild\_complex.HochschildComplex method), 240  
 algebraic\_topological\_model() (in module sage.homology.algebraic\_topological\_model), 257  
 algebraic\_topological\_model() (sage.homology.cell\_complex.GenericCellComplex method), 224  
 algebraic\_topological\_model() (sage.homology.cubical\_complex.CubicalComplex method), 129  
 algebraic\_topological\_model() (sage.homology.delta\_complex.DeltaComplex method), 111  
 algebraic\_topological\_model() (sage.homology.simplicial\_complex.SimplicialComplex method), 50  
 algebraic\_topological\_model() (sage.homology.simplicial\_set.SimplicialSet\_finite method), 170  
 algebraic\_topological\_model\_delta\_complex() (in module sage.homology.algebraic\_topological\_model), 258  
 all\_degeneracies() (in module sage.homology.simplicial\_set), 173  
 all\_n\_simplices() (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 148  
 ambient() (sage.homology.simplicial\_set\_constructions.QuotientOfSimplicialSet method), 193  
 ambient\_space() (sage.homology.simplicial\_set\_constructions.SubSimplicialSet method), 197  
 an\_element() (sage.homology.simplicial\_complex\_homset.SimplicialComplexHomset method), 89  
 an\_element() (sage.homology.simplicial\_set\_morphism.SimplicialSetHomset method), 209  
 apply\_degeneracies() (sage.homology.simplicial\_set.AbstractSimplex\_class method), 145  
 associated\_chain\_complex\_morphism() (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 80  
 associated\_chain\_complex\_morphism() (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 213  
 automorphism\_group() (sage.homology.simplicial\_complex.SimplicialComplex method), 50

## B

BarnetteSphere() (in module sage.homology.examples), 94  
 barycentric\_subdivision() (sage.homology.delta\_complex.DeltaComplex method), 111  
 barycentric\_subdivision() (sage.homology.simplicial\_complex.SimplicialComplex method), 51

`base_as_subset()` (sage.homology.simplicial\_set\_constructions.ConeOfSimplicialSet\_finite method), 179  
`base_ring()` (sage.homology.homology\_morphism.InducedHomologyMorphism method), 262  
`basis()` (sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis method), 254  
`beti()` (sage.homology.cell\_complex.GenericCellComplex method), 224  
`beti()` (sage.homology.chain\_complex.ChainComplex\_class method), 6  
`beti()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 148  
`boundary()` (sage.homology.chains.Chains.Element method), 20  
`boundary()` (sage.homology.hochschild\_complex.HochschildComplex method), 240  
`BrucknerGrunbaumSphere()` (in module sage.homology.examples), 94

## C

`cartesian_product()` (sage.homology.chain\_complex.ChainComplex\_class method), 6  
`cartesian_product()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 149  
`cell_complex()` (sage.homology.chains.CellComplexReference method), 19  
`CellComplexReference` (class in sage.homology.chains), 19  
`cells()` (sage.homology.cell\_complex.GenericCellComplex method), 225  
`cells()` (sage.homology.cubical\_complex.CubicalComplex method), 130  
`cells()` (sage.homology.delta\_complex.DeltaComplex method), 112  
`cells()` (sage.homology.simplicial\_complex.SimplicialComplex method), 51  
`cells()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 150  
`Chain_class` (class in sage.homology.chain\_complex), 15  
`chain_complex()` (sage.homology.cell\_complex.GenericCellComplex method), 225  
`chain_complex()` (sage.homology.chains.Chains method), 21  
`chain_complex()` (sage.homology.cubical\_complex.CubicalComplex method), 130  
`chain_complex()` (sage.homology.delta\_complex.DeltaComplex method), 112  
`chain_complex()` (sage.homology.simplicial\_complex.SimplicialComplex method), 51  
`chain_complex()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 151  
`chain_complex()` (sage.homology.simplicial\_set.SimplicialSet\_finite method), 171  
`ChainComplex()` (in module sage.homology.chain\_complex), 3  
`ChainComplex_class` (class in sage.homology.chain\_complex), 5  
`ChainComplexHomospace` (class in sage.homology.chain\_complex\_homospace), 38  
`ChainComplexMorphism` (class in sage.homology.chain\_complex\_morphism), 27  
`ChainContraction` (class in sage.homology.chain\_homotopy), 31  
`ChainHomotopy` (class in sage.homology.chain\_homotopy), 33  
`Chains` (class in sage.homology.chains), 20  
`Chains.Element` (class in sage.homology.chains), 20  
`ChessboardComplex()` (in module sage.homology.examples), 94  
`CHomP` (class in sage.interfaces.chomp), 267  
`ClassifyingSpace()` (in module sage.homology.simplicial\_set\_examples), 201  
`coboundary()` (sage.homology.chains.Cochains.Element method), 23  
`coboundary()` (sage.homology.hochschild\_complex.HochschildComplex method), 241  
`cochain_complex()` (sage.homology.chains.Cochains method), 25  
`Cochains` (class in sage.homology.chains), 22  
`Cochains.Element` (class in sage.homology.chains), 23  
`coefficients()` (sage.homology.hochschild\_complex.HochschildComplex method), 242  
`coequalizer()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 213  
`cohomology()` (sage.homology.cell\_complex.GenericCellComplex method), 226  
`cohomology()` (sage.homology.hochschild\_complex.HochschildComplex method), 242  
`cohomology()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 152  
`cohomology_ring()` (sage.homology.cell\_complex.GenericCellComplex method), 227

CohomologyRing (class in sage.homology.homology\_vector\_space\_with\_basis), 247  
 CohomologyRing.Element (class in sage.homology.homology\_vector\_space\_with\_basis), 248  
 complex() (sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis method), 254  
 ComplexProjectivePlane() (in module sage.homology.examples), 95  
 ComplexProjectiveSpace() (in module sage.homology.simplicial\_set\_examples), 201  
 cone() (sage.homology.cubical\_complex.CubicalComplex method), 131  
 cone() (sage.homology.delta\_complex.DeltaComplex method), 113  
 cone() (sage.homology.simplicial\_complex.SimplicialComplex method), 52  
 cone() (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 152  
 cone\_vertices() (sage.homology.simplicial\_complex.SimplicialComplex method), 53  
 ConeOfSimplicialSet (class in sage.homology.simplicial\_set\_constructions), 177  
 ConeOfSimplicialSet\_finite (class in sage.homology.simplicial\_set\_constructions), 178  
 connected\_component() (sage.homology.simplicial\_complex.SimplicialComplex method), 53  
 connected\_sum() (sage.homology.cubical\_complex.CubicalComplex method), 131  
 connected\_sum() (sage.homology.delta\_complex.DeltaComplex method), 113  
 connected\_sum() (sage.homology.simplicial\_complex.SimplicialComplex method), 53  
 constant\_map() (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 153  
 constant\_map() (sage.homology.simplicial\_set\_morphism.SimplicialSetHomset method), 210  
 contraction() (sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis method), 254  
 coproduct() (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 154  
 coproduct() (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 214  
 Cube (class in sage.homology.cubical\_complex), 124  
 Cube() (sage.homology.cubical\_complex.CubicalComplexExamples method), 135  
 CubicalComplex (class in sage.homology.cubical\_complex), 127  
 CubicalComplexExamples (class in sage.homology.cubical\_complex), 135  
 cup\_product() (sage.homology.chains.Cochains.Element method), 23  
 cup\_product() (sage.homology.homology\_vector\_space\_with\_basis.CohomologyRing.Element method), 249

## D

decone() (sage.homology.simplicial\_complex.SimplicialComplex method), 54  
 defining\_map() (sage.homology.simplicial\_set\_constructions.PullbackOfSimplicialSets method), 186  
 defining\_map() (sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets method), 190  
 degeneracies() (sage.homology.simplicial\_set.AbstractSimplex\_class method), 145  
 degree() (sage.homology.chains.CellComplexReference method), 19  
 degree\_of\_differential() (sage.homology.chain\_complex.ChainComplex\_class method), 7  
 degree\_on\_basis() (sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis method), 255  
 delta\_complex() (sage.homology.simplicial\_complex.SimplicialComplex method), 54  
 DeltaComplex (class in sage.homology.delta\_complex), 107  
 DeltaComplexExamples (class in sage.homology.delta\_complex), 118  
 dhswnf() (in module sage.homology.matrix\_utils), 265  
 diagonal\_morphism() (sage.homology.simplicial\_complex\_homset.SimplicialComplexHomset method), 90  
 diagonal\_morphism() (sage.homology.simplicial\_set\_morphism.SimplicialSetHomset method), 210  
 differential() (sage.homology.chain\_complex.ChainComplex\_class method), 7  
 dimension() (sage.homology.cell\_complex.GenericCellComplex method), 228  
 dimension() (sage.homology.cubical\_complex.Cube method), 125  
 dimension() (sage.homology.simplicial\_complex.Simplex method), 44  
 dimension() (sage.homology.simplicial\_set.AbstractSimplex\_class method), 145  
 disjoint\_union() (sage.homology.cell\_complex.GenericCellComplex method), 228  
 disjoint\_union() (sage.homology.cubical\_complex.CubicalComplex method), 132

`disjoint_union()` (sage.homology.delta\_complex.DeltaComplex method), 114  
`disjoint_union()` (sage.homology.simplicial\_complex.SimplicialComplex method), 54  
`disjoint_union()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 154  
`DisjointUnionOfSimplicialSets` (class in sage.homology.simplicial\_set\_constructions), 179  
`DisjointUnionOfSimplicialSets_finite` (class in sage.homology.simplicial\_set\_constructions), 180  
`dual()` (sage.homology.chain\_complex.ChainComplex\_class method), 8  
`dual()` (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 27  
`dual()` (sage.homology.chain\_homotopy.ChainContraction method), 32  
`dual()` (sage.homology.chain\_homotopy.ChainHomotopy method), 34  
`dual()` (sage.homology.chains.Chains method), 22  
`dual()` (sage.homology.chains.Cochains method), 25  
`DunceHat()` (in module sage.homology.examples), 95

## E

`Element` (sage.homology.chain\_complex.ChainComplex\_class attribute), 5  
`elementary_subdivision()` (sage.homology.delta\_complex.DeltaComplex method), 114  
`Empty()` (in module sage.homology.simplicial\_set\_examples), 202  
`equalizer()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 214  
`euler_characteristic()` (sage.homology.cell\_complex.GenericCellComplex method), 229  
`euler_characteristic()` (sage.homology.simplicial\_set.SimplicialSet\_finite method), 171  
`eval()` (sage.homology.chains.Cochains.Element method), 24

## F

`f_triangle()` (sage.homology.simplicial\_complex.SimplicialComplex method), 55  
`f_vector()` (sage.homology.cell\_complex.GenericCellComplex method), 229  
`f_vector()` (sage.homology.simplicial\_set.SimplicialSet\_finite method), 172  
`face()` (sage.homology.cubical\_complex.Cube method), 125  
`face()` (sage.homology.simplicial\_complex.Simplex method), 44  
`face()` (sage.homology.simplicial\_complex.SimplicialComplex method), 55  
`face()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 155  
`face_data()` (sage.homology.simplicial\_set.SimplicialSet\_finite method), 172  
`face_degeneracies()` (in module sage.homology.simplicial\_set), 173  
`face_iterator()` (sage.homology.simplicial\_complex.SimplicialComplex method), 55  
`face_poset()` (sage.homology.cell\_complex.GenericCellComplex method), 229  
`face_poset()` (sage.homology.delta\_complex.DeltaComplex method), 115  
`faces()` (sage.homology.cubical\_complex.Cube method), 125  
`faces()` (sage.homology.simplicial\_complex.Simplex method), 45  
`faces()` (sage.homology.simplicial\_complex.SimplicialComplex method), 56  
`faces()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 156  
`faces_as_pairs()` (sage.homology.cubical\_complex.Cube method), 126  
`facets()` (sage.homology.simplicial\_complex.SimplicialComplex method), 56  
`facets_for_K3()` (in module sage.homology.examples), 105  
`facets_for_K3()` (in module sage.homology.simplicial\_complex), 75  
`facets_for_RP4()` (in module sage.homology.examples), 105  
`facets_for_RP4()` (in module sage.homology.simplicial\_complex), 75  
`factor()` (sage.homology.simplicial\_set\_constructions.Factors method), 181  
`factor()` (sage.homology.simplicial\_set\_constructions.ProductOfSimplicialSets method), 183  
`Factors` (class in sage.homology.simplicial\_set\_constructions), 181  
`factors()` (sage.homology.simplicial\_set\_constructions.Factors method), 181  
`fat_wedge_as_subset()` (sage.homology.simplicial\_set\_constructions.ProductOfSimplicialSets\_finite method), 184

[fiber\\_product\(\)](#) (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 82  
[fixed\\_complex\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 56  
[flip\\_graph\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 57  
[free\\_module\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 8  
[free\\_module\(\)](#) (sage.homology.hochschild\_complex.HochschildComplex method), 242  
[free\\_module\\_rank\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 9  
[fundamental\\_group\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 58

## G

[g\\_vector\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 59  
[generated\\_subcomplex\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 59  
[GenericCellComplex](#) (class in sage.homology.cell\_complex), 223  
[grading\\_group\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 9  
[graph\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 229  
[graph\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 132  
[graph\(\)](#) (sage.homology.delta\_complex.DeltaComplex method), 115  
[graph\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 60  
[graph\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 156

## H

[h\\_triangle\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 60  
[h\\_vector\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 60  
[have\\_chomp\(\)](#) (in module sage.interfaces.chomp), 268  
[help\(\)](#) (sage.interfaces.chomp.CHomP method), 268  
[HochschildComplex](#) (class in sage.homology.hochschild\_complex), 239  
[HochschildComplex.Element](#) (class in sage.homology.hochschild\_complex), 239  
[homchain\(\)](#) (in module sage.interfaces.chomp), 268  
[homcubes\(\)](#) (in module sage.interfaces.chomp), 269  
[homology\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 230  
[homology\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 9  
[homology\(\)](#) (sage.homology.hochschild\_complex.HochschildComplex method), 242  
[homology\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 157  
[homology\\_with\\_basis\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 231  
[HomologyGroup\(\)](#) (in module sage.homology.homology\_group), 245  
[HomologyGroup\\_class](#) (class in sage.homology.homology\_group), 245  
[HomologyVectorSpaceWithBasis](#) (class in sage.homology.homology\_vector\_space\_with\_basis), 251  
[HomologyVectorSpaceWithBasis.Element](#) (class in sage.homology.homology\_vector\_space\_with\_basis), 253  
[homsimpl\(\)](#) (in module sage.interfaces.chomp), 270  
[HopfMap\(\)](#) (in module sage.homology.simplicial\_set\_examples), 202  
[Horn\(\)](#) (in module sage.homology.simplicial\_set\_examples), 203

## I

[identity\(\)](#) (sage.homology.simplicial\_complex\_homset.SimplicialComplexHomset method), 90  
[identity\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 157  
[identity\(\)](#) (sage.homology.simplicial\_set\_morphism.SimplicialSetHomset method), 211  
[image\(\)](#) (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 82  
[image\(\)](#) (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 215  
[in\\_degree\(\)](#) (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 28  
[in\\_degree\(\)](#) (sage.homology.chain\_homotopy.ChainHomotopy method), 34  
[inclusion\\_map\(\)](#) (sage.homology.simplicial\_set\_constructions.DisjointUnionOfSimplicialSets\_finite method), 181



`inclusion_map()` (sage.homology.simplicial\_set\_constructions.SubSimplicialSet method), 197

`inclusion_map()` (sage.homology.simplicial\_set\_constructions.WedgeOfSimplicialSets\_finite method), 200

`induced_homology_morphism()` (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 83

`induced_homology_morphism()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 215

`InducedHomologyMorphism` (class in sage.homology.homology\_morphism), 261

`intersection()` (sage.homology.simplicial\_complex.SimplicialComplex method), 61

`iota()` (sage.homology.chain\_homotopy.ChainContraction method), 32

`is_acyclic()` (sage.homology.cell\_complex.GenericCellComplex method), 232

`is_algebraic_gradient_vector_field()` (sage.homology.chain\_homotopy.ChainHomotopy method), 35

`is_balanced()` (sage.homology.simplicial\_complex.SimplicialComplex method), 61

`is_bijective()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 216

`is_boundary()` (sage.homology.chain\_complex.Chain\_class method), 16

`is_boundary()` (sage.homology.chains.Chains.Element method), 20

`is_ChainComplexHomospace()` (in module sage.homology.chain\_complex\_homospace), 38

`is_ChainComplexMorphism()` (in module sage.homology.chain\_complex\_morphism), 29

`is_coboundary()` (sage.homology.chains.Cochains.Element method), 24

`is_cocycle()` (sage.homology.chains.Cochains.Element method), 24

`is_cohen_macaulay()` (sage.homology.simplicial\_complex.SimplicialComplex method), 62

`is_connected()` (sage.homology.cell\_complex.GenericCellComplex method), 232

`is_connected()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 158

`is_constant()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 216

`is_contiguous_to()` (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 84

`is_cycle()` (sage.homology.chain\_complex.Chain\_class method), 16

`is_cycle()` (sage.homology.chains.Chains.Element method), 21

`is_degenerate()` (sage.homology.simplicial\_set.AbstractSimplex\_class method), 145

`is_empty()` (sage.homology.simplicial\_complex.Simplex method), 45

`is_face()` (sage.homology.cubical\_complex.Cube method), 126

`is_face()` (sage.homology.simplicial\_complex.Simplex method), 45

`is_flag_complex()` (sage.homology.simplicial\_complex.SimplicialComplex method), 62

`is_homology_gradient_vector_field()` (sage.homology.chain\_homotopy.ChainHomotopy method), 35

`is_identity()` (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 28

`is_identity()` (sage.homology.homology\_morphism.InducedHomologyMorphism method), 263

`is_identity()` (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 85

`is_identity()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 216

`is_immutable()` (sage.homology.simplicial\_complex.SimplicialComplex method), 63

`is_injective()` (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 28

`is_injective()` (sage.homology.homology\_morphism.InducedHomologyMorphism method), 263

`is_injective()` (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 85

`is_injective()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 217

`is_isomorphic()` (sage.homology.simplicial\_complex.SimplicialComplex method), 63

`is_mutable()` (sage.homology.simplicial\_complex.SimplicialComplex method), 63

`is_nondegenerate()` (sage.homology.simplicial\_set.AbstractSimplex\_class method), 146

`is_partitionable()` (sage.homology.simplicial\_complex.SimplicialComplex method), 64

`is_pointed()` (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 217

`is_pseudomanifold()` (sage.homology.simplicial\_complex.SimplicialComplex method), 65

`is_pure()` (sage.homology.cubical\_complex.CubicalComplex method), 132

`is_pure()` (sage.homology.simplicial\_complex.SimplicialComplex method), 65

`is_reduced()` (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 158

`is_shellable()` (sage.homology.simplicial\_complex.SimplicialComplex method), 66



[is\\_shelling\\_order\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 66  
[is\\_SimplicialComplexHomset\(\)](#) (in module sage.homology.simplicial\_complex\_homset), 90  
[is\\_SimplicialComplexMorphism\(\)](#) (in module sage.homology.simplicial\_complex\_morphism), 86  
[is\\_subcomplex\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 132  
[is\\_surjective\(\)](#) (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 29  
[is\\_surjective\(\)](#) (sage.homology.homology\_morphism.InducedHomologyMorphism method), 263  
[is\\_surjective\(\)](#) (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 86  
[is\\_surjective\(\)](#) (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 218

## J

[join\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 233  
[join\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 133  
[join\(\)](#) (sage.homology.delta\_complex.DeltaComplex method), 115  
[join\(\)](#) (sage.homology.simplicial\_complex.Simplex method), 45  
[join\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 67  
[join\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 158

## K

[K3Surface\(\)](#) (in module sage.homology.examples), 95  
[KleinBottle\(\)](#) (in module sage.homology.examples), 96  
[KleinBottle\(\)](#) (in module sage.homology.simplicial\_set\_examples), 203  
[KleinBottle\(\)](#) (sage.homology.cubical\_complex.CubicalComplexExamples method), 136  
[KleinBottle\(\)](#) (sage.homology.delta\_complex.DeltaComplexExamples method), 119  
[KoszulComplex](#) (class in sage.homology.koszul\_complex), 237

## L

[lattice\\_paths\(\)](#) (in module sage.homology.simplicial\_complex), 76  
[link\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 68

## M

[map\\_from\\_base\(\)](#) (sage.homology.simplicial\_set\_constructions.ConeOfSimplicialSet\_finite method), 179  
[map\\_from\\_base\(\)](#) (sage.homology.simplicial\_set\_constructions.ReducedConeOfSimplicialSet\_finite method), 196  
[mapping\\_cone\(\)](#) (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 218  
[mapping\\_torus\(\)](#) (sage.homology.simplicial\_complex\_morphism.SimplicialComplexMorphism method), 86  
[matching\(\)](#) (in module sage.homology.examples), 106  
[MatchingComplex\(\)](#) (in module sage.homology.examples), 96  
[maximal\\_cells\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 133  
[maximal\\_faces\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 68  
[minimal\\_nonfaces\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 68  
[module\(\)](#) (sage.homology.hochschild\_complex.HochschildComplex method), 243  
[MooreSpace\(\)](#) (in module sage.homology.examples), 96

## N

[n\\_cells\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 233  
[n\\_cells\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 158  
[n\\_chains\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 233  
[n\\_chains\(\)](#) (sage.homology.delta\_complex.DeltaComplex method), 116  
[n\\_chains\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 159  
[n\\_cubes\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 134  
[n\\_faces\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 69

`n_skeleton()` (`sage.homology.cell_complex.GenericCellComplex` method), 234  
`n_skeleton()` (`sage.homology.cubical_complex.CubicalComplex` method), 134  
`n_skeleton()` (`sage.homology.delta_complex.DeltaComplex` method), 116  
`n_skeleton()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 69  
`n_skeleton()` (`sage.homology.simplicial_set.SimplicialSet_finite` method), 172  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.ConeOfSimplicialSet` method), 178  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.DisjointUnionOfSimplicialSets` method), 180  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.ProductOfSimplicialSets` method), 184  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.PullbackOfSimplicialSets` method), 186  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets` method), 191  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet` method), 193  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.ReducedConeOfSimplicialSet` method), 195  
`n_skeleton()` (`sage.homology.simplicial_set_constructions.SuspensionOfSimplicialSet` method), 198  
`n_skeleton()` (`sage.homology.simplicial_set_examples.Nerve` method), 204  
`n_skeleton()` (`sage.homology.simplicial_set_morphism.SimplicialSetMorphism` method), 219  
`Nerve` (class in `sage.homology.simplicial_set_examples`), 203  
`nondegenerate()` (`sage.homology.simplicial_set.AbstractSimplex_class` method), 146  
`nondegenerate_intervals()` (`sage.homology.cubical_complex.Cube` method), 126  
`nondegenerate_simplices()` (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 159  
`NonDegenerateSimplex` (class in `sage.homology.simplicial_set`), 146  
`nonzero_degrees()` (`sage.homology.chain_complex.ChainComplex_class` method), 11  
`NotIConnectedGraphs()` (in module `sage.homology.examples`), 97

## O

`one()` (`sage.homology.homology_vector_space_with_basis.CohomologyRing` method), 249  
`ordered_degrees()` (`sage.homology.chain_complex.ChainComplex_class` method), 11

## P

`pi()` (`sage.homology.chain_homotopy.ChainContraction` method), 33  
`PoincareHomologyThreeSphere()` (in module `sage.homology.examples`), 97  
`Point()` (in module `sage.homology.simplicial_set_examples`), 204  
`process_generators_chain()` (in module `sage.interfaces.chomp`), 271  
`process_generators_cubical()` (in module `sage.interfaces.chomp`), 271  
`process_generators_simplicial()` (in module `sage.interfaces.chomp`), 272  
`product()` (`sage.homology.cell_complex.GenericCellComplex` method), 234  
`product()` (`sage.homology.cubical_complex.Cube` method), 126  
`product()` (`sage.homology.cubical_complex.CubicalComplex` method), 134  
`product()` (`sage.homology.delta_complex.DeltaComplex` method), 117  
`product()` (`sage.homology.simplicial_complex.Simplex` method), 46  
`product()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 69  
`product()` (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 160  
`product()` (`sage.homology.simplicial_set_morphism.SimplicialSetMorphism` method), 219  
`product_on_basis()` (`sage.homology.homology_vector_space_with_basis.CohomologyRing` method), 250  
`ProductOfSimplicialSets` (class in `sage.homology.simplicial_set_constructions`), 182  
`ProductOfSimplicialSets_finite` (class in `sage.homology.simplicial_set_constructions`), 184  
`projection_map()` (`sage.homology.simplicial_set_constructions.ProductOfSimplicialSets_finite` method), 184  
`projection_map()` (`sage.homology.simplicial_set_constructions.PullbackOfSimplicialSets_finite` method), 187  
`projection_map()` (`sage.homology.simplicial_set_constructions.WedgeOfSimplicialSets_finite` method), 200  
`ProjectivePlane()` (in module `sage.homology.examples`), 97  
`PseudoQuaternionicProjectivePlane()` (in module `sage.homology.examples`), 98

[pullback\(\)](#) (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 162  
[pullback\(\)](#) (`sage.homology.simplicial_set_morphism.SimplicialSetMorphism` method), 220  
[PullbackOfSimplicialSets](#) (class in `sage.homology.simplicial_set_constructions`), 185  
[PullbackOfSimplicialSets\\_finite](#) (class in `sage.homology.simplicial_set_constructions`), 187  
[pushout\(\)](#) (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 163  
[pushout\(\)](#) (`sage.homology.simplicial_set_morphism.SimplicialSetMorphism` method), 220  
[PushoutOfSimplicialSets](#) (class in `sage.homology.simplicial_set_constructions`), 188  
[PushoutOfSimplicialSets\\_finite](#) (class in `sage.homology.simplicial_set_constructions`), 191

## Q

[quotient\(\)](#) (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 164  
[quotient\\_map\(\)](#) (`sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet_finite` method), 194  
[QuotientOfSimplicialSet](#) (class in `sage.homology.simplicial_set_constructions`), 192  
[QuotientOfSimplicialSet\\_finite](#) (class in `sage.homology.simplicial_set_constructions`), 194

## R

[random\\_element\(\)](#) (`sage.homology.chain_complex.ChainComplex_class` method), 12  
[RandomComplex\(\)](#) (in module `sage.homology.examples`), 98  
[RandomTwoSphere\(\)](#) (in module `sage.homology.examples`), 99  
[rank\(\)](#) (`sage.homology.chain_complex.ChainComplex_class` method), 12  
[RealProjectivePlane\(\)](#) (in module `sage.homology.examples`), 99  
[RealProjectivePlane\(\)](#) (`sage.homology.cubical_complex.CubicalComplexExamples` method), 136  
[RealProjectivePlane\(\)](#) (`sage.homology.delta_complex.DeltaComplexExamples` method), 119  
[RealProjectiveSpace\(\)](#) (in module `sage.homology.examples`), 100  
[RealProjectiveSpace\(\)](#) (in module `sage.homology.simplicial_set_examples`), 204  
[reduce\(\)](#) (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 166  
[ReducedConeOfSimplicialSet](#) (class in `sage.homology.simplicial_set_constructions`), 194  
[ReducedConeOfSimplicialSet\\_finite](#) (class in `sage.homology.simplicial_set_constructions`), 195  
[remove\\_face\(\)](#) (`sage.homology.simplicial_complex.SimplicialComplex` method), 70  
[remove\\_faces\(\)](#) (`sage.homology.simplicial_complex.SimplicialComplex` method), 71  
[rename\\_latex\(\)](#) (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 166  
[rename\\_vertex\(\)](#) (in module `sage.homology.simplicial_complex`), 77  
[restriction\\_sets\(\)](#) (`sage.homology.simplicial_complex.SimplicialComplex` method), 71  
[RudinBall\(\)](#) (in module `sage.homology.examples`), 101

## S

[sage.homology.algebraic\\_topological\\_model](#) (module), 257  
[sage.homology.cell\\_complex](#) (module), 223  
[sage.homology.chain\\_complex](#) (module), 3  
[sage.homology.chain\\_complex\\_homspace](#) (module), 37  
[sage.homology.chain\\_complex\\_morphism](#) (module), 27  
[sage.homology.chain\\_homotopy](#) (module), 31  
[sage.homology.chains](#) (module), 19  
[sage.homology.cubical\\_complex](#) (module), 123  
[sage.homology.delta\\_complex](#) (module), 107  
[sage.homology.examples](#) (module), 93  
[sage.homology.hochschild\\_complex](#) (module), 239  
[sage.homology.homology\\_group](#) (module), 245  
[sage.homology.homology\\_morphism](#) (module), 261  
[sage.homology.homology\\_vector\\_space\\_with\\_basis](#) (module), 247

`sage.homology.koszul_complex` (module), 237  
`sage.homology.matrix_utils` (module), 265  
`sage.homology.simplicial_complex` (module), 41  
`sage.homology.simplicial_complex_homset` (module), 89  
`sage.homology.simplicial_complex_morphism` (module), 79  
`sage.homology.simplicial_set` (module), 139  
`sage.homology.simplicial_set_catalog` (module), 207  
`sage.homology.simplicial_set_constructions` (module), 177  
`sage.homology.simplicial_set_examples` (module), 201  
`sage.homology.simplicial_set_morphism` (module), 209  
`sage.interfaces.chomp` (module), 267  
`set()` (`sage.homology.simplicial_complex.Simplex` method), 46  
`set_immutable()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 72  
`shift()` (`sage.homology.chain_complex.ChainComplex_class` method), 12  
`ShiftedComplex()` (in module `sage.homology.examples`), 101  
`shrink_simplicial_complex()` (in module `sage.homology.simplicial_set`), 174  
`Simplex` (class in `sage.homology.simplicial_complex`), 43  
`Simplex()` (in module `sage.homology.examples`), 102  
`Simplex()` (in module `sage.homology.simplicial_set_examples`), 205  
`Simplex()` (`sage.homology.delta_complex.DeltaComplexExamples` method), 119  
`simplicial_data_from_kenzo_output()` (in module `sage.homology.simplicial_set_examples`), 206  
`SimplicialComplex` (class in `sage.homology.simplicial_complex`), 47  
`SimplicialComplexHomset` (class in `sage.homology.simplicial_complex_homset`), 89  
`SimplicialComplexMorphism` (class in `sage.homology.simplicial_complex_morphism`), 80  
`SimplicialSet` (in module `sage.homology.simplicial_set`), 147  
`SimplicialSet_arbitrary` (class in `sage.homology.simplicial_set`), 147  
`SimplicialSet_finite` (class in `sage.homology.simplicial_set`), 169  
`SimplicialSetHomset` (class in `sage.homology.simplicial_set_morphism`), 209  
`SimplicialSetMorphism` (class in `sage.homology.simplicial_set_morphism`), 211  
`SmashProductOfSimplicialSets_finite` (class in `sage.homology.simplicial_set_constructions`), 196  
`Sphere()` (in module `sage.homology.examples`), 102  
`Sphere()` (in module `sage.homology.simplicial_set_examples`), 205  
`Sphere()` (`sage.homology.cubical_complex.CubicalComplexExamples` method), 136  
`Sphere()` (`sage.homology.delta_complex.DeltaComplexExamples` method), 120  
`Sq()` (`sage.homology.homology_vector_space_with_basis.CohomologyRing.Element` method), 248  
`standardize_degeneracies()` (in module `sage.homology.simplicial_set`), 174  
`standardize_face_maps()` (in module `sage.homology.simplicial_set`), 175  
`stanley_reisner_ring()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 72  
`star()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 73  
`stellar_subdivision()` (`sage.homology.simplicial_complex.SimplicialComplex` method), 73  
`structure_map()` (`sage.homology.simplicial_set_constructions.PullbackOfSimplicialSets_finite` method), 187  
`structure_map()` (`sage.homology.simplicial_set_constructions.PushoutOfSimplicialSets_finite` method), 191  
`subcomplex()` (`sage.homology.delta_complex.DeltaComplex` method), 117  
`subcomplex()` (`sage.homology.simplicial_set_constructions.QuotientOfSimplicialSet` method), 194  
`subsimplicial_set()` (`sage.homology.simplicial_set.SimplicialSet_arbitrary` method), 166  
`SubSimplicialSet` (class in `sage.homology.simplicial_set_constructions`), 196  
`sum_indices()` (in module `sage.homology.homology_vector_space_with_basis`), 255  
`SumComplex()` (in module `sage.homology.examples`), 102  
`summand()` (`sage.homology.simplicial_set_constructions.DisjointUnionOfSimplicialSets` method), 180  
`summand()` (`sage.homology.simplicial_set_constructions.WedgeOfSimplicialSets` method), 199

[summands\(\)](#) (sage.homology.simplicial\_set\_constructions.DisjointUnionOfSimplicialSets method), 180  
[summands\(\)](#) (sage.homology.simplicial\_set\_constructions.WedgeOfSimplicialSets method), 199  
[SurfaceOfGenus\(\)](#) (in module sage.homology.examples), 104  
[SurfaceOfGenus\(\)](#) (sage.homology.cubical\_complex.CubicalComplexExamples method), 136  
[SurfaceOfGenus\(\)](#) (sage.homology.delta\_complex.DeltaComplexExamples method), 120  
[suspension\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 134  
[suspension\(\)](#) (sage.homology.delta\_complex.DeltaComplex method), 118  
[suspension\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 74  
[suspension\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 167  
[suspension\(\)](#) (sage.homology.simplicial\_set\_morphism.SimplicialSetMorphism method), 221  
[SuspensionOfSimplicialSet](#) (class in sage.homology.simplicial\_set\_constructions), 197  
[SuspensionOfSimplicialSet\\_finite](#) (class in sage.homology.simplicial\_set\_constructions), 198

## T

[tensor\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 13  
[to\\_complex\(\)](#) (sage.homology.chains.Chains.Element method), 21  
[to\\_complex\(\)](#) (sage.homology.chains.Cochains.Element method), 24  
[to\\_cycle\(\)](#) (sage.homology.homology\_vector\_space\_with\_basis.HomologyVectorSpaceWithBasis.Element method), 253  
[to\\_matrix\(\)](#) (sage.homology.chain\_complex\_morphism.ChainComplexMorphism method), 29  
[to\\_matrix\(\)](#) (sage.homology.homology\_morphism.InducedHomologyMorphism method), 263  
[torsion\\_list\(\)](#) (sage.homology.chain\_complex.ChainComplex\_class method), 15  
[Torus\(\)](#) (in module sage.homology.examples), 104  
[Torus\(\)](#) (in module sage.homology.simplicial\_set\_examples), 205  
[Torus\(\)](#) (sage.homology.cubical\_complex.CubicalComplexExamples method), 137  
[Torus\(\)](#) (sage.homology.delta\_complex.DeltaComplexExamples method), 120  
[trivial\\_module\(\)](#) (sage.homology.hochschild\_complex.HochschildComplex method), 243  
[tuple\(\)](#) (sage.homology.cubical\_complex.Cube method), 127  
[tuple\(\)](#) (sage.homology.simplicial\_complex.Simplex method), 46

## U

[UniqueSimplicialComplex](#) (class in sage.homology.examples), 104  
[universal\\_property\(\)](#) (sage.homology.simplicial\_set\_constructions.PullbackOfSimplicialSets\_finite method), 188  
[universal\\_property\(\)](#) (sage.homology.simplicial\_set\_constructions.PushoutOfSimplicialSets\_finite method), 192

## V

[vector\(\)](#) (sage.homology.chain\_complex.Chain\_class method), 16  
[vector\(\)](#) (sage.homology.hochschild\_complex.HochschildComplex.Element method), 240  
[vertices\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 75

## W

[wedge\(\)](#) (sage.homology.cell\_complex.GenericCellComplex method), 234  
[wedge\(\)](#) (sage.homology.cubical\_complex.CubicalComplex method), 135  
[wedge\(\)](#) (sage.homology.delta\_complex.DeltaComplex method), 118  
[wedge\(\)](#) (sage.homology.simplicial\_complex.SimplicialComplex method), 75  
[wedge\(\)](#) (sage.homology.simplicial\_set.SimplicialSet\_arbitrary method), 168  
[wedge\\_as\\_subset\(\)](#) (sage.homology.simplicial\_set\_constructions.ProductOfSimplicialSets\_finite method), 185  
[WedgeOfSimplicialSets](#) (class in sage.homology.simplicial\_set\_constructions), 198  
[WedgeOfSimplicialSets\\_finite](#) (class in sage.homology.simplicial\_set\_constructions), 200

## Z

ZieglerBall() (in module `sage.homology.examples`), [105](#)